


Excel

效率手册 早做完，不加班

(精华版 函数篇)

(升级版)

陈锡卢 著



清华大学出版社
北京

Excel效率手册

——早做完，不加班

(精华版 函数篇)升级版

陈锡卢 著

清华大学出版社
北京

内 容 简 介

这是一本有趣的Excel书，基于作者的职场经历改编而成！通过讲故事的形式，讲解Excel中的函数与公式，让读者在轻松、愉快的环境中学到各种技能。跟着书中主角卢子前行，所有Excel函数与公式的疑难、所有复杂的数据以及表格问题都将得到解决。

本书为读者展示了在Excel 2013中运用函数与公式解决疑难问题的实战技能，包含函数与公式的基础理论及操作技能、各类函数的实际应用。在函数语法描述中，全部采用中文表述与函数图解，对Excel 2013帮助文件进行提炼和修正，帮助读者在理解函数时少走弯路。同时提供知识扩展和配套的相关练习题目，可以让读者更全面地学习和巩固技能。

本书能有效地帮助职场新人提升竞争力，也能帮助财务、品质分析、人力资源管理等人员解决实际问题。

本书封面贴有清华大学出版社防伪标签，无标签者不得销售。

版权所有，侵权必究。侵权举报电话：010-62782989 13701121933

图书在版编目(CIP)数据

Excel效率手册：早做完，不加班：精华版：升级版．函数篇 / 陈锡卢著．—北京：清华大学出版社，2017

ISBN 978-7-302-47686-3

I. ①E… II. ①陈… III. ①表处理软件—手册 IV. ①TP391.13-62

中国版本图书馆CIP数据核字(2017)第155281号

责任编辑：秦 甲

封面设计：梁鸿雁

责任校对：吴春华

责任印制：

出版发行：清华大学出版社

网 址：<http://www.tup.com.cn>，<http://www.wqbook.com>

地 址：北京清华大学学研大厦A座

邮 编：100084

社 总 机：010-62770175

邮 购：010-62786544

投稿与读者服务：010-62776969，c-service@tup.tsinghua.edu.cn

质 量 反 馈：010-62772015，zhiliang@tup.tsinghua.edu.cn

印 装 者：

经 销：全国新华书店

开 本：180mm×210mm

印 张：12.25

字 数：392千字

版 次：2017年8月第1版

印 次：2017年8月第1次印刷

印 数：1~4000

定 价：55.00元

产品编号：074997-01

推荐语

作为中国会计视野论坛的管理员，我和卢子版主有很多接触。他一直为网友解答各种有关Excel的问题，无论是低级的还是复杂的，都孜孜不倦地为人答疑解惑。他也是一个乐于分享的人。本书的很多内容来自视野论坛数万点击量的热帖“听故事，学Excel”，随着故事的讲述让大家由浅入深地学习Excel的各种功能，非常有趣，值得一读。

——胡晓栋


上海国家会计学院中国会计视野论坛
(bbs.esnai.com) 管理员

“学会计，有方法”一直以来都是会计网的建站宗旨，会计越学越活，方法越用越新。作为一名会计财务人员，熟练掌握实用的Excel技巧，将大大提升工作效率。卢子的这本书，深入浅出地讲解了Excel的各种技巧，图文并茂，如同真人现身指导，颇有特色。书中很多案例都来自卢子与网友的互动，让人边看边学，很有参与感。更难得的是卢子非常系统地为一章都做了小结，有方法总结，也有学习感悟，都是实战干货。认真看，经常用，您就能够发现一个简单、有趣的Excel世界。

——李海明

会计网 (kuaiji.com) 联合创始人





公式与函数是我们借助Excel完成工作使用频率最高，也是Excel非常强大的一个功能，因此每个Excel用户都应很好地掌握它。

本书的前一个版本《Excel效率手册 早做完，不加班》已获得巨大成功。如果您看过《Excel效率手册 早做完，不加班》，再看本书，您获取的Excel知识会更加系统和连贯。

本书依然延续了前一个版本的文风，采用“引入情景案例 → 提出解决方案 → 涉及Excel关键知识点剖析讲解 → 完整操作步骤 → 扩展知识点讲述”的方法讲解Excel函数与公式的精髓。

——杨阳

IT部落窝（www.itblw.com）创办者

本书是在第一本《Excel效率手册 早做完，不加班》的基础上对函数的拓展讲解，结合了作者本人的实际工作经验和在帮助他人过程中遇到的问题。本书对于常用函数都有比较详细的分解，并且结合实际中的由于制表样式问题使得统计难度提升，再通过作者对函数理解的基础上对表格及公式的修改，减少了令人战栗的长公式，使得公式简单易懂。希望读者通过阅读本书，提升自己的对公式的理解和运用水平，减少自身的统计工作量，可以快快乐乐上班，轻轻松松回家，而不用加班。



——李应钦

IT部落窝论坛版主



这是一本很有特色的Excel函数书。书中以卢子的学习之路为主线，介绍了对函数认识的四个阶段，以故事的形式讲述Excel函数知识，让读者在轻松阅读中见识函数公式的神奇。相信认真阅读此书不仅会让读者提升函数的应用能力，更能培养一些分析问题、解决问题的能力。提高效率，远离加班！

——李云龙
自然美国际事业集团



前言

经常有朋友问：函数难学吗？不会英语能学好函数吗？

对于这种问题我也深有感触，因为我曾经就有过这样的疑问。经过长时间的摸索，现在终于找到了答案！

市面上有关函数的书籍非常多，但大都写得太过专业并且死板，看起来枯燥无味。对于这样的书籍，大多数人都很难坚持看完全书。通常都是随手翻看几页，就束之高阁。

当初我也一直在寻找一本通俗易懂的书，能让我有兴趣看完全书，但遗憾的是，找了很多年，至今都没有找到，为此我很苦恼。在学习的过程中很吃力，为了避免大家重蹈我的覆辙，我决定将自己这十年来的成长经验通过故事的形式讲给大家听。

如果您觉得看起来很轻松，千万别误以为是一本小说，其实这是一本讲述我当初学习函数心路历程的书。

请大声地对自己说：我有信心学好函数！然后继续看下去。

Excel中内置的函数就有几百个，很多人看到这么多的函数头都大了，其实只要掌握十分之一就是高手。对于普通人来说，学会20个左右的函数即可完成绝大部分工作。抓住重点，才能事半功倍。

学习函数其实并不难，即使不会英语照样可以学好函数。为什么这样说呢？因为我本人就是从菜鸟走过来的，当初什么都不懂，而且我本人英语水平就很差，在我的印象中除了作弊以外，没有一次考试及格。像我这种水平的人都能学好函数，您有什么理由学不好呢？只要您对自己多一点信心，就一定可以学会！下面就让我们一起来学习函数，这一路走来，相信您会有意想不到的收获。



致谢

感谢Excel，我的恋人。感谢好友“无言的人”对我技术上的支持，让我的Excel更添光彩。
感谢ExcelHome站长周庆麟，是他的帮助让我获得了学习Excel的快乐。感谢中国会计视野论坛、IT部落窝所有的Excel网友，是你们让我感受到更广泛地分享的快乐。还要感谢您，拿起这本书的读者——您的阅读，将我对Excel的爱传向广阔的世界。

另外，还得感谢以下参与本书编写的人员：郑晓芬、李应钦、邱显标、龚思兰、邓丹、陆超男、杨明辉、刘宋连、邓海南。

编 者





升级版说明

“Excel效率手册 早做完，不加班”系列书自2014年4月出版以来得到了20万读者的支持，对于只有高中学历的我而言，简直就是奇迹！其实这套书写得并不完美，有很多地方写得不够通俗易懂，也不够全面。长时间在售后群和读者交流沟通后，征集了读者大量的反馈意见，从而对本书进行升级。升级篇最大的特点就是采用了大量函数图解让书籍更加通俗易懂，还增加了大量的知识扩展，对相关知识点进行补充说明。以前是站在自己的角度看问题，现在是站在读者的角度看问题。

售后服务

为了让读者能够更好地学习函数，本书赠送大量视频教程，同时还提供QQ交流群在线交流，群号：576517640。这是第5个2000人的交流群，我们的交流群非常人性化，除了交流Excel和工作，同时还允许偶尔胡侃！

我们的口号是什么！



本书Excel版本说明

本书的操作版本为Excel 2013版，让你在了解新功能的同时，感叹新版本的强大，同时会让您爱不释手。对于旧版本的朋友也不用担心，Excel的兼容性很强，2007版、2010版与2013版基本一致。



目录

CONTENTS

第1章

意外的升职

- 1.1 人生第一次培训2
 - 1.1.1 快速准确录入数据 3
 - 1.1.2 图片处理小妙招 8
 - 1.1.3 排版打印细节 10
- 1.2 尝试制作模板 11
 - 1.2.1 不良记录统计表 11
 - 1.2.2 餐费报销表 14
 - 1.2.3 车费报销表 15
 - 1.2.4 产品报价表 16
 - 1.2.5 获取Excel提供的精美表格模板 18
- 1.3 小结与练习 20

第2章

高手变菜鸟

- 2.1 小插曲，转学Photoshop失败 22
- 2.2 好奇害死猫，走上Excel不归路 23

第3章

改变

- 2.3 为何你至今还是一个Excel菜鸟 26
- 2.4 小结与练习 29

- 3.1 勇敢踏出第一步 32
- 3.2 原来统计并非手工活 36
- 3.3 小结与练习 39

第4章

全面扫盲

- 4.1 学习前的准备 42

4.1.1	基本概念	42
4.1.2	初识函数	44
4.1.3	快速、准确地输入函数的两种方法	45
4.1.4	编辑公式三招	46
4.1.5	快速填充公式小妙招	48
4.1.6	输入的公式为什么不参与计算	50
4.1.7	删除公式只保留计算结果	52
4.1.8	保护公式不被修改	53
4.1.9	定义名称的三种方法	57
4.1.10	获取帮助的方法和渠道	61
4.2	实战	62
4.2.1	四则运算	62
4.2.2	比较运算符	64
4.2.3	相对引用、绝对引用、混合引用	66
4.3	小结与练习	68



5.1	SUM函数家族	70
5.1.1	SUM函数三种实用的求和用法	70
5.1.2	SUMIF(SUMIFS)函数按条件求和及优势解读	75
5.1.3	DSUM函数多条件求和的优缺点	79
5.2	VLOOKUP函数家族	85
5.2.1	VLOOKUP函数按条件查询对应值	86
5.2.2	LOOKUP函数各种方向查询对应值的经典公式	87
5.2.3	OFFSET函数的经典运用	88

5.2.4	初识INDEX、MATCH函数组合	93
5.3	IF函数家族	95
5.4	其他常用函数	98
5.4.1	LEFT、MID、RIGHT字符提取三兄弟	98
5.4.2	SUBSTITUTE、REPLACE函数	102
5.4.3	COUNT函数家族	104
5.4.4	极值与均值	106
5.4.5	日期与时间的处理	107
5.5	小试身手	111
5.5.1	加入第一帮派——ExcelHome	111
5.5.2	凭借VLOOKUP函数解疑难	112
5.5.3	LOOKUP函数的二分法只是个传说	118
5.5.4	七夕闲聊	123
5.5.5	随便VS随机	125
5.5.6	四舍五入，卢子已奔三	128
5.5.7	数字舍入，存在即合理	130
5.6	小结与练习	131



6.1	数组的基本概念	134
6.2	数组公式的优势	137
6.3	手把手教你学数组公式	141
6.4	有趣的自动扩展现象	144



6.5 小结与练习	146
-----------------	-----



7.1 像玩游戏一样玩SUM函数	148
7.1.1 闲聊SUM函数	148
7.1.2 SUM函数取代SUMIF(SUMIFS)函数 ..	151
7.1.3 SUM函数取代COUNTIF (COUNTIFS)函数	157
7.1.4 SUM函数价值的体现	161
7.1.5 SUM函数优势尽显的一面	166
7.1.6 SUMPRODUCT函数取代 SUM函数的数组公式	169
7.2 万金油公式组合的妙用	171
7.2.1 按条件提取对应值	172
7.2.2 筛选符合条件的所有对应值	175
7.3 小结与练习	177



8.1 从疯狂下载者变成真正学习者 ...	180
-----------------------	-----

8.2 为公式消得人憔悴	181
--------------------	-----

8.2.1 三大隐藏函数的用法	181
8.2.2 两大让你望而却步的高级函数	184
8.2.3 与名称共存亡的宏表函数	190

8.3 小结与练习	196
-----------------	-----



9.1 深思，迷途知返	198
-------------------	-----

9.2 学以致用	199
----------------	-----

9.2.1 优化出差费用统计表	199
9.2.2 优化检查日报	205
9.2.3 优化月度总结表	208
9.2.4 挑战月度不良统计表	211
9.2.5 月度数据核对	224

9.3 疑难汇集	226
----------------	-----

9.3.1 身份证问题汇集	227
9.3.2 字符提取之奇思妙想	232
9.3.3 将查找进行到底	235
9.3.4 认识各种各样的排名	239

9.4 小结与练习	241
-----------------	-----





10.1 我的前任是极品..... 244

10.2 表格大改革..... 247

10.2.1 来料 247

10.2.2 生产 248

10.2.3 销售 250

10.2.4 进销存 259

10.3 左手系统，右手Excel 262

10.3.1 巧借考勤系统 262

10.3.2 考勤数据再整理 267

10.3.3 统计工资 269

10.3.4 制作工资条 270

10.4 小结与练习..... 274

写在后面的话..... 276



第1章

意外的升职



毕业并不代表学习的结束，而是另一种学习的开始。卢子听了香姐的话后，用心学习Excel，转眼已经两年过去了。这两年卢子成长很快，学了很多数据录入的技能，工作效率也因此提高很多。

机会永远是给有准备之人的！也许是上天对卢子的眷顾，品质组长要调往日本学习，卢子就是品质组长的最佳人选。机会来了，卢子一定要好好把握，争取得到领导的认可。

1.1 人生第一次培训

当了品质组长以后，卢子手下就有3个组员，这些人都是刚毕业的学生。品质方面的技能肯定要教会，但这不是一时半会儿就可以掌握的，可以先缓一缓。当务之急是教他们做报告，报告是直接呈现给领导看的，其他工作做得好不好领导并不能直接看到，但报告做得不好领导可是看在眼里，挨批的还是卢子本人。

想法已定，卢子就召集了所有组员进行了一次简单的培训，内容是《出货检查不良报告》的制作。主要包含三大方面的内容。

◎ 数据录入的一些技巧，如图1-1所示，在模板内录入内容。

	A	B	C	D	E	F	G	H
1	联络：隆成（小天使）/技术课							2009年6月10日
2	发行号：--							Richell品质保证课
3	供应商：隆成（小天使）殿							陈锡卢
4	出货检查不良报告							
5	以下不良内容请技术课协助解决							
6								出货日期：
7								检查日期：090610
9	产品番号	品名	出货数	检查数	良品数	不良数	Lot.No.	
10	98133	HKワイリッシュユロイド	233	245		12		
11						不良数	不良数	
12	不良项目：		1、Richell盖装反		B	4	245	
13			2、收车座过松		B	3	245	
14			3、一道锁，二道锁用拉力器测试超出基准		B	3	245	
15			4、塑胶件水口残留		B	2	245	

图1-1 数据录入模板

◎ 图片处理的技巧，如图1-2所示。




图1-2 图片处理

◎ 打印的技巧，如图1-3所示。



图1-3 打印

一份表格只有经过这三大步骤才算完整，原始数据和图片都是根据检查现场获得，不能随意捏造！

1.1.1 快速准确录入数据

卢子的话刚说完，就听到组员潘小姐在发牢骚。

潘小姐：这些有什么好说的，以前读计算机专业的时候都学过。

卢子：很好，就由你先来制作第一张表格。

潘小姐：做就做，谁怕谁！



说完，就看见潘小姐来到电脑前操作起来。卢子仔细观察着她的录入过程，里面内容怎么写她就跟着写，没有涉及任何快捷键和单元格设置。比如2009年6月10日，就照着这个日期录入，就连“检查数”全部都是245也逐一录入。

看到这里，卢子不由得打断了还在操作的潘小姐。

卢子：先停一停，你知道“快捷键”这个词吗？

潘小姐：我打字速度很快，从来不记这些快捷键。

卢子：(◎o◎)啊！记住常用的几个快捷键能够使你事半功倍，靠笨方法不可取。就如输入2009年6月10

日，这个是当天日期。其实用快捷键Ctrl+;就可以快速输入，这样还能保证数字录入的准确性，如图1-4所示。




图1-4 快速录入当天日期

突然，另一个组员杨小姐打断我的话：这样输入的是2009-6-10，跟2009年6月10日并不一样啊。

卢子：没错。其实这只是第一步而已，还剩下一个关键的步骤，就是设置单元格格式。按快捷键Ctrl+1，弹出“设置单元格格式”对话框，切换到“数字”选项卡，然后在“分类”列表框中选择“日期”选项，并在“类型”列表框中选择包含年月日的类型，观察示例的变化，如图1-5所示。




图1-5 设置日期格式



如果没有设置为“文本”格式，在其他任何单元格中输入“090610”，按Enter键，首个“0”就会自动消失，如图1-8所示，不信你们可以自己动手试试。

组员们尝试了几次后，终于都承认了这个事实。

其实这里也可以直接在单元格中输入“0, 090610”，也就是在数字前面增加一个“，”，这样就等同于文本格式。

看着组员们终于有一点点改变，卢子趁热打铁：大家有没有注意到“不良等级”这里，其实可以通过下拉菜单选择，如图1-9所示。通过设置下拉菜单，可以更准确地输入等级，防止出错。

对于“不良等级”可能体现不出价值，但如果对长字符串的内容就非常明显，比如输入供应商名称，如图1-10所示。




图1-8 自动消失的“0”




图1-9 下拉菜单选择



图1-10 输入供应商名称

下拉菜单的制作，如图1-11所示，选择单元格区域F13:F16，切换到“数据”选项卡，再单击“数据验证”（注：低版本叫“数据有效性”）按钮，在“允许”下拉列表框中选择“序列”选项，来源设置为A,B,C，最后单击“确定”按钮。切记，要在英文状态下输入内容。




图1-11 序列的设置



知识扩展

有时候需要生成A~Z这样的字母，直接输入A并下拉得到的将全部是A，并不会自动变成B、C，如果借助公式就可以轻松实现，如图1-12所示。

=CHAR(64+ROW(A1))

	A	B	C	D	E
1					
2		直接下拉		公式生成	
3		A		A	
4		A		B	
5		A		C	
6		A		D	
7		A		E	
8		A		F	
9		A		G	
10		A		H	
11					

图1-12 输入字母

A的ASCII码是65，B的ASCII码是66，以此类推。而ROW函数是获取1~N这种序号，ROW(A1)是1，ROW(A2)是2。64+ROW(A1)是65，这样嵌套CHAR函数就得到了相对应的字母。

我们在录入检查数的时候，检查数都是相同的数字，一个个录入效率很低，这时可以尝试下这个方法：选择单元格区域H13:H16，输入245，最后按快捷键Ctrl+Enter，如图1-13所示。

245						
	D	E	F	G	H	I
			2017年6月7日			
发行号：--			Richell品质保证课			
			陈锡卢			
出货检查不良报告						
以下不良内容请技术课协助解决						
			出货日期：未定			
			检查日期：090610			
名	出货数	检查数	良品数	不良等级	不良数	检查数
シュロイド	233	245	233			
				B	4	245
				B	3	245
常用拉力器测试超出基准				B	3	245
留				B	2	245

图1-13 快速录入相同数据



知识扩展

这里的检查数和E10单元格中的检查数其实是一样的，也就是说，可以直接引用这个单元格来自动生成。在H13单元格中输入公式，并向下填充公式，如图1-14所示。

=E\$10

fx =E\$10						
C	D	E	F	G	H	I
					2017年6月7日	
发行号：--					Richell品质保证课	
股					陈锡卢	
出货检查不良报告						
以下不良内容请技术课协助解决						
				出货日期：未定		
				检查日期：090610		
品名		出货数	检查数	良品数	不良数	Lot.No.
イリッシュロイド		233	245	233	12	090603
				不良等级	不良数	检查数
1盖装反				B	4	245
2过松				B	3	245
3二道锁用拉力器测试超出基准				B	3	245
4水口残留				B	2	245

图1-14 填充公式

1.1.2 图片处理小妙招

卢子：制作报告有一个很重要的环节，就是图片的处理。通过照相机拍摄的图片只是原始图片，我们还需要对图片进行简单的描述，这样别人才能够清楚地知道问题出在哪里。现在以其中的“Richell盖装反”这项不良信息进行说明。

切换到“插入”选项卡，再单击“图片”按钮，浏览到图片所在的位置，选中如图1-15所示的两张图片，最后单击“插入”按钮。图片插入后，大家怎样调整大小呢？




图1-15 插入图片



其实，卢子的组内还有一个成员，那就是卢子的同学，就叫他陈先生吧。

这回终于轮到陈先生发言了：如图1-16所示，将鼠标指针移到图片4个顶点的中间，会出现一个能上下拖动大小的双箭头，可以利用它进行调整。

卢子：这是一种可行的方法，但这样调整处理的图片容易变形。我们可以找另外4个点进行调整。如图1-17所示，在图片的4个边角进行拖动，这样图片的放大缩小就保持固定的比例而不会导致变形。



图1-16 中间双箭头调整




图1-17 边角双箭头调整

更多的时候，卢子是利用“图片工具”输入数据进行大小的调整，如图1-18所示。这样有一个好处，可以将所有图片的大小都设置成一样的，比用双箭头调整更快捷。




图1-18 输入调整

图片处理还有很多技巧，我再简单介绍下，其他的还得靠大家平时多去学习。图片大小调整后，需要将一些内容重点圈出来并进行说明。如图1-19所示，切换到“插入”选项卡，单击“形状”按钮，在弹出的下拉菜单中选择长方形，当然也可以选择其他的形状。插入形状后对形状进行一些简单的设置，这个就留给大家去摸索，一口吃不成胖子，慢慢来。

组员应声：没想到小小表格要学习的东西还挺多的。




图1-19 插入形状



知识扩展

要养成一个好习惯，就是将图片和形状组合起来，这样移动时才不会导致形状错位。按住Ctrl键选中每一个图形，右击，在弹出的下拉菜单中选择“组合”命令，如图1-20所示。



图1-20 选择“组合”命令

1.1.3 排版打印细节

卢子：现在讲一下排版打印的问题，因为数据录入、图片处理最终都是为了打印出来交给领导。虽说这个是模板，事先已经设置成一页打印，但有时一些误操作会导致一页内容容纳不了，比如在表格插入一行后的打印预览效果，打印预览的快捷键为Ctrl+P。如图1-21所示，这样打印出来的表格很不美观，就得查看原始表格，看看是什么原因。在表格中插入行列或者图片太大等，应重新调整大小，设置成一页打印。

其实做表格与检查产品一样，都需要细心，只有细心才能做好工作。




图1-21 打印预览



组员：以后我们会用心做好表格的，请你放心。

卢子：这样最好了。今天的培训就到此结束，我只是教大家一些简单的方法，要靠大家多动手操作才能掌握好，以后在实际操作中有问题可以随时找我。

1.2 尝试制作模板

很多事情只有亲身经历才知道，虽然只是一个组长，但需要考虑的问题很多，如图1-22所示。

序号	事项	进展情况
1	新人培训	已完成
2	各种报告模板的制作	进行中
3	安排每个人的工作	进行中
4	跟供应商、办公室、组员之间的沟通	进行中
5	处理各种问题	进行中

图1-22 处理事项

卢子过去习惯了依赖别人，现在必须逼着自己改变。新人培训已告一段落，现在可以做的就是尝试制作各种报告模板，其他的问题慢慢解决。如果此时卢子有三头六臂就好了，可惜没有。

1.2.1 不良记录统计表

在日企必须及时地向领导反馈信息，这点很重要。每天将所做的主要事情跟领导报告，如果出差在外面，可以通过发送邮件把事情简要说明，必要时添加Excel文档。如图1-23所示，



发送前一定要确认称呼妥当，无错别字，附件有没有添加，确认无误后再发送。如果是重要邮件，一定要打电话确认领导有没有收到。




图1-23 发送电子邮件

说句实话，出差对于卢子而言就是家常便饭，一个月会有半个月的时间在外面出差。跟领导沟通最多的是靠邮件，但邮件所能反馈的信息毕竟有限，还需要借助一些表格进行辅助说明。为此卢子考虑了一段时间，终于开始尝试制作“不良记录统计表”，如图1-24所示。

	A	B	C	E	F	G	H	I	J
									不良记录
1									
2									
3	序号	检查日期	番号	良品数	不良数	良品率			
4							项目	不良项1	
5	1	090712	41321 (BR)	130	3	98%	名称	漏警告标	
6							数量	1	
7	2	090713	41321 (BR)	250	5	98%	名称	轮盖没装到位	
8							数量	4	
9	3	090714	41322 (BK)	60	2	97%	名称	漏折叠方法卡	
10							数量	2	

图1-24 不良记录统计表

这里涉及三条简单的公式，说明一下。

检查数：因为是实施全数检查，检查数=良品数+不良数。

=E5+F5



不良数：就是根据右边的每个项目进行累计不良。

```
=SUM(J6:S6)
```

如图1-25所示，切换到“公式”选项卡，再单击“自动求和”按钮，就自动出现SUM函数，只需用鼠标选取求和区域即可。




图1-25 自动求和

良品率：就是良品数占检查数的比例。

```
=E5/D5
```

这样得到的是0.97这样的数字，还需要设置单元格格式将其转换成百分比。操作很简单，只需单击%按钮即可，如图1-26所示。



图1-26 设置百分比格式

第一份表格模板终于制作完成，成就感不由得倍增。以后卢子只需将相关数据录入，然后发送邮件即可。





知识扩展

良品率其实也可以不用设置单元格格式，而是借助TEXT函数来实现，这个函数俗称万能函数，在以后的章节中会多次提到。

=TEXT(E5/D5,"0%")



1.2.2 餐费报销表

出差在外除了工作还得生活，工作上有表格，生活也得有表格。比如统计每天的餐费，然后汇总，出差回来可以报销。公私分明，消费多少就记录多少，不能多也不能少。多了是贪污，少了亏自己。

正常情况下都有单据，只要细心记录下来，统计出错的概率很低。Excel是个非常强大的计算器，比原始的计算器强一万倍。如图1-27所示，出差费用统计表是卢子个人使用，所以没有太多讲究，怎么做都可以，只要最终能够正确地统计金额即可。

	A	B	C	D	E
1	日期 ▼	类型 ▼	人数 ▼	人员 ▼	金额 ▼
2	2009-9-6	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	27
3	2009-9-6	午餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	70
4	2009-9-6	晚餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	48
5	2009-9-7	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	28
6	2009-9-7	午餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	78
7	2009-9-7	晚餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	71
8	2009-9-8	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	20
9	2009-9-8	午餐	3	郑春琴、宋晓媛、陈锡卢	50
10	2009-9-8	晚餐	3	郑春琴、宋晓媛、陈锡卢	50
11	汇总				442

图1-27 出差费用统计表

这份表格涉及一个很重要的功能：“表格”。“表格”可以实现自动隔行填充颜色，防止看错行，当然利用“表格”可以实现很多统计。其实操作起来很简单，选中单元格A1，切换到“插入”选项卡，单击“表格”按钮，弹出“创建表”对话框。默认情况下会智能地帮你选择好区域，单击“确定”按钮即可，如图1-28所示。

有了“表格”后，怎么统计呢？其实也同样简单，如图1-29所示，只需选中“汇总行”复选框即可。






图 1-28 插入表格



图 1-29 汇总行

原来可以汇总是因为自动帮我们添加了公式，如图1-30所示。

=SUBTOTAL(109,[金额])

“汇总行”的作用可不仅仅是求和而已，还包含各种各样的统计。单击其下拉按钮，可以看到平均值、最大值、最小值等计算功能，如图1-31所示。

E11	:	X	✓	fx	=SUBTOTAL(109,[金额])
1	A	B	C	D	E
1	日期	类型	人数	人员	金额
2	2009-9-6	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	27
3	2009-9-6	午餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	70
4	2009-9-6	晚餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	48
5	2009-9-7	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	28
6	2009-9-7	午餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	78
7	2009-9-7	晚餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	71
8	2009-9-8	早餐	4	杨晓凤、潘振美、陈焕锐、陈锡卢	20
9	2009-9-8	午餐	3	郑春琴、宋晓媛、陈锡卢	50
10	2009-9-8	晚餐	3	郑春琴、宋晓媛、陈锡卢	50
11	汇总				442
12					

图 1-30 自动添加公式

郑春琴、宋晓媛、陈锡卢	50
	442
	无
	平均值
	计数
	数值计数
	最大值
	最小值
	求和
	标准偏差
	方差
	其他函数...

图 1-31 各种统计

1.2.3 车费报销表

正常情况下出门都有专车接送，偶尔司机有事就会坐公交（大巴），这时就会产生一些小



费用。卢子也做了一个简单的模板来进行记录，如图1-32所示。

	A	B	C	D	E	F
1	车费报销明细表					
2						
3	制作日期：	2009-12-10	部门：	品质保证课	制表人：	陈锡卢
4						
5	日期	出发地	目的地	单程/往返	金额	备注
6	2009-12-2	公司	天河区科信局	单程	2	
7	2009-12-5	公司	平安大厦	往返	6	
8	2009-12-7	公司	市交运	往返	12	
9						
10						
11						
12						
13	合计				20	
14						

图1-32 车费报销明细表

制作日期使用函数，获取动态日期。

```
=TODAY()
```

合计也使用公式汇总金额。

```
=SUM(E6:E12)
```

后话：因为卢子每次都能够在及时地将工作信息反馈给领导，费用统计没有出过差错，领导对卢子还是挺满意的。

1.2.4 产品报价表

前面三份表格都是卢子使用自己的表格模板，这一天项目部的人突然要卢子帮忙设计一个产品报价表。

对于产品报价表，卢子以前并没接触过，跟项目部的人沟通后，制作出如图1-33所示的表格模板。

J17							
	A	B	C	D	E	F	G
1	A学校产品报价表						
2	序号	名称	品牌/型号	单位	数量	单价	金额
3	1	300W红外筒机	海康威视	个	34	1150	39100
4	2	200W红外半球	海康威视	个	9	1250	11250
5	3	摄像机电源	海康威视	个	43	55	2365
6	4	摄像机支架	海康威视	个	34	42	1428
7	5	防雷器	天盾	个	43	320	13760
8	6	设备箱	现代	个	43	80	3440
9	7	硬盘录像机	海康威视	台	2	4800	9600
10	8						
11	9						
12	10						
13		总计					80943
14							
15							

图1-33 产品报价表



这份表格其实挺简单的，唯一的难点就是要获取大写的金额，这种高难度的东西卢子哪里会啊？

有这么一句话：内事不决找百度，外事不决找谷歌。不会大写金额，那就上百度，搜索了一阵子，终于找到了一条公式：

```
= "合计 "&IF(I19<0,"无效数值",IF(I19=0,"",IF(I19<1,"",TEXT(INT(I19),"[dbnum2]")&
"元")&IF(INT(I19*10)-INT(I19)*10=0,IF(INT(I19)*(INT(I19*100)-INT(I19*10)*10)=0,"",
"零"),IF(AND((INT(I19)-INT(I19/10)*10)=0,INT(I19)>0),"零"&TEXT(INT(I19*10)-INT(I19)*10,"[dbnum2]")&
"角",TEXT(INT(I19*10)-INT(I19)*10,"[dbnum2]")&"角"))&IF((INT(I19*100)-INT(I19*10)*10)=0,
"整",TEXT(INT(I19*100)-INT(I19*10)*10,"[dbnum2]")&"分")))
```

这么长的公式，差点把卢子吓死，卢子强装镇定地看了下，这里除了涉及I19这个单元格，没有其他单元格，也就是把I19这个单元格换成实际需要的单元格即可。

借助快捷键Ctrl+H打开“查找和替换”对话框，输入查找内容“I19”，替换内容“G13”，单击“全部替换”按钮，在弹出的对话框中单击“确定”按钮，如图1-34所示。



图1-34 替换公式中的单元格

替换完成后，复制公式，选中B14单元格，在编辑栏中粘贴公式，按Enter键就得到了大写金额。

```
= "合计 "&IF(G13<0,"无效数值",IF(G13=0,"",IF(G13<1,"",TEXT(INT(G13),"[dbnum2]")&
"元")&IF(INT(G13*10)-INT(G13)*10=0,IF(INT(G13)*(INT(G13*100)-INT(G13*10)*10)=0,"",
"零"),IF(AND((INT(G13)-INT(G13/10)*10)=0,INT(G13)>0),"零"&TEXT(INT(G13*10)-INT(G13)*10,"[dbnum2]")&
"角",TEXT(INT(G13*10)-INT(G13)*10,"[dbnum2]")&"角"))&IF((INT(G13*100)-INT(G13*10)*10)=0,
"整",TEXT(INT(G13*100)-INT(G13*10)*10,"[dbnum2]")&"分")))
```

虽然公式不是卢子写的，但搞定这么长的一条公式还是很有成就感的。



1.2.5 获取Excel提供的精美表格模板

在使用Excel几年后，卢子发现原来默认就有提供一些表格模板，可以从中借鉴。新建的时候，可以看到很多模板，如图1-35所示。



图1-35 表格模板

表格模板太多，很难第一时间找到，这时就要借助搜索功能，比如搜索“费用报销”，就出现相应的表格模板，如图1-36所示。



图1-36 费用报销模板



打开费用报销模板，里面设计得很精美，就是颜色设置得比较浅，对于视力不好的人真是一个挑战，如图1-37所示。

A

B

C

D

E

F

G

H

I

J

1

公司名称

2

费用报表

3

4

员工：Kim Abercrombie

签发方：

5

部门：销售

6

周末：2013-3-7

日期：

7

里程运价率：¥6.70

8

9

星期五

星期六

星期日

星期一

星期二

星期三

星期四

10

交通

2013-3-1

2013-3-2

2013-3-3

2013-3-4

2013-3-5

2013-3-6

2013-3-7

总计

11

行驶里程

145

145

12

英里报销

¥971.50

¥0.00

¥0.00

¥0.00

¥0.00

¥0.00

¥0.00

¥971.50

13

停车收费

¥170.00

¥170.00

14

汽车租赁

¥790.00

¥790.00

¥1,580.00

15

出租车/轿车

¥0.00

16

其他（火车或公共汽车）

¥0.00

17

机票

¥2,350.00

¥2,350.00

18

总计

¥1,761.50

¥0.00

¥3,310.00

¥0.00

¥0.00

¥0.00

¥5,071.50

图1-37 精美的模板

仔细查看还能发现设置了不少公式，非常智能。如图1-38所示，是其中的一个公式，和用“表格”功能汇总的公式差不多。

118

:

✕

✓

f_x

=SUBTOTAL(109,I12:I17)

其中一个公式

	B	C	D	E	G	H	I	J	K
4	员工：Kim Abercrombie				签发方：				
5	部门：销售								
6	周末：2013-3-7				日期：				
7	里程运价率：¥6.70								
8									
9		星期五	星期六	星期日	星期一	星期二	星期三	星期四	
10	交通	2013-3-1	2013-3-2	2013-3-3	2013-3-4	2013-3-5	2013-3-6	2013-3-7	总计
11	行驶里程	145							145
12	英里报销	¥971.50	¥0.00	¥0.00	¥0.00	¥0.00	¥0.00	¥0.00	¥971.50
13	停车收费				¥170.00				¥170.00
14	汽车租赁	¥790.00			¥790.00				¥1,580.00
15	出租车/轿车								¥0.00
16	其他（火车或公共汽车）								¥0.00
17	机票				¥2,350.00				¥2,350.00
18	总计	¥1,761.50	¥0.00		¥3,310.00	¥0.00	¥0.00	¥0.00	¥5,071.50

图1-38 设置的公式

在你经验不足的情况下，模仿这些精美的模板是一个不错的选择。先模仿，后创新！

1.3 小结与练习

截止到目前，Excel对卢子而言跟Word大致没啥两样，只是做一些简单的数据处理。被别人夸奖几次，卢子就飘飘然，以为自己真的是高手。其实，这两年的学习，只是让卢子明白了一些菜单的使用，知道它的存在，而不知其精髓。就如函数而言，它的强大是卢子在以后的路上万万没有想到的。

1. 如图1-39所示，直接用公式下拉，当没有记录“良品数”和“不良数”时“检查数”会显示为0，“良品率”会显示错误值“#DIV/0!”，不太美观，该如何处理？

2. 如图1-40所示，怎么求出“良品数”的最大值和最小值？

	A	B	C	D	E	F
1	检查日期	番号	检查数	良品数	不良数	良品率
2	090712	41321(BR)	131	130	1	99%
3	090713	41321(BR)	252	250	2	99%
4			0			#DIV/0!
5			0			#DIV/0!
6			0			#DIV/0!
7			0			#DIV/0!
8			0			#DIV/0!
9						

图1-39 屏蔽错误值

	A	B	C	D	E	F
1	检查日期	良品数		如何获取最大最小值？		
2	090712	403		最大值	462	
3	090713	368				
4	090714	206		最小值	206	
5	090715	420				
6	090716	440				
7	090717	462				
8	090718	345				
9						

图1-40 获取最大值和最小值



第2章

高手变菜鸟



学习Excel两年，越来越觉得不过如此，Excel不就是加了边框的Word，会打字就会用Excel。当初卢子就差点被这个想法害惨，还好能够迷途知返，重新学习。后来悟出一个道理：不要整天跟水平比你差的人在一起，这样你永远得不到提高，要多读书，多跟比你厉害的人接触。

2.1 小插曲，转学Photoshop失败

那时候对Excel开始有了一点兴趣，公司内部又开始招聘统计员，待遇虽然比原来的职位差很多，但卢子还是有一点点冲动，准备转岗，因为做统计可以接触更多的Excel，就向老乡HR主管咨询。HR主管建议不要从事统计员，这个职位没有前途，平常有空余时间可以学点Photoshop，用途比较大；而Excel没什么用，不能当饭吃。

因为这个建议，卢子准备放弃学习Excel，转学Photoshop，到书店购买了一本《Photoshop从入门到精通》，开始学习。厚厚的一本书，当枕头刚好。翻看十几页，发觉并不难，还有模有样地跟着操作起来。也许是软件相通的缘故，卢子学习Photoshop进步很快，书里的很多知识卢子都能很快掌握。

但学了Photoshop以后，工作上用不上，平常又不喜欢拍照，要不然拿来处理照片也不错。兴趣是有了，但光有兴趣是不行的，学得快忘得也快。好不容易学习了半个月，居然把前面所学的技能忘得一干二净！

卢子做深思状，考虑当初是什么原因才学习Photoshop？最后发现当初好像就是一时冲动，并没有其他原因。既然这样，还是果断放弃。

题外话：如果当初杨阳的IT部落窝网络学院提前两年开办，卢子估计那时就有兴趣有动力学习Photoshop。毕竟杨阳是卢子的搭档，她又是学设计的，是讲课老师还是十大认证讲师。不过正因为学院晚开两年，卢子学习Excel才有了现在的成就。很喜欢IT部落窝网络学院里面的一句话：你可以拒绝学习，但你的竞争对手不会，如图2-1所示。



图2-1 IT部落窝网络学院宣传语



2.2 好奇害死猫，走上Excel不归路

凡事如果都按计划走，就没有了现在的卢子。忽然有一天，无聊地打开领导的表格，发现一件让卢子很惊讶的事情：在“品名”列单元格，出现了一条很长很长的公式，对于当时的卢子而言，无疑是超级公式，如图2-2所示。

C2

✕

✓

fx

=IF(B2="", "", VLOOKUP(B2, 'C:\Users\chenxilu\Desktop\[产品清单.xlsx]Sheet1'!\$B\$1:\$D\$11000, 2, 0))

A

B

C

D

E

F

G

H

I

J

1

日期

产品番号

品名

检查数

良品数

不良数

2

2009-1-26

98133

HKワイリッシュロイド

60

60

0

3

2009-1-27

98133

HKワイリッシュロイド

159

157

2

4

2009-2-22

98133

HKワイリッシュロイド

33

33

0

5

2009-3-26

93657

テーブルチェアα フットレスト付

710

702

8

图2-2 超级公式

这是什么鬼啊？跟天书一样。

更改了产品番号，品名居然自动更改，真的很神奇！跟领导打听，才知道这是日本总公司那边设置的公式。

有句老话叫“好奇害死猫”，卢子就是被这条公式害死了，本来以为自己是高手，结果门都还没进，完全是菜鸟一个。为了学习更多的Excel知识，从此走上Excel不归路。

卢子一下班就捣鼓IF函数，试了三天，貌似看懂一点点名堂。如图2-3所示，是IF函数的语法。




图2-3 IF函数语法

今天温度已超过35℃，如果没有风扇（当然条件好的话有空调更佳），晚上要睡觉纯属痴心妄想。用IF可以这么表示：

=IF(A1="风扇", "睡得着", "睡不着")

说明：文本需要加英文状态下的双引号，数字则不需要加双引号。

A1="风扇"这个条件只有两种结果，A1如果不是风扇就返回FALSE（假），如果是风扇就返回TRUE（真）。条件为FALSE（假）就返回第三参数“睡不着”，条件为TRUE（真）就返回第二参数“睡得着”，如图2-4所示。

	A	B	C	D	E	F	G	H
1	没有风扇		FALSE		睡不着			
2	风扇		TRUE		睡得着			
3								
4								
5			=A1="风扇"		=IF(A1="风扇","睡得着","睡不着")			
6								

图2-4 IF函数运算示意图

对于IF函数大概知道语法，可是对于VLOOKUP函数始终看不懂，参数好多，让人怎么能记住呢？愁死卢子了。

如果能学会这个函数，以后就省事了，可以不用每次都找领导输入日语品名。想到这里，卢子就有动力学习VLOOKUP函数。卢子这人有一个缺点，就是不善于沟通，有时宁愿多走点弯路，也不想一步到位。

几年以后才知道，原来当初走的弯路实际上却是捷径，如图2-5所示（来自微博）。

从百度搜索了“根据产品番号，自动生成品名”，终于找到了一条公式，如图2-6所示。

=VLOOKUP(D4,Sheet1A:B,2,0)




图2-5 最好的学习方法



图2-6 百度搜索





知识扩展

正常情况下查询是用下面的公式：

=VLOOKUP(A3,\$D\$3:\$E\$7,2,0)

这里的区域\$D\$3:\$E\$7怎么可以写成人海？

其实人海是一个事先定义好的名称，方便理解和简化公式。如图2-9所示，选择区域，在名称框中输入“人海”，按Enter键，就完成了定义名称的操作。

	A	B	C	D	E
1					
2	人员	性别		人员	性别
3	我是菜鸟	男		张三	男
4				笑着今朝	男
5				王姐	女
6				我是菜鸟	男
7				李四	女
8					

图2-9 定义名称

定义好名称后就可以调用了，可以直接写，也可以借助功能键F3粘贴名称，如图2-10所示。




图2-10 粘贴名称

2.3 为何你至今还是一个Excel菜鸟

这里引用卢子几年后写的一篇文章，很多人之所以成为菜鸟是有原因的，方法不对，即使



努力也不一定有结果。别再当井底之蛙了，走出去才有更广阔的天地。

为何你至今还是一个Excel菜鸟？

前几天看到我曾经教过的一个网友，在Excel交流群内，问了一个很基础的问题，心里挺替他难过。两年来丝毫没有进步，还停留在过去的水平。

这其实是一个很普遍的现象，大家有没有考虑过是什么原因导致的呢？

一个致命的原因：你喜欢跟比你水平差不多或者水平比你差的人交往。这样你就会认为自己的水平还不错，很多人都比你差，学不学习不重要，因此你只能是原地踏步。

你有没有想到，别人怎么进步那么快，才一两年就成为一个小高手？

如果想让自己变得更优秀，不妨参考下面几个方法。

◎ 多看一流高手的文章（帖子）

（1）技术论坛的版主水平都很高，你可以留意他用心整理总结的帖子。就如下面两篇帖子：

① 听故事，学Excel原创连载（适合初学者学习）。

② Excel各种问题汇聚（函数版）（适合有一定水平的人学习）。

（2）新浪微博的加V大号（@ExcelHome、@Excel之恋），他们很多都是行业的精英或者是机构，有空留心他们发的微博。

（3）微信的公众号，跟微博有点像，关注后都会经常看到推出的一些优秀文章。

以上三种都是一流高手，通过他们的文章（帖子），你可以学到很多知识，但跟他们交流会比较困难，一流高手一般都会比较忙。

◎ 多跟二流高手交流

这种人潜伏在各大论坛、各交流群内，经常替人解答疑问。因为他们的时间比较充裕，所以给了我们一个很好的机会，随时都可以找他们交流，这样一来就可以发现自己本身的局限性，从而根据不足想出改善对策。

◎ 多读书（阅读积累）

通过上面两种方式，你可以学到很多知识，但这些知识会比较零散。而书籍一般是高手好多年的心血总结而来，会比较集中全面。有的书籍还会提供一些学习方法，通过这些你就可以少走很多弯路。



知识扩展

最近看到阿何（清华高材生）写的一篇文章不错，推荐一读。



如何迅速提高自己的技能水平？

如何迅速提高自己的技能水平？我是采用下面的方法去做的。

1. 找到最好的学习资料

通过豆瓣、博客、知乎、微博、百度等网站，寻找跟技能相关的好教程、书籍、视频，能下载的下下载来，能买的全部买（放心，花不了多少钱）。

还有一个很好的方法：找到这个领域顶尖的那些人，看他们写的一篇一篇文章，把他们推荐的书都买回来。

买回来之后，怎么使用这些资料？

一开始千万不要一本一本地看。

在你对一个领域缺乏整体认知之前，就按照传统读书的方法，一本一本地看，一页一页地看，是非常危险的。你会一下子掉进无边的细节里面，感觉自己进步超级慢，最终导致放弃。

正确的方法是，找到最经典的那本教程，用最快的速度翻完它。然后用思维导图工具，把这个领域的知识结构先画出来，建立整体知识框架。

之后，才是细致化的学习过程。也就是把知识框架里每一个原来只是模糊认知的点，进行深入系统的学习。

2. 短期高强度连续学习

我认为，如果时间允许的话，短期集中式的高强度学习，是快速提升技能水平的最有效方法。

举个我本人的例子吧。

第二次创业的时候，公司一个设计师突然辞职了，留下一大堆项目，正好那段时间招聘人又特别困难，我只能自己顶上。

在这之前，我从来没有接触过Photoshop软件，也不了解平面设计。怎么办呢？先通过步骤1，找到大量的学习资源。然后就进入了疯狂学习操练的节奏。

一个月时间里，除了必要的工作外，我推掉了一切事情，放弃了一切娱乐，死磕Photoshop。

先从基本的功能操作开始，对着软件菜单一个一个点过去看效果，一边记笔记。结果只用了3天，就熟悉了大部分简单功能的使用。然后对着一份视频教程，对着300多个效果图，一个一个地照着教程步骤亲手操作。

这个过程持续了大半个月，结束之后，随意看到一张平面设计图，我大概都知道应该怎样做出这个效果。

再后面，是技巧整合。用学到的所有方法，去做出一份完整的作品。我临摹了几十个网站的主要页面，因为有了前面扎实的基础，这个过程反而很快。



再后面，就是慢慢提升的过程了，属于水磨功夫。

如果你想快速从门槛外进入门槛内的话，最好的方式就是像我这样，对自己狠一点，短时间高强度学习和操练。

3. 向高手请教

任何一项技能，有很多细节、注意点，都是书里、教程里学不到的。遇到问题的时候，你琢磨好几天可能都解决不了，但有经验的人的一句话可能就帮你搞定了。

从入门到高手，很重要的一点是找到自己的老师。可以在现实生活中找，也可以从网上找。

问题在于，高手凭啥帮你，指导你呢？

两种情况：一种是你能提供高手没有的资源，而正好是他需要的，两人进行交换；另一种是付费学习。

很多人舍不得花这个钱。这样的人，基本也不会有什么大的出息。一个人连投资自己都不舍得，小钱不花，还指望挣大钱？

我统计了一下。自己一年买书的钱最少几千块上万块（很多是电子书），至于报名各类付费课程、拜师的费用，就更多了。

但是通过这些投入，技能提升而带来的回报远不止十倍。

2.4 小结与练习

很多自认为高手的人，其实往往都是菜鸟。之所以会产生这个误区，是因为他身边都是一群菜鸟。别做井底之蛙，走出来。向比你强的人学习，记住一个公式：你经常接触的五个人的平均水平就是你的水平。如果想让自己变得更强大，只有不断接触比你更优秀的人，除此别无他法。

1. 如图2-11所示，每一款产品都有一个订单数，出了几次货。如果订单数跟出货数一



样，就是出货OK，否则就是出货NG。

	A	B	C	D	E	F	G	H
1	产品	订单数	第1回 出货	第2回 出货	第3回 出货	第4回 出货	第5回 出货	产品 出货情况
2	R102	3000	700	800	500	300	700	OK
3	H126	1000	200	300	100	50		NG
4	H276	500	300	100	20	80		OK
5								

图2-11 如何判断产品的出货情况

2. 如图2-12所示，如何根据合并单元格的产品，填充成每个单元格都有内容？

	A	B	C
1	效果图	产品	不良内容
2	R102	R102	内外箱印刷不良、破损、脏污
3	R102		内外箱（产品）漏贴条形码、日期印
4	R102		管件划伤、凹痕、坑洼
5	R102		防滑垫划伤、倾斜、裂
6	R102		布地破损、开线
7	R102		带子破损、开线
8	R102		涂装不良
9	R102		脏污、油污，胶水残留
10	H126	H126	管件划伤、凹痕、坑洼
11	H126		防滑垫划伤、倾斜、裂
12	H126		布地破损、开线
13	H126		带子破损、开线
14	H126		涂装不良
15	H126		脏污、油污，胶水残留

图2-12 填充合并单元格内容



第3章

改变



初步见识到函数与公式的强大功能后，卢子开始慢慢改变自己。一方面完善表格模板，另一方面请教同事。沟通是很重要的，有的事对你而言很难，但对于某些人而言就是不值一提的小事。勇于改变自己，相信不久的将来，一定能更好地掌握Excel。

3.1 勇敢踏出第一步

卢子虽然学了几个函数，但工作依旧没有太大的变化。每次制作“出货检查不良报告”的时候，总是无意识地对“品名”列多看几眼。来日企两年了，到现在卢子都没学会日语输入，想想也挺可笑的。报告每次做完99%，就剩品名让领导输入，时间久了领导也就习以为常。不过卢子却在想解决之策，怎么将VLOOKUP函数运用到这里，如图3-1所示。

	A	B	C	D	E	F	G	H	I
1	联络：	隆成（新厂）/技术课						2009年3月14日	
2			发行号：					Richell品质保证课	
3		供应商：	隆成（新厂）					陈锡卢	
4			出货检查不良报告						
5			以下不良内容请技术课协助解决						
6							出货日期：	090314	
7							检查日期：		
9		产品番号	品 名	出货数	检查数	良品数	不良数	Lot.No.	
10		40061		220	231	220	11	090311	
11		4006		534	535	534	1	090312	

图3-1 空着的日语品名

时间一天天地过去了，但事情依然没进展。现在难点就是找到一份产品清单，里面列举所有产品返回对应的品名。卢子向领导打听，部门内是否有我们公司所有产品的清单？领导的答复让卢子的心情一下子陷入低谷，曾经试图找这一份清单，翻遍了各台电脑依旧没有发现，后来得知只有日本总公司才有。没有清单，会用VLOOKUP函数也没用，此事就告一段落。

忽然有一天，卢子收到日本的Excel文档成绩书，里面有十多款产品。卢子看到这些成绩书，突然脑袋里闪出这样的念头：自己制作产品清单。当然这个清单只是针对隆成公司这边而已，全部供应商的产品清单想都不用想，几万款产品。

有想法就得尝试，卢子将隆成公司的所有产品番号一一罗列出来，将成绩书上有的品名复



制、粘贴到产品清单里，这样就完成了一小半。还好卢子想到了自己的老乡会写日语，就麻烦她将剩余的品名输入进去。写日语对我而言很难，但对于她而言就是不值一提的小事。经过半天的时间，如图3-2所示，终于完成了产品清单，太难能可贵了。

F21			
产品清单			
	A	B	C
1	编号	番号	品名
2	1	40401	ファインアールカルネオアッパー (BK)
3	2	40402	ファインアールカルネオアッパー (BR)
4	3	40403	ファインアールカルネオアッパー (R)
5	4	40404	ファインアールカルネオアッパー (O)
6	5	40405	ファインアールカルネオアッパー (PU)
7	6	40406	ファインアールカルネオアッパー (GR)
8	7	40411	マルシェ (P)
9	8	40412	マルシェ (B)
10	9	40571	ワイリッシュロイド (BK)
11	10	40572	ワイリッシュロイド (BE)
12	11	40981	リベラシェスタ (BR)
13	12	40983	リベラシェスタ (R)
14	13	43010	テーブルチェアー
15	14	48896	アルキュートMC (BR)
16	15	48897	アルキュートMC (B)
17	16	48906	アルキュートLT-R (BR)
18	17	48907	アルキュートLT-R (B)

图3-2 产品清单

不会日语的人真伤不起！

有了这份清单，设置品名查找公式就变得轻而易举。根据以前的表格，依样画葫芦，如图3-3所示。

C10			
=IFERROR(VLOOKUP(B10,产品清单!B:C,2,0),**)			
	A	B	C
1	联络：隆成（新厂）/技术课		
2	发行号：		
3	供应商：隆成（新厂）限		
4	出货检查不良报告		
5	以下不良内容请技术课协助解决		
6			
7			
9	产品番号	品 名	出货数 检查数 良品数 不良数
10	40061	ファインアールカルネオモキュア (BK)	220 231 220 11
11	4006		534 535 534 1

图3-3 设置查询公式

其实这里直接用VLOOKUP函数也可以，之所以嵌套IF函数是为了在单元格没输入产品番号的时候不显示错误值#N/A，如图3-4所示。

C10			
=VLOOKUP(B10,产品清单!B:C,2,0)			
	A	B	C
1	联络：隆成（新厂）/技术课		
2	发行号：		
3	供应商：隆成（新厂）限		
4	出货检查不良报告		
5	以下不良内容请技术课协助解决		
6			
7			
9	产品番号	品 名	出货数 检查数 良品数 不良数
10	40061	ファインアールカルネオモキュア (BK)	220 231 220 11
11		#N/A	534 535 534 1

图3-4 没嵌套IF函数的效果

当然错误值#N/A在这里本来是没多大影响的，只是看起来不太美观而已。

后来有一天学到了一个新函数IFERROR，有人也许会问，这个是干什么用的，跟前面又有何关系？

函数语法：

=IFERROR(表达式 如果第一参数的表达式返回错误值则返回第二个参数的值)

就是让错误值显示成你想显示的任意值，不是显示错误值本身，如图3-5所示。




图3-5 IFERROR函数的用法

最终，卢子将公式改成：

=IFERROR(VLOOKUP(B10,产品清单!B:C,2,0),"")

虽然在这里进行容错处理意义不大，但起码能将所学在第一时间用上。另一种解释就是显示错误值不好看，显示成空白好看点。

在不良报告设置公式，虽然对卢子而言没什么，但对领导而言是一种解脱，让领导省去无数次输入品名的麻烦。后来这个模板在公司内部悄悄地流行起来，卢子也因此受到领导小小的表扬。

知识扩展



Excel 2013版本以上提供了一个专门针对错误值#N/A的函数IFNA，语法和IFERROR函数一样，也就是说可以将公式改成：

=IFNA(VLOOKUP(B10,产品清单!B:C,2,0),"")



说到错误值#N/A，顺便介绍一下Excel中几种常见的错误值，有些你可能见过，有些可能你还没有见过，并结合实例进行简单的说明。

NO.1: “#####” 错误

当单元格由于不够宽显示，比如“日期”列宽度不够，只要调到相应的宽度即可，如图3-6所示。

	A	B	C	D
1	日期			日期
2	#####			2010-1-12
3	#####			2010-1-13
4	#####			2010-1-14
5	#####			2010-1-15
6	#####			2010-1-16
7	#####			2010-1-17
8	#####			2010-1-18
9	#####			2010-1-19
10	#####			2010-1-20
11	#####			2010-1-21
12				

图3-6 “#####” 错误

NO.2: “#DIV/0!” 错误

当一个数除以零(0)的时候或者不包含任何值的单元格时，Excel将显示此错误，如图3-7所示。

	A	B	C	D
1	产品	检查数	良品数	良品率
2	A			#DIV/0!
3	B	2	1	50%
4	C	0	0	#DIV/0!
5				

图3-7 “#DIV/0!” 错误

用IF函数进行一个判断即可。

```
=IF(B2=0,"",C2/B2)
```

NO.3: “#NAME?” 错误

当Excel无法识别公式中的文本时将显示此错误，由于在公式中，字符串没有

添加英文双引号或函数名称拼写错误，如图3-8所示。

错误	使用公式	说明
#NAME?	=IF(A1=风扇,睡得着,睡不着)	文本没有加英文状态双引号
#NAME?	=VLOOKUP(A1,M:N,2,0)	少一个O，正确的是VLOOKUP

图3-8 “#NAME?” 错误

NO.4: “#NULL!” 错误

当指定两个不相交区域的交集时(交集运算符是分隔公式中引用的空格字符)而导致的错误值，它其实是一种值的返回结果，例如，区域A1:B2和C6:D7不相交，因此，输入公式=SUM(A1:B2 C6:D7)将返回#NULL!错误，如图3-9所示。

	A	B	C	D	E
1	8	6	2	1	
2	5	5	5	3	
3	6	7	5	5	
4	8	4	6	3	
5	10	7	3	4	
6	4	6	8	9	
7	4	3	10	9	
8					
9					
10					
11			#NULL!		
12					

图3-9 “#NULL!” 错误

NO.5: “#NUM!” 错误

当公式或函数包含无效数值时，如求负数的平方根，导致出错，如图3-10所示。

	A	B	C	D
1				
2				
3				
4			#NUM!	
5				

图3-10 “#NUM!” 错误

NO.6: “#REF!” 错误

引用的区域被删除后，如良品数的区域

被删除，导致出错，如图3-11所示。

NO.7: “#VALUE!” 错误

单元格内含有文本，运算导致出错，如图3-12所示。

H2		:				= #REF! + G2		
	A	B	C	D	E	F	G	H
1	产品	良品数	不良数	检查数		产品	不良数	检查数
2	A	12	1	13		A	 1	#REF!
3	B	2	1	3		B	1	#REF!
4	C	3	0	3		C	0	#REF!
5								

图3-11 “#REF!” 错误





D2		:				=B2+C2
	A	B	C	D		
1	产品	良品数	不良数	检查数		
2	A	12	零 	#VALUE!		
3	B	2	1	3		
4	C	3	0	3		
5						

图3-12 “#VALUE!” 错误

3.2 原来统计并非手工活

卢子虽然学了几个函数，但依然还是菜鸟。其实，严格来说只能算路上菜，也就是走在通往菜鸟的路上。

Richell公司每年都组织十个人左右去日本学习，郭小姐也申请去日本。她一去日本，卢子就得顶替她的工作。卢子这人最不擅长沟通，所以对经常需要跟供应商沟通的组长一职并不感兴趣，正所谓无官一身轻。卢子本想推脱，但这一职位卢子是最合适的人选，除了卢子没人熟悉中山那边的情况。

既来之，则安之。还是用点心在交接的事儿上面才是正道，要不然郭小姐一走，卢子就只有哭的份儿。

常用表格交接

只见郭小姐熟练地打开《隆成每天不良记录表》，这是中山隆成的不良记录表，每天都要记录下每一款产品的不良数量，有颜色填充部分是事先设置好的公式，会自动帮你统计，如图3-13所示。





知识扩展

文中郭小姐向卢子介绍用COUNTA统计数字次数，其实是错误的，这个是统计非空单元格的个数。统计数字用COUNT函数，这是很久以后卢子才发现的问题。

卢子跟着试操作一遍，基本就记住了。以前一直用计算器，统计数据非常慢，有了这两个函数，统计又上了一个台阶。

郭小姐看卢子明白了，就打开了第二张表《隆成月度统计表》，如图3-17所示。这里面有填充色的已经设置好公式，还有最小、最大、合计、平均这些都设置好公式，你只要填入一些基础数据就行。

52期	4月	5月	6月	7月	8月	9月	10月	11月	12月	1月	2月	3月	最小数	最大数	合计	平均
检查回数	11	8	4	11	11	2	7	10	15	5	4	0	0	15	88	7.3
总人数	45	31	18	38	45	8	22	32	49	18	12	0	0	49	318	26.5
实绩时间	79	48	24	63	74	11	41	72	125	37	26	0	0	125	600	50
平均每回人数(人)	4	4	5	3	4	4	3	3	3	4	3	0	#####	#####	#####	4
平均每回实绩时间(H)	7.2	6.0	6.0	5.7	6.5	2.2	3.2	3.2	8.3	7.4	6.5	0	#####	#####	#####	6.8
检查批数	25	11	10	17	20	10	10	10	26	7	6	0	0	26	143	11.9
不合格批数	25	9	10	17	20	10	10	10	26	7	6	0	0	26	141	12
合格批数	0	2	0	0	0	0	0	0	0	0	0	0	0	2	4	0.16567
批合格率	0.0%	18.2%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	18.2%	18.2%	3%
累计检查批数	25	36	46	63	83	85	94	104	130	137	143	143	——	——	——	——
累计不合格批	25	34	44	61	81	83	92	102	128	135	141	141	——	——	——	——
累计合格批数	0	2	2	2	2	2	2	2	2	2	2	2	——	——	——	——
累积批合格率	0.0%	5.6%	4.3%	3.2%	2.4%	2.4%	2.1%	1.9%	1.5%	1.5%	1.4%	1.4%	——	——	——	——
出货数(pcs)	8,025	2,696	1,340	3,287	4,650	278	389	900	4,381	2,255	1,121	0	0	8,025	29,322	2443.5
检查数(pcs)	8,235	2,244	1,371	3,364	4,697	278	447	1,783	4,476	2,298	1,238	0	0	8235	30,431	2535.92
良品率	97%	120%	98%	98%	99%	100%	87%	50%	98%	98%	91%	#####	#####	#####	——	96%
出货检工程异常(件)	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
受入检查工程异常(件)	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

图3-17 月度统计表

卢子粗略看了下，也没问什么。

郭小姐就继续说了一些工作注意事项、生活注意事项，说了好久。不过卢子基本上也没记住她说什么，就如读书的时候，左耳进，右耳出，管老师说得天花乱坠。

不过对这两张表格，卢子却很感兴趣，利用下班的时间，又反复看了几次。无意间单击“自动求和”下拉按钮，如图3-18所示，居然还有这么多有关的函数。估计很多人都跟卢子一样，看到这些无限感慨，以为这里

就只有SUM函数而已。




图3-18 各种函数

卢子尝试单击了最大值，默认情况下区



域会出现错误，自己重新用鼠标选择正确的区域，如图3-19所示。

输入公式以后，按Enter键就得到了最大销售额，用同样的方法获取了最小销售额，如图3-20所示。

	A	B	C	D	E	F	G
1	人员	销售额					
2	张三	500					
3	李四	400					
4	王五	700					
5							
6							

	A	B	C	D	E	F
1	人员	销售额				
2	张三	500		最大销售额	700	=MAX(B2:B4)
3	李四	400		最小销售额	400	=MIN(B2:B4)
4	王五	700				
5						

图3-19 更改区域

	A	B	C	D	E	F
1	人员	销售额				
2	张三	500		最大销售额	700	=MAX(B2:B4)
3	李四	400		最小销售额	400	=MIN(B2:B4)
4	王五	700				
5						

图3-20 获取最大最小销售额

其实这一系列函数的语法都一样，学会了一个其他也就会了，函数其实也没那么难学。

3.3 小结与练习

菜鸟并不可怕，可怕的是你甘心做菜鸟。认识自己的不足，勇于改变自己。自己能学会自然最好，但别忽视了身边的资源，适当的时候请教同事也不失为一种好方法，可以让你学得更快。

1. 如图3-21所示，如何根据姓名简称获取性别？

	A	B	C	D	E	F
1	姓名	性别		姓名	性别	
2	笑着今朝	男		今朝	男	
3	李云龙	男		老乡	女	
4	微博老乡	女				
5	无言的人	男				
6	漂亮MM	女				
7						
8						

图3-21 根据姓名简称获取性别

2. 如图3-22所示，如何根据右边的成绩判定表获取等级？

	A	B	C	D	E	F	G
1	成绩	等级			区间	等级	
2	50	不及格		0	0-59	不及格	
3	80	良好		60	60-70	及格	
4	90	优秀		71	71-85	良好	
5	67	及格		86	86-100	优秀	
6							
7							
8							

成绩判定表

图3-22 根据成绩判断等级

3. 如图3-23所示，分别统计数字个数、非空单元格个数、空单元格个数。

	A	B	C	D	E
1	字符串				
2	今朝				
3		今朝： 含有空格			
4	12		数字个数	3	
5	713		非空单元格个数	5	
6			空单元格个数	1	
7	34				
8					

统计结果

图3-23 统计各种单元格个数



第4章

全面扫盲



基础知识很枯燥，也很重要。万丈高楼平地起，要想直接盖顶层而没有地基肯定行不通。基础不牢以后是要吃苦头的，请用心学好。基础知识需要怎么学呢？一般我都是先粗略地看全部内容，然后从中挑选感兴趣的内容，再细读。对基础知识有一个初步的了解后，就暂时先放一边，等实际遇到相关问题时，再打开来看这部分，无限循环。在不知不觉中就记牢这些知识。切忌贪多，一次性无法记住全部基础知识，除非你的记忆力超级强。

4.1 学习前的准备

“实用为王”才是学习的目的，毕竟我们学习技术为的就是要解决问题。不要一味贪多，现在用不上的函数就不要学习。到现在我对工程函数、财务函数还一无所知，但这并不妨碍我解决实际问题，等到需要时再学习也不迟。

“助人为乐”各人的思维习惯与角度不尽相同，导致水平有高低，思维有局限。帮助别人时候就是帮助自己。

如图4-1所示，请牢记三大快捷键，当你熟练掌握后，对你会有莫大的好处！




图4-1 三大快捷键

4.1.1 基本概念

什么是公式？

有人也许会说，公式有什么好学的，从小学开始我们就一直在接触“公式”，如圆的面积、勾股定理等公式，如图4-2所示。但这是数学公式，跟Excel的公式是不同的。Excel中的公式是以“=”号开始，通过使用运算符将数据、函数等元素按一定顺序连接在一起，从而实现对工作表的数值执行计算的等式。说白了，只要以“=”号开头有计算功能的都是公式。






图4-2 数学公式

什么是函数？

简单来说，函数就是预先定义好的公

式。如图4-3所示，如对E2:E10区域的数据进行求和，利用SUM函数可以轻松汇总，但用常规公式很麻烦，还容易出错。

	A	B	C	D	E	F
1	名称	单位	数量	单价	金额	
2	三位大板开关	个	2	15	30	
3	二位大板开关	个	1	12	12	
4	手套	对	12	2.5	30	
5	电工胶布	卷	20	1.5	30	
6	螺丝	盒	1	15	15	
7	斯耐德 1P/20A	个	7	35	245	
8	20A漏电	套	4	120	480	
9	2P/63A漏电	套	1	438	438	
10	填缝剂	盒	2	24	48	
11				合计	1328	=SUM(E2:E10)
12					1328	=E2+E3+E4+E5+E6+E7+E8+E9+E10
13						

图4-3 函数与常规公式的比较

使用函数除了可以简化公式，同时也可以实现公式无法完成的功能。如图4-4所示，查询人员属于哪部影片？

=VLOOKUP(D2,A:B,2,0)

	A	B	C	D	E
1	人员	影片		人员	影片
2	曹雪芹	红楼梦		宋江	水浒传
3	高鹗	红楼梦			
4	宋江	水浒传			
5	施耐庵	水浒传			
6	李逵	水浒传			
7	林冲	水浒传			
8	燕青	水浒传			

图4-4 查询对应值

知识扩展

关于“=”号的位置小故事

很久以前，卢子在刚学Excel的时候，想要用它来进行计算。在单元格中输入“1+2”没反应，输入“1+2=”依然没反应，不禁叹息，原来Excel是不能计算的。如图4-5所示，直到有一天，无意间单击“自动求和”按钮，出现“=SUM(C2:C6)”，才恍然大悟，原来是将“=”放在最前面，输入“=20+30”，立马出现50，输入“=2*9”，立马出现18。老外的想法就跟我们中国人不同，就如称呼Mr.Chen，完全颠倒思维。得出一条理论，要学老外的东西就得按老外的想法去做，否则会出错。



图4-5 无反应的输入

说明：允许不输入“=”，而用“+”符号替代。

你也敢跟计算器比速度

有很多人喜欢用计算器计算数据，老是说电脑录入数据速度比不上计算器。真的吗？NO！因为“=”号的位置跟小键盘隔得远，第一次录入=1……这一步比计算器慢。但其实可以不用按“=”号，而利用“+”

号取代“=”，效果也是一样的。如输入“+1+1”，按Enter键就会得到2，查看公式单元格的公式变成“=1+1”。如图4-6所示，下回遇到有人说，他计算器按得快，你也有胆量跟他比速度。

说明：可以用“+”取代“=”，但不能用“-”取代“=”，切记！




图4-6 用“+”取代“=”

4.1.2 初识函数

函数可以看成是一个黑箱，我们只需关心它的输入与输出，对里面的具体实现无须关心。这些黑箱能对输入进行某种处理，最后输出结果。这些输入就叫参数，各参数之间用英文状态下的逗号“,”来划分；输出就是函数返回值，而黑箱可以用一对英文状态下的括号“(,,)”来表示，它比较形象，是将参数包裹起来。人有姓，树有名，为了对这些黑箱起个标识作用，就在括号之前贴上了“标签”，这个标签就是函数名。函数名除用于标识外还起到一定的功能描述作用，如图4-7所示。




图4-7 函数理解示意图



每个函数都有自己的名称，如果你叫错名称的话就会出现“#NAME?”。

如函数名称少了一个O，如图4-8所示。




图4-8 出错提示

=VLOOKUP(D2,A:B,2,0)

正确写法：

=VLOOKUP(D2,A:B,2,0)

4.1.3 快速、准确地输入函数的两种方法

函数名输入错误，一般发生在2003版，高版本输入错误的可能性很小。怎么快速、准确地输入函数呢？

方法一：插入函数

如图4-9所示，单击“公式”选项卡中的“插入函数”按钮，弹出“插入函数”对话框。在“或选择类别”下拉列表框中选择“查找与引用”选项，并选择VLOOKUP函数，单击“确定”按钮。




图4-9 “插入函数”对话框

这种方法适用于前期刚学习的阶段，当对函数熟悉后就得学会用第二种方法。

方法二：公式记忆式键入

不要担心Excel的记忆力，它远比你想象的还要好。只要输入前一两个字母，就会出现跟这些字母有关的所有函数。如图4-10所示，在单元格中输入“=v”，函数不区分大小写。选择VLOOKUP，双击即可。




图4-10 公式记忆式键入



知识扩展

如果有一天，你在工作表中输入前几个字母后没有任何提示的话，不要怀疑Excel老了，记忆力减退，而是你的工作表被别人动了手脚，重新更改过来即可。

说明：2003版本并不存在这个功能，所以手工输入的时候经常出错。

选择“文件”选项卡中的“选项”命令，打开“Excel选项”对话框。选择“公式”选项，选中“公式记忆式键入”复选框，单击“确定”按钮，如图4-11所示。




图4-11 设置公式记忆式键入

设置好后，又可以重新提示。这个功能很好用，我经常用，感觉很方便。特别是出现函数嵌套的时候，你用插入函数很难完成，而用“公式记忆式键入”就很轻松。

4.1.4 编辑公式三招

A列数字的总计减去C列数字的总计，统计结果，如图4-12所示。

```
=SUM(A1:A10)-SUM(C1:C10)
```



	A	B	C	D	E
1	1		2		
2	2		2		
3	3		2		
4	4		2		
5	5	减去	3	结果	29
6	6		3		
7	7		3		
8	8		3		
9	9		3		
10	10		3		

图4-12 两列数字差异的和

不要以为这些都是手工书写的，这样会很烦琐。懒人原则，借助软件内置功能，帮你输入完整正确的公式。

第一招 鼠标选取

输入“=SUM(”，然后用鼠标选取A1:A10单元格区域，如图4-13所示。

图4-13 鼠标选取单元格区域

第二招 复制相同内容

当输入到“=SUM(A1:A10)-”时，停下你的手，不要再次输入“SUM(”，而是直接复制SUM(A1:A10)，然后用鼠标更改区域，如图4-14所示。

=SUM(A1:A10)-SUM(A1:A10)

图4-14 复制相同内容

你也许对这个不以为然，“我输入这些内容都比复制要快”，真的如此吗？

那假如是下面的情况，又该如何？难不成你真的不用复制？

```
=SUMIF($A$1:$A$10,">5")-SUMIF($A$1:$A$10,">9")
=SUMIF($A$1:$A$10,">5")/COUNTIF($A$1:$A$10,">5")
```

第三招 在“=”号前面录入空格

写公式能一气呵成，那是最理想的了。但有时出于某些原因无法一次性完成。如写到一半时，突然朋友找你有点事，而你已经录入了一大半公式，直接单击“√”会出现错误，单击“×”公式又前功尽弃，如图4-15所示。




图4-15 错误提示

在左右为难之时，你肯定会怪朋友，早不来，晚不来，偏偏这个时候来。其实只要在“=”号前面录入一个空格，就可以搞定这个问题。现在的公式变成了文本，直接按Enter键就可以。等到事情忙完后，重新去掉空格，编辑公式即可，如图4-16所示。




图4-16 录入空格

技巧虽小，但挺实用。

4.1.5 快速填充公式小妙招

日常工作中，我们经常要大批量地填入公式，如何能更高效地完成公式的填充就显得尤为重要。如图4-17所示，最常见的办法就是输入第一个公式，将鼠标放在单元格右下角，出现“+”字箭头的时候，向下拖动公式。

	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	2	15	30
3	二位大板 开关	个	1	12	12
4	手套	对	12	2.5	30
5	电工胶布	卷	20	1.5	30
6	螺丝	盒	1	15	15
7	斯耐德 1P/20A	个	7	35	245
8	20A漏电	套	4	120	480
9	2P/63A漏电	套	1	438	438
10	填缝剂	盒	4	12	48
11					
12					

图4-17 下拉填充公式

如果涉及的单元格比较多，下拉公式很浪费时间。曾经有一个网友因为有1万多行内容，用下拉方法足足用了几分钟才完成，还拉得手酸。其实还有一种比较高效率的做法，就是输入第一个公式，将鼠标放在单元格右下角，出现“+”字箭头的时候，双击填充公式，如图4-18所示。

	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	2	15	30
3	二位大板 开关	个	1	12	12
4	手套	对	12	2.5	30
5	电工胶布	卷	20	1.5	30
6	螺丝	盒	1	15	15
7	斯耐德 1P/20A	个	7	35	245
8	20A漏电	套	4	120	480
9	2P/63A漏电	套	1	438	438
10	填缝剂	盒	2	24	48
11					

图4-18 双击填充公式





知识扩展

公式计算功能很方便，相信很多初学者会预设公式，如图4-19所示，这样方便数据输入后能够自动计算。由于公式所引用的都是空单元格，所以计算结果为0，因此不够美观。不仅如此，预设公式还占用资源，若要在新输入数据的行中自动填充上一行的公式，该如何实现呢？

E11	:	X	✓	fx	=C11*D11
	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	2	15	30
3	二位大板 开关	个	1	12	12
4	手套	对	12	2.5	30
5	电工胶布	卷	20	1.5	30
6	螺丝	盒	1	15	15
7	斯耐德 1P/20A	个	7	35	245
8	20A漏电	套	4	120	480
9	2P/63A漏电	套	1	438	438
10	填缝剂	盒	2	24	48
11					0
12					0
13					

图4-19 预设公式

正常默认情况下，公式超过4行只要重新填写数据，公式都会自动扩展。假如不会，可以进行设置。如图4-20所示，在“文件”选项卡中选择“选项”命令，在打开的“Excel选项”对话框中选择“高级”选项，在“编辑选项”区域中选中“扩展数据区域格式及公式”复选框，单击“确定”按钮。




图4-20 “Excel选项”对话框

正常情况下我都是采用“表格”实现公式自动扩展。如图4-21所示，将鼠标指针放在区域中任意单元格，如A1单元格，单击“插入”选项卡中的“表格”按钮，默认情况下会帮你自动选取区域，保持默认设置不变，单击“确定”按钮。

设置完后，只要增加新内容，就会自动填充公式，如图4-22所示。




图4-21 插入“表格”

	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	2	15	30
3	二位大板 开关	个	1	12	12
4	手套	对	12	2.5	30
5	电工胶布	卷	20	1.5	30
6	螺丝	盒	1	15	15
7	斯耐德 1P/20A	个	7	35	245
8	20A漏电	套	4	120	480
9	2P/63A漏电	套	1	438	438
10	填缝剂	盒	2	24	48
11	Excel效率手册	本	1	35	35
12					

图4-22 自动填充公式

4.1.6 输入的公式为什么不参与计算

有时明明是设置好了公式，公式也没发现问题，但就是不参与计算，如图4-23所示，这是怎么回事呢？

	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	2	15	=C2*D2
3	二位大板 开关	个	1	12	
4	手套	对	12	2.5	
5	电工胶布	卷	20	1.5	
6	螺丝	盒	1	15	
7	斯耐德 1P/20A	个	7	35	
8	20A漏电	套	4	120	
9	2P/63A漏电	套	1	438	
10	填缝剂	盒	2	24	
11					

图4-23 公式不参与计算

这种情况其实是公式这一列单元格被设置为“文本”格式，将其重新设置成“常规”格式，再输入公式即可，如图4-24所示。






图4-24 更改单元格格式



知识扩展

有时单元格显示公式而非计算结果，并不是因为单元格设置的问题，而是不小心单击了“显示公式”按钮，如图4-25所示。针对这种情况，直接再单击一次“显示公式”按钮就可以恢复正常。




图4-25 显示公式

对于既想提供计算结果，又想显示公式的情况，Excel 2013以上版本提供了一个新函数FORMULATEXT，借助它作为辅助列就可以获取公式，如图4-26所示。




图4-26 FORMULATEXT函数的运用

还有一种情况更奇葩，明明单元格格式是常规的，公式也没问题，就是计算错误，如图4-27所示，这是怎么回事呢？



图4-27 计算错误

切换到“公式”选项卡，单击“计算选项”按钮，将“手动”命令前的勾选取消，而选择“自动”命令，如图4-28所示。更改完后就会自动重新计算得到正确的结果。手工重算一般是针对数据比较多的情况，计算太卡，这样不会计算就不会影响效率，在关闭工作表的时候才一次完成计算。



图4-28 自动重算

4.1.7 删除公式只保留计算结果

如图4-29所示，工作表中设置了公式，尤其是有些公式引用了其他表格的值，发给别人时，怎样只保留计算结果？






图4-29 引用其他表格的公式

如图4-30所示，选择公式区域B2:B4，右击，在弹出的快捷菜单中单击“123”按钮，即将公式粘贴成值，如图4-31所示。

低版本的Excel没有“123”这个功能，可右击，在弹出的快捷菜单中选择“选择性粘贴”命令，在弹出的对话框中选中“数值”单选按钮，单击“确定”按钮，如图4-32所示。




图4-30 复制、粘贴




图4-31 保留值的效果




图4-32 “选择性粘贴”对话框

4.1.8 保护公式不被修改

表格模板因为要发给别人填写，里面的公式是不允许别人改动的，要不然容易造成一些不必要的麻烦。如图4-33所示，是一个设置了很多公式的表格模板，如何才能保护公式不被修改呢？

STEP 01 单击工作表左上角的三角形，可以选中整个工作表。按快捷键Ctrl+1，打开“设置单元格格式”对话框，切换到“保护”选项卡，取消选中“锁定”复选框，单击“确定”按钮，如图4-34所示。

	A	B	C	D	E	F	G	H	I
1	序号	名称	单位	数量	单价	金额	成本单价	成本金额	毛利率
2	1	书籍	本	2	35	70	30	60	14.29%
3	2	水	瓶	3	3	9	2	6	33.33%
4	3								
5	4								
6	5								
7	6								
8	7								
9	8								
10	9								
11	10								
12		总计				79		66	16.46%
13		合计：柒拾玖元整							
14									

图4-33 设置了很多公式的表格模板




图4-34 取消锁定

STEP 02 按功能键F5打开“定位条件”对话框，选中“公式”单选按钮，单击“确定”按钮，如图4-35所示。






图4-35 定位公式

STEP 03 按快捷键Ctrl+1，打开“设置单元格格式”对话框，切换到“保护”选项卡，选中“锁定”和“隐藏”复选框，单击“确定”按钮，如图4-36所示。



图4-36 选中“锁定”和“隐藏”复选框

STEP 04 切换到“审阅”选项卡，单击“保护工作表”按钮，在弹出的“保护工作表”对话框中输入密码，单击“确定”按钮。在弹出的“确认密码”对话框中，再次输入密码，单击“确定”按钮，如图4-37所示。



图4-37 设置保护

经过以上4个步骤后，只要修改公式所在单元格，就弹出警告对话框，如图4-38所示。在没有公式的单元格中照样允许输入内容。




图4-38 警告对话框



知识扩展

设置了保护公式不被修改以后，假如原来的公式设置错误，需要重新修改，则可以取消工作表保护。切换到“审阅”选项卡，单击“撤销工作表保护”按钮，输入原来的密码，单击“确定”按钮，如图4-39所示。






图4-39 撤销保护工作表

4.1.9 定义名称的三种方法

在公式中经常定义名称，借助名称可以让公式更容易理解。

=VLOOKUP(D2,对应表,2,0)

对应表就是一个名称，是引用A:B两列的数据，如图4-40所示。

	A	B	C	D	E
1	人员	影片		人员	影片
2	曹雪芹	红楼梦		曹雪芹	红楼梦
3	高鹗	红楼梦			
4	宋江	水浒传			
5	施耐庵	水浒传			
6	李逵	水浒传			
7	林冲	水浒传			
8	燕青	水浒传			
9	鲁智深	水浒传			

根据人员查找影片

图4-40 查询对应值

名称在公式比较长的时候更能体现出价值。下面详细介绍三种定义名称的方法。

1. 使用名称框

最快捷的方法就是利用名称框。如图4-41所示，选择区域，输入名称，按Enter键即可。




图4-41 名称框

2. 以选定区域的值创建名称

当需要创建很多名称时，以选定区域的值创建名称是最好的办法。选择区域，单击“公式”选项卡中的“根据所选内容创建”按钮，在弹出的对话框中取消选中“最左列”复选框，再单击“确定”按钮，如图4-42所示。




图4-42 以选定区域的值创建名称



知识扩展

正常情况下根据所选内容创建名称是一种很好用的办法，但是当数据源的内容行数不一样的时候，定义出来的名称区域包含空白单元格，不是很理想，如图4-43所示。




图4-43 区域包含空白单元格



面对这种情况，需要多增加一步操作才能解决。

选择区域，按F5键打开“定位条件”对话框，选中“常量”单选按钮，单击“确定”按钮，如图4-44所示。



图4-44 定位常量

定位后就帮你智能选好区域，单击“公式”选项卡中的“根据所选内容创建”按钮，在弹出的对话框中选中“首行”复选框，再单击“确定”按钮，如图4-45所示。




图4-45 根据所选内容创建名称

这样定义出来的名称区域都是有内容的单元格，如图4-46所示。




图4-46 定义的名称区域都是有内容的单元格

3. 使用功能区命令

这种方法适合创建动态名称，当数据源经常变动时，此方法是首选。单击“公式”选项卡中的“定义名称”按钮，在弹出对话框的“名称”文本框中输入“动态区域”，在“引用位置”文本框中输入下面的公式，然后单击“确定”按钮，如图4-47所示。

=OFFSET(定义名称!\$A\$1,,,COUNTA(定义名称!\$A:\$A),2)




图4-47 使用功能区命令

这里没有强制性要求用哪一种方法，只要你觉得哪种顺手就选用哪一种。正常情况下我都是使用第三种方法定义名称，只因习惯而已。

怎么给名称命名呢？

原则：越简单、越容易记，越好。



如果你不嫌麻烦的话,也可以起一个长长的名字:我吃饱了撑着你拿我怎么办。

管理名称

单击“公式”选项卡中的“名称管理器”按钮,打开“名称管理器”对话框,这时所有定义的名称就展示在我们面前。我们可以对名称进行编辑、删除。如刚才那个“我吃饱了撑着你拿我怎么办”的名称是开玩笑的,没有意义,选择这个名称,单击“删除”按钮,在弹出的对话框中再单击“确定”按钮,如图4-48所示。




图4-48 删除名称

4.1.10 获取帮助的方法和渠道

无意间看见好友刘苏整理的获取帮助的方法和渠道,写得挺好的,大家可以作为参考,如图4-49所示。




图4-49 获取帮助

下面重点讨论F1键。

在Excel里,有一个万金油的按键就是F1。

不论何时,按下F1键,都可以打开微软为你准备的帮助。只需以下两步,就能快速找到详细的函数说明。

STEP 01 按下F1键,输入要搜索的函数,如图4-50所示。



图4-50 搜索函数

STEP 02 单击函数词条,查看具体内容,如图4-51所示,可以看到函数的语法说明和视频教程。



图4-51 查看帮助

一个函数下面提供了很多知识点，现在随便单击打开其中一个知识点——使用SUM的最佳做法，就可以看到详细的案例说明，如图4-52所示。



图4-52 查看详细的知识点

如果能全部吃透，你就是函数大神了！
最后再强调一句：独学而无友，则孤陋寡闻。在学习的过程中要多跟别人交流，切记！

4.2 实战

4.2.1 四则运算

卢子一想到以后就要负责中山隆成那边的工作，就发愁起来了。以前需要处理的数据很少，用计算器勉强能应付过去，而现在需要处理的数据越来越多，不靠Excel根本做不好。考虑了半天，卢子打算豁出去了，买本《Excel函数手册》来学习。



对于英语水平超级差劲的卢子来说，学函数实在是一种挑战。卢子思索着：我能学好函数吗？顾虑归顾虑，但难得第一次买Excel书籍，怎么也要花点心思来学习。

这段时间一直出差，一下班，女同事就看电视，卢子对电视不感兴趣，基本上很少跟她们一起看。在这边属于断网状态，那时候还没有微博，手机能做的事情实在有限，上一会儿QQ后就不知干什么。突然，卢子想起了前段时间购买的《Excel函数手册》，所以拿出来准备学习。为了不打击自己的自信心，卢子选择了以前接触过的函数，开始看起来。有很多知识，如果学过后没有复习的话，很容易就忘记了，重复看有助于记忆。就像很多人考试前一样，都会花点心思重新翻开书籍复习，这样以前的知识点又被牢记于心。

这样的日子持续了两个月，卢子每天抽出一段时间翻开书籍，对有兴趣的知识点先了解，不感兴趣的不看，如工程函数、财务函数等。没想到卢子还学会了很多函数的基本用法，为此兴奋了一小段时间。

还因此得出结论：不会英语也可以学函数！

学习的目的是什么？不就是为了能学以致用吗？为此，卢子开始尝试自己制作更多的表格模板，虽然很粗糙，但也是一种进步。

经常出差，有时关心的仅仅是每回可以拿多少补助费而已，人有时就是这么现实，不要见笑。卢子还算有点基础，如图4-53所示，不一会儿就制作了“出差登记表”，这

样一来，每一回出差就知道住宿天数，能获取多少补助。

人员	开始日期	结束日期	日补助金额	住宿天数	补助小计
陈锡卢	7月26日	7月30日	30		
	8月5日	8月13日	30		
	8月16日	8月20日	30		
	8月24日	8月25日	30		
				合计	¥0

图4-53 出差登记表

怎么获取住宿天数呢？首先来了解下日期知识点。日期其实就是一组序列，1900-1-1就是日期的起点，也就是序列1；同理1900-1-2就是序列2。如果将日期的单元格格式设置为“常规”，这就是日期的本身序列号，如图4-54所示。

开始日期	结束日期	日补助金额	住宿天数	补助小计
40020	7月30日	30		

图4-54 设置日期为常规格式

我们知道数字是可以直接进行四则运算的，日期是特殊的数字，住宿天数就是结束日期减去开始日期。如图4-55所示，默认情况下，相减单元格会自动变成日期格式。当你看到日期格式的时候不要慌，只要明白日期其实也就是普通的数字，设置为常规格式就变回来了。




图4-55 将日期格式改为常规格式

补助小计的公式为：

=E5*F5

知识扩展

四则运算符号数学表示法与Excel表示法不同，不能直接用数学表示法，否则会提示错误，如图4-56所示。

数学表示法	Excel表示法
+	+
-	-
×	*
÷	/

→ =1×1




图4-56 四则运算符号

4.2.2 比较运算符

以前没有出差登记表的时候，卢子也没在意补助金额，反正领导给多少就是多少。现在



有了这份表格，就知道领导给的补助是否准确，如图4-57所示。

	A	B	C	D	E	F	G
1	人员	开始日期	结束日期	补助小计	领取补助	核对	
2	陈锡卢	7月26日	8月1日	180	150	FALSE	
3		8月5日	8月13日	240	240	TRUE	
4		8月16日	8月20日	120	120	TRUE	
5		8月24日	8月25日	30	30	TRUE	
6							

图4-57 核对金额

话说领导是怎么计算补助的：掰手指算天数，7-26,7-27...8-1，一般持续十天以上的出差很少，所以手指还够用，统计好天数后，比如5天就敲计算器=5×30，逐个统计。这样计算其实也没什么不好，就是偶尔会出一点点小差错而已，如果能够核对两次以上估计没问题。

领导的做法卢子不敢干预，也没有领导那么有时间，所以能够借助Excel统计的数据，还是尽量借助Excel。

接着一起看Excel怎么表示区间。

作为质检人员，接触最多的是AQL抽样基准表，如图4-58所示，在某个范围内的出货数量要抽检多少数量。

	A	B
1	出货数量	抽检数量 (n)
2	1~500	20
3	501~1200	32
4	1201~3200	50
5	3201~10000	80
6	10001以上	125

图4-58 AQL抽样基准表

那么区间该怎么表示呢？如图4-59所示，在数学中经常用 $1 \leq X \leq 500$ 这种表示方法。在Excel中的表示方法是否也一样？卢子尝试了，却是不成立的。

A1	:	X	✓	f _x	=1<=30<=500
1	A	B	C	D	E
2	FALSE				
3					
4					

目标成立，但按Enter
键后却不成立

图4-59 数学区间表示法

很多时候，理想与现实会存在很大的差距。在Excel中首先执行 $1 \leq 30$ 的判断，这个明显成立，返回TRUE。接着用 $TRUE \leq 500$ ，逻辑值大于数字，得到FALSE。

Excel中数据的排序依据：错误值>逻辑值>文本>数字，如图4-60所示。

	A	B	C
1	数据	备注	
2	#DIV/0!	错误值	
3	TRUE	逻辑值	
4	FALSE	逻辑值	
5	今朝	文本	
6	100000	数字	
7	900	数字	
8			
9			

降序排序

图4-60 数据排序依据

既然这样不行，正确的做法应该是怎样的呢？处于某个区间也就是大于等于最小值，小于等于最大值，只有满足这两个条件才能成立。

=AND(1<=30,30<=500)

AND函数就是需要所有条件都满足才行，就如手机要连接Wi-Fi上网，需要同时满足所有条件。

- ① 手机支持Wi-Fi;
- ② Wi-Fi已覆盖;
- ③ 有秘钥。

三个缺一不可，否则就不能用Wi-Fi上网，如图4-61所示。




图4-61 AND函数说明图

前面说过四则运算，其实也可以用在这个地方。

$$=(1 \leq 30) * (30 \leq 500)$$

满足条件显示TRUE，不满足条件显示FALSE，然后两者相乘。在这里TRUE可以看作是1，FALSE可以看作是0，只要都满足就显示1，否则就是0。

$$\begin{aligned} \text{TRUE} * \text{TRUE} &= 1 \\ \text{TRUE} * \text{FALSE} &= 0 \\ \text{FALSE} * \text{FALSE} &= 0 \end{aligned}$$

4.2.3 相对引用、绝对引用、混合引用

这一天卢子没事在办公室，回头看到覃美女在做报表。卢子发现覃美女现在懂得用电脑处理数据，而没用计算器，心中暗自夸奖：不错，有进步。可是过了几秒后，发生了一件让卢子大跌眼镜的事情，如果卢子有眼镜的话早就碎了一地。

只见覃美女，如图4-62所示，输入“=”，接着选中G6单元格，输入“/”，接着再用鼠标选中F3单元格。看到这里还算正常，接着直接按Enter键，到H7单元格，又重复上一步操作。

序号	不良描述	不良等级	不良数	不良率
1	塑胶件毛边、水口残留	B	27	=G6/F3
2	行走振动、不走直线，四脚不平	B	23	
3	轮盖装不到位	C	22	
4	后轮歪	B	14	
5	手把套不良	C	8	
6	其它		19	
	合计		113	

图4-62 没有做任何处理的引用

啊？这样也行？

卢子半调侃问：怎么不用计算器算呢？

覃美女倒也回答得爽快：用Excel计算比较准。

卢子：Excel确实挺适合用来计算，但你为什么不用绝对引用？

覃美女：这个以前读书时学过，不过太久没用了，都不知道怎么用。

卢子：你输入第一条公式后，别急着按Enter键，用鼠标选中F3单元格，然后按F4键。注意观察编辑栏的变化，这时自动添加了两美元（\$），如图4-63所示。最后向下填充公式，如图4-64所示。

序号	不良描述	不良等级	不良数	不良率
1	塑胶件毛边、水口残留	B	27	=G6/\$F\$3
2	行走振动、不走直线，四脚不平	B	23	
3	轮盖装不到位	C	22	
4	后轮歪	B	14	
5	手把套不良	C	8	
6	其它		19	
	合计		113	

图4-63 改变单元格的引用方式



H6		=G6/\$F\$3					
A	B	C	D	E	F	G	H
H126不良明细表							
出货总数：		3,702	检查总数：		3,815		
序号	不良描述			不良等级	不良数	不良率	
1	塑胶件毛边、水口残留			B	27	0.7%	
2	行走振动、不走直线，四脚不平			B	23	0.6%	
3	轮盖装不到位			C	22	0.6%	
4	后轮歪			B	14	0.4%	
5	手把套不良			C	8	0.2%	
6	其它				19	0.5%	

图4-64 填充公式

覃美女：还真的是这样，这样就快捷多了。原来是按F4键，这回我得记牢它。



知识扩展

相对引用：当向右或向下填充时，引用的单元格也会随之向右或向下移动。这就好像飞机在空中加油，虽然两架飞机都在飞，并且它们离地面的距离也在发生变化，但是两架飞机之间的距离是保持不变的。

绝对引用：当填充时，引用的单元格位置不发生变化。

混合引用：当填充时，引用的单元格位置部分发生变化，要么行变，要么列变。

相对引用与绝对引用的切换可以使用功能键F4来进行，如图4-65所示。

	A	B	C	D	E	F	G	H
1	赵默笙	美女1	美女2		=A1	赵默笙	美女1	美女2
2	美女3	美女4	美女5			美女3	美女4	美女5
3	美女6	美女7	美女8			美女6	美女7	美女8
4	美女9	美女10	美女11			美女9	美女10	美女11
5								
6					=A\$1	赵默笙	赵默笙	赵默笙
7						赵默笙	赵默笙	赵默笙
8						赵默笙	赵默笙	赵默笙
9						赵默笙	赵默笙	赵默笙
10								
11					=A\$1	赵默笙	赵默笙	赵默笙
12						美女3	美女3	美女3
13						美女6	美女6	美女6
14						美女9	美女9	美女9
15								
16					=A\$1	赵默笙	美女1	美女2
17						赵默笙	美女1	美女2
18						赵默笙	美女1	美女2
19						赵默笙	美女1	美女2
20								

图4-65 引用说明图

相对引用：通过使用相对引用对单元格填充，行标和列标都相应地进行了变化引用。

绝对引用：复制公式时行号和列标都不发生改变。

混合引用：复制公式时要么行号变，要么列变。

如果你下拉单元格时希望行不发生改变，那么请在行号前加上美元符号(\$)，有钱就走不动路了(如A\$1)。

如果你右拉时希望列不发生改变，那么请在列标前加上美元符号(\$)，有钱她就不会被别人拐跑了(如\$A1)。

如果想上双保险，行号列标都不想发生改变，行号列标前都加上\$，全方位锁定你(如\$A\$1)。

4.3 小结与练习

本章主要是学习函数前的一些准备，把这些基本知识学好，有助于后面函数的学习。概念还是很重要的。

1. 如图4-66所示，根据出差登记表，如何获取住宿天数和补助金额？
2. 如图4-67所示，列标变成数字，该怎么复原？

	A	B	C	D	E	F	G
1		出差登记表					
2							
3				日补助金额		30	
4		人员	开始日期	结束日期	住宿天数	补助小计	
5		陈锡卢	7月26日	7月30日			
6			8月6日	8月13日			
7			8月16日	8月20日			
8			8月24日	8月25日			
9							
10							
11							
12				合计		¥0	
13							

图4-66 出差登记表

	1	2	3	4	5
1					
2					
3					
4					
5					
6					
7					
8					
9					
10					

图4-67 列标变成数字



第5章

神奇的函数



函数与公式是Excel的精髓，每天都有无数人在讨论其用法。基本上每天都有一些精妙的公式被发掘出来。学公式靠的是逻辑思维、思考问题的方法（角度），不像技巧那样死记硬背。不经过认真思考、举一反三，永远学不好公式。学好公式，绝大部分的工作都可以轻松搞定。曾经有人说过，你只要学好公式，再配合一些技巧，你就可以成为高手。让我们一起向高手的方向迈进一小步！

5.1 SUM函数家族

函数中最受欢迎的有三大家族，一是以SUM函数为首的求和家族，二是以VLOOKUP函数为首的查找引用家族，三就是以IF函数为首的逻辑函数家族。根据二八定律，学好这三大家族的函数，就能完成80%的工作。

SUM函数家族最常见的成员有5个，如图5-1所示。

函数	说明
SUM	对区域进行求和
SUMIF	对区域中符合指定条件的值求和
SUMIFS	对区域中符合多个指定条件的值求和
DSUM	返回列表或数据库中满足指定条件的记录字段（列）中的数字之和
SUMPRODUCT	在给定的几组数组中，将数组间对应的元素相乘，并返回乘积之和

图5-1 SUM函数家族

5.1.1 SUM函数三种实用的求和用法

SUM函数的求和有很多种，这里主要通过三方面进行详细说明。

- ① 普通求和；
- ② 快捷键求和；
- ③ 多表格求和；

下面通过案例进行详细说明，如图5-2所示，一起来看看SUM函数家族怎么完成各种汇总。

	A	B	C
1	日期	产品	数量
2	2012-1-10	电脑	10
3	2013-1-10	电脑	28
4	2013-1-11	电视机	36
5	2013-1-17	电冰箱	80
6	2013-1-20	电脑	54
7	2013-2-1	电脑	55
8	2013-2-2	电脑	82
9			

图5-2 产品销售明细

1. 普通求和

现在要知道所有产品的销售数量。



=SUM(C2:C8)

对于SUM函数估计大家都很熟悉，就是对区域的数据进行求和。需要注意的是，如果区域存在文本，将自动被忽略。如现在将求和区域变成B2:C8，包含了产品名称，直接用SUM求和不会有任何影响。

=SUM(B2:C8)

SUM函数的区域可以是一行、一列、区域、单元格等，如图5-3所示。

	E	F	G	H	I	J	K
1							
2						结果	公式
3	1	2	3	4		10	=SUM(E3:H3)
4	5	6	7	8			
5	9	10	11	12		78	=SUM(E3:H5)
6							
7						5	=SUM(E3,H3)
8							

图5-3 求和区域

2. 快捷键求和

如图5-4所示，将鼠标放在“自动求和”的按钮上面，就会提示求和的快捷键为Alt+=。




图5-4 自动求和提示的快捷键

也就是说，借助快捷键Alt+=可以轻松实现对区域进行求和，如图5-5所示。




图5-5 快捷键求和

利用快捷键Alt+=不仅可以对一列的数据求和，还能对行列的数据进行同时求和。选择区域，按快捷键Alt+=瞬间就对区域进行求和，如图5-6所示。

	A	B	C	D	E
1	部门	姓名	工资	奖金	工资合计
2	生产部	姓名1	3082	385	
3	生产部	姓名2	3076	400	
4	生产部	姓名3	3165	173	
5	生产部	姓名4	3273	330	
6	生产部	姓名5	3519	465	
7	生产部	姓名6	4274	237	
8	生产部	姓名7	4666	456	
9	包装部	姓名12	4919	265	
10	包装部	姓名13	4079	229	
11	包装部	姓名14	3805	166	
12	包装部	姓名15	3979	416	
13	包装部	姓名16	4205	323	
14	合计		46042	3845	49887

图5-6 对行列数据进行同时求和

还有一种就是将数据分成几段求和，如图5-7所示，分别对各个部门统计工资。

	A	B	C
1	部门	姓名	工资
2	生产部	姓名1	3082
3	生产部	姓名2	3076
4	生产部	姓名3	3165
5	生产部	姓名4	3273
6	小计		
7	销售部	姓名5	3519
8	销售部	姓名6	4274
9	销售部	姓名7	4666
10	小计		
11	包装部	姓名12	4919
12	包装部	姓名13	4079
13	包装部	姓名14	3805
14	包装部	姓名15	3979
15	包装部	姓名16	4205
16	小计		
17			

图5-7 分段求和

当你输入完第一个公式后，你会发现很

像这种分段求和，跳跃式的求和，而且每次求和求的都是上方连续区域的数据，我们可以巧妙利用功能区上的自动求和功能加上定位条件来快速实现。先用鼠标将C列有数据的区域选中（不能把整个C列选完，只选中C列有数据的区域），再按F5键，打开“定位”对话框，单击“定位条件”按钮，在弹出的“定位条件”对话框中，选中“空值”单选按钮，单击“确定”按钮，如图5-9所示。



图5-9 定位空值

这时C列的空单元格就被选好了，只需按一下快捷键Alt+=就完成统计，如图5-10所示。

纠结，公式没法下拉，如果是一个个输入公式又非常麻烦，当部门非常多的时候还容易出错，如图5-8所示。

C6			=SUM(C2:C5)	
A	B	C	D	E
1	部门	姓名	工资	
2	生产部	姓名1	3082	
3	生产部	姓名2	3076	
4	生产部	姓名3	3165	
5	生产部	姓名4	3273	
6	小计		12596	
7	销售部	姓名5	3519	
8	销售部	姓名6	4274	
9	销售部	姓名7	4666	
10	小计			
11	包装部	姓名12	4919	
12	包装部	姓名13	4079	
13	包装部	姓名14	3805	
14	包装部	姓名15	3979	
15	包装部	姓名16	4205	
16	小计			
17				

图5-8 没法下拉



C6				
=SUM(C11:C15)				
A	B	C	D	E
1	部门	姓名	工资	
2	生产部	姓名1	3082	
3	生产部	姓名2	3076	
4	生产部	姓名3	3165	
5	生产部	姓名4	3273	
6	小计		12596	
7	销售部	姓名5	3519	
8	销售部	姓名6	4274	
9	销售部	姓名7	4666	
10	小计		12459	
11	包装部	姓名12	4919	
12	包装部	姓名13	4079	
13	包装部	姓名14	3805	
14	包装部	姓名15	3979	
15	包装部	姓名16	4205	
16	小计		20987	

图5-10 快捷键求和

3. 多表格求和

SUM函数有一种特殊的表示方法，可以快速对多个表格进行汇总。如图5-11所示，对汇总外的所有表格的A1单元格进行汇总。

汇总	Sheet1	Sheet2	Sheet3	1月	2月	3月
----	--------	--------	--------	----	----	----

图5-11 SUM函数多表汇总

=SUM('*!A1)

“*”是通配符，代表所有字符，用在这里代表除了公式所在表格（汇总）的其他所有表格。

例如，将包含“月”的所有表格的A1单元格进行汇总。

=SUM('*月!A1)

通配符一共有两个，一个是“*”（代表所有字符），另一个是“?”（代表1个字符）。如图5-12所示，统计包含月份数字的和，一个是用“*”，一个是用“?”，可以看到用“*”是将所有月份都统计起来，而用“?”只是统计1~9月，10月以上的数字是两位数，所以统计不了。

	A	B	C	D	E
1	月份	数量		结果	公式
2	1月	1		15	=SUMIF(A:A,"*月",B:B)
3	2月	1		9	=SUMIF(A:A,"?月",B:B)
4	3月	1			
5	4月	1			
6	5月	1			
7	6月	1			
8	7月	1			
9	8月	1			
10	9月	1			
11	10月	2			
12	11月	2			
13	12月	2			
14					

图5-12 通配符的比较

有很多人喜欢将每月的数据分别存放在各个表格，然后汇总。如图5-13所示，有6个月的数据，表格格式相同。

汇总	1月	2月	3月	4月	5月	6月
	A	B	C			
1	销售人员	商品	销售量			
2	曹泽鑫	彩电	43			
3	刘敬堃	冰箱	88			
4	周德宇	电脑	94			
5	周德宇	相机	72			
6	曹泽鑫	彩电	100			
7	王腾宇	冰箱	57			
8	周德宇	彩电	14			
9	王学敬	电脑	81			
10	周德宇	相机	6			
11	周德宇	彩电	5			

图5-13 格式相同的多表汇总

如图5-14所示，现在想在汇总表依次获取每月的销售量，该怎么做呢？

	A	B	C
1	月份	销售量	
2	1月		
3	2月		
4	3月		
5	4月		
6	5月		
7	6月		
8			

图5-14 汇总每个月的销售量

有人也许会说，这还不简单，直接用SUM函数然后分别引用表格不就可以了。

```
=SUM('1月'!C:C)
=SUM('2月'!C:C)
...
=SUM('6月'!C:C)
```

当然这也是种办法，不过用Excel就得考虑扩展性，假如是12个月呢？甚至是按销售人员分表，那估计就得用几百个表格，这样还能继续更改公式吗？

如果你的时间比较充足，要这么做也无可厚非。但对于懒惰成性的卢子而言，这种事情是绝不允许发生的。现在每个表格名已知，区域也已知，只要引用起来不就好了。

```
=A2&"!C:C"
```

这样就可以获得1~6月C列区域，那是不是这样就能直接求和呢？

```
=SUM(A2&"!C:C")
```

如图5-15所示，得到的是#VALUE!，警告你出错了，原因是里面的区域仅仅是文本，而不是真正的区域，那该怎么办呢？

	A	B
1	月份	销售量
2	1月	#VALUE!
3	2月	#VALUE!
4	3月	#VALUE!
5	4月	#VALUE!
6	5月	#VALUE!
7	6月	#VALUE!
8		

图5-15 直接引用求和出错

别急，先看下INDIRECT函数，这个函数

可以对文本进行引用，一般叫间接引用。

有直接引用，自然有间接引用。比如有三个人，分别叫甲、乙、丙，现在甲要知道丙的事情，可以直接去问丙，也可以通过乙间接去了解丙的事情。也就是说，直接引用直接输入区域就行，不通过第三者，通常的引用都是直接引用，如区域C:C。间接引用就是通过第三者才能获得的，如INDIRECT("C:C")。也就是说，只要嵌套INDIRECT函数就可以。

```
=SUM(INDIRECT(A2&"!C:C"))
```

再来看一个例子，加深理解。如图5-16所示，要获取A1单元格的内容，可以利用INDIRECT函数间接引用。

	A	B	C	D
1	我是牵线的		结果	公式
2			我是牵线的	=INDIRECT("A1")
3				
4				

图5-16 INDIRECT函数的间接引用

当然这样体现不出INDIRECT函数的价值，但如果A1“我是牵线的”代表一个表格名称，而且是变动的，就非常有用。现在我要间接获取“我是牵线的”这个表格的A1，如图5-17所示。

C1 : <input type="button" value="X"/> <input type="button" value="✓"/> <input type="button" value="fx"/> =INDIRECT(A1&"!A1")				
	A	B	C	D
1	我是牵线的		介绍TA给你认识	
2				
3				

图5-17 间接引用其他表格名称

当A1的表格名称变成“帅哥”的时候，间接引用的内容也跟着改变，这样就避免每



次改动表格名称就要改变公式的麻烦,如图5-18所示。




图5-18 改变表格名称

5.1.2 SUMIF(SUMIFS)函数按条件求和及优势解读

SUMIF(SUMIFS)函数的作用就是按条件进行求和,这里主要介绍三种用途。

- ① 基本的条件求和;
 - ② 有趣的第三参数;
 - ③ 查询数据的优势;
- 下面通过实例进行说明。

1. 条件求和

如果想知道电脑的销售数量。

=SUMIF(B2:B8,"电脑",C2:C8)

SUMIF→SUM+IF, IF就是“如果”的意思,也就是如果满足条件就对区域中的数据

进行求和。

语法:

=SUMIF(条件区域,条件,求和区域)

SUMIF函数通俗语法示意图如图5-19所示。




图5-19 SUMIF函数通俗语法示意图

条件区域产品为“电脑”,对满足这些条件的求和区域数量进行求和,求和过程如图5-20所示。

	A	B	C	D	E
1	日期	产品	数量		
2	2012-1-10	电脑	10		
3	2013-1-10	电脑	28		
4	2013-1-11	电视机	36		
5	2013-1-17	电冰箱	80		
6	2013-1-20	电脑	54		
7	2013-2-1	电脑	55		
8	2013-2-2	电脑	82		
9					

图5-20 SUMIF函数求和过程

如果公式需要下拉,区域要加绝对引用,否则会出错。如图5-21所示,因为区域用相对引用,所以下拉的时候自动变了,本来电脑的数量是229,但出来的结果是191。

	A	B	C	D	E	F	G
1	日期	产品	数量		产品	数量	
2	2012-1-10	电脑	10		电视机	36	=SUMIF(B2:B8,E2,C2:C8)
3	2013-1-10	电脑	28		电冰箱	80	=SUMIF(B3:B9,E3,C3:C9)
4	2013-1-11	电视机	36		电脑	191	=SUMIF(B4:B10,E4,C4:C10)
5	2013-1-17	电冰箱	80				
6	2013-1-20	电脑	54				
7	2013-2-1	电脑	55				
8	2013-2-2	电脑	82				
9							

图5-21 错误的引用

正确的公式是下面两个。

=SUMIF(\$B\$2:\$B\$8,E2,\$C\$2:\$C\$8)
=SUMIF(B:B,E2,C:C)

SUMIF函数有一个隔列求和的经典案例，刚好有网友在群内提到这个问题，在这里说明一下。如图5-22所示，每个日期下面都有收支，最后统计本月的收支。

	AY	AZ	BA	BB	BC	BD	BE	BF	BG	BH	BI	BJ	BK	BL	BM	BN	BO	BP	BQ	BR	BS	BT	BU	BV
1																								
2		22日	23日	24日	25日	26日	27日	28日	29日	30日	31日	本月合计												
3		收	支	收	支	收	支	收	支	收	支	收	支	收	支	收	支	收	支	收	支	余	金额	
4			1	2	1					2	156	2				7				216	61	163	1059.5	
5		36	6	8	4	4	10		9	24	2		3							156	165	47	470	
6				7							6									72	51	65	214.5	
7		2							7											24	20	52	130	
8																				24	26	16	40	
9		4				5							4							0	36	43	176.3	
10		9	2			2	11	3		18	3									120	137	36	41.76	
11		48	4			2	13	14	48	7	2	6								144	162	111	821.4	
12										2										12	3	25	162.5	

图5-22 收支统计

网友对收支分别设置公式。

收：

```
=I4+K4+M4+O4+Q4+S4+U4+W4+Y4+AA4+AC4+AE4+AG4+AI4+AK4+AM4+AO4+AQ4+AS4+AU4+AW4+AY4+BA4+BC4+BE4+BG4+BI4+BK4+BM4+BO4+BQ4
```

支：

```
=J4+L4+N4+P4+R4+T4+V4+X4+Z4+AB4+AD4+AF4+AH4+AJ4+AL4+AN4+AP4+AR4+AT4+AV4+AX4+AZ4+BB4+BD4+BF4+BH4+BJ4+BL4+BN4+BP4+BR4
```

对于这样的公式，单元格太多了，即使错了也不知道。其实这个数据源很有特点，“收”和“支”可以作为条件，然后用SUMIF函数进行条件求和就行。

在BS4单元格中输入公式，并向右和向下填充公式，就完成了本月收支情况。

```
=SUMIF($I$3:$BR$3,BS$3,$I4:$BR4)
```

前面提到的都是单条件求和，如果是多条件求和就得用SUMIFS函数。

对2013-1-16之前的电脑的销售数量进行求和。

```
=SUMIFS(C2:C8,A2:A8,"<2013-1-16",B2:B8,"电脑")
```

SUMIFS函数语法：

```
=SUMIFS(求和区域,条件区域1,条件1,条件区域2,条件2,条件区域n,条件n)
```

在SUMIF函数的基础上理解SUMIFS函数就变得非常简单，SUMIFS函数的条件区域和



条件是成双成对的，只是有一个很明显的差异就是SUMIF函数的求和区域在后面，SUMIFS函数的求和区域在前面，如图5-23所示。

	A	B	C
1	日期	产品	数量
2	2012-1-10	电脑	10
3	2013-1-10	电脑	28
4	2013-1-11	电视机	36
5	2013-1-17	电冰箱	80
6	2013-1-20	电脑	54
7	2013-2-1	电脑	55
8	2013-2-2	电脑	82

图5-23 SUMIFS函数示意图

2. 有趣的第三参数

第三参数省略的用法：

SUMIF函数的语法包含三个参数，但这个并不是硬性规定，允许只存在两个参数。

如求数量大于50的和。

```
=SUMIF(C2:C8,">50",C2:C8)
```

缩写：

```
=SUMIF(C2:C8,">50")
```

如图5-24所示，标准写法跟省略写法得到的结果一样，也就是说，当条件区域与求和区域相同时，求和区域可以不用写。

	A	B	C	D	E	F
1	日期	产品	数量			
2	2012-1-10	电脑	10		271	=SUMIF(C2:C8,">50",C2:C8)
3	2013-1-10	电脑	28			
4	2013-1-11	电视机	36		271	=SUMIF(C2:C8,">50")
5	2013-1-17	电冰箱	80			
6	2013-1-20	电脑	54			
7	2013-2-1	电脑	55			
8	2013-2-2	电脑	82			

图5-24 第三参数省略

第三参数简写的用法：

如图5-25所示，第三参数用标准的C2:C8与C2得到的结果是一样的，也就是说，SUMIF函数支持简写。在使用简写时，第三参数的区域大小会自动扩展成与第一参数的区域大小一样。

	A	B	C	D	E	F
1	日期	产品	数量			
2	2012-1-10	电脑	10		229	=SUMIF(B2:B8,"电脑",C2:C8)
3	2013-1-10	电脑	28			
4	2013-1-11	电视机	36		229	=SUMIF(B2:B8,"电脑",C2)
5	2013-1-17	电冰箱	80			
6	2013-1-20	电脑	54			
7	2013-2-1	电脑	55			
8	2013-2-2	电脑	82			
9						

图5-25 第三参数简写

在只有一个求和区域时使用简写看不出明显效果，但是如果是多个求和区域就非常明显。如图5-26所示，现在有三个区域，要对三个区域的所有电脑的数量进行汇总。

	A	B	C	D	E	F	G	H	I
1	产品	数量		产品	数量		产品	数量	
2	电脑	10		电脑	20		电冰箱	10	
3	电脑	28		电视机	28		电脑	28	
4	电视机	36		电视机	36		电视机	36	
5	电冰箱	80		电冰箱	80		电冰箱	80	
6	电脑	54		电脑	54		电脑	54	
7	电脑	55		电脑	55		电冰箱	55	
8	电脑	82		电视机	82		电脑	82	
9									

图5-26 多个区域求和

正常情况下，因为有三个区域，使用三个SUMIF函数进行统计。

```
=SUMIF(A2:A8,"电脑",B2:B8)+SUMIF(D2:D8,"电脑",E2:E8)+SUMIF(G2:G8,"电脑",H2:H8)
```

这样写公式很烦琐，以前就帮一个做财务的朋友处理过类似的问题。每个月一个区域，一年就是12个区域进行条件求和，如果按照这种思路要写12个SUMIF函数，显然不明智。其实SUMIF函数有一个很神奇的功能，不管有多少个区域，只要格式相同都能够自动识别。在这里只使用一个SUMIF函数就能轻易解决。

```
=SUMIF(A2:G8,"电脑",B2)
```

3. 查询数据的优势

查询数据正常情况下都是使用VLOOKUP函数，但这个函数有一点不好，就是在查询不到对应值的时候显示错误值#N/A，如图5-27所示，93666没有对应值。如果能让查询不到的结果显示0，看起来就好点。

因为查询返回的值是数值，所以可以用SUMIF函数进行查询。SUMIF函数是条件求和，没有符合条件的，统计的结果就等于0，如图5-28所示。



F2 : =VLOOKUP(E2,A:C,3,0)						
	A	B	C	D	E	F
1	番号	俗称	订单数		番号	订单数
2	93657	R102	1500		93667	1200
3	43010	R102	3000		40406	800
4	40981	H276	750		93666	#N/A
5	93668	H276	750			
6	93667	H216	1200			
7	40406	H126	800			
8	40401	H126	1000			
9	40405	H126	200			
10	93669	H216	1200			
11						

图5-27 查询时返回错误值

F2 : =SUMIF(A:A,E2,C:C)						
	A	B	C	D	E	F
1	番号	俗称	订单数		番号	订单数
2	93657	R102	1500		93667	1200
3	43010	R102	3000		40406	800
4	40981	H276	750		93666	0
5	93668	H276	750			
6	93667	H216	1200			
7	40406	H126	800			
8	40401	H126	1000			
9	40405	H126	200			
10	93669	H216	1200			
11						

图5-28 SUMIF函数查询

VLOOKUP函数查找条件要求很严，需要格式一样才行。例如，2016-2-3为日期格式，如果用“2016-2-3”文本格式就查不到，而用SUMIF函数却可以，在这里也能体现出SUMIF函数的便利性，如图5-29所示。

	A	B	C	D	E	F	G
1	日期	番号	俗称	订单数			
2	2016-2-2	93657	R102	1500		#N/A	=VLOOKUP("2016-2-3",A:D,4,0)
3	2016-2-3	43010	R102	3000		3000	=SUMIF(A:A,"2016-2-3",D:D)
4	2016-2-4	40981	H276	750			
5	2016-2-5	93668	H276	750			
6	2016-2-6	93667	H216	1200			
7	2016-2-7	40406	H126	800			
8	2016-2-8	40401	H126	1000			
9	2016-2-9	40405	H126	200			
10	2016-2-10	93669	H216	1200			
11							

图5-29 日期格式不同

如果要采用VLOOKUP函数获取正确值，可以借助--或者DATE函数。

```
=VLOOKUP("--2016-2-3",A:D,4,0)
=VLOOKUP(DATE(2016,2,3),A:D,4,0)
```

5.1.3 DSUM函数多条件求和的优缺点

有了SUMIFS函数以后再多条件查询都能够解决。这个函数是从2007版才开始有的，而使用2003版的朋友看到这里估计得感叹，对于多条件查询高版本有SUMIFS函数，我能用什么？

虽然2003版即将退出历史舞台，但并不能忽视现在使用人群的庞大，十年前微软公司就是靠2003版争取到90%的用户。

神秘的D字头函数，DSUM闪亮登场！D字头函数使用者很少，但并不能因此就忽略了它们，它们是非常强大的，如图5-30所示。



G2		X ✓ fx		=DSUM(A1:C8,3,E1:F2)			
	A	B	C	D	E	F	G
1	日期	产品	数量		日期	产品	数量
2	2012-1-10	电脑	10		<2013-1-16	电脑	38
3	2013-1-10	电脑	28				
4	2013-1-11	电视机	36				
5	2013-1-17	电冰箱	80				
6	2013-1-20	电脑	54				
7	2013-2-1	电脑	55				
8	2013-2-2	电脑	82				
9							

图5-30 DSUM函数的运用

当你看到这个公式，你会不会怀疑自己的眼睛看错了，居然这么简单，不太可能啊！但事实就摆在眼前，不容置疑。

DSUM函数的语法：

=DSUM(数据源区域,求和列号,条件区域)

DSUM函数通俗语法示意图如图5-31所示。

VLOOKUP ▾		: ✕ ✓ <i>fx</i>		=DSUM(A1:C8,3,E1:F2)				
	A	B	C	D	E	F	G	H
1	日期	产品	数量		日期	产品	数量	
2	2012-1-10	电脑	10		<2013-1-16	电脑	=DSUM(A	
3	2013-1-10	电脑	28					
4	2013-1-11	电视机	36					
5	2013-1-17	电冰箱	80					
6	2013-1-20	电脑	54					
7	2013-2-1	电脑	55					
8	2013-2-2	电脑	82					
9								
10								
11								

条件区域

数据源区域

对数据源区域的第3列
即数量进行求和

图5-31 DSUM函数通俗语法示意图

第二参数的“3”也可以直接用列名表示，现在数据源第3列是数量，直接写“数量”。

=DSUM(A1:C8,"数量",E1:F2)

=DSUM(A1:C8,G1,E1:F2)

DSUM函数跟其他函数不一样，必须要有标题才能使用，否则出错。如图5-32所示，不管是条件区域的标题，还是数据源区域的标题，只要任意一个没标题，统计就出错。



G2

:

✖

✓

fx

=DSUM(A1:C8,3,E1:F2)

	A	B	C	D	E	F	G
1	日期	产品	数量			产品	数量
2	2012-1-10	电脑	10		<2013-1-16	电脑	0
3	2013-1-10	电脑	28				
4	2013-1-11	电视机	36				
5	2013-1-17	电冰箱	80				
6	2013-1-20	电脑	54				
7	2013-2-1	电脑	55				
8	2013-2-2	电脑	82				
9							

✖

G2

:

✖

✓

fx

=DSUM(A1:C8,3,E1:F2)

	A	B	C	D	E	F	G
1	日期	产品	数量		日期	产品	数量
2	2012-1-10	电脑	10		<2013-1-16	电脑	0
3	2013-1-10	电脑	28				
4	2013-1-11	电视机	36				
5	2013-1-17	电冰箱	80				
6	2013-1-20	电脑	54				
7	2013-2-1	电脑	55				
8	2013-2-2	电脑	82				
9							

✖

图5-32 缺少标题统计出错

这个函数有个好处就是，不管有多少个条件，都一样处理，想出错都难。
DSUM函数之三条件求和，如图5-33所示。




I2	:	  	=DSUM(A1:D8,3,F1:H2)						
	A	B	C	D	E	F	G	H	I
1	日期	产品	数量	备注		日期	产品	备注	数量
2	2012-1-10	电脑	10	已结款		<2013-1-16	电脑	未结款	28
3	2013-1-10	电脑	28	未结款					
4	2013-1-11	电视机	36	未结款					
5	2013-1-17	电冰箱	80	未结款					
6	2013-1-20	电脑	54	未结款					
7	2013-2-1	电脑	55	未结款					
8	2013-2-2	电脑	82	未结款					

图5-33 DSUM函数之三条件求和

DSUM函数虽然很好用，但也存在一些缺陷，如多条件求和时，向下复制公式会出现问题。不能单独汇总，而是累计汇总。这是个大问题，莫非DSUM函数因这个问题而变成垃圾函数？如图5-34所示。

H2	:				=DSUM(\$A\$1:\$D\$20,4,\$F\$1:G2)				
	A	B	C	D	E	F	G	H	
1	销售人员	商品	销售量	销售额		销售人员	商品	销售额	
2	曹泽鑫	彩电	13	29900		曹泽鑫	彩电	108100	
3	刘敬堃	冰箱	27	70200		刘敬堃	冰箱	219900	
4	周德宇	电脑	40	344000		周德宇	电脑	563900	
5	周德宇	相机	42	154980					
6	曹泽鑫	彩电	34	78200					
7	王腾宇	冰箱	24	62400					
8	周德宇	彩电	32	73600					
9	王学敏	电脑	13	111800					
10	周德宇	相机	34	125460					
11	周德宇	彩电	20	46000					
12	周德宇	相机	43	158670					
13	房天琦	空调	45	126000					
14	王腾宇	空调	34	95200					
15	刘敬堃	冰箱	16	41600					
16	郝宗泉	彩电	23	52900					
17	周德宇	彩电	30	69000					
18	房天琦	冰箱	38	98800					
19	曹泽鑫	冰箱	31	80600					
20	王学敏	空调	42	117600					

累计汇总，错误！

图5-34 DSUM函数累计汇总错误

一个函数的功能是有限的，适当的时候要借助其他办法。既然是累计汇总，H2就是等于本身，H3也就多了H2的销售额，H4多了H2+H3的销售额，只要能将这多出来的去掉就行。

也就是说，分别对H2,H2:H3,H2:H4这三个区域求和，第一个区域只有一个单元格，变成区域的形式可以写为H2:H2。仔细观察第一个H2始终不变，采用绝对引用，第二个H2会逐渐改变，采用相对引用。为了加深理解，先看下面的例子，对A列的数据进行累加。

=SUM(\$A\$2:A2)

固定第一个A2，也就是加\$符号，这样当公式下拉的时候，只改变第二个A2的值，到B6的时候，变成A2:A6，从而实现累加，如图5-35所示。

	A	B	C
1	数据	累加	
2	1	1	=SUM(\$A\$2:A2)
3	2	3	=SUM(\$A\$2:A3)
4	3	6	=SUM(\$A\$2:A4)
5	4	10	=SUM(\$A\$2:A5)
6	5	15	=SUM(\$A\$2:A6)
7			

图5-35 SUM函数累计求和

所以，多条件求和可以用：

=DSUM(\$A\$1:\$D\$20,4,\$F\$1:G2)-SUM(\$H\$1:H1)

SUM函数忽略文本，所以可以直接引用第一个单元格。



其实都是低版本惹的祸，在高版本里这个会很简单。

```
=SUMIFS(D:D,A:A,F2,B:B,G2)
```

只能说每个函数都有其优缺点，要好好利用其优点。



知识扩展

D字头函数除了DSUM函数外，还有一系列函数：DMAX、DMIN和DCOUNT(DCOUNTA)，它们的语法都是一样的，学会了一个就等于学会了所有。

如图5-36所示，根据销售人员明细表，统计三个销售人员销售的最大金额、最小金额、计数。

	A	B	C	D	E	F	G
1	销售人员	销售额		销售人员	最大	最小	计数
2	曹泽鑫	29900		曹泽鑫			
3	刘敬堃	70200		刘敬堃			
4	周德宇	344000		周德宇			
5	周德宇	154980					
6	曹泽鑫	78200					
7	王腾宇	62400					
8	周德宇	73600					
9	王学敏	111800					
10	房天琦	126000					
11	王腾宇	95200					
12	刘敬堃	41600					
13	郝宗泉	52900					
14	周德宇	69000					
15	房天琦	98800					
16	曹泽鑫	80600					
17	王学敏	117600					
18							

图5-36 统计销售人员的最大金额、最小金额、计数

最大金额：

```
=DMAX(A1:B17,2,D1:D4)
```

最小金额：

```
=DMIN(A1:B17,2,D1:D4)
```

计数：

```
=DCOUNT(A1:B17,2,D1:D4)
```



=DCOUNTA(A1:B17,2,D1:D4)

下面介绍SUMPRODUCT函数对区域乘积求和。

看到这里也许细心的朋友会发现，还少了一个重要的函数SUMPRODUCT的说明，这个低版本的朋友应该用得非常多，特别是多条件求和。

SUMPRODUCT函数是对区域的乘积再求和。毕竟是出自一家人，帮助都差不多，非常好理解，如图5-37所示，统计总金额。

=SUMPRODUCT(C2:C10,D2:D10)




D11	:				=SUMPRODUCT(C2:C10,D2:D10)		
	A	B	C	D	E	F	G
1	名称	单位	数量	单价			
2	三位大板 开关	个	2	15			
3	二位大板 开关	个	1	12			
4	手套	对	12	2.5			
5	电工胶布	卷	20	1.5			
6	螺丝	盒	1	15			
7	斯耐德 1P/20A	个	7	35			
8	20A漏电	套	4	120			
9	2P/63A漏电	套	1	438			
10	填缝剂	盒	2	24			
11			总金额	1328			
12							

图5-37 统计总金额

这个函数的作用就相当于利用一个辅助列获取每个产品的金额，然后再汇总，如图5-38所示。

A	B	C	D	E	F
1	名称	单位	数量	单价	辅助列
2	三位大板 开关	个	2	15	30 =C2*D2
3	二位大板 开关	个	1	12	12 =C3*D3
4	手套	对	12	2.5	30 =C4*D4
5	电工胶布	卷	20	1.5	30 =C5*D5
6	螺丝	盒	1	15	15 =C6*D6
7	斯耐德 1P/20A	个	7	35	245 =C7*D7
8	20A漏电	套	4	120	480 =C8*D8
9	2P/63A漏电	套	1	438	438 =C9*D9
10	填缝剂	盒	2	24	48 =C10*D10
11		总金额	1328		1328 =SUM(E2:E10)

图5-38 辅助列统计金额

强调一句，SUMPRODUCT函数将非数值型的数组元素作为0处理，如C2现在的值是文本的“零”。零在这里被识别为0，用SUMPRODUCT函数统计就正确，而用辅助列统计反而出错，这就是SUMPRODUCT函数的一个好处，如图5-39所示。

=SUMPRODUCT(C2:C10,D2:D10)

A	B	C	D	E	F
1	名称	单位	数量	单价	辅助列
2	三位大板 开关	个	零	15	#VALUE! =C2*D2
3	二位大板 开关	个	1	12	12 =C3*D3
4	手套	对	12	2.5	30 =C4*D4
5	电工胶布	卷	20	1.5	30 =C5*D5
6	螺丝	盒	1	15	15 =C6*D6
7	斯耐德 1P/20A	个	7	35	245 =C7*D7
8	20A漏电	套	4	120	480 =C8*D8
9	2P/63A漏电	套	1	438	438 =C9*D9
10	填缝剂	盒	2	24	48 =C10*D10
11		总金额	1298	#VALUE!	=SUM(E2:E10)

图5-39 包含文本统计金额

说明：SUMPRODUCT函数如果只有一个参数，作用就跟SUM函数一样，对区域进行求和，如图5-40所示。

E11	:				=SUMPRODUCT(E2:E10)	
	A	B	C	D	E	F
1	名称	单位	数量	单价	金额	
2	三位大板 开关	个	2	15	30	
3	二位大板 开关	个	1	12	12	
4	手套	对	12	2.5	30	
5	电工胶布	卷	20	1.5	30	
6	螺丝	盒	1	15	15	
7	斯耐德 1P/20A	个	7	35	245	
8	20A漏电	套	4	120	480	
9	2P/63A漏电	套	1	438	438	
10	填缝剂	盒	2	24	48	
11				总金额	1328	
12						

图5-40 只有一个参数的情况下





知识扩展

SUMPRODUCT函数一般跟加权统计有关，借助这个函数，2013年帮我解决了一次工作抉择。

好工作其实就是：在离家很近的地方做自己喜欢的事，并得到高薪。但现实中这种工作少之又少，所以要把握好兴趣、亲戚朋友、工资三面对你的重要程度。为自己做了个小小的权重分析，发现目前留在潮州比去深圳发展占有微弱的优势，高出0.02分。最终决定还是暂时留在潮州发展，如图5-41所示。

	A	B	C	D	E	F
	城市	兴趣 (36%)	亲戚朋友 (34%)	工资 (30%)	得分	
1						
2	潮州	3	5	4	3.98	
3	深圳	4	3	5	3.96	
4						
5						
6	=SUMPRODUCT(B2:D2,{0.36,0.34,0.3})					
7						
8						

@Excel之恋
weibo.com/u/1936136972

图5-41 加权统计

5.2 VLOOKUP函数家族

2014年的某一天，卢子（@Excel之恋）在微博发了一个VLOOKUP函数的运用，引来33万人次的阅读量，860次的转发，可见大家对VLOOKUP函数的偏爱有多深。后面会重点介绍这个函数的用法，这里只做简单说明。

VLOOKUP家族很庞大，常用成员有5个函数，如图5-42所示。

函数	说明
VLOOKUP	在数组第一列中查找，然后在行之间移动以返回单元格的值
LOOKUP	在向量或数组中查找值
OFFSET	从给定引用中返回引用偏移量
INDEX	使用索引从引用或数组中选择值
MATCH	在引用或数组中查找值

图5-42 VLOOKUP函数家族

5.2.1 VLOOKUP函数按条件查询对应值

既然这里提到了2014年，就顺便以IT部落窝的人员作为数据源，并作为查找依据，这些人曾经也为IT部落窝做出了很多贡献。如图5-43所示，根据员工姓名查找职务。

数据查找，首推VLOOKUP函数。

```
=VLOOKUP(E2,B:C,2,0)
```

VLOOKUP函数查找的示意图如图5-44所示。

	A	B	C	D	E	F
1	员工号	员工姓名	职务		员工姓名	职务
2	A101	卢子	超级版主		simple	论坛元老
3	A102	久久	版主			
4	A103	冷逸	版主			
5	A104	闲池	版主			
6	A105	simple	论坛元老			
7	A106	传递	金牌会员			
8	A107	lsxlsl	金牌会员			
9	A108	西单	高级会员			
10	A109	hooyi	中级会员			
11						

图5-43 根据员工姓名查找职务

	A	B	C	D	E	F
1	员工号	员工姓名	职务		员工姓名	职务
2	A101	卢子	超级版主		simple	=VLOOKU
3	A102	久久	版主			
4	A103	1	2			
5	A104	simple	论坛元老			
6	A105	传递	金牌会员			
7	A106	lsxlsl	金牌会员			
8	A107	西单	高级会员			
9	A108	hooyi	中级会员			
10						
11						
12						

图5-44 VLOOKUP函数查找示意图



知识扩展

这个函数的查找值必须在查找区域的首列，否则出错，如图5-45所示，现在查找区域的首列变成了A列。







F2	:				=VLOOKUP(E2,A:C,2,0)	
	A	B	C	D	E	F
1	员工号	员工姓名	职务		员工姓名	职务
2	A101	卢子	超级版主		simple	#N/A
3	A102	久久	版主			
4	A103	冷逸	版主			
5	A104	闲池	版主			
6	A105	simple	论坛元老			
7	A106	传递	金牌会员			
8	A107	lsxlsk	金牌会员			
9	A108	西单	高级会员			
10	A109	hooyi	中级会员			
11						

图5-45 查找值不在查找区域首列出错

这里的查找区域和数据源区域是两个概念，如果写为A:C，查找区域跟数据源区域是一样的，但如果写为B:C，查找区域跟数据源区域就不一样。返回的列数也是根据查找区域进行确定，千万别搞混乱了。

5.2.2 LOOKUP函数各种方向查询对应值的经典公式

现在人的阅读习惯都是从左到右，跟古代不同。VLOOKUP函数很好用，如果要逆序查找，也就是从右到左，就相对比较麻烦。可以借用IF({1,0},)组合来实现，不过要花费九牛二虎之力，吃力不讨好。这时它的兄弟LOOKUP函数就派上用场，借助这个函数轻而易举就办到。LOOKUP函数不区分正常顺序跟逆序，如图5-46所示，根据员工姓名，查找员工号，用在这里再合适不过。

	A	B	C	D	E	F	G
1	员工号	员工姓名	职务		员工姓名	员工号	
2	A101	卢子	超级版主		simple	A105	
3	A102	久久	版主				
4	A103	冷逸	版主				
5	A104	闲池	版主				
6	A105	simple	论坛元老				
7	A106	传递	金牌会员				
8	A107	lsxlsk	金牌会员				
9	A108	西单	高级会员				
10	A109	hooyi	中级会员				
11							

图5-46 根据员工姓名，查找员工号

LOOKUP函数有一个经典查找公式，这里先记住，至于怎么得到的，暂时可以不用管。

=LOOKUP(1,0/((条件1)*(条件2)*...(条件n)),返回区域)

根据这条公式，可以得到：

=LOOKUP(1,0/(E2=B2:B10),A2:A10)

条件就是E2=B2:B10，条件可以是一个或者多个。

返回区域就是A2:A10，因为要返回员工号，也就是A列的区域。

如果公式要下拉的话，记得区域加绝对引用。

=LOOKUP(1,0/(E2=\$B\$2:\$B\$10),\$A\$2:\$A\$10)

LOOKUP函数除了不区分正反向查询外，同时不管是竖向还是横向照样可以搞

定，这也是该函数的神奇之处。如图5-47所示，将数据源转置后，根据员工姓名查询职务。

[illegible]

图5-47 数据源转置



VLOOKUP其实有两兄弟，VLOOKUP是老大，LOOKUP是老二，老三是被人遗忘的HLOOKUP。因为审美观的问题，导致了很少有几行无数列的表格，也就使HLOOKUP无用武之地。VLOOKUP函数是垂直查询，HLOOKUP函数是水平查询，除了方向不同，其他用法一样。这里通过一个简单案例介绍HLOOKUP函数，根据员工姓名查询职务，如图5-48所示。

[illegible]

图5-48 水平查询

5.2.3 OFFSET函数的经典运用

既然提到论坛的朋友，怎么可以将真正的朋友忘记，下面来看OFFSET函数之卢子访友。

卢子有三个朋友：无言、安安、胖纸，朋友间就得互相来往。每个朋友的距离都挺远，卢子住在潮州，无言离卢子最近，在汕头那边；安安次之，在深圳；胖纸最远，在东莞。同属广东，坐车半天内都能到，如图5-49所示。



	A	B	C	D	E
1	卢子	*	无言	*	
2	*	*	*	*	
3	*	*	*	*	
4	*	*	*	*	
5	*	*	*	*	
6	安安	*	*	*	
7	*	胖纸	*	*	
8					

图5-49 OFFSET函数之卢子访友

假如OFFSET函数就是卢子的车，如何到每个朋友那里？

有人说OFFSET函数会轻功，那速度当然不比车子慢。先来看看语法：

=OFFSET(起点,偏移行,偏移列,行高,列宽)

注：行高、列宽为可选参数。

卢子要去无言那边，只需向右坐2站就到，如图5-50所示。

	A	B	C	D	E
1	卢子	*	无言	*	
2	*	*	*	*	
3	*	*	*	*	
4	*	*	*	*	

图5-50 右移

=OFFSET(A1,0,2)

如果要去安安那边，需向下坐5站才能到，如图5-51所示。

	A	B
1	卢子	*
2	*	*
3	*	*
4	*	*
5	*	*
6	安安	*
7	*	胖纸
8		

图5-51 下移

=OFFSET(A1,5,0)

去胖纸那边就稍微麻烦点，要向下坐6站，再向右坐1站才能到，如图5-52所示。

	A	B	C
1	卢子	*	无言
2	*	*	*
3	*	*	*
4	*	*	*
5	*	*	*
6	安安	*	*
7	*	胖纸	*
8			

图5-52 下移再右移

=OFFSET(A1,6,1)

看到这里大概知道OFFSET函数是干什么用的了，如果偏移的行数为正数就是向下偏移，偏移的列数为正数就是向右偏移。相反，如果偏移的行数为负数就是向上偏移，偏移的列数为负数就是向左偏移。

假如卢子现在在胖纸家里，想要回到自己的家，就得向上坐6站，就是-6，向左坐1站，也就是-1，如图5-53所示。

	A	B	C
1	卢子	*	无言
2	*	*	*
3	*	*	*
4	*	*	*
5	*	*	*
6	安安	*	*
7	*	胖纸	*
8			

图5-53 上移左移

=OFFSET(B7,-6,-1)

既然知道怎么去，就得知道怎么回，卢子还不至于路痴到忘记回来的路。

安安跟胖纸离得很近，卢子想知道她们两家合并的范围有多大，也就是深圳跟东莞的范围。卢子就得先到安安这里，然后将这里的行高设置为2，列宽设置为2，这样就可以知道这两地的范围，如图5-54所示。

	A	B	C
1	卢子	*	无言
2	*	*	*
3	*	*	*
4	*	*	*
5	*	*	*
6	安安	*	*
7	*	胖纸	*
8			

图5-54 范围

```
=OFFSET(A1,5,0,2,2)
```

但这只是划分范围，没有统计。统计可以用COUNTA函数，得到这两地的范围为4。

```
=COUNTA(OFFSET(A1,5,0,2,2))
```

OFFSET函数有两个经典的案例：隔行引用和动态引用。

(1) 如图5-55所示，根据销售明细表，获取每个季度的销售情况。

	A	B	C	D	E
1	月份	销售额		季度	销售额
2	1月	355			
3	2月	563			
4	3月	345			
5	一季度	1263			
6	4月	4546			
7	5月	4464			
8	6月	3245			
9	二季度	12255			
10	7月	4354			
11	8月	4556			
12	9月	3464			
13	三季度	12374			
14	10月	3256			
15	11月	6785			
16	12月	7437			
17	四季度	17478			

图5-55 隔行引用

在D2单元格中输入公式，并向下和向右复制。

```
=OFFSET(A$1,4*ROW(A1),0)
```

季度的数据在第5行、第9行、第13行和第17行，也就是每个数据都隔着4行。这时只要构架一个相差4的等差序列，然后进行偏移获取即可。ROW函数可以获取行号，下拉就得到1、2、3、4。行号再乘以4就得到相差4的等差序列，也就是4*ROW就得到4、8、12、16，最后在第1行进行偏移，就获取了相对应的值。



(2) 如图5-56所示, 数据透视表是非常强大的统计工具, 但直接根据数据源创建透视表后, 当数据源新增加数据的时候, 并没有自动统计进去。如何才能动态获取一个数据源? 这样当数据源新增加数据的时候, 数据透视表也能自动更新区域。

	A	B	C	D	E	F	G	H	I
1	番号	俗称	订单数	订单号	尾数	完成情况		行标签	求和项: 订单数
2	93657	R102	1500	100429	12	NG		H126	2000
3	43010	R102	3000	100429	0	OK		H216	2400
4	40981	H276	750	100517	0	OK		H276	1500
5	93668	H276	750	100522	123	NG		R102	4500
6	93667	H216	1200	100522	0	OK		总计	10400
7	40406	H126	800	100602	0	OK			
8	40401	H126	1000	100601	0	OK			
9	40405	H126	200	100603	100	OK			
10	93667	H216	1200	100605	0	NG			
11	10000	YD	50	111111	40	NG	新增加		
12									

图5-56 数据源新增加数据统计出错

这个问题需借助OFFSET函数定义一个动态区域的名称, 然后数据透视表引用这个动态名称即可解决。

STEP 01 如图5-57所示, 切换到“公式”选项卡, 单击“定义名称”按钮, 在弹出的“新建名称”对话框的“名称”文本框中输入“动态区域”, 在“引用位置”文本框中输入下面的公式, 单击“确定”按钮。

=OFFSET(动态引用!\$A\$1,,,COUNTA(动态引用!\$A:\$A),6)



图5-57 动态区域

数据源从A1单元格开始不需要偏移行列，也就是第2、3参数都是0（用逗号占位也代表0），行高不确定，用COUNTA函数统计非空的单元格就获取行高，列宽为6列。这样就用OFFSET函数获取了一个动态区域。

STEP 02 如图5-58所示，在“插入”选项卡中单击“数据透视表”按钮，在弹出的“创建数据透视表”对话框中，设置“表/区域”为“动态区域”，选中“选择放置数据透视表的位置”选项组中的“现有工作表”单选按钮，在“位置”文本框中输入“动态引用!\$H\$1”，单击“确定”按钮。




图5-58 数据透视表引用动态区域

STEP 03 如图5-59所示，重新布局数据透视表，在数据源更新数据后，刷新数据透视表，就能自动统计。

	A	B	C	D	E	F	G	H	I
1	序号	俗称	订单数	订单号	尾数	完成情况		行标签	求和项: 订单数
2	93657	R102	1500	100429	12	NG		H126	2000
3	43010	R102	3000	100429	0	OK		H216	2400
4	40981	H276	750	100517	0	OK		H276	1500
5	93668	H276	750	100522	123	NG		R102	4500
6	93667	H216	1200	100522	0	OK		YD	50
7	40406	H126	800	100602	0	OK		LT	40
8	40401	H126	1000	100601	0	OK		总计	10490
9	40405	H126	200	100603	100	OK			
10	93667	H216	1200	100605	0	NG			
11	10000	YD	50	111111	40	NG			
12	20000	LT	40	222222	40	NG			
13									

图5-59 自动统计



5.2.4 初识INDEX、MATCH函数组合

访友总不能两手空空，至少得买点水果表示一下吧。如图5-60所示，水果种类很多，有苹果、香蕉等，每一种的单价都是不同的。

	A	B	C	D
1	水果	单价	数量	
2	苹果	6.9	40	
3	香蕉	3.4	38	
4	柠檬	5.5	15	
5	橙子	2.5	25	
6	梨	5.9	40	
7	杏	28	10	
8	腰果	35.5	16	
9	花生	12.5	20	
10	胡桃	17.5	12	
11				

图5-60 水果明细表

现在要获取区域中第3行第2列的对应值，就可以用INDEX函数，得到3.4。

```
=INDEX(A1:C10,3,2)
```

想知道A列第3行的对应水果，就可以用INDEX函数，得到香蕉。

```
=INDEX(A1:A10,3)
```

第1行第2列的对应值，就可以用INDEX函数，得到单价。

```
=INDEX(A1:C1,2)
```

摸清了水果的情况，才好挑选。不过对于卢子而言，很多时候都只是知道大概要买什么而已，没有实际概念。假如现在要获取最后5种水果的情况，怎么办？

一般情况下记录都是随时增加的，你不可能去数哪几条是最后的，数到的数据即使现在可以用，过几天就变化了。最后一条非空记录的行号，可以用COUNTA函数获取，

这个函数用于统计非空单元格的个数。

```
=COUNTA(A:A)
```

现在知道非空单元格一共有10条，那么最后5条记录就是：10,9,8,7,6。一般情况下获取连续数字首先考虑到的就是用ROW函数实现。

```
=COUNTA(A:A)-ROW(A1)
```

这样就可以得到9~6这4条记录，但缺少第1条记录，也就是说在第1行的时候只能减去0。

```
=COUNTA(A:A)-ROW(A1)+1
```

现在已经知道行号，只需嵌套INDEX函数就可以，如图5-61所示。

	A	B	C	D	E	F	G	H
1	水果	单价	数量		最后购买的5种产品			
2	苹果	6.9	40		胡桃	17.5	12	
3	香蕉	3.4	38		花生	12.5	20	
4	柠檬	5.5	15		腰果	35.5	16	
5	橙子	2.5	25		杏	28	10	
6	梨	5.9	40		梨	5.9	40	
7	杏	28	10					
8	腰果	35.5	16					
9	花生	12.5	20					
10	胡桃	17.5	12					
11								

图5-61 获取最后5种产品

但有的人觉得还是按原来顺序排序好，记录显示是6~10。也就是第1行减去4，第2行减去3...这时只需做小小的变形就行。

```
=COUNTA(A:A)-5+ROW(A1)
```

现在就得到6~10，再重新嵌套INDEX函数。

```
=INDEX(A:A,COUNTA(A:A)-5+ROW(A1))
```


如果想要改成引用A:C这种区域的形式，可以用：

```
=INDEX($A:$C,COUNTA(A:A)-5+ROW(A1),COLUMN(A1))
```



知识扩展

INDEX函数更多的是结合MATCH函数实现各种查询引用，其作用跟VLOOKUP函数一样。如图5-62所示，根据水果查询单价。

```
=INDEX(B:B,MATCH(E2,A:A,0))
```

F2						
1	水果	单价	数量		水果	单价
2	苹果	6.9	40		柠檬	5.5
3	香蕉	3.4	38		苹果	6.9
4	柠檬	5.5	15		梨	5.9
5	橙子	2.5	25			
6	梨	5.9	40			
7	杏	28	10			
8	腰果	35.5	16			
9	花生	12.5	20			
10	胡桃	17.5	12			
11						

图5-62 查询水果的单价

MATCH函数用于获取排位，如柠檬在A列排第4位，就返回4。详细用法参考“5.5.2 凭借VLOOKUP函数解疑难”。

INDEX、MATCH函数组合有一个经典的查询模式，适用于单条件或者多条件查询。查询模式为：

```
=INDEX(区域,MATCH,MATCH)
```

注：单条件查询，第三参数的MATCH可以省略。

如图5-63所示，根据水果查询数量和单价的对应值。

```
=INDEX($A:$C,MATCH($E2,$A:$A,0),MATCH(F$1,$A$1:$C$1,0))
```

F2						
1	水果	单价	数量		水果	数量
2	苹果	6.9	40		柠檬	15
3	香蕉	3.4	38		苹果	40
4	柠檬	5.5	15		梨	40
5	橙子	2.5	25			
6	梨	5.9	40			
7	杏	28	10			
8	腰果	35.5	16			
9	花生	12.5	20			
10	胡桃	17.5	12			
11						

图5-63 查询数量和单价



5.3 IF函数家族

据网络调查，SUM、VLOOKUP、IF是使用频率最高的三个函数，初步了解了前面两个函数家族，现在开始了解IF函数。IF函数家族既简单，又重要。

IF函数家族常用成员有3个，如图5-64所示。

函数	说明
IF	对条件进行判断
OR	检测任意一项条件是否为真
AND	用于检测所有的条件是否为真

图5-64 IF函数家族

只要留心观察，生活中到处充满IF，我们每天都在跟IF打交道。

如果明天下暴雨，我就不去上班。

如果有你陪在我身边，我会很开心。

如果有网络，我就上网，否则睡觉。

如果2007年毕业不去东莞，我就不会接触Excel；如果不学习Excel，我就不会去论坛、交流群；如果不交流，我就不会认识这么多Excel爱好者；如果……否则一切都是空谈。

太多这样的例子，说上三天三夜都说不完。

卢子以前喜欢记账，就是每天将收支记录下来，然后月底汇总，如图5-65所示，看消费是否超出预算，如果超过预算就要查找原因。看钱花得是不是值得，不值得就得控制消费。

	A	B	C	D	E	F
1	项目	收入	支出	备注	支出预算	
2	1月份	2240	638.5		正常开支	
3	2月份	3640	4991	春节	超出预算	
4	3月份	1880	851		超出预算	
5	4月份	1880	553.5		正常开支	
6	5月份	1700	535.5		超出预算	
7	6月份	2470	309		正常开支	
8	7月份	2620	425.5		正常开支	
9	8月份	2680	378.5		正常开支	
10	9月份	2740	651		正常开支	
11	10月份	2770	382.5		正常开支	
12	11月份	2470	577		正常开支	
13	12月份	2450	818		超出预算	
14						

图5-65 收支预算表

卢子的预算是灵活的，挣得多就花得多，将收入的30%作为支出标准，其他的存进银行卡。如果收入乘以30%大于支出，就是正常开支，否则就是超出预算。

```
=IF(B2*30%>C2,"正常开支","超出预算")
```

通过计算公式，发现居然有4个月超出预算，看来卢子的控制能力还有待提高。仔细一看，2月份是春节，春节期间的支出比较多也情有可原。在这里就得再加一个条件，春节不管消费多少都是正常开支。

```
=IF(OR(B2*30%>C2,D2="春节"),"正常开支","超出预算")
```

“OR”在这里的意思就是或者。

语法：

```
=OR(条件1,条件2,条件N)
```

只要满足其中一个条件就显示TRUE，否则显示FALSE。比如电脑在杀毒的过程中，只有一个盘中毒（TRUE）就会提示电脑中毒，只有所有盘都不中毒（FALSE），才提示安全，如图5-66所示。




图5-66 OR函数示意图

“OR”可以换成“+”，作用一样。

```
=IF((B2*30%>C2)+(D2="春节"),"正常开支","超出预算")
```

说到OR函数，就不得不说到AND函数，两者的语法一样。AND函数表示并且的意思，只有所有条件同时满足才显示TRUE，否则显示FALSE。

对于宅男卢子而言，只要有电脑、网络、Excel三者就满足，三者若缺一，让卢子待在家里就有点困难，会产生不满。如图5-67所示，这时可以设置公式。

	A	B	C	D	E
1	电脑	网络	Excel	满足	
2	电脑		Excel	不满	
3	电脑	网络		不满	
4					

图5-67 AND函数

```
=IF(AND(A1="电脑",B1="网络",C1="Excel"),"满足","不满")
```

OR可以用“+”取代，AND则用“*”取代。

```
=IF((A1="电脑")*(B1="网络")*(C1="Excel"),"满足","不满")
```

IF函数家族更多的是起到一个辅助作用，所以也无须学习太深，知道常用的方法就行。



可千万别为了使用IF函数家族而导致公式比长城还长，那样就违背初衷了，如图5-68所示，是一个网友写的超级公式。

我这个是简单的了，有一个更多。。。=IF(OR(AND(D641=100,E641=105),AND(D641=100,E641=120),AND(D641=100,E641=140),AND(D641=100,E641=190),AND(D641=102,E641=145),AND(D641=102,E641=195),AND(D641=105,E641=114),AND(D641=105,E641=130),AND(D641=105,E641=140),AND(D641=105,E641=150),AND(D641=105,E641=175),AND(D641=110,E641=170),AND(D641=110,E641=195),AND(D641=113,E641=200),AND(D641=115,E641=140),AND(D641=60,E641=150),AND(D641=115,E641=175),AND(D641=118,E641=170),AND(D641=110,E641=175),AND(D641=75,E641=250),AND(D641=75,E641=290),AND(D641=75,E641=110),AND(D641=75,E641=190),AND(D641=120,E641=170),AND(D641=120,E641=200),AND(D641=80,E641=190),AND(D641=92,E641=110),AND(D641=85,E641=110),AND(D641=88,E641=110),AND(D641=85,E641=165),AND(D641=95,E641=105),AND(D641=95,E641=170),AND(D641=95,E641=165),AND(D641=100,E641=170),AND(D641=90,E641=120),AND(D641=90,E641=160),AND(D641=95,E641=220),AND(D641=60,E641=140),AND(D641=90,E641=170),AND(D641=130,E641=90),AND(D641=140,E641=125),AND(D641=60,E641=240),AND(D641=95,E641=155),AND(D641=100,E641=160),AND(D641=93,E641=115),AND(D641=38,E641=205),AND(D641=95,E641=115),AND(D641=40,E641=205),AND(D641=38,E641=158),AND(D641=70,E641=140),AND(D641=140,E641=160),AND(D641=70,E641=100),AND(D641=70,E641=130),AND(D641=85,E641=170),AND(D641=110,E641=145),AND(D641=70,E641=175)),3.5,IF(OR(AND(D641=120,E641=25),AND(D641=125,E641=21),AND(D641=150,E641=25)),2,IF(OR(AND(D641=145,E641=45),AND(D641=140,E641=140),AND(D641=130,E641=130)),5,IF(OR(AND(D641=90,E641=40),AND(D641=100,E641=40),AND(D641=90,E641=20),AND(D641=90,E641=45),AND(D641=100,E641=50),AND(D641=120,E641=60),AND(D641=100,E641=20)),1,IF(OR(AND(D641=64,E641=83),AND(D641=70,E641=50)),1.5,IF(AND(D641=80,E641=40),0.8,IF(ROUND(D641*D641*E641/1000000000)*IF(D641>400,1500,IF(D641>170,2000,IF(D641>130,3000,3500))),2)<0.8,0.8,ROUND(D641*D641*E641/1000000000)*IF(D641>400,1500,IF(D641>170,2000,IF(D641>130,3000,3500))),2)))))))))

图5-68 超级公式

知识扩展



正常情况下嵌套三层IF函数以上就尽量采用其他函数，如判定成绩的等级，如图5-69所示。

=IF(A2>=86,"优秀",IF(A2>=71,"良",IF(A2>=60,"及格","不及格")))

	A	B	C	D	E	F
1	成绩	等级			区间	等级
2	45	不及格			0-59	不及格
3	87	优秀			60-70	及格
4	67	及格			71-85	良
5	98	优秀			86-100	优秀
6	23	不及格				
7					成绩判定对应表	
8						

图5-69 判定成绩等级

IF函数一多就容易乱，对付区间的判断最适合用LOOKUP函数，看起来比较简洁。

```
=LOOKUP(A2,{0,60,71,86},{"不及格","及格","良","优秀"})
```

如果你是一个数学高手，还可以直接获取区间的下限，从而引用，公式就更加简洁。如图5-70所示，借助IMREAL函数获取下限，IMREAL("0-59i")，复数0-59i的实系数为0，同理IMREAL("60-70i")，复数60-70i的实系数为60。对于大多数人而言，手工写下限，然后进行查找也是一种不错的办法。




D2		:				=IMREAL(E2&"i")
	A	B	C	D	E	F
1	成绩	等级			区间	等级
2	45	不及格		0	0-59	不及格
3	87	优秀		60	60-70	及格
4	67	及格		71	71-85	良
5	98	优秀		86	86-100	优秀
6	23	不及格				
7						
8						

图5-70 IMREAL函数

最终可使用下面两条公式中的任意一条。

```
=LOOKUP(A2,$D$2:$F$5)
```

```
=LOOKUP(A2,$D$2:$D$5,$F$2:$F$5)
```

5.4 其他常用函数

跟三大家族的光芒四射比，其他函数就显得逊色很多，但并不代表其他函数就不重要，要学好Excel函数，需要掌握50个左右函数才行。

5.4.1 LEFT、MID、RIGHT字符提取三兄弟

字符提取函数属于次要函数，使用频率远远比不上上面的三大家族函数。但只要是能提高我们工作效率的函数，都了解一点，多学习总是有好处的。



根据产品编码求型号和颜色，产品编码规则是前6位代表型号，后两位代表颜色，如图5-71所示。

	A	B	C
1	产品编码	型号	颜色
2	85611211		
3	85611312		
4	85611313		
5	85611414		
6			

图5-71 根据产品编码求型号和颜色

求型号使用LEFT函数，求颜色使用RIGHT函数就可轻松解决，如图5-72所示。

	A	B	C
1	产品编码	型号	颜色
2	85611211	856112	11
3	85611312	856113	12
4	85611313	856113	13
5	85611414	856114	14
6			
7	使用公式：		
8	=LEFT(A2,6)		
9	=RIGHT(A2,2)		
10			

图5-72 提取后效果

这两个函数的语法是一样的，唯一的差别是一个从左面提取，另一个从右面提取。

=LEFT(文本字符串,从左开始截取几位字符)
=RIGHT(文本字符串,从右开始截取几位字符)

函数提取字符串的示意图如图5-73所示。

856112	← 85611211 →	11
LEFT	6 2	RIGHT

图5-73 LEFT、RIGHT提取字符串示意图

LEFT与RIGHT函数如果遇到了从中间提取字符，这两个函数就不管用了，如图5-74所示。

示，将金额的百、十、个位数字依次提取填写在单元格中。

	A	B	C	D
1	金额	百	十	个
2	998			
3	108			
4	223			
5	238			
6	470			
7	295			
8	134			
9				

图5-74 提取金额的百、十、个位数字

当然也有人会说，用这两个函数设置3条公式也能勉强提取出来。

=LEFT(A2)
=RIGHT(LEFT(A2,2))
=RIGHT(A2)

当提取左边或者右边为1位数的时候，第2参数可以省略。

其实它们有一个兄弟叫MID函数，这个函数可以从任意位置提取，是三兄弟中能力最强的一个。

=MID(\$A2,1,1)
=MID(\$A2,2,1)
=MID(\$A2,3,1)

通过不断地改变第2参数，从而获取每个数字。

MID函数语法：

=MID(文本字符串,开始位置,提取字符数)

现在要从字符串123456789中提取，开始位置为第4位，提取字符数为5位，就提取到45678，如图5-75所示。

回到刚刚提取金额的例子，虽然可以不断地改变第2参数，但是这么做明显不智

能。假如要分离身份证，那就要改18次，非常烦琐，有没有更好的办法呢？答案是肯定的，那就是借助COLUMN函数，这个函数是获取列号，A列就返回1，B列就返回2，以此类推，如图5-76所示。


123456789

 =MID("123456789",4,5) =COLUMN(A1) =COLUMN(B1) =COLUMN(C1) =COLUMN(D1) =COLUMN(E1)

图5-75 MID函数提取示意图

图5-76 获取列号

综合起来，最终公式为：

```
=MID($A2,COLUMN(A1),1)
```

公式通过向右复制和向下复制，一次性提取所有金额的数字。




知识扩展

在实际中，很多情况下金额的位数都不确定，对1~3位的，如果直接用这个公式会出现问题，如图5-77所示。

	A	B	C	D
1	金额	百	十	个
2	998	9	9	8
3	18	1	8	
4	223	2	2	3
5	2	2		
6	470	4	7	0
7	34	3	4	
8	134	1	3	4
9				

图5-77 位数不定的情况下出错

我们知道自定义单元格格式为000的时候，金额会显示3位数，不足位数的以0代替，在函数中有一个类似于自定义单元格格式功能的函数：TEXT，借助它可以让所有金额统一成3位，如图5-78所示。

E2 :  =TEXT(A2,"000")

	A	B	C	D	E	F
1	金额	百	十	个		
2	998	9	9	8	998	
3	18	1	8		018	
4	223	2	2	3	223	
5	2	2			002	
6	470	4	7	0	470	
7	34	3	4		034	
8	134	1	3	4	134	
9						

图5-78 统一位数



TEXT函数语法:

=TEXT(源数据,要显示的方式代码)

用TEXT函数结合前面的MID函数,就可以完美解决这个问题,如图5-79所示。




B2		:	  			=MID(TEXT(\$A2,"000"),COLUMN(A1),1)		
	A	B	C	D	E	F	G	
1	金额	百	十	个				
2	998	9	9	8				
3	18	0	1	8				
4	223	2	2	3				
5	2	0	0	2				
6	470	4	7	0				
7	34	0	3	4				
8	134	1	3	4				
9								
10								

图5-79 不定位数分离

接着通过分离电话号码的区号与号码来巩固下用法,如图5-80所示。

	A	B	C
1	电话号码	区号	号码
2	028-86647596	028	86647596
3	0813-2561587	0813	2561587
4	0552-2046547	0552	2046547
5	028-85145841	028	85145841
6	0831-8212021	0831	8212021
7			

图5-80 分离电话号码的区号与号码

区号是从左面提取,明显就是用LEFT函数,但区号位数不确定,也就是说N不确定。怎么确定N呢?细心的朋友可能会发现,区号跟号码中间用“-”连接起来,也就是说,只要能确定“-”的位置就能够确定N。

判断字符串的位置可以用FIND函数。

=FIND("-",A2)

通过FIND函数知道“-”在A2的位置是

第4位,那么“-”前面1位就是区号的总位数,也就是N。

=FIND("-",A2)-1

所以结合LEFT函数,可得到:

=LEFT(A2,FIND("-",A2)-1)

再看看号码,号码是在“-”后面,也就是从“-”的位置+1位就是号码的起始位置,然后提取N位,MID函数的N可以写一个很大的数字,如果没有字符提取,就不提取。

=MID(A2,FIND("-",A2)+1,99)

这个公式并没有提取99位,只是把“-”后面的所有内容都提取出来,也就是说,只要写一个比号码长度大的任意数字,都可以将号码提取出来。

=MID(A2,FIND("-",A2)+1,9)

当然在这里,号码是在后面,也可以用RIGHT函数提取,N就是总字符减去“-”的位置。总字符是多少?喜欢取巧的朋友可能会看到都是12个字符。

=RIGHT(A2,12-FIND("-",A2))

可是我们现在是在学习,以全面理解公式为主。统计字符数函数是LEN。

=RIGHT(A2,LEN(A2)-FIND("-",A2))

通过这些例子,基本用法都学得差不多了,剩下的就是灵活运用。



5.4.2 SUBSTITUTE、REPLACE函数

SUBSTITUTE函数类似查找替换功能，可以将旧字符替换成新字符。这个函数有一个著名的游戏——青蛙变王子。只要能玩好这个游戏，基本上就能熟练掌握这个函数。

如图5-81所示，将文中的青蛙替换成王子，实现各种效果。

	A	B	C
1	问题	源数据	效果
2	1	我是青蛙！	我是王子！
3	2	我是青蛙，你是青蛙！	我是王子，你是王子！
4	3	你是青蛙，我是青蛙！	你是青蛙，我是王子！
5	4	我是青蛙，你是青蛙？	我是王子，你是？
6	5	我是青蛙，你是王子！	我是王子，你是青蛙！
7			

图5-81 青蛙变王子

SUBSTITUTE函数语法：

`=SUBSTITUTE(文本字符串,旧内容,新内容,[替换第N个])`

问题1：将青蛙变成王子，因为只有一个青蛙，所以不考虑N的问题。

`=SUBSTITUTE(B2,"青蛙","王子")`

问题2：将所有青蛙变成王子，等同于替换一个。

`=SUBSTITUTE(B3,"青蛙","王子")`

问题3：将第2个青蛙变成王子，也就是N等于2。

`=SUBSTITUTE(B4,"青蛙","王子",2)`

问题4：将第1个青蛙变成王子，第2个青蛙变成空白。SUBSTITUTE函数一次只能替换一个青蛙，所以需要两次替换才可以。

`=SUBSTITUTE(SUBSTITUTE(B5,"青蛙","王子",1),"青蛙","")`

问题5：将青蛙变成王子，王子变成青蛙。先进行一次替换，将青蛙变成王子，现在就有2个王子，再将第2个王子变成青蛙，也就是N等于2。

`=SUBSTITUTE(SUBSTITUTE(B6,"青蛙","王子"),"王子","青蛙",2)`

这个函数功能比较单一，只要掌握N的用法，基本上就能学好。



实际运用:

如图5-82所示,有些人喜欢在输入人员名字的时候输入空格,把传统的手工记录思维用到Excel上。如果对人员不做任何后期处理的话,那倒没什么。但实际上很多数据都要进行汇总分析,这时就导致汇总出现问题。

	A	B	C
1	销售人员	销售量	
2	张 三	65	
3	李 四	80	
4	周德宇	86	
5	张 三	83	
6	曹泽鑫	80	
7	王腾宇	63	
8	周德宇	6	
9	李 四	69	
10			

图5-82 中文名多录入空格

借助SUBSTITUTE函数替换掉所有空格。

```
=SUBSTITUTE(A2," ","")
```

如图5-83所示,在录入英文名的时候,中间要用空格隔开,但只需要一个就行。

	A	B
1	姓名	
2	Alan Jones	
3	Bob Smith	
4	Carol Williams	
5		
6		

图5-83 英文名多录入空格

如果用SUBSTITUTE函数替换的话,会将所有空格都替换掉。这时TRIM函数就派上用场,这个可以去除多余的空格,中间只保留一个空格。

```
=TRIM(A2)
```

知识扩展



如图5-84所示,有的时候需要从一堆字中找出某个不同的字,如在“未”里找“未”,考验人的眼力。你需要花多长时间才能数出来里面有多少个“未”?



图5-84 找字游戏

对于我这种眼神不好的,找半天都找不出来,不过我借助公式能分分钟计算出来。



```
=LEN(A1)-LEN(SUBSTITUTE(A1,"未",""))
```

原理分析：

SUBSTITUTE(A1,"未","")将“未”全部替换成空白，也就是剩下的都是“未”。

LEN(SUBSTITUTE(A1,"未","")), 就是统计有多少个“未”。

LEN(A1)就是总字符数。

总字符数减去剩下的“未”数，也就是得到被替换掉了多少个“未”，被替换掉的次数也就是原来“未”的个数。

REPLACE函数跟SUBSTITUTE函数功能差不多，都是对字符串进行替换，下面介绍一下这个函数的用法。在公布获奖的手机号的时候，经常会隐藏中间4位号码，用“*”表示，如图5-85所示。

	A	B
1	源数据	格式化手机号码
2	13355550811	133****0811
3	15911181118	159****1118
4	15858581118	158****1118
5		

图5-85 格式化手机号码

使用公式：

```
=REPLACE(A2,4,4,"****")
```

REPLACE函数语法：

```
=REPLACE(文本字符串,起始位置,替换字符的个数,新文本)
```

格式化也就是将中间的数字替换成“*”，开始位置为4，要替换掉的数字为4个，替换成的新文本是“****”。

REPLACE函数有替换功能，同时也有插入功能，替换与插入由第三个参数决定，如果第三参数为0，也就是插入字符，如图5-86所示，从第3位开始，插入字符“和卢子”。

	A	B
1	源数据	需要的效果
2	我爱学Excel	我爱和卢子学Excel
3		

图5-86 插入字符

使用公式为：

```
=REPLACE(A2,3,0,"和卢子")
```

5.4.3 COUNT函数家族

COUNT函数家族成员还真不少，网上有人对这五兄弟做了一个精彩的点评。



老大COUNT是个数字控，只要看见数字，不管它是数值、日期还是时间，就开始统计个数。

老二COUNTA是个直肠子，看见单元格里有东西，它就开始数个数。

老三COUNTIF是个好孩子，只要符合规定它就开始计数。

老四COUNTIFS一直以三哥为榜样，但比它更挑剔，非得满足很多条件才肯统计个数。

老五COUNTBLANK很纯洁也很天真，必须是干干净净的单元格才肯计数。

如图5-87所示，是建筑物品供应商不同产品在每个月的销量。

	A	B	C	D	E
1	项目	1月	2月	3月	
2	砖	1000		500	
3	木材		5000		
4	玻璃	2000	1000		
5	钢材	1000			
6					

图5-87 建筑物品供应商不同产品在每个月的销量

问题1：统计各月份各有多少种建筑物品有销量。

```
=COUNT(B2:B5)
```

COUNT函数是统计数字的个数，如1月份有3种建筑物品有销量。

问题2：统计各月份各有多少种建筑物品没销量。

```
=COUNTBLANK(B2:B5)
```

COUNTBLANK函数是统计空单元格个数，如1月份有1种建筑物品没有销量。

问题3：统计供应商总共有多少种建筑

物品。

```
=COUNTA(A2:A5)
```

COUNTA函数是统计非空单元格个数，项目那里有4个建筑物品就返回4。

问题4：统计各月份各有多少种建筑物品销量大于1000。

```
=COUNTIF(B2:B5,">1000")
```

COUNTIF函数是统计满足条件的值的个数。

函数语法：

```
=COUNTIF(条件区域,条件)
```

再来介绍一个例子巩固下COUNTIF函数的用法。

问题5：统计建筑物品有多少种是两个字符的。

```
=COUNTIF(A2:A5,"??")
```

通配符的说明：星号（*）代表所有字符，问号（?）代表一个字符。

问题6：统计各月份各有多少种建筑物品销量大于1000并且小于3000。

```
=COUNTIFS(B2:B5,">1000",B2:B5,"<3000")
```

COUNTIFS函数是统计满足多个条件的值的个数。

函数语法：

```
=COUNTIFS(条件区域1,条件1, 条件区域2,条件2, 条件区域N,条件N)
```

COUNT函数家族与SUM函数家族差不多，最大的不同就是一个是计数，一个是求和。



5.4.4 极值与均值

在读书的时候经常听到：第一名成绩××，最后一名成绩××，平均成绩××。其实这些就是极值和平均值。

如图5-88所示，是一份各销售人员销售金额汇总表。

	A	B	C
1	销售人员	销售额	
2	曹泽鑫	188700	
3	房天琦	224800	
4	郝宗泉	52900	
5	刘敬堃	111800	
6	王腾宇	157600	
7	王学敏	229400	
8	周德宇	971710	
9			

图5-88 销售数据汇总表

问题1：获取最大销售额。

```
=MAX(B2:B8)
```

问题2：获取最小销售额。

```
=MIN(B2:B8)
```

问题3：获取销售提成，按销售额的1%计算，最大不得超过3000，最小不得低于1000。

```
=MIN(3000,MAX(1000,B2*1%))
```

首先将销售额乘以1%与1000比较，使用MAX函数获取最大值，当1%销售额低于1000时取1000，即给销售提成设置了下限。

将MAX函数返回的值与3000比较，使用MIN函数提取最小值，当MAX超过3000时取3000，即给提成设置了上限。

通过设置上下限，限制了提成在1000～3000之间。

这样说起来有点绕，其实Excel提供了一个可以返回中间值的函数MEDIAN。

```
=MEDIAN(1000,B2*1%,3000)
```

问题4：获取平均销售额。

```
=AVERAGE(B2:B8)
```

问题5：去除最大、最小值求平均值，在体育比赛中经常会出现。

```
=(SUM(B2:B8)-MAX(B2:B8)-MIN(B2:B8))/  
(COUNT(B2:B8)-2)
```

直接求和后依次减去最大值、最小值，然后除以（总数-2）个，这样就获取平均值。不过这样挺烦琐的，其实Excel提供了一个去除首尾的函数：TRIMMEAN。

```
=TRIMMEAN(B2:B8,2/COUNT(B2:B8))
```

函数语法：

```
=TRIMMEAN(区域,比例)
```

比如有10个值，去除1个最大值、1个最小值，也就是比例为0.2。如果去除2个最大值、2个最小值，也就是比例为0.4。也就是说去除的总数除以实际的数就是比例。

问题6：获取第二大销售额。

```
=LARGE(B2:B8,2)
```

问题7：获取倒数第二销售额。



```
=SMALL(B2:B8,2)
```

这两个函数语法一样，第一参数为区域，第二参数为N。N就是第几个的意思。很多时候我们需要的是将销售额从大到小进行降序排序，这时可以用：

```
=LARGE(B$2:B$8,ROW(A1))
```

ROW函数可以获取1~N的序号。如果需要从小到大升序排序，将LARGE函数换成SMALL函数即可。

5.4.5 日期与时间的处理

日期与时间可以说是与我们密切相关的问题，基本上每天都会有人问下面这些问题。

今天是几号？

今天是星期几？

这个月有没有31天？

离国庆节还有几天？

现在是几点？

.....

如图5-89所示，是某公司各人员销售清单。

	A	B	C	D	E	F	G
1	某公司各人员销售清单						
2					2009-1-10	星期六	
3							
4	日期	年	月	日	姓名	金额	
5	2009-1-1	2009	1	1	缪诗星	3200	
6	2009-1-2	2009	1	2	郭美珊	3200	
7	2009-1-2	2009	1	2	赵艳红	6400	
8	2009-1-4	2009	1	4	王茂东	840	
9	2009-1-5	2009	1	5	谷玉翔	17512	
10	2009-1-5	2009	1	5	李春华	12800	
11	2009-1-6	2009	1	6	赵艳红	6400	
12	2009-1-8	2009	1	8	张树森	6400	
13	2009-1-8	2009	1	8	李春华	3200	
14	2009-1-9	2009	1	9	周希涛	12800	
15	2009-1-9	2009	1	9	王慧颖	12800	
16	2009-1-9	2009	1	9	刘琼	3200	
17							

图5-89 某公司各人员销售清单

问题1：在制作表格的时候，都希望制表日期是动态的，而不用每天更改，最好还能显示星期几。

获取当前日期：

```
=TODAY()
```

如果需要精确到时分，可用：

```
=NOW()
```

精确到时分有一个好处，就是别人要问你时间的话，你可以轻松地得出，现在是北京时间9点或者9:30等。

```
=HOUR(NOW())&"点"  
=TEXT(NOW(),"h:mm")
```

将日期转换成星期几：

```
=TEXT(TODAY(),"aaaa")  
=TEXT(NOW(),"aaaa")
```

TEXT函数的第二参数是怎么得到的？其实这个函数跟自定义单元格格式很像，无须记忆，需要的时候自定义单元格格式，查看一下就知道。



如图5-90所示，将单元格格式设置为日期格式，选择“星期三”选项，观察示例的变化。




图5-90 设置单元格为星期几

如图5-91所示，选择“自定义”选项，就能查看设置为星期几的代码。




图5-91 星期几的代码



[\$-804]aaaa;@

[\$-804]代表中文简体，aaaa代表星期几，@代表文本。也就是说，如果是日期就返回中文简体的星期几，文本就显示本身。正常情况下，都直接写4个a就行。

对于喜欢用英文显示的朋友，也可以用dddd作为第2参数，这样就显示英文的星期几。如果要显示英文缩写，可用ddd，如图5-92所示。

效果	公式
Wednesday	=TEXT(TODAY(),"dddd")
Wed	=TEXT(TODAY(),"ddd")

图5-92 英文星期几

Excel中还有一个函数能返回星期几，比如星期六就返回数字6。

=WEEKDAY(TODAY(),2)

参数2有很多种选择，如图5-93所示，设置为2符合我们中国人的习惯用法：数字1（星期一）、数字7（星期日）。

RETURN_TYPE	返回的数字
1 或省略	数字 1 (星期日) 到 7 (星期六)。同 Microsoft Excel 早期版本。
2	数字 1 (星期一) 到 7 (星期日)。
3	数字 0 (星期一) 到 6 (星期日)。
11	数字 1 (星期一) 到 7 (星期日)。
12	数字 1 (星期二) 到数字 7 (星期一)。
13	数字 1 (星期三) 到数字 7 (星期二)。
14	数字 1 (星期四) 到数字 7 (星期三)。
15	数字 1 (星期五) 到数字 7 (星期四)。
16	数字 1 (星期六) 到数字 7 (星期五)。
17	数字 1 (星期日) 到 7 (星期六)。

图5-93 WEEKDAY参数2对应表

问题2：如何获取年月日？

=YEAR(A5)
=MONTH(A5)

=DAY(A5)

日期可以拆分，当然也就能组合。DATE函数可以将年月日重新组合成日期，如图5-94所示，计算2016年各月付款日。

	A	B	C
1	计算2016年各月付款日		
2	付款年份	付款月份	
3	2016	6	
4			
5			
6	付款明细	固定付款日	当月付款日
7	交燃气费	1	
8	交电话费	5	
9	交税	10	
10	发工资	15	
11			

图5-94 计算2016年各月付款日

使用公式为：

=DATE(\$A\$3,\$B\$3,B7)

这个函数还能求2月份有多少天，如图5-95所示。

	A	B	C
13			
14	求2月份有多少天		
15	年份	最后一天	2月有多少天
16	2016	2016-2-29	29
17			

图5-95 求2月份有多少天

最后一天的公式：

=DATE(A16,3,0)

2月有多少天的公式：

=DAY(B16)

有时候我们不确定2月有多少天，这里有个巧妙的做法，我们可以求3月份0号的值，DATE函数知道0号不存在，于是便返回3月1号前一天的日期，这样我们就可以求得2月



的最后一天，从而求得2月有多少天。

日期函数还有一个比较常用的函数EDATE。

我们经常会听到类似的话：

我两个月前去过广州玩。

我半年后要去潮州。

这合同还有一个月就到期。

.....

这时EDATE函数就派上用场，这个函数可以返回指定日期前（后）N个月的日期。如图5-96所示，获取每个合同的结束日期，每份合同的签订期为12个月。

	A	B	C
1	合同	开始日期	结束日期
2	合同1	2016-1-2	
3	合同2	2016-2-11	
4	合同3	2016-8-26	
5	合同4	2016-10-22	
6	合同5	2016-2-16	
7	合同6	2015-8-29	
8	合同7	2016-1-24	
9			

图5-96 合同表格

输入下面的公式，并将单元格设置为日期格式，如图5-97所示。

=EDATE(B2,12)




图5-97 获取结束日期

EDATE函数语法：

=EDATE(开始日期,之前/之后的月份)

参数2如果是负数就代表前N个月，是正数就代表后N个月。

如要获取十年前今日的日期，可用下面的公式，并将单元格设置为日期格式。

=EDATE(TODAY(),-120)

1年为12个月，十年就是120个月，之前的日期也就是-120。

知识扩展

用EDATE函数的时候需要设置单元格格式，其实日期格式也可以直接用TEXT函数来设置。

=TEXT(EDATE(TODAY(),-120),"e-m-d")

5.5 小试身手

学了一些函数的基本用法，卢子开始尝试帮网友解答问题，在帮助别人的过程中认识到自己有很多不足，通过慢慢地学习积累，卢子懂得的知识越来越全面。其实这就是传说中的助人式学习，对于巩固自己的知识非常有用。网友实际工作中会遇到各种各样的问题，而你在帮助他们解决问题的时候就会从多种角度考虑，这样会对Excel更加深刻。

5.5.1 加入第一帮派——ExcelHome

天下武功出少林，天下Excel技巧出ExcelHome。在我的印象中，每一个高手都曾经是ExcelHome会员，如果你是高手但不是ExcelHome会员的，请告诉我。

说句实话，现在的书籍很多都是枯燥无味的，能写得通俗易懂的基本没有，卢子利用两个月的时间还是没法看完整本书。有一天晚上，卢子在一个朋友的空间发现了ExcelHome这个论坛，出于好奇打开了论坛。在论坛的首页提供了好多的视频，卢子如获至宝，下载了一个晚上。

第二天就开始尝试看这些视频，不看还没什么，一看就觉得自己什么都不懂。不过还是勉强了一些，在视频结束的时候提到可以在论坛交流之类的话。为此卢子注册了一个账号——卢子。卢子→炉子，感觉这个比较容易记。这就是卢子名字的由来。注册后就在论坛到处闲逛，无意间发现有一些精华的帖子可以下载，这一发现可以说浪费了卢子无数个晚上的时间。卢子变成了疯狂的下载者，从这里下载，从那里下载，电脑的磁盘内容每日剧增。资料这么多，卢子自喜了好一段时间。以为有这些，以后就是高手，其实下载≠学习。有一天，卢子在逛论坛的时候，发现一个帖子“从疯狂的下载者变成真正的学习者”，一帖说到了卢子的心。下载一万个附件，还不如用心看好一个帖子，卢子一狠心将电脑上所有与Excel有关的资料全部删除，开始静下心来学习。当然这是很久以后的事儿。

卢子开始在ExcelHome潜水学习，这一潜水就是几个月，这期间功力大增。但独学而无友



是一件很痛苦的事情，卢子尝试在ExcelHome发表了一个帖子，感叹没有朋友这件事。没想到很快就得到ExcelHome站长的回复，鼓励卢子多跟大家交流，交流多了自然就会有朋友。看到老大的回复，心里很受鼓舞，从此走上另一种学习方式——助人式学习。

在论坛尝试回答了一些自己会的问题，逐渐懂得越来越多。也就是在这一年，加入了很多Excel交流群。

5.5.2 凭借VLOOKUP函数解疑难

截至目前，卢子虽然懂了不少函数，但真正研究透彻的就只有VLOOKUP函数。VLOOKUP函数是一个很神奇的函数，有了它查找数据变得非常轻松。

ExcelHome的函数版块每天都热闹非凡，提问者一个接一个，也正因为如此，才给卢子提供了一个大展身手的好机会。其实，准确点应该叫“Excel门弄斧”。

在这众多问题中，卢子只关注一类问题——查找。只要标题包含关键字“查找”，卢子就会打开链接，尝试解答。一个晚上解答三五个疑难也属于正常的事，随着解答问题的增多，卢子也越来越了解VLOOKUP函数。在2010年的某一天，卢子总结了一个帖子“VLOOKUP函数问题汇集”，里面涉及9个方面，常见的问题基本上都有，如图5-98所示。

VLOOKUP函数问题汇集

序号	内容
1	模糊查找与精确查找
2	处理错误值#N/A
3	利用错误值#N/A
4	查找的结果返回多列
5	反向查找1
6	反向查找2
7	多个条件查找
8	通配符查找
9	转置区域查找
10

图5-98 VLOOKUP函数问题汇集

卢子根据这个重新整理了8个最实用的问题，在这里跟大家一起分享。

(1) 如图5-99所示，根据番号精确查找俗称。

	A	B	C	D	E	F
1	番号	俗称		番号	俗称	
2	40411	H276		40403	H126	
3	40412	H276		40572	H142	
4	40571	H142		93657	R102	
5	40572	H142				
6	40401	H126				
7	40402	H126				
8	40403	H126				
9	43010	R102				
10	93657	R102				
11	93667	H216				
12	93668	H276				
13	40981	H276				
14	40983	H276				
15						

图5-99 根据番号精确查找俗称

=VLOOKUP(D2,A:B,2,0)

VLOOKUP函数语法：

=VLOOKUP(查找值,查找区域,返回查找区域第N列,查找模式)

VLOOKUP函数语法示意图如图5-100所示。






图5-100 VLOOKUP函数语法示意图

(2) 如图5-101所示, 屏蔽错误值查找。

	A	B	C	D	E	F
1	番号	俗名		番号	俗名	
2	40401	H126		40402		
3	40412	H276		40412	H276	
4	93657	R102		93656		
5	40572	H142		40572	H142	
6						

图5-101 屏蔽错误值查找

```
=VLOOKUP(D2,A:B,2,0)
```

VLOOKUP函数如果查找不到对应值会显示错误值#N/A, 这个看起来很不美观。这时可以在外面加容错函数IFERROR, 如果是2013版本就更好, 可以用IFNA函数, 它专门处理#N/A这种错误值。

```
=IFERROR(VLOOKUP(D2,A:B,2,0),"")
=IFNA(VLOOKUP(D2,A:B,2,0),"")
```

函数语法:

```
=IFERROR(表达式,错误值要显示的结果)
```

说白了就是将错误值显示成你想要的结果, 不是错误值就返回原来的值。IFNA函数的作用也是一样, 只是IFERROR函数是针对

所有错误值, 而IFNA函数只针对#N/A。

(3) 如图5-102所示, 按顺序返回多列对应值。

	A	B	C	D	E	F	G
1	番号	俗名	订单数	订单号	尾数	完成情况	
2	93657	R102	1500	100429	12	NG	
3	43010	R102	3000	100429	0	OK	
4	40981	H276	750	100517	0	OK	
5	93668	H276	750	100522	123	NG	
6	93667	H216	1200	100522	0	OK	
7	40406	H126	800	100602	0	OK	
8	40401	H126	1000	100601	0	OK	
9	40405	H126	200	100603	100	OK	
10	93667	H216	1200	100605	0	NG	
11							
12	番号	俗名	订单数				
13	93668	H276	750				
14	93667	H216	1200				
15	40406	H126	800				
16							

图5-102 按顺序返回多列对应值

通过上面的例子, 我们知道可以通过更改第3参数, 返回各项对应值, 例如:

```
=VLOOKUP($A13,$A$1:$F$10,2,0)
=VLOOKUP($A13,$A$1:$F$10,3,0)
```

如果项目少, 更改几次参数也没什么, 但项目多时, 肯定不方便。如图5-103所示, 可以通过ROW、COLUMN函数产生行列号, 从而得到1,2,...,n的值。

N	O	P	Q	R	S
	=COLUMN(A1)				
	1	2	3	4	5
=ROW(A1)	1				
	2				
	3				
	4				
	5				
	6				
	7				

图5-103 通过ROW、COLUMN函数产生行列号

```
=VLOOKUP($A13,$A$1:$F$10,COLUMN(B1),0)
```



因为这里是在同一行产生序号，所以用COLUMN函数。

(4) 如图5-104所示，按不同顺序返回对应值。

	A	B	C	D	E	F	G
1	序号	俗称	订单数	订单号	尾数	完成情况	
2	93657	R102	1500	100429	12	NG	
3	43010	R102	3000	100429	0	OK	
4	40981	H276	750	100517	0	OK	
5	93668	H276	750	100522	123	NG	
6	93667	H216	1200	100522	0	OK	
7	40406	H126	800	100602	0	OK	
8	40401	H126	1000	100601	0	OK	
9	40405	H126	200	100603	100	OK	
10	93667	H216	1200	100605	0	NG	
11							
12	序号	订单数	俗称				
13	93668	750	H276				
14	93667	1200	H216				
15	40406	800	H126				
16							

图5-104 按不同顺序返回对应值

这回看来只能手动更改第3参数了，COLUMN函数完全派不上用场。

NO! 每当你觉得操作烦琐时，就要停下来思考，也许Excel本身存在这个功能，只是自己一时想不到或者不知道而已。列号不管千变万化，在数据源中的位置始终不变，利用这个特点可以去搜索一下看看用什么函数可以解决。

如图5-105所示，在“搜索函数”文本框中输入“位置”，单击“转到”按钮，就会出现与位置有关的函数。查看每个函数的说明，找到我们需要的，如MATCH函数，返回符合特定值特定顺序的项在数组中的相应位置，单击“确定”按钮。

如图5-106所示，在弹出的“函数参数”对话框中尝试填写相应的参数，每个参数的作用下面都有相关说明，填写后会出现计算结果3，也就是订单数在区域中是第3列。尝试更改第1参数为C12（俗称），计算结果

是2，也就是区域中第2列。经过尝试，知道这个函数是我们要找的那个函数，单击“取消”按钮，返回工作表。




图5-105 搜索函数的用法



图5-106 填写相应的参数

如图5-107所示，在单元格再做最后一次验证。

B17	:	X	✓	fx	=MATCH(B\$12,\$A\$1:\$F\$1,0)
12	序号	订单数	俗称		
16					
17		3	2		
18					

图5-107 验证MATCH函数用法



到这一步已经十拿九稳了，将公式设置为：

```
=VLOOKUP($A13,$A$1:$F$10,MATCH(B$12,$A$1:$F$1,0),0)
```

(5) 如图5-108所示，根据番号逆序查找俗称。

	A	B	C	D	E
1	俗称	番号	订单数		
2	R102	93657	1500		
3	R102	43010	3000		
4	H276	40981	750		
5	H276	93668	750		
6	H216	93667	1200		
7	H126	40406	800		
8	H126	40401	1000		
9	H126	40405	200		
10	H216	93667	1200		
11					
12	番号	俗称			
13	93657	R102			
14	93668	H276			
15	93667	H216			
16					

图5-108 根据番号逆序查找俗称

帮助文件提到VLOOKUP函数只能按首列查找，不能逆向查找，既然如此，那就得想办法将非首列的区域转换成首列。怎么转换区域呢？这时IF函数就派上用场，一步步来了解IF函数的转换。

看看好友传递如何趣聊IF函数，吃货的福音。

IF函数其实只有一个参数来判断是否符合条件，返回FALSE或TRUE两种结果。

如图5-109所示，当菜只有甜的或咸的两种口味时，甜味是红烧肉，咸味是酱油肉。

	A	B	C
1	咸的	甜的	
2	酱油肉	红烧肉	
3			

图5-109 两种口味的菜

盲人吃饭时，看不到是什么菜。当别人问盲人：“你现在吃的什么菜？是咸的吗？如果是咸的，就是酱油肉，如果不是咸的，就是红烧肉。”（给定判断条件：咸味）盲人刚好在吃红烧肉，于是就咂着嘴说：“嗯，好吃，不是咸的！是红烧肉。”（根据提问的要求，不符合咸的）假如盲人当时在吃酱油肉，一定回答：“是的，咸的，是酱油肉。”（条件为真，是！TRUE）。盲人根据口感，结合提问者说的条件，就知道自己吃的是红烧肉还是酱油肉了。

把这段话用公式来写：

```
=IF(A1="咸的",A2,B2)
```

翻译：是咸的吗？要是（TRUE），就是酱油肉，要不是咸的（FALSE），就是甜的红烧肉。

A1="咸的"这个条件也可以直接换成TRUE或者FALSE。

```
=IF(TRUE,A2,B2)
```

因为满足条件，所以返回A2的对应值酱油肉。

```
=IF(FALSE,A2,B2)
```

因为不满足条件，所以返回B2的对应值红烧肉。



因为TRUE=1, FALSE=0, 所以可以直接用1跟0表示。

```
=IF(1,A2,B2)
=IF(0,A2,B2)
```

IF函数不仅可以返回一个单元格的值，还可以返回多个单元格的值。

```
=IF({1,0},A2,B2)
=IF({0,1},A2,B2)
```

选择两个单元格输入，按Ctrl+Shift+Enter组合键结束。条件为{1,0}，返回A2:B2的对应值，顺序不变；条件为{0,1}，返回A2:B2的对应值，顺序对换。也就是说，通过改变1跟0的位置，可以调换两单元格的前后位置，如图5-110所示。

正常顺序	酱油肉	红烧肉
逆序	红烧肉	酱油肉

图5-110 调换单元格

看到这里，知道IF函数通过改变1,0可以调换单元格的顺序，改变区域的顺序也是可以实现。

用IF函数重新构造的区域，如图5-111所示，是多单元格数组公式，记得按Ctrl+Shift+Enter组合键结束，否则会出错。关于数组公式更详细的知识参考第6章和第7章。

新区域：

```
=IF({1,0},B2:B10,A2:A10)
```

所以公式可以变成：

```
=VLOOKUP(A13,新区域,2,0)
```

两个公式合并，大功告成。

```
=VLOOKUP(A13,IF({1,0},B2:B10,A2:A10),2,0)
```




E2	:				{=IF({1,0},B2:B10,A2:A10)}	
	A	B	C	D	E	F
1	俗称	番号	订单数			
2	R102	93657	1500		93657	R102
3	R102	43010	3000		43010	R102
4	H276	40981	750		40981	H276
5	H276	93668	750		93668	H276
6	H216	93667	1200		93667	H216
7	H126	40406	800		40406	H126
8	H126	40401	1000		40401	H126
9	H126	40405	200		40405	H126
10	H216	93667	1200		93667	H216

图5-111 重新构造的区域

(6) 如图5-112所示，根据俗称和订单号两个条件查询完成情况。

	A	B	C	D	E	F
1	俗称	订单号	订单号	尾数	完成情况	
2	R102	1500	100429	12	NG	
3	H276	750	100517	0	OK	
4	H276	750	100522	123	NG	
5	H216	1200	100522	0	OK	
6	H126	800	100602	0	OK	
7	H126	1000	100601	0	OK	
8	H126	200	100603	100	OK	
9	H216	1200	100605	0	NG	
10						
11	俗称	订单号	完成情况			
12	R102	100429	NG			
13	H276	100522	NG			
14	H126	100603	OK			
15						

图5-112 根据俗称和订单号两个条件查询完成情况

正常情况下，VLOOKUP函数不能实现多条件查询，通过IF函数的学习，我们知道IF函数可以重新构造区域，这里就再次用IF构成一个区域，如图5-113所示。



新区域:

```
=IF({1,0},A2:A9&C2:C9,E2:E9)
```

所以公式可以变成:

```
=VLOOKUP(A12&B12,新区域,2,0)
```

两个公式合并,大功告成,记得按Ctrl+Shift+Enter组合键结束。

```
=VLOOKUP(A12&B12,IF({1,0},$A$2:$A$9&$C$2:$C$9,$E$2:$E$9),2,0)
```

(7) 如图5-114所示,根据俗称的第一个字符查找番号。

G2					
=IF({1,0},A2:A9&C2:C9,E2:E9)					
	A	B	C	D	E
1	俗称	订单数	订单号	尾数	完成情况
2	R102	1500	100429	12	NG
3	H276	750	100517	0	OK
4	H276	750	100522	123	NG
5	H216	1200	100522	0	OK
6	H126	800	100602	0	OK
7	H126	1000	100601	0	OK
8	H126	200	100603	100	OK
9	H216	1200	100605	0	NG

图5-113 重新构造区域

	A	B	C	D	E	F
1	俗称	番号		俗称	番号	
2	MC	48897		M	48897	
3	LT-R	48906		R	93657	
4	R102	93657				
5	H142	40572				
6						

图5-114 根据俗称的第一个字符查找番号

```
=VLOOKUP(D2&"*",A:B,2,0)
```

星号(*)是通配符,代表所有字符,问号(?)代表一个字符。D2&"*"就是开头包含D2的意思。

(8) 如图5-115所示,根据区间判断成绩的等级。

	A	B	C	D	E	F	G
1		区间	等级		成绩	等级	
2	1	1-60	差		65	中	
3	61	61-70	中		90	优	
4	71	71-85	良		35	差	
5	86	86-100	优				
6							

图5-115 根据区间判断成绩等级

借助辅助列的话,很容易查询等级,只需将VLOOKUP函数的第四参数设置为1或者省略即可。

```
=VLOOKUP(E2,A:C,3)
```

如果不用辅助列,需结合前面所有函数知识才能完成,有兴趣的朋友可以去研究。

```
=VLOOKUP(E2,IF({1,0},--LEFT(B$2:B$5,FIND("-",B$2:B$5)-1),C$2:C$5),2)
```



通过这8个疑难，基本的查询问题都能够解决。



知识扩展

几年以后卢子无意间发现了IMREAL函数，所以不用辅助列的数组公式可以稍微简单一点。
`=VLOOKUP(E2,IF({1,0},IMREAL(B$2:B$5&"")),C$2:C$5),2)`



5.5.3 LOOKUP函数的二分法只是个传说

在“VLOOKUP函数问题汇集”这个帖子发出去当天，帖子上升为24人气榜首。接着看到，下载量超过1000次。这对于名人而言，或许没有什么，但对于初出茅庐的卢子而言，却是莫大的鼓舞。帖子的回复中有一个人叫周义坤，一看资料是老乡，所以留了点心。

卢子又像往常一样回答有关查找的问题，依然使用VLOOKUP函数解决，而当卢子再次查看帖子的时候，后面总有一个叫周义坤的人用LOOKUP函数提供另一种答案。连续好多问题都是这样，也许这就是缘分吧。

周义坤首先在论坛加卢子为好友，卢子同意并用QQ添加他为好友。两人开始在QQ聊起关于查找的问题，他反复强调LOOKUP函数好用，让卢子一定要去学习，并推荐一个帖子“VLOOKUP与LOOKUP——过招”。其实，卢子与周义坤不就是在上演一场真实版的两函数过招吗？

卢子抽空看了这个帖子，很受启发，同时查找了很多有关LOOKUP函数的资料，开始学习起来。不过在这些帖子中提到了二分法，这个说法不知道吓怕了多少初学者。卢子也是其中一员，所以在很长的一段时间，对这个函数依然不太了解，甚至有意逃避它。直到有一天跟提出这个二分法的函数高手gouweicao78聊天，才知道这个二分法可以不掌握，这只不过是个传说，即使不会照样可以学好LOOKUP函数。聊天的时候还涉及LOOKUP函数的另一种方法——经典查找模式，这个实在太好用了，有了它，查找就是这么简单。

gouweicao78是谁？有函神之称（Excel函数之神），跟另一个人山菊花合称花草，ExcelHome的两大高手。gouweicao78的帖子富有逻辑性，山菊花的帖子幽默有趣，各有其优点。他们两人的精华帖，卢子基本上都用心看了多次。

扯远了，继续聊LOOKUP函数的话题。gouweicao78就是靠着LOOKUP函数而一举成名的。学习LOOKUP函数看他的帖子最好。不要迷恋二分法，二分法只是一个传说。抛开这个传说，一起来见证LOOKUP函数的神奇。

这个LOOKUP函数有什么好学的，帮助都提到，如果区域没有升序可能会导致出错，既然这样，其作用明摆着就很小。



LOOKUP函数喜欢以大欺小，用1查找0，也就是查找小于或者等于它的值，因为在查找的时候，自动忽略错误值，所以能找到0，如图5-119所示。

=LOOKUP(1,{#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;0;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!;#DIV/0!},\$B\$2:\$B\$14)

图5-119 用1查找0



LOOKUP函数在查找的时候，如果有多个对应值，会返回最后一个满足条件的对应值。也就是说用1查找0时，如果有多个0，会查找到最后一个0所对应的值。例如，用俗称来查找番号，H276对应了多个番号，但查找的结果只返回最后一个，也就是40983，如图5-120所示。

The diagram shows two tables side-by-side. The left table has columns A (俗称) and B (番号). The right table has columns G (俗称) and H (番号). A red dashed arrow points from the formula bar, which contains `=LOOKUP(1,0/(A2:A14=D2),B2:B14)`, to the right table's row 14. In this row, column D (H276) matches column G (H276), so the result in column H is 40983. A cloud-shaped callout box contains the text "将公式转换成单元格理解公式".

	A	B	D	E	G	H
1	俗称	番号	俗称	番号	俗称	番号
2	H276	40411	H276	40983	0	40411
3	H276	40412			0	40412
4	H142	40571			#DIV/0!	40571
5	H142	40572			#DIV/0!	40572
6	H126	40401			#DIV/0!	40401
7	H126	40402			#DIV/0!	40402
8	H126	40403			#DIV/0!	40403
9	R102	43010			#DIV/0!	43010
10	R102	93657			#DIV/0!	93657
11	H216	93667			#DIV/0!	93667
12	H276	93668			0	93668
13	H276	40981			0	40981
14	H276	40983			0	40983

将公式转换成
单元格理解公式

图5-120 查找最后满足条件的值

如果还不理解，没关系，先记住这个模式！以后用多了自然会理解，就如书读百遍，其义自见一样。

(2) 如图5-121所示,屏蔽错误值查找。

	A	B	C	D	E	F
1	番号	俗称		番号	俗称	
2	40401	H126		40402		
3	40412	H276		40412	H276	
4	93657	R102		93656		
5	40572	H142		40572	H142	
6						

图5-121 屏蔽错误值查找

=IFERROR(LOOKUP(1,0/(\$A\$2:\$A\$5=D2),\$B\$2:\$B\$5),"")

这个与VLOOKUP函数一样，都是用IFERROR函数进行容错处理。

(3) 如图5-122所示, 按顺序返回多列对应值。



	A	B	C	D	E	F	G
1	番号	俗称	订单数	订单号	尾数	完成情况	
2	93657	R102	1500	100429	12	NG	
3	43010	R102	3000	100429	0	OK	
4	40981	H276	750	100517	0	OK	
5	93668	H276	750	100522	123	NG	
6	93667	H216	1200	100522	0	OK	
7	40406	H126	800	100602	0	OK	
8	40401	H126	1000	100601	0	OK	
9	40405	H126	200	100603	100	OK	
10	93667	H216	1200	100605	0	NG	
11							
12	番号	俗称	订单数				
13	93668	H276	750				
14	93667	H216	1200				
15	40406	H126	800				
16							

图5-122 按顺序返回多列对应值

=LOOKUP(1,0/(\$A\$2:\$A\$10=\$A13),B\$2:B\$10)

利用混合引用的特点，将B\$2:B\$10向右拖动，就变成C\$2:C\$10，从而改变返回区域。

(4) 如图5-123所示，按不同顺序返回对应值。

	A	B	C	D	E	F	G
1	番号	俗称	订单数	订单号	尾数	完成情况	
2	93657	R102	1500	100429	12	NG	
3	43010	R102	3000	100429	0	OK	
4	40981	H276	750	100517	0	OK	
5	93668	H276	750	100522	123	NG	
6	93667	H216	1200	100522	0	OK	
7	40406	H126	800	100602	0	OK	
8	40401	H126	1000	100601	0	OK	
9	40405	H126	200	100603	100	OK	
10	93667	H216	1200	100605	0	NG	
11							
12	番号	订单数	俗称				
13	93668	750	H276				
14	93667	1200	H216				
15	40406	800	H126				
16							

图5-123 按不同顺序返回对应值

这个就有点麻烦，怎样才能确定返回的区域呢？

用OFFSET函数可以通过偏移得到一个区域，比如求订单数就是将A列偏移2列即可得到。

=OFFSET(A2:A10,0,2)

求俗称就是将A列偏移1列即可得到。

=OFFSET(A2:A10,0,1)

前面提到MATCH函数可以获取字段的排位，通过排位就可知道需要偏移多少列。

=MATCH(B\$12,\$A\$1:\$F\$1,0)

如订单数是排第3位，却只需偏移2列就可以得到，也就是说排位-1就得到偏移列。

=MATCH(B\$12,\$A\$1:\$F\$1,0)-1

这样返回区域就确定下来了。

=OFFSET(\$A\$2:\$A\$10,,MATCH(B\$12,\$A\$1:\$F\$1,0)-1)

套用以大欺小法，得到：

=LOOKUP(1,0/(\$A\$2:\$A\$10=\$A13),OFFSET(\$A\$2:\$A\$10,,MATCH(B\$12,\$A\$1:\$F\$1,0)-1))

其实使用公式说难也难，说易也易。只要掌握窍门，就跟堆积木一样，将一个积木组合成你需要的模型。

(5) 如图5-124所示，根据番号逆序查找俗称。

	A	B	C	D	E
1	俗称	番号	订单数		
2	R102	93657	1500		
3	R102	43010	3000		
4	H276	40981	750		
5	H276	93668	750		
6	H216	93667	1200		
7	H126	40406	800		
8	H126	40401	1000		
9	H126	40405	200		
10	H216	93667	1200		
11					
12	番号	俗称			
13	93657	R102			
14	93668	H276			
15	93667	H216			
16					

图5-124 根据番号逆序查找俗称



在LOOKUP函数的字典中，没有“逆向”这个词。不管什么方向，对LOOKUP函数都一样。

```
=LOOKUP(1,0/($B$2:$B$10=A13),$A$2:$A$10)
```

(6) 如图5-125所示，根据俗称和订单号两个条件查询完成情况。

	A	B	C	D	E	F
1	俗称	订单号	订单号	尾数	完成情况	
2	R102	1500	100429	12	NG	
3	H276	750	100517	0	OK	
4	H276	750	100522	123	NG	
5	H216	1200	100522	0	OK	
6	H126	800	100602	0	OK	
7	H126	1000	100601	0	OK	
8	H126	200	100603	100	OK	
9	H216	1200	100605	0	NG	
10						
11	俗称	订单号	完成情况			
12	R102	100429	NG			
13	H276	100522	NG			
14	H126	100603	OK			
15						

图5-125 根据俗称和订单号两个条件查询完成情况

同样多条件和单条件对LOOKUP函数而言都是一样的，一个通式闯天下，只需将条件用“&”组合起来即可。

```
=LOOKUP(1,0/($A$2:$A$9&$C$2:$C$9=A12&B12),$E$2:$E$9)
```

当然也可以直接套用通用公式，多个条件用“*”连接起来。

```
=LOOKUP(1,0/((($A$2:$A$9=A12)*($C$2:$C$9=B12)),$E$2:$E$9)
```

(7) 如图5-126所示，根据俗称的第一个字符查找番号。

	A	B	C	D	E	F
1	俗称	番号		俗称	番号	
2	MC	48897		M	48897	
3	LT	48906		R	93657	
4	R102	93657				
5	H142	40572				
6						

图5-126 根据俗称的第一个字符查找番号

从左边提取字符用LEFT函数，第一个也就是提取1位。

```
=LEFT(A2,1)
```

如果仅仅是提取1位，这个1可以省略，默认情况下就是1。

```
=LEFT(A2)
```

组合起来就是：

```
=LOOKUP(1,0/(LEFT($A$2:$A$5)=D2),$B$2:$B$5)
```


最后记住这个通用公式：

```
=LOOKUP(1,0/((条件1)*(条件2)*...*(条件n)),返回区域)
```

有了这个经典查找的通用公式，什么查找都变得简单了！

通过学习两大查找函数LOOKUP与VLOOKUP，确实让卢子在工作上风光了一把，但学习仍需继续。





知识扩展

有的时候查找的值并不都是在开头，而是位置不确定，只是包含在里面。如根据镇流器ID查询数量，多个ID用/隔开，输入其中一个ID查找数量，这时就不能用LEFT函数，要用FIND函数进行判断是否包含在里面，如图5-127所示。

E2 : =LOOKUP(1,0/FIND(D2,\$A\$2:\$A\$17), \$B\$2:\$B\$17)

	A	B	C	D	E	F
1	镇流器ID	数量		镇流器ID	数量	
2	2060300297	41		1010210060	39	
3	2060301946	100		2060310942	30	
4	2060310892	100		2060301760	27	
5	2060301739/1010210060	39				
6	2060301760	27				
7	2060311396	20				
8	2060301205	50				
9	2060311198	50				
10	2060300993	126				
11	2060310110	40				
12	2060300075	40				
13	2060301203	30				
14	2060311503	20				
15	2060310255	48				
16	2060310942/2060310062	30				
17	2060300057/2060300058	26				

图5-127 包含查询

5.5.4 七夕闲聊

明天就是七夕了，这个时候除了爱情的事儿，聊别的都感觉不合时宜。

卢子从来不相信一见钟情，所谓的一见钟情不过就是“一见钟脸”而已，而日久生情更符合实际情况。感情这事儿挺奇妙，有人因为感情毁了一生，有人因为感情成就了一生。每个成功男人的背后都有一个默默无闻的女人，可见女人对男人影响很大！当然我现在还不能算男人，只能说是个男孩，呵呵。

据说每个Excel学得好的人，都有一段不为人知的往事，而卢子的往事却是因为感情。以前读书时只顾读书，俨然一书呆子，读了十多年书居然没谈过恋爱，挺悲剧的。直到工作两年的时候，才谈了一场恋爱。这段感情的时间不算很长，但深深地改变了卢子。

那女孩叫F，在她还没开始上班的时候，香姐就跟卢子说这人挺可爱的，不知道是有意还是无意这么说。不管怎么样，卢子把这话记在心中。等到F来上班的时候，卢子特意多看了她



两眼，感觉也没什么吸引人的。可能人需要对比才能看得出来，就像电影《唐伯虎点秋香》中的秋香，唐伯虎在没见到秋香之前，别人都跟他说她是美女，当他第一次见到时，感觉很一般。最后他的朋友说：你需要跟其他女孩一起对比看。其他女孩长得实在不敢恭维，这时很普通的秋香就变成一美女。

继续说F的事儿。跟F一起来上班的还有Z，两个人一对比，优缺点就摆在眼前。对这事卢子本来也没太当回事，可是令人意想不到的是往后连续几天下班后的巧遇。正因为这几次巧遇，让卢子跟F越来越熟。

卢子尝试第一次约F出来，没想到没有听到任何推托的话语，真的是太顺利。我们开始逛街，聊一些可有可无的话儿，有人说过，正因为有这些废话存在，才大大增进人的感情。有了第一次就有第二次，第三次……对每一次约会卢子都非常珍惜。事情来得太顺利也许并非好事，某一天，公司内部聚餐，发生了一件不愉快的事情。也正因为这件事情，卢子伤心地提前离开，关掉手机，不让任何人联系到。当第二天打开手机看到F发来的很多短信，卢子回了一句：“我们分手吧。”卢子回了这句，心都在滴血。

.....

为了忘掉这段感情，卢子一有空闲就到处游玩，在短短的一个月内，卢子玩遍了大半个东莞。不过这样终究不是办法，人总得积极进取。也正因为如此，卢子把业余的90%的时间用在了学习Excel上。每天不断学习，不断交流，直到把这段感情深藏内心。不管初恋怎么样，都是无法忘掉的。也许就是这个原因，使卢子的Excel操作水平突飞猛进。

聊了那么多，突然有点伤感。还是不说了，回到Excel话题中。

人最难忘记的往往是第一次和最后一次，例如在Excel中，如何分别获取人员呢？如图5-128所示。

	A	B	C	D
1	关系	人员		第一次喜欢的人
2	喜欢	F		F
3	一般	Z		
4	好感	K		最后一次喜欢的人
5	喜欢	S		H
6	好感	B		
7	喜欢	H		
8				

图5-128 查找第一次和最后一次喜欢的人

如果找不到正确的方法，此题很难。一旦了解VLOOKUP和LOOKUP两个函数，解这题轻而易举。当有多个对应值，VLOOKUP函数是查找首次出现的对应值，LOOKUP函数是查找最后一次出现的对应值，利用这个特点，可以将公式设置为：

```
=VLOOKUP("喜欢",A2:B7,2,0)
```



```
=LOOKUP(1,0/(A2:A7="喜欢"),B2:B7)
```

如图5-129所示，用我的真心编写六国语言，只想对你说，我爱你。不过这话截止到目前，卢子还没对任何女孩子说过，只能藏在内心深处。

D2		=INDEX(A1:A6,RANDBETWEEN(1,6))			
	A	B	C	D	E
1	あいしてる	日		我很想对你说：	
2	I love you	英		52946 45230 50403	
3	Scusa ma ti chiamo amore	法			
4	사랑해요	韩			
5	我爱你	中			
6	52946 45230 50403	E			
7					

图5-129 六国语言

5.5.5 随便VS随机

“随便”这个词我们会经常听到，比如你问朋友想吃什么，她往往回答“随便”。又如，你问她想看什么电影，她的回答依然是“随便”。

有些人就会由“随便”联想到“随机”，但它们俩真的不是一回事！“随机”在很多时候体现的是一种专业。卢子从事品质检查，经常会利用AQL对产品进行随机抽样，也就是根据一个标准从总体中抽取一小部分进行检查。这时你不能随便，那样犯错误的可能性非常大。有经验的人会进行随机抽样，比如对每个卡板的产品从各个方向抽取部分产品；又或者每过一个时间段，从生产的产品中挑选部分进行检验。

在Excel中有两个随机函数：RANDBETWEEN和RAND。在你犹豫的时候，可以让它们帮你做决定。如果现在有十个产品，编号为1到10，你想随机挑选一个的话，就可以用RANDBETWEEN，这个函数返回位于两个指定数之间的一个随机整数，其语法如下：

```
=RANDBETWEEN(起始值,终点值)
```

因为是从1到10，所以可以设置为：

```
=RANDBETWEEN(1,10)
```

前面提到F9键可以解读公式，其实它还可以让公式重新计算。轻按几下F9键，从中选择一个产品作为抽样的对象。

如果知道产品的名称为A,B,C,D,E,F,G,H,I,J，想从中挑选三个怎么办呢？RANDBETWEEN函数只能产生随机整数，即使可以产生字母，也有可能出现重复。因为是随机数，不可能



保证在这么小的范围内，所有的数字都不重复。

有句话叫作：正难则反。正常的办法很难做到，我们就可以变通下。先把产品名称罗列在A列，然后在B列利用RAND函数生成随机数，最后进行排序，从中选择前三个即可。RAND这个函数没有参数，其作用就是产生0~1之间的随机数，如图5-130所示。

B2 : X ✓ fx =RAND()				
	A	B	C	D
1	产品	随机数		
2	A	0.947655		
3	I	0.393443		
4	F	0.190608		
5	C	0.689195		
6	G	0.945117		
7	E	0.801261		
8	D	0.952384		
9	J	0.660444		
10	H	0.298778		
11	B	0.047426		
12				

图5-130 利用随机数排序

也许有人会问，如果产品名为A~Z，一共26个字母，手工输入很麻烦，有没有更加快捷的方法？答案是肯定的，不过先不急，先来看三个数字密码。52946 45230 50403，如果把这三个数字分开放，也许没人知道含义，但在情人节这一天说，即使是傻瓜都能大概猜到意思。如果你想从谐音上获取含义，根本就破解不了。

一起来认识CODE和CHAR两兄弟，两者如影随形，有其中一个就会有另外一个。每一个字符都有唯一的数字码，而每个

数字码都能组成唯一的字符。52946 45230 50403，这三个就是数字码，通过这些可以找到由这三个数字码组成的字符，如图5-131所示。

A2 : X ✓ fx =CHAR(A1)				
	A	B	C	D
1	52946	45230	50403	
2	我	爱	你	
3				

图5-131 CHAR的用法

数字码通过CHAR函数可以转换成字符，那么字符就可以通过CODE函数转换成数字码。如图5-132所示，知道A的数字码为65，Z的数字码为90，每个字母之间的数字码都是递增的，步长值为1。

B2 : X ✓ fx =CODE(A2)				
	A	B	C	D
1	字母	数字码		
2	A	65		
3	B	66		
4	Z	90		
5				

图5-132 CODE函数的用法

前面已经提过ROW函数可以产生行号，COLUMN函数可以产生列号。在这里就派上用途了。利用ROW函数可以产生65到90的数字码。

```
=CHAR(ROW(A65))
```

知道了字母的对应数字码，现在如果要随机生成A~Z就变得很简单，无须用辅助列，直接输入公式即可。

```
=CHAR(RANDBETWEEN(65,90))
```





知识扩展

年终时，有些企业会组织职工联欢会，可能还会搞个抽奖的小节目增添乐趣，下面就介绍一个用Excel做的简单的抽奖方法。如图5-133所示，先将人员的名字制作成表格。

	A	B	C	D
1	姓名			中奖者3名
2	王建军			
3	陈锡卢			
4	郑晓芬			
5	李应钦			
6	邱显标			
7	龚思兰			
8	邓丹			
9	陆超男			
10	杨明辉			
11	刘宋连			
12	邓海南			
13				

图5-133 人员

在B2单元格中输入公式，并下拉填充公式。

=RAND()

在D2单元格中输入公式，并下拉填充公式。

=INDEX(\$A\$2:\$A\$12,RANK(B2,\$B\$2:\$B\$12))

完成操作后，按F9键就可以进行抽奖，如图5-134所示。




D2	:				=INDEX(\$A\$2:\$A\$12,RANK(B2,\$B\$2:\$B\$12))		
	A	B	C	D	E	F	G
1	姓名			中奖者3名			
2	王建军	0.384393		陆超男			
3	陈锡卢	0.122791		邓海南			
4	郑晓芬	0.720962		陈锡卢			
5	李应钦	0.643105					
6	邱显标	0.368278					
7	龚思兰	0.54225					
8	邓丹	0.493997					
9	陆超男	0.277112					
10	杨明辉	0.99209					
11	刘宋连	0.624535					
12	邓海南	0.702741					
13							

图5-134 抽奖



5.5.6 四舍五入，卢子已奔三

网友 如图5-135所示，用ROUND函数对数据进行四舍五入，怎么统一一位数，即让12显示12.0？

	A	B	C	D	E
1	数据	实际效果		理想效果	
2	12	12		12.0	
3	33.5	33.5		33.5	
4	12.45	12.5		12.5	
5	32.11	32.1		32.1	
6					

图5-135 对数据进行四舍五入并统一一位数

卢子 可以用TEXT函数取代ROUND函数。

```
=TEXT(A2,"0.0")
```

TEXT函数的功能其实跟自定义单元格格式有点相似，先在“设置单元格格式”对话框中，设置“小数位数”为小数点后1位，如图5-136所示。




图5-136 更改小数点位



选择“自定义”选项查看代码，这时显示0.0_，如图5-137所示。




图5-137 自定义单元格获取格式代码

这个下划线“_”有什么作用呢？其实是显示一个和下划线一样长度的空白，当然也可以直接用空白代替。看到这里知道这个“_”在这里没有作用，可以去掉。也就是最终的自定义格式代码为0.0。

为什么不使用自定义单元格格式而用TEXT函数代替呢？自定义单元格格式只是欺骗我们的眼睛罢了，实际数据的性质并没有改变。用TEXT函数得到的是真正的数据。这就像化妆（自定义单元格格式）与整容（TEXT），化妆后看起来很好看，卸妆后依然还是本来的相貌，没有任何改变。整容后实际相貌已发生改变。

聊到四舍五入，突然想起了一个问题，明天就是卢子的生日，也就是说卢子已26周

岁了，四舍五入已奔三。岁月不饶人啊！

也许有人会说，这样也行？

Excel可以这样四舍五入，当然现实也可以这么用。

ROUND函数的第二参数，不仅可以是正数，也可以是零，甚至是负数。-1就是对个位进行四舍五入后进位。6大于5，也就是进1，如图5-138所示。

奔三的算法

$$26 \text{ --- } \rightarrow 30 = \text{ROUND}(A1, -1)$$

图5-138 ROUND参数设置

如果换成-2的话，卢子就返老还童了，重新走进娘胎里。十位数为2，小于5，直接舍去，就等于0。

=ROUND(A1,-2)

5.5.7 数字舍入，存在即合理

说到ROUND函数，不得不说它的两兄弟：ROUNDUP函数和ROUNDDOWN函数。既然是三兄弟，那么语法也就一样，只是作用稍有差别而已。UP就是向上的意思，DOWN就是向下的意思，也就是说，ROUNDUP就是向上舍入数字，ROUNDDOWN就是向下舍入数字。

李云龙用下面的公式测试卢子的年龄，如果实际是这么多岁，卢子已成仙。

=ROUNDUP(A1,-2)

-2也就是对十位上的数字进行判断，UP不管什么数都只进位，不舍去。也就是说，卢子已成百岁仙人。

同理，如果用DOWN，不管什么数都只舍不进，卢子也就还没出生。

=ROUNDDOWN(A1,-2)

这两个函数有什么用呢？

比如公司最近生意好，老板在发工资的时候就大方点，是2120元就发2200元。百元以下通通进位，这时ROUNDUP就派上用场，只需把第二参数改成-2就行。又如别人在买东西，126元就算120元，零头不要，这时就设置ROUNDDOWN的第二参数为-1，如图5-139所示。

接着看另外一种舍入，在很多时候，称重量并不需要非常准确，尾数小于0.25就算0，大于等于0.75则算1，其余就算0.5。也就是说，最小精确度为半斤，不足半斤按数量

的大小进行舍入。

	A	B	C
1	发工资		
2	金额	显示	公式
3	2130	2200	=ROUNDUP(A3,-2)
4	2230	2300	=ROUNDUP(A4,-2)
5	3490	3500	=ROUNDUP(A5,-2)
6	2300	2300	=ROUNDUP(A6,-2)
7			
8	买东西		
9	金额	显示	公式
10	126	120	=ROUNDDOWN(A10,-1)
11	125	120	=ROUNDDOWN(A11,-1)
12	221	220	=ROUNDDOWN(A12,-1)
13	123	120	=ROUNDDOWN(A13,-1)
14			

图5-139 ROUNDUP函数和ROUNDDOWN函数的比较

粗略这么一看，上面的三兄弟都派不上用场，难道还存在第4个这样的函数？卢子学Excel这么多年了，至今没发现过，也许在未来的某个版本中就会出现也说不定。

仔细观察，0.25跟0.75两个数字乘以2，尾数都是0.5，这样就可以用ROUND进行四舍五入。当然有借有还，现在借了*2，自然要/2，这样才能回到原来，如图5-140所示。

B2	:	x	✓	f _x	=ROUND(A2*2,0)/2
	A	B	C	D	E
1	数据	实际效果			
2	12.25	12.5			
3	33.5	33.5			
4	12.45	12.5			
5	32.76	33			
6					

图5-140 ROUND的特殊用法





知识扩展

数字舍入除了ROUND函数三兄弟外，还有三个常用的函数，分别是INT、CEILING和FLOOR。

INT函数就是将数字舍掉小数部分，保留整数，如图5-141所示。

=INT(A2)

	A	B
1	数据	效果
2	12.25	12
3	33.5	33
4	12.45	12
5	32.76	32
6	1.74	1
7	1.56	1
8		

图5-141 INT函数的用法

CEILING函数是向上舍入指定基数的倍数，而FLOOR函数刚好相反，是向下舍入指定基数的倍数，如图5-142所示，是将基数设置为0.5跟1，两个函数的对比效果。

	A	C	D	E	F
1	数据	基数0.5		基数1	
2	12.25	12.5	12	13	12
3	33.5	33.5	33.5	34	33
4	12.45	12.5	12	13	12
5	32.76	33	32.5	33	32
6	1.74	2	1.5	2	1
7	1.56	2	1.5	2	1
8					
9	=CEILING(A2,基数)				
10	=FLOOR(A2,基数)				
11					

图5-142 CEILING和FLOOR函数对比

5.6 小结与练习

卢子的水平在加入ExcelHome之前很低，虽然懂得很多函数，但只是对函数最基本的了解而已。因为在这之前没有实战经验，更多的是从书上获取。读万卷书，不如实际操作学得更



快。ExcelHome给了卢子一个平台，让卢子发挥。帮人答疑也是一种学习的好方法，而且进步会很快。当你学到一定程度的时候，发觉用函数其实就跟玩游戏一样，想怎么玩就怎么玩。

1. 如图5-143所示，怎么统计所有表格的第二字符为T的A1的和，也就是红色填充色的表格？

Sheet1	ST1	BT111	Sheet2	STV	汇总
--------	-----	-------	--------	-----	----

图5-143 统计第二字符为T的A1的和

2. 如图5-144所示，求所有项目的累计时间，总时间超过24小时，直接用SUM函数求和出错，该如何解决？

B7	:	x	✓	f _x	=SUM(B2:B6)
	A	B	C	D	E
1	项目	时间			
2	A	5:20			
3	B	10:30			
4	C	14:10			
5	D	8:50			
6	E	20:10			
7	合计	11:00	✗		
8					

图5-144 对超出24小时的时间求和

3. 如图5-145所示，LOOKUP函数不支持通配符，如何利用制造商的简称查找产品？

	A	B	C	D	E	F
1	产品	制造商		制造商	产品	
2	冷冻甜椒	日照恒宝食品有限公司		佳商	冷冻甜椒	
3	冷冻甜椒	厦门佳商食品有限公司		中冷	冷冻芦笋	
4	生鲜生姜	北京东方种业有限公司				
5	冷冻芦笋	福建中冷食品有限公司				
6	生鲜葱	潍坊万通食品有限公司				
7						

图5-145 通配符查询

4. 如图5-146所示，大于10000按百位数进行四舍五入，大于5000按十位数进行舍入，大于100按个位数进行舍入。这怎么做呢？

注：不考虑小于100这种情况。

E14		:			
	A	B	C		
1	数据	实际效果			
2	172	170			
3	445	450			
4	5600	5600			
5	10210	10000			
6	430	430			
7	49012	49000			
8					

图5-146 数据舍入问题



第6章

了解另外一片天地 ——数组公式



假如Excel划分成九级，你学好普通公式最多也就到第三级，而学习数组公式即使再差也是第四级初。别看两者相差不多，每往上一级能力相差就越大。学会数组公式，你就会知道很多普通公式都搞不定的问题，在这里都能轻松解决。

6.1 数组的基本概念

什么是数组？

如四大名著是一个整体，“西游记”就是其中一个组。如图6-1所示，只要我一说“西游记”，就会一次性出现：唐僧、孙悟空、猪八戒、沙僧，再如点“三国演义”，里面的四个人也会一起出现，而不用逐个点名，这就是与普通公式的差异。

一句话：数组一次搞定，普通公式多次搞定。

F2	:	✕	✓	<i>f_x</i>	{=西游记}	
	A	B	C	D	E	F
1	西游记	红楼梦	水浒传	三国演义		
2	唐僧	曹雪芹	李逵	诸葛亮		唐僧
3	孙悟空		林冲	张飞		孙悟空
4	猪八戒		燕青	刘备		猪八戒
5	沙僧		鲁智深	关羽		沙僧
6			柴进			
7			武松			
8			吴用			
9			宋江			

图6-1 西游记组

注：“西游记”等为事先定义好的名称，对名称有疑问可参考“4.1.9定义名称的三种方法”。

什么是数组公式？

利用数组公式可以执行多项计算并返回一个或多个结果。输入数组公式后必须按Ctrl+Shift+Enter组合键结束，在输入数组公式时，Excel会自动在大括号{}之间插入该公式。

一般情况下，能力和脾气是成正比的，能力越强，脾气越大。正因为数组公式的能力比普通公式强，所以脾气大点，动不动就要脾气不干。



如图6-2所示，你不是说一喊“西游记”所有人就出来吗，怎么只出来一个唐僧，其他人跑哪去了？

你把我当成普通人，只按Enter键结束，当然不理你啦。正确的做法是选择区域，按Ctrl+Shift+Enter组合键结束。刚开始学习数组公式的时候经常会忘记按组合键结束，平常有空的时候多按几次慢慢就养成习惯了。




F2		:				=西游记		
	A	B	C	D	E	F	G	
1	西游记	红楼梦	水浒传	三国演义				
2	唐僧	曹雪芹	李逵	诸葛亮		唐僧		
3	孙悟空		林冲	张飞				
4	猪八戒		燕青	刘备				
5	沙僧		鲁智深	关羽				
6			柴进					
7			武松					
8			吴用					
9			宋江					
10								

图6-2 没按组合键

西游记有四个人，选择4个单元格，输入“=西游记”，注意编辑栏的变化，用三个组合键结束就会自动生成{}，如图6-3所示。直接按Enter键结束的话就没有出现正确内容。

F2

:

✕

✓

fx

{=西游记}

	A	B	C	D	E	F	G
1	西游记	红楼梦	水浒传	三国演义			
2	唐僧	曹雪芹	李逵	诸葛亮		唐僧	
3	孙悟空		林冲	张飞		孙悟空	
4	猪八戒		燕青	刘备		猪八戒	
5	沙僧		鲁智深	关羽		沙僧	
6			柴进				
7			武松				
8			吴用				
9			宋江				
10							

Ctrl

Shift

Enter

图6-3 按三个组合键

如图6-4所示，西游记只有四人，如果多选区域的话也会出错。

:				{=西游记}
H	I	J		
唐僧				
孙悟空				
猪八戒				
沙僧				
#N/A				

多选一个单元格

图6-4 区域不匹配

如图6-5所示，如果要删除其中一个数组内容，就弹出“不能更改数组的某一部分”的警告对话框。西游记四人是一个整体，不能更改其中一个内容，要改就得全部改。



图6-5 删除其中一个数组内容

如果在编辑栏单独修改一个数组内容，如图6-6所示，这种情况下连让你按Enter键都不允许，这时可以单击“×”按钮。




图6-6 单击“×”按钮

在知道多单元格数组的使用区域情况下，可以用鼠标选中整个区域进行修改公式。如果不确定区域，可以单击包含多单元格数组的任意单元格，使用快捷键Ctrl+/就可以快速选中整个数组区域，如图6-7所示。



图6-7 快速选中数组区域

只要你摸清数组公式的脾气，以后注意点，就能越来越得心应手。



6.2 数组公式的优势

在很多时候常规公式和数组公式都可以完成同一功能，只是有的时候常规公式一步完成，有的时候多步完成，而数组公式是一步到位解决问题，省去中间的过程。下面通过几个例子来比较说明。

如图6-8所示，在购买产品的时候，有的产品的单价是包含小数点的，也就是几毛钱，而实际在算金额的时候会对金额进行四舍五入精确到元，这样该如何处理？

	A	B	C	D	E
1	名称	单位	数量	单价	金额
2	三位大板 开关	个	3	15.3	
3	二位大板 开关	个	2	12.6	
4	手套	对	12	2.5	
5	电工胶布	卷	20	1.5	
6	螺丝	盒	1	15	
7	斯耐德 1P/20A	个	7	35	
8	20A漏电	套	4	120	
9	2P/63A漏电	套	1	438	
10	填缝剂	盒	6	24.4	
11					

图6-8 对金额进行四舍五入

利用常规公式，在E2单元格输入公式，并向下填充公式。

```
=ROUND(C2*D2,0)
```

利用数组公式，直接选择区域E2:E10输入公式，按Ctrl+Shift+Enter组合键结束。

```
=ROUND(C2:C10*D2:D10,0)
```

这就是一步到位解决问题，数组公式没什么特殊的优势。但有一点需要注意的是，在这里使用常规公式可以直接改动任意单元格，而使用数组公式不允许改动任意单元格，只能全部改动，这样有一个好处就是防止别人不小心改动公式，导致最终结果错误。

这个比较简单，体现不出数组的价值，来一道稍微难点的问题，体现数组的魅力。

如图6-9所示，如何根据目标产品求总数量？

常规法是用SUMIF函数依次获取每个目标产品的数量，然后用SUM函数汇总。而数组公式能够省去辅助列，一步到位。

```
=SUM(SUMIF(A:A,D4:D7,B:B))
```



	A	B	C	D	E	F
1	产品	零售数量		如何根据目标产品求总数量？		
2	20947D0	2				
3	21095A1	3				
4	21539A0	1				
5	21667D0	1				
6	21668D0	1				
7	21697A0	9				
8	21701A0	6				
9	21701A1	2				
10	21765A0	1				
11	21767A0	2				
12	21769A0	2				
13	21771D0	1				
14	21797D0	1				
15	21819D1	3				

目标产品	常规法	
21697A0	9	=SUMIF(A:A,D4,B:B)
21701A0	6	=SUMIF(A:A,D5,B:B)
21828A0	2	=SUMIF(A:A,D6,B:B)
21829A0	7	=SUMIF(A:A,D7,B:B)
总数量	24	=SUM(E4:E7)

数组法	24	{=SUM(SUMIF(A:A,D4:D7,B:B))}
-----	----	------------------------------

图6-9 根据目标产品求总数量

首先来看看这个是怎么运算的，F9键的功能应该没忘记吧，利用这个来看下运算过程。

```
=SUM(SUMIF(A:A,D4:D7,B:B))
```

在编辑栏选择SUMIF(A:A,D4:D7,B:B)这部分，抹黑，按F9键得到{9;6;2;7}。

也就是说，SUMIF函数第二参数选择区域的话，可以分别统计，然后在最外面嵌套SUM函数将统计的结果汇总起来。

也就是说，不管过程有几步，数组公式都能一步到位，省去中间步骤。

当然数组公式的作用不仅仅如此，还能实现普通公式无法实现的功能。如图6-10所示，要获取不重复的职业。

	A	B	C	D	E	F	G	H
1	姓名	性别	公司名称	职业	身份证号		获取不重复的职业	
2	徐新凯	女	王山实业有限公司	行政	xxxxxx110820420002		职业	
3	杨爱丽	女	东南实业	行政	xxxxxx227821020342			
4	王军	男	坦森行贸易	财务	xxxxxx227811114261			
5	吴晓静	女	国顶有限公司	行政	xxxxxx227821208002			
6	周慧慧	女	通恒机械	品质	xxxxxx114810610006			
7	张根红	女	森通	品质	xxxxxx227820316282			
8	李峰	男	国皓	财务	xxxxxx227820316287			
9	陆国峰	男	迈多贸易	财务	xxxxxx227731006161			
10	吴国辉	男	祥通	财务	xxxxxx227551216043			
11	孔祥龙	男	广通	生产	xxxxxx221730305283			
12	顾青青	男	光明奶业	生产	xxxxxx197601053817			
13	徐洪奎	男	威航货运有限公司	财务	xxxxxx197010244056			
14	顾炳年	男	三捷实业	行政	xxxxxx227480307363			
15	朱美蓉	女	浩天旅行社	财务	xxxxxx227640924102			
16	金春梅	女	同恒	品质	xxxxxx227650206422			
17								

图6-10 获取不重复的职业

这个用普通公式非常难实现，如果用数组公式还是可以实现的。在G3单元格输入公式，



按Ctrl+Shift+Enter组合键结束，并向下填充公式，就能提取到不重复的职业，如图6-11所示。

```
=INDEX(D:D,SMALL(IF(MATCH($D$2:$D$16,$D$2:$D$16,0)=ROW($1:$15),ROW($2:$16),4^8),ROW(A1)))&"
```

G3														:	{=INDEX(D:D,SMALL(IF(MATCH(\$D\$2:\$D\$16,\$D\$2:\$D\$16,0)=ROW(\$1:\$15),ROW(\$2:\$16),4^8),ROW(A1)))&"}													
	A	B	C	D	E	F	G	H	I	J	K	L	M															
1	姓名	性别	公司名称	职业	身份证号		获取不重复的职业																					
2	徐新凯	女	王山实业有限公司	行政	xxxxxx110820420002		职业																					
3	杨爱丽	女	东南实业	行政	xxxxxx227821020342		行政																					
4	王军	男	坦森行贸易	财务	xxxxxx227811114261		财务																					
5	吴晓静	女	国顶有限公司	行政	xxxxxx227821208002		品质																					
6	周慧慧	女	通恒机械	品质	xxxxxx114810610006		生产																					
7	张银红	女	森通	品质	xxxxxx227820316282																							
8	李峰	男	国皓	财务	xxxxxx227820316287																							
9	陆国峰	男	迈多贸易	财务	xxxxxx227731006161																							
10	吴国辉	男	祥通	财务	xxxxxx227551216043																							

图6-11 提取到不重复的职业

这个数组公式比较复杂，在后面的章节会解释里面的含义，在这里只要知道数组公式可以实现普通公式很难实现的效果就行。

数组还有一种比较特殊的形式，叫常量数组，不需要按Ctrl+Shift+Enter组合键结束，跟普通公式差不多。

如图6-12所示，成绩分成四个区间，查询成绩属于哪个等级。

	A	B	C	D	E	F	G
1	成绩	等级			区间	等级	
2	65	中		1	1-60	差	
3	90	优		61	61-70	中	
4	35	差		71	71-85	良	
5				86	86-100	优	
6							
7							

图6-12 查询成绩属于哪个等级

正常遇到这种情况，都是用普通公式进行查询。

用VLOOKUP函数查询：

```
=VLOOKUP(A2,D:F,3)
```

用LOOKUP函数查询：

```
=LOOKUP(A2,$D$2:$F$5)
```

这两个公式很好用，但公式会借助一个区域作为辅助，假如这个区域被删除了，公式就会显示错误，如图6-13所示。

B2 : =LOOKUP(A2,#REF!)						
	A	B	C	D	E	F
1	成绩	等级				
2	65	#REF!				
3	90	#REF!				
4	35	#REF!				
5						
6						

图6-13 区域被删除后的效果

这时常量数组就派上用场了，无须借助其他区域就可以搞定问题。

```
=LOOKUP(A2,{1;61;71;86},{ "差"; "中"; "良"; "优" })
```

这里面的{}是手工输入的，不是按Ctrl+Shift+Enter组合键结束生成的。如果数据源本身有“优良中差”这种文字，也可以直接引用单元格区域F2:F5，然后在编辑栏按F9

键生成，这样就免去了手工输入的麻烦，如图6-14所示。

	A	B	C	D	E	F	G
1	成绩	等级			区间	等级	
2	65	=LOOKUP(A2,{1;61;71;86},F2:F5)		1	1-60	差	
3	90	优		61	61-70	中	
4	35	差		71	71-85	良	
5				86	86-100	优	

	A	B	C	D	E	F	G	H
1	成绩	等级			区间	等级		
2	65	=LOOKUP(A2,{1;61;71;86},{"差";"中";"良";"优"})		1	1-60	差		
3	90	优		61	61-70	中		
4	35	差		71	71-85	良		
5				86	86-100	优		

图6-14 生成常量数组

常量数组还有一个经典用法，那就是计算个人所得税的公式。

`=ROUND(MAX((A3-3500)*{0.03,0.1,0.2,0.25,0.3,0.35,0.45}-{0,105,555,1005,2755,5505,13505},0),2)`

个税公式，拿来即用即可，里面的A3是指应发工资扣除五险以后的余额。



知识扩展

常量数组可以是一维的，也可以是二维的。一维数组可以是垂直的，也可以是水平的。

一维水平数组中的元素用逗号（,）分开，比如下面：

`{1,61,71,86}`

一维垂直数组中的元素用分号（;）分开，比如下面：

`{1;61;71;86}`

水平数组就相当于在同一行显示，垂直数组就相当于在同一列显示。

对于二维数组，用逗号将一行内的元素分开，用分号将各行分开，比如下面：

`{1,"差";61,"中";71,"良";86,"优"}`
`{1,61,71,86,"差","中","良","优"}`

即二维数组就是水平数组和垂直数组合并起来，如图6-15所示。

	A	B	C	D	E	F	G
1							
2							
3							
4							
5							
6							
7							
8							
9							

图6-15 一维数组和二维数组



注意：不可以用数组公式中使用列出常数的方法列出单元引用、名称或公式。例如： $\{2*3,3*3,4*3\}$ 因为列出了多个公式，是不可用的。 $\{A1,B1,C1\}$ 因为列出多个引用，也是不可用的。不过可以使用一个区域，例如 $\{A1:C1\}$ 。

6.3 手把手教你学数组公式

光说不练假把式，边说边练才是真本事。

我们知道用ROW函数和COLUMN函数可以生成行列序号，如果要生成1~10可以通过这两个函数获得，但能否一次生成呢？答案是肯定的。

如图6-16所示，选择区域A1:A10，输入=ROW(1:10)，按Ctrl+Shift+Enter组合键结束，这时就一次生成1~10。这个也叫多单元格数组公式。




图6-16 创建第一个数组公式

同理，如图6-17所示，要生成水平1~10，可以在编辑栏输入=COLUMN(A:J)，按Ctrl+Shift+Enter组合键结束。




A1												:							{=COLUMN(A:J)}																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																																								

图6-17 创建水平数组公式

动手的朋友会发现，用COLUMN函数创建数组很麻烦，还得去数字母才知道多少，很不直观，如果要创建1~1000，估计得数哭了。所以绝大多数情况下都是借助ROW函数生成，这个比较直观。

这时有的朋友可能会说，ROW函数不是只能垂直生成1~N吗，水平生产不是得用COLUMN函数？

话是没错，但别忘了Excel中还存在一个转置函数TRANSPOSE，这个函数跟选择性粘



贴的转置一样，可以将行列互相转置。

=TRANSPOSE(ROW(1:10))

它有什么作用呢？估计很多朋友现在还是疑惑，别急，以后你就会知道。先透露下MMULT函数，这是一个高级函数，这个函数很多时候就得用TRANSPOSE+ROW才能完成系列运算。不过这个函数太难了，轻易不会使用，留在后面的章节再详细了解。

来一个实际案例自己尝试下，如图6-18所示，获取总金额。

	A	B	C	D
1	名称	单位	数量	单价
2	三位大板 开关	个	3	15.3
3	二位大板 开关	个	2	12.6
4	手套	对	12	2.5
5	电工胶布	卷	20	1.5
6	螺丝	盒	1	15
7	斯耐德 1P/20A	个	7	35
8	20A漏电	套	4	120
9	2P/63A漏电	套	1	438
10	填缝剂	盒	6	24.4
11			总金额	
12				

图6-18 获取总金额

如图6-19所示，输入SUM函数，然后用鼠标引用区域，填写相关参数。

D2	:	X	✓	fx	=SUM(C2:C10*D2:D10)	
	A	B	C	D	E	F
1	名称	单位	数量	单价		
2	三位大板 开关	个	3	15.3		
3	二位大板 开关	个	2	12.6		
4	手套	对	12	2.5		
5	电工胶布	卷	20	1.5		
6	螺丝	盒	1	15		
7	斯耐德 1P/20A	个	7	35		
8	20A漏电	套	4	120		
9	2P/63A漏电	套	1	438		
10	填缝剂	盒	6	24.4		
11			总金额	=SUM(C2:C10*D2:D10)		
12						

图6-19 填写相关参数

这时切记不能直接按Enter键，否则就出

错，因为直接按Enter键会被当成普通公式处理，如图6-20所示。




D11	:				=SUM(C2:C10*D2:D10)	
	A	B	C	D	E	F
1	名称	单位	数量	单价		
2	三位大板 开关	个	3	15.3		
3	二位大板 开关	个	2	12.6		
4	手套	对	12	2.5		
5	电工胶布	卷	20	1.5		
6	螺丝	盒	1	15		
7	斯耐德 1P/20A	个	7	35		
8	20A漏电	套	4	120		
9	2P/63A漏电	套	1	438		
10	填缝剂	盒	6	24.4		
11			总金额	#VALUE!		
12						

图6-20 直接按Enter键出错

由于习惯的问题，按Enter键了也没关系，重新进入编辑栏，再按Ctrl+Shift+Enter组合键结束，这时就成功了，如图6-21所示。

D11	:	X	✓	\sum	{=SUM(C2:C10*D2:D10)}	
	A	B	C	D	E	F
1	名称	单位	数量	单价		
2	三位大板 开关	个	3	15.3		
3	二位大板 开关	个	2	12.6		
4	手套	对	12	2.5		
5	电工胶布	卷	20	1.5		
6	螺丝	盒	1	15		
7	斯耐德 1P/20A	个	7	35		
8	20A漏电	套	4	120		
9	2P/63A漏电	套	1	438		
10	填缝剂	盒	6	24.4		
11			总金额	1455.5		
12						

图6-21 生成数组公式

公式是否输入成功，只需看编辑栏是否有自动生成的{}，如果有就代表成功了。数组公式最外面的{}千万不能自己手工输入，而要通过按Ctrl+Shift+Enter组合键生成才行。这样就完成了人生第一个数组公式，是不是很有成就感？

接下来就是挑战多单元格数组的输入，在使用VLOOKUP函数逆向查找的时候，IF函数重新构造区域，其实就是多单元格数组的输入。



输入这个多单元格数组很容易出错，因为在选择区域的时候，多一行少一行都是不允许的，否则就出错，如图6-22所示。

	A	B	C	D	E	F	G	H	I	J
1	职业	姓名	性别	公司名称		姓名	职业		姓名	职业
2	行政	徐新凯	女	王山实业有限公司		徐新凯	行政		徐新凯	行政
3	行政	杨爱丽	女	东南实业		杨爱丽	行政		杨爱丽	行政
4	财务	王军	男	坦森行贸易		王军	财务		王军	财务
5	行政	吴晓静	女	国顶有限公司		吴晓静	行政		吴晓静	行政
6	品质	周慧慧	女	通恒机械		周慧慧	品质		周慧慧	品质
7	品质	张银红	女	森通		张银红	品质		张银红	品质
8	财务	李峰	男	国皓		李峰	财务		李峰	财务
9	财务	陆国峰	男	迈多贸易		陆国峰	财务		陆国峰	财务
10	财务	吴国辉	男	祥通		吴国辉	财务		吴国辉	财务
11	生产	孔祥龙	男	广通		孔祥龙	生产		孔祥龙	生产
12	生产	顾青青	男	光明奶业		顾青青	生产		顾青青	生产
13	财务	徐洪奎	男	威航货运有限公司		徐洪奎	财务		徐洪奎	财务
14	行政	顾炳年	男	三捷实业		顾炳年	行政		顾炳年	行政
15	财务	朱美蓉	女	浩天旅行社		朱美蓉	财务		朱美蓉	财务
16	品质	金春梅	女	同恒		金春梅	品质			
17						#N/A	#N/A			
18						#N/A	#N/A			

图6-22 区域没选对

选择一个行列区域一模一样的区域，在编辑栏输入公式，如图6-23所示。

OR

:

✖

✓

fx

=IF({1,0},B1:B16,A1:A16)

	A	B	C	D	E	F	G
1	职业	姓名	性别	公司名称		A1:A16)	
2	行政	徐新凯	女	王山实业有限公司			
3	行政	杨爱丽	女	东南实业			
4	财务	王军	男	坦森行贸易			
5	行政	吴晓静	女	国顶有限公司			
6	品质	周慧慧	女	通恒机械			
7	品质	张银红	女	森通			
8	财务	李峰	男	国皓			
9	财务	陆国峰	男	迈多贸易			
10	财务	吴国辉	男	祥通			
11	生产	孔祥龙	男	广通			
12	生产	顾青青	男	光明奶业			
13	财务	徐洪奎	男	威航货运有限公司			
14	行政	顾炳年	男	三捷实业			
15	财务	朱美蓉	女	浩天旅行社			
16	品质	金春梅	女	同恒			
17							

图6-23 多单元格数组操作



如图6-24所示，最后按Ctrl+Shift+Enter组合键结束，就完成了多单元格数组的输入。这时你会发现一个很有意思的现象，多单元格数组的每一个单元格的公式都是一样的，这一点与普通公式有所不同，普通公式在下拉的时候会随着单元格的改变而改变。

F1	:				{=IF({1,0},B1:B16,A1:A16)}			
	A	B	C	D	E	F	G	H
1	职业	姓名	性别	公司名称		姓名	职业	
2	行政	徐新凯	女	王山实业有限公司		徐新凯	行政	
3	行政	杨爱丽	女	东南实业		杨爱丽	行政	
4	财务	王军	男	坦森行贸易		王军	财务	
5	行政	吴晓静	女	国顶有限公司		吴晓静	行政	
6	品质	周慧慧	女	通恒机械		周慧慧	品质	
7	品质	张银红	女	森通		张银红	品质	
8	财务	李峰	男	国皓		李峰	财务	
9	财务	陆国峰	男	迈多贸易		陆国峰	财务	
10	财务	吴国辉	男	祥通		吴国辉	财务	
11	生产	孔祥龙	男	广通		孔祥龙	生产	
12	生产	顾青青	男	光明奶业		顾青青	生产	
13	财务	徐洪奎	男	威航货运有限公司		徐洪奎	财务	
14	行政	顾炳年	男	三捷实业		顾炳年	行政	
15	财务	朱美蓉	女	浩天旅行社		朱美蓉	财务	
16	品质	金春梅	女	同恒		金春梅	品质	
17								
18								

图6-24 多单元格数组输入后的效果

6.4 有趣的自动扩展现象

在用多单元格数组公式统计补助小计的时候，会发现一个奇怪的现象，F3只是一个单元



格，而E5:E8是一个区域，这样居然也能统计正确，如图6-25所示。

出差登记表					
			日补助金额	30	
人员	开始日期	结束日期	住宿天数	补助小计	
陈锡卢	7月26日	7月30日	4	120	
	8月5日	8月13日	8	240	
	8月16日	8月20日	4	120	
	8月24日	8月25日	1	30	
合计				¥510	

图6-25 多单元格数组公式统计补助小计

数组公式究竟是如何运算的呢？

这类似于孙悟空变出小猴子，当妖怪很多的时候，可以变出无数个猴子出来跟他们一对一战斗。也就是说当只有一个30的时候，而有4个数字要与这个30进行运算，一个30就会变成4个30，这样就对等，从而进行运算，如图6-26所示。

	A	B	C	D	E	F	G
1		30					
2	4			4	30		120
3	8			8	30		240
4	4			4	30		120
5	1			1	30		30

图6-26 以1变4

如果有两个数字要跟4个数字进行运算，同样道理，区域会自动扩展到一模一样从而进行运算，如图6-27所示。

是不是挺有趣呢？

不过这种区域自动扩展并不是任意时候都管用，如图6-28所示，在同一列的4个数字跟2个数字进行运算，这样同一行的就会进行相应的运算。因为从第3个数字开始就没有对

应值，这时找不到与它进行计算的数字，就返回错误值#N/A。如果是1个数字，扩展肯定是这个数字，现在有2个数字，Excel不懂如何选择哪个数字进行扩展，最终就放弃了扩展。

	A	B	C	D	E	F	G	H	I	J
1										
2	4	30	40		4	30	40		120	160
3	8				8	30	40		240	320
4	4				4	30	40		120	160
5	1				1	30	40		30	40

图6-27 多个数字扩展

	A	B	C	D	E	F
1						
2	4	30		120		
3	8	40		320		
4				#N/A		
5	1			#N/A		

图6-28 区域自动扩展失效

了解完数组公式的自动扩展，现在要统计合计，理解公式就变得很简单，输入公式后，按Ctrl+Shift+Enter组合键结束，转换过程如图6-29所示。

=SUM(E5:E8*F3)


{=SUM(E5:E8*F3)}
{=SUM(E5:E8*30)}
{=SUM(E5:E8*{30;30;30;30})}

图6-29 合计公式的区域自动扩展

6.5 小结与练习

本章主要对数组进行初步了解，掌握一些概念性的东西。了解什么是数组及数组公式，数组公式与普通公式的差别。一定要牢记Ctrl+Shift+Enter组合键，学习数组脱离不了这三键。以前卢子刚学习的时候也老是忘记，平常没事多按几次，用不了多久就习惯成自然。


1. 如图6-30所示，将总计的普通公式改成数组公式，输入在区域G2:G9。



	A	B	C	D	E	F	G
1	项目	商场	类别	数量	单位	单价	总计
2	橙子	日用杂货	农产品	0.9	千克	299	269.1
3	苹果	果园	农产品	1.35	千克	199	268.65
4	香蕉	日用杂货	农产品	1	捆	399	399
5	莴苣	市场	农产品	2	颗	229	458
6	番茄	市场	农产品	1.8	千克	349	628.2
7	南瓜	市场	农产品	2	个	150	300
8	芹菜	日用杂货	农产品	2	捆	199	398
9	黄瓜	市场	农产品	0.45	千克	229	103.05

图6-30 普通公式改成数组公式

2. 如图6-31所示，借助转置函数TRANSPOSE+VLOOKUP实现水平查找项目的对应商场。



	A	B	C	D	E	F	G	H	I
1	项目	橙子	苹果	香蕉	莴苣	番茄	南瓜	芹菜	黄瓜
2	商场	日用杂货	果园	日用杂货	市场	市场	市场	日用杂货	市场
3	类别	农产品	农产品	农产品	农产品	农产品	农产品	农产品	农产品
4	数量	0.9	1.35	1	2	1.8	2	2	0.45
5	单位	千克	千克	捆	颗	千克	个	捆	千克
6	单价	299	199	399	229	349	150	199	229
7	总计	269.1	268.65	399	458	628.2	300	398	103.05
8									
9									
10	项目	橙子							
11	商场	日用杂货							
12									

图6-31 水平查找对应值



第7章

骗你爱上数组公式



学习数组公式也有段时间，但卢子始终在数组公式的门外徘徊。听到Ctrl+Shift+Enter组合键犹如老鼠看到猫，但心中又存在幻想，希望有朝一日神人出现或者天降宝典，那时卢子就能够轻松学好数组公式。但始终没等到这一天，后来卢子借用玩游戏的时间来学习数组公式，学了很久，忽然有一天发觉会了好多，真是意外的惊喜。

7.1 像玩游戏一样玩SUM函数

SUM函数看似简单，实则蕴含高深用法。什么是高手？

网上流传一段关于Excel函数大腕儿的话：一定得选最复杂的函数，请最牛的大师，写最多嵌套的公式！在编辑栏直接写入，公式最少也得四百字符，数组、多维、循环，能写的全写上，引用要有INDIRECT，里头要有ADDRESS，不是OFFSET就是INDIRECT，要是直接用VLOOKUP，你都不好意思跟人家打招呼。

其实这是一个误区，高手并不一定要写出复杂的公式，而是能将最平常的函数变幻出神奇莫测的功能，解决掉你想都不敢想的问题。

7.1.1 闲聊SUM函数

Q 如果我们要求1~10的和怎么办？

A 可以通过辅助列，在A1~A10分别输入1~10，然后用SUM函数求和，如图7-1所示。

```
=SUM(A1:A10)
```

	A	B	C	D
1	1		总计	
2	2		55	
3	3			
4	4			
5	5			
6	6			
7	7			
8	8			
9	9			
10	10			
11				

图7-1 统计1~10的和

不过，如果数据庞大的话，如输入1~



10000，用这个办法恐怕行不通，你光输入就得输入到哭。这时就可以通过ROW函数来生成1~N这样的序号。

```
=SUM(ROW(1:10000))
```

这时习惯用普通公式的人，一按Enter键就会发现问题，如图7-2所示，怎么结果才是1呢？

图7-2 统计出错

卢子刚开始学习数组公式的时候，就经常犯这种低级错误。记住了，输入数组公式需要按Ctrl+Shift+Enter组合键结束才可以，否则出错。三键一按，立马乖乖变成50005000，可以用数学速算法验证下，答案是正确的。

```
=(10000+1)*10000/2
```

这个公式虽好，但假如某个人误操作你的表格，将其中两行删除，这时你就傻眼了，公式自己会变，如图7-3所示。

图7-3 删除行导致出错

看来这个公式并不完美，仍需改善，这

时INDIRECT函数就派上用场了。

```
=SUM(ROW(INDIRECT("1:10000")))
```

如图7-4所示，不管你怎么删除行，效果始终不变。

图7-4 删除行后效果始终不变

加上INDIRECT函数就相当于引用一个固定的1:10000，而不用借助单元格引用，直接用ROW(1:10000)其实还是要借助单元格才能实现。

现在不讨论这种万一的情况，继续聊新的话题。

如果要求1~10的奇数跟偶数又该如何？

其实我们可以利用MOD（数据,2），如果是奇数就是1、如果是偶数就是0这个特点来处理。

奇数：

```
=SUM(MOD(ROW(1:10),2)*ROW(1:10))
```

偶数：

```
=SUM((MOD(ROW(1:10),2)=0)*ROW(1:10))
=SUM(MOD(ROW(1:10)-1,2)*ROW(1:10))
```

以上都是数组公式，需要按Ctrl+Shift+Enter组合键结束。





知识扩展

在高版本中一般用ISODD函数来判断是否为奇数，用ISEVEN来判断是否为偶数，所以求奇数跟偶数的和也可以改成下面的数组公式。

```
=SUM(ISODD(ROW(1:10))*ROW(1:10))
```

```
=SUM(ISEVEN(ROW(1:10))*ROW(1:10))
```



如果要求1~10的前5大、前5小的和要怎么做呢？



求前几大可以用LARGE函数，求前几小可以用SMALL函数。

前5大的和：

```
=SUM(LARGE(ROW(1:10),ROW(1:5)))
```

前5小的和：

```
=SUM(SMALL(ROW(1:10),ROW(1:5)))
```

以上都是数组公式，需要按Ctrl+Shift+Enter组合键结束。

LARGE函数跟SMALL函数的语法一样，学习一个就行。LARGE函数的语法：

```
=LARGE 数字,N
```

如图7-5所示，求A列最大和第二大的数字，最大即N是1，第二大即N是2，依次类推。

	A	B	C	D	E
1	491		548	=LARGE(\$A\$1:\$A\$10,1)	
2	371		491	=LARGE(\$A\$1:\$A\$10,2)	
3	475				
4	142				
5	302				
6	122				
7	123				
8	322				
9	548				
10	422				
11					

图7-5 第N大



7.1.2 SUM函数取代SUMIF(SUMIFS)函数

SUM函数谁都知道可以求和，但它还可以取代SUMIF函数和SUMIFS函数实现条件求和，你知道吗？当初就是被这一用法深深吸引，觉得SUM函数很神奇，卢子才不知不觉喜欢上数组公式。

如图7-6是某学校各人员成绩，现在需要统计各学部的总成绩。

	A	B	C	D	E	F	G	H	I
1	姓名	年龄	性别	成绩	学部		学部	总成绩	
2	张三	22	女	86	财经		财经	363	
3	李四	21	男	73	网络技术应用		网络技术应用	304	
4	王五	18	男	90	计算机		计算机	277	
5	马六	19	女	100	财经		多媒体技术	364	
6	周七	23	男	93	多媒体技术				
7	丽萍	17	女	76	网络技术应用				
8	海华	18	男	65	网络技术应用				
9	方海	20	女	92	财经				
10	凯丽	17	女	77	多媒体技术				
11	李阳峰	20	男	85	财经				
12	蒋海坡	19	男	90	网络技术应用				
13	蔡明波	22	男	95	多媒体技术				
14	姜大雨	19	男	87	计算机				
15	马华	21	女	99	多媒体技术				
16	呈浩歌	19	男	100	计算机				
17									

图7-6 统计各学部总成绩

一般用SUMIF函数就可以解决，SUMIF函数就是条件求和的意思。

```
=SUMIF(E:E,G2,D:D)
```

其实我们可以将SUMIF拆分成两个函数SUM+IF，先进行条件判断，然后求和。以G2的学部财经为例，需要判断的条件就是E列学部是不是等于财经，如果是就返回成绩，不是就显示0。如图7-7所示，选择区域在K2:K16输入公式，然后按Ctrl+Shift+Enter组合键结束。可以清楚地看到所有符合财经的都返回本身的成绩，不是财经的都

返回0。

```
=IF(E2:E16=G2,D2:D16,0)
```

	A	B	C	D	E	F	K	L
1	姓名	年龄	性别	成绩	学部			
2	张三	22	女	86	财经		86	
3	李四	21	男	73	网络技术应用		0	
4	王五	18	男	90	计算机		0	
5	马六	19	女	100	财经		100	
6	周七	23	男	93	多媒体技术		0	
7	丽萍	17	女	76	网络技术应用		0	
8	海华	18	男	65	网络技术应用		0	
9	方海	20	女	92	财经		92	
10	凯丽	17	女	77	多媒体技术		0	
11	李阳峰	20	男	85	财经		85	
12	蒋海坡	19	男	90	网络技术应用		0	
13	蔡明波	22	男	95	多媒体技术		0	
14	姜大雨	19	男	87	计算机		0	
15	马华	21	女	99	多媒体技术		0	
16	呈浩歌	19	男	100	计算机		0	
17								

图7-7 用IF函数判断是不是财经

IF函数的区域判断跟单元格单独判断是一样的，其实这个公式等同于下面公式下拉的结果。唯一的差别在于，一个是选择区域按三键结束，一个是下拉获取结果而已。

```
=IF(E2=$G$2,D2,0)
```

到这一步已经完成了90%的工作，只需在最外面嵌套SUM函数进行求和即可。

```
=SUM(IF($E$2:$E$16=G2,$D$2:$D$16))
```

后面的所有数组公式的输入都是需要按三键结束，在这里再强调一下，为了方便说明，不再重复。

照着这个思路，看看SUM函数如何取代SUMIFS函数。如图7-8所示，现在增加了一个条件性别，变成多条件求和。

[illegible]

图7-8 多条件求和

我们都知道在普通公式里面AND函数表示并且的意思，多条件设置都是用AND函数，按常规想法就这样设置公式。

=IF(AND(E2:E16=G2,C2:C16=H2),D2:D16,0)

如图7-9所示，输入公式后的结果让人大跌眼镜，居然全部都是0，怎么回事呢？

[illegible]

图7-9 多单元格数组使用AND函数出错



如图7-10所示，这时用常规公式，下拉填充公式，发现并没有异常，怎么回事呢？




图7-10 用常规公式结果正常

这时有“独孤九剑”之称的F9键就派上用场了。如图7-11所示，在编辑栏选择AND部分按F9键，得到FALSE，只有一个结果。也就是说只要有任何一个条件不满足就是FALSE，这样一来肯定有问题。必须得到15个结果才行。




图7-11 解读AND部分

那有没有其他函数可以取代AND函数？答案是肯定的，那就是“*”。

=IF((E2:E16=G2)*(C2:C16=H2),D2:D16,0)

现在用F9键同样对条件进行测试。如图7-12所示，也就是说同时满足两个条件的返回1，否则返回0。1就是TRUE，所以返回D列对应值，0就是FALSE，所以返回0。




图7-12 解读*部分

如果对F9键还不熟练，没关系，继续回到单元格内进行测试，如图7-13所示，与我们的判断一致。

M2 : {=IF((E2:E16=G2)*(C2:C16=H2),D2:D16,0)}

	A	B	C	D	E	F	G	H	I	J	M
1	姓名	年龄	性别	成绩	学部		学部	性别	总成绩		
2	张三	22	女	86	财经		财经	女	278		86
3	李四	21	男	73	网络技术应用		网络技术应用	男	228		0
4	王五	18	男	90	计算机		计算机	男	277		0
5	马六	19	女	100	财经		多媒体技术	女	176		100
6	周七	23	男	93	多媒体技术						0
7	丽萍	17	女	76	网络技术应用						0
8	海华	18	男	65	网络技术应用						0
9	方海	20	女	92	财经						92
10	凯丽	17	女	77	多媒体技术						0
11	李阳峰	20	男	85	财经						0
12	蒋海坡	19	男	90	网络技术应用						0
13	蔡明波	22	男	95	多媒体技术						0
14	姜大雨	19	男	87	计算机						0
15	马华	21	女	99	多媒体技术						0
16	呈浩歌	19	男	100	计算机						0

.....→

图7-13 单元格验证

现在只需在最外面嵌套SUM函数即可。

```
=SUM(IF(($E$2:$E$16=G2)*($C$2:$C$16=H2),$D$2:$D$16))
```

用“*”可以将各个条件连接起来，其实求和区域也可以看成一个条件，将公式变成：

```
=SUM(($E$2:$E$16=G2)*($C$2:$C$16=H2)*$D$2:$D$16)
```

这样SUM函数条件求和的通用公式就推出来了。

```
=SUM((条件1)*(条件2)*(条件3)*...求和区域)
```

大胆假设，小心求证。虽然通用公式推出来了，但最好还是再通过一个案例进行验证，只有这样自己才敢放心使用这个通用公式。

如图7-14所示，根据购买产品明细表，统计每个月份的购买数量和购买金额。

	A	B	C	D	E	F	G	H	I	J
1	日期	名称	单位	数量	单价	金额		月份	数量	金额
2	2016-2-21	三位大板 开关	个	3	15.3	46		2		
3	2016-2-22	二位大板 开关	个	2	12.6	25		3		
4	2016-2-23	手套	对	12	2.5	30		4		
5	2016-2-27	电工胶布	卷	20	1.5	30				
6	2016-3-2	螺丝	盒	1	15	15				
7	2016-3-3	耐德 1P/20A	个	7	35	245				
8	2016-3-4	20A漏电	套	4	120	480				
9	2016-4-7	2P/63A漏电	套	1	438	438				
10	2016-4-8	填缝剂	盒	6	24.4	146				

图7-14 统计每个月份的购买数量和购买金额



明细表的日期是按天算的，而统计表是按月算的，这时就要将日统一成月份。获取月份可以用MONTH函数实现，最终完整的数组公式为：

```
=SUM((MONTH($A$2:$A$10)=$H2)*($D$1:$F$1=I$1)*$D$2:$F$10)
```

统计数据最牛的工具其实是数据透视表，用这个来验证效果最好。如图7-15所示，是用数据透视表汇总出来的效果和函数汇总的效果，发现统计出来是一样的，也就是说，我们验证正确。至于用数据透视表如何统计，这个靠自己去摸索，这里不做说明。

12											
	A	B	C	D	E	F	G	H	I	J	
1	日期	名称	单位	数量	单价	金额		月份	数量	金额	
2	2016-2-21	三位大板 开关	个	3	15.3	46		2	37	131	
3	2016-2-22	二位大板 开关	个	2	12.6	25		3	12	740	
4	2016-2-23	手套	对	12	2.5	30		4	7	584	
5	2016-2-27	电工胶布	卷	20	1.5	30					
6	2016-3-2	螺丝	盒	1	15	15					
7	2016-3-3	斯耐德 1P/20A	个	7	35	245		日期	数量	金额	
8	2016-3-4	20A漏电	套	4	120	480		2月	37	131	
9	2016-4-7	2P/63A漏电	套	1	438	438		3月	12	740	
10	2016-4-8	填缝剂	盒	6	24.4	146		4月	7	584	
11								总计	56	1455	
12											

图7-15 数据透视表验证公式结果

SUM函数不仅仅可以实现条件求和，还可以进行各种各样的求和，下面通过一些例子来见识下。

如图7-16所示，数量是用VLOOKUP函数查询引用过来的，找不到对应值就显示#N/A，如果直接求和的话出错，该如何处理？

```
=SUM(A2:A11)
```

	A	B	C	D
1	数量		总数量	
2	86		#N/A	
3	73			
4	90			
5	100			
6	#N/A			
7	76			
8	65			
9	92			
10	#N/A			
11	85			
12				

图7-16 含错误值求和

解决方案：

(1) 加一个条件来判断数量是不是数字，就可以解决。

```
=SUM(IF(ISNUMBER(A2:A11),A2:A11))
```

ISNUMBER函数只有一个参数，作用就是判定单元格是不是数字，如果是就返回TRUE，否则返回FALSE。



IS类函数还有好多个，语法都一样，输入“=IS”就出现IS类的所有函数，如图7-17所示。
IS类函数的作用对应表如图7-18所示。




图7-17 IS类函数

函数	如果符合以下条件，则返回 TRUE
ISBLANK	值为空白单元格。
ISERR	值为任意错误值（除去 #N/A）。
ISERROR	值为任意错误值（#N/A、#VALUE!、#REF!、#DIV/0!、#NUM!、#NAME? 或 #NULL!）。
ISLOGICAL	值为逻辑值。
ISNA	值为错误值 #N/A（值不存在）。
ISNONTEXT	值为不是文本的任意项。（请注意，此函数在值为空单元格时返回 TRUE）。
ISNUMBER	值为数字。
ISREF	值为引用。
ISTEXT	值为文本。

图7-18 IS函数作用对应表

（2）借助高版本函数进行容错，再求和会更简单。

```
=SUM(IFERROR(A2:A11,0))
=SUM(IFNA(A2:A11,0))
```

这两个函数的用法在前面说过，这里不再重复解释。

如图7-19所示，统计各季度的销售额。

	A	B	C	D	E
1	月份	销售额		统计各季度的销售额	
2	1月	355		43370	
3	2月	563			
4	3月	345			
5	一季度	1263			
6	4月	4546			
7	5月	4464			
8	6月	3245			
9	二季度	12255			
10	7月	4354			
11	8月	4556			
12	9月	3464			
13	三季度	12374			
14	10月	3256			
15	11月	6785			
16	12月	7437			
17	四季度	17478			
18					

图7-19 统计各季度的销售额

```
=SUM(IF(ISNUMBER(FIND("季度",A2:A17)),
B2:B17))
```

SUM函数不能直接用通配符“*”，但FIND函数也可以实现查找“季度”这两个字符是不是存在，如果存在就返回字符所在位置，否则返回错误值。

ISNUMBER函数用于判断是否是数字，也就是如果包含季度，就返回B列的销售额。

也可以直接套用前面介绍的通用公式：

```
=SUM((ISNUMBER(FIND("季度",A2:A17))) *
B2:B17)
```

当然这里只是为了说明用法，因为数据源本身就很有规律，每个季度都等于3个月的合计，所以也可以采取取巧法。

```
=SUM(B2:B17)/2
```



7.1.3 SUM函数取代COUNTIF(COUNTIFS)函数

SUM函数能实现各种各样求和，其实也是情理之中，如果说SUM函数能取代COUNTIF函数和COUNTIFS函数实现条件计数，也许很多人会觉得是天方夜谭。SUM函数强大得令人难以想象，很多人掌握的知识仅仅是冰山一角。

如图7-20所示，统计各姓氏的人数。

	A	B	C	D	E	F
1	姓名	年龄		姓氏	人数	
2	杨成桂	22		杨	6	
3	杨成松	21		陈	4	
4	杨成萌	18				
5	杨开珍	19				
6	杨成宽	23				
7	袁昌军	17				
8	杨继群	18				
9	陈文才	20				
10	陈友才	17				
11	陈元桃	20				
12	陈友群	19				
13						

图7-20 统计各姓氏的人数

提取姓名中的第一个字符也就是姓氏，可以用LEFT函数。

```
=LEFT(A2,1)
```

第二参数省略不写的话，默认就是提取1位，所以可简化为：

```
=LEFT(A2)
```

结合前面SUM+IF的用法可用：

```
=SUM(IF(LEFT(A$2:A$12)=D2,1))
```

让符合条件的显示1，然后嵌套SUM函数进行求和。

当然也可以借助通用公式：

```
=SUM(1*(LEFT(A$2:A$12)=D2))
```

如图7-21所示，统计各学部男女的人数。

	A	B	C	D	E	F	G	H	I	J
1	姓名	年龄	性别	成绩	学部		学部	男	女	
2	张三	22	女	86	财经		财经	1	3	
3	李四	21	男	73	网络技术应用		网络技术应用	3	1	
4	王五	18	男	90	计算机		计算机	3	0	
5	马六	19	女	100	财经		多媒体技术	2	2	
6	周七	23	男	93	多媒体技术					
7	丽萍	17	女	76	网络技术应用					
8	海华	18	男	65	网络技术应用					
9	方海	20	女	92	财经					
10	凯丽	17	女	77	多媒体技术					
11	李阳峰	20	男	85	财经					
12	蒋海坡	19	男	90	网络技术应用					
13	蔡明波	22	男	95	多媒体技术					
14	姜大雨	19	男	87	计算机					
15	马华	21	女	99	多媒体技术					
16	呈浩歌	19	男	100	计算机					
17										

图7-21 统计各学部男女的人数



利用SUM+IF的做法，依样画葫芦也能做到，但这里不做说明。读者是否还记得SUM函数求和的通用公式：

```
=SUM((条件1)*(条件2)*(条件3)*...*求和区域)
```

其实省略掉求和区域就是条件计数的通用公式：

```
=SUM((条件1)*(条件2)*(条件3)*...(条件n))
```

有了这个通用公式，条件计数如此简单！

```
=SUM(($E$2:$E$16=$G2)*($C$2:$C$16=H$1))
```

计数还有一个经典的问题，就是不重复计数。如图7-22所示，统计不重复学部的个数。

	A	B	C	D	E	F	G	H
1	姓名	年龄	性别	成绩	学部		不重复个数	
2	张三	22	女	86	财经		4	
3	李四	21	男	73	网络技术应用			
4	王五	18	男	90	计算机			
5	马六	19	女	100	财经			
6	周七	23	男	93	多媒体技术			
7	丽萍	17	女	76	网络技术应用			
8	海华	18	男	65	网络技术应用			
9	方海	20	女	92	财经			
10	凯丽	17	女	77	多媒体技术			
11	李阳峰	20	男	85	财经			
12	蒋海坡	19	男	90	网络技术应用			
13	蔡明波	22	男	95	多媒体技术			
14	姜大雨	19	男	87	计算机			
15	马华	21	女	99	多媒体技术			
16	呈浩歌	19	男	100	计算机			
17								

图7-22 统计不重复学部的个数

```
=SUM(1/COUNTIF(区域,区域))
```

是计算区域不重复个数的经典公式，需要好好理解。

```
=SUM(1/COUNTIF(E2:E16,E2:E16))
```

如果区域很多的话，可以改小，这样便于理解。

观察

```
=SUM(1/COUNTIF(E2:E16,E2:E16))
```



F9键抹黑:

```
=SUM(1/{4;4;3;4;4;4;4;4;4;4;3;4;3})
```

Esc键返回:

```
=SUM(1/COUNTIF(E2:E16,E2:E16))
```

F9键抹黑:

```
=SUM({0.25;0.25;0.333333333333333;0.25;0.25;0.25;0.25;0.25;0.25;0.25;0.25;0.25;0.333333333333333;0.25;0.333333333333333})
```

Esc键返回, 在单元格按三建结束看到结果: 4。

分析

用F9键观察有时并不太直观, 回到工作表中继续看看, 如图7-23所示。

1	2	3
4	1/4	4
4	1/4	
3	1/3	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
4	1/4	
3	1/3	
4	1/4	
3	1/3	

图7-23 辅助理解

=COUNTIF(E2:E16,E2:E16)是多单元格数组, 等同于=COUNTIF(E\$2:E\$16,E2)下拉的结果, 也就是统计每个单元格本身出现的次数, 如1。

=1/COUNTIF(E2:E16,E2:E16)是多单元格数组, 等同于=1/COUNTIF(E\$2:E\$16,E2)下拉的结果, 也就是1/每个单元格本身出现的次数。为了让数据更直观转换成分数形式, 如2, 出现4次就变成1/4, 出现3次就变成1/3, $1/3+1/3+1/3=3*(1/3)=1$, $1/N+\dots+1/N=N*(1/N)=1$, 不管出现几次, 相加都等于1。

最后将这些相加就得到不重复的数量, 如3。



解读公式的一些习惯如下。

- (1) 把区域改小，这样便于查看，如A1:A1000改成A1:A9。
- (2) 用F9键配合Ctrl+Z快捷键或者Esc键不断地看运算过程再返回，重复到理解为止。
- (3) 输入公式回到单元格查看运算过程，这种相对比较直观。
- (4) 分析。

第(2)，(3)点可选，看你对公式运用的熟练程度，如果不熟练，选择(3)；熟练的话，选择(2)。

学好SUM函数不过是为了打开数组之门，知道数组的一些基本用法，要学好数组还需要更多的知识支撑才可以。



知识扩展

统计不重复的个数公式更经典，但并不完美，数据源存在空单元格时，统计会出错，如图7-24所示。

G2	:	X	✓	fx	{=SUM(1/COUNTIF(E2:E16,E2:E16))}		
1	A	B	C	D	E	F	G
	姓名	年龄	性别	成绩	学部		不重复个数
2	张三	22	女	86	财经		#DIV/0!
3	李四	21	男	73	网络技术应用		
4	王五	18	男	90	计算机		
5	马六	19	女	100	财经		
6	周七	23	男	93	多媒体技术		
7	丽萍	17	女	76	网络技术应用		
8	海华	18	男	65			
9	方海	20	女	92	财经		
10	凯丽	17	女	77	多媒体技术		
11	李阳峰	20	男	85	财经		
12	蒋海坡	19	男	90	网络技术应用		
13	蔡明波	22	男	95	多媒体技术		
14	姜大雨	19	男	87	计算机		
15	马华	21	女	99	多媒体技术		
16	呈浩歌	19	男	100	计算机		
17							

图7-24 数据源存在空单元格时统计出错

其实只需要将数组公式稍微改变一下就可以，通过E2:E16&""来解决空单元格的问题，不过这样一来就多算了一个空单元格的数目，所以再减去1就行。

```
=SUM(1/COUNTIF(E2:E16,E2:E16&""))-1
```

这是一个解决方法，其实还有一个更完美的公式，就是发现公式出错的根源。



如图7-25所示，用COUNTIF函数统计空单元格的时候，本来应该得到1，但实际上统计的结果等于0。

G8				=COUNTIF(E2:E16,E8)			
	A	B	C	D	E	F	G
1	姓名	年龄	性别	成绩	学部		不重复个数
2	张三	22	女	86	财经		4
3	李四	21	男	73	网络技术应用		
4	王五	18	男	90	计算机		
5	马六	19	女	100	财经		
6	周七	23	男	93	多媒体技术		
7	丽萍	17	女	76	网络技术应用		
8	海华	18	男	65			0
9	方海	20	女	92	财经		
10	凯丽	17	女	77	多媒体技术		
11	李阳峰	20	男	85	财经		
12	蒋海坡	19	男	90	网络技术应用		
13	蔡明波	22	男	95	多媒体技术		
14	姜大雨	19	男	87	计算机		
15	马华	21	女	99	多媒体技术		
16	呈浩歌	19	男	100	计算机		
17							

图7-25 COUNTIF统计空单元格出错

从而导致1/COUNTIF得到错误值#DIV/0!。

=1/COUNTIF(E2:E16,E8)

最终SUM函数在求和的时候遇到错误值统计也出错，也就是解决这部分1/COUNTIF，让错误值显示0，统计就不会出现问题。容错函数首选IFERROR函数，最终的数组公式为：

=SUM(IFERROR(1/COUNTIF(E2:E16,E2:E16),0))

这个公式可以看成通用公式，不管数据源是否存在空单元格，统计都是正确的。

7.1.4 SUM函数价值的体现

截止到目前SUM函数仅仅是取代SUMIF(SUMIFS)函数和COUNTIF(COUNTIFS)函数，还不能真正体现出价值，后面看看它如何实现其他函数做不到的事儿。从两个方面进行说明：针对不规范数据的统计和针对特殊布局的统计。

1. 针对不规范数据的统计

如图7-26所示，是一份含有数据单位的人员销售清单，直接求和得不出正确答案，怎么才能让含单位的销售额可以求和呢？



B8				
	A	B	C	D
1	人员	销售额		
2	王平	600元		
3	任风英	120元		
4	任风英	1000元		
5	周远碧	210元		
6	王庆	129元		
7	易诗琼	123元		
8	总销售额	0		

图7-26 含数据单位的人员销售清单

帮助文件提到，SUM函数会自动忽略文本，“600元”就是文本，不属于数字。最简单的做法就是将“元”替换成空格，然后自定义单元格格式G/通用格式“元”。很多人就是搞不明白一格一属性的道理，才会造成汇总数据困难重重。正确的做法应该是将“元”写在字段名那里变成“销售额（元）”，这样别人一看便知道单位。

刚才提到了“替换”这个词，函数中也

有属于自己的替换函数。

SUBSTITUTE函数语法：

=SUBSTITUTE(文本字符串,旧字符,新字符,[替换第N个])

SUBSTITUTE函数通俗语法如图7-27所示。

WORKD...							
	A	B	C	D	E	F	H
1	人员	销售额					
2	王平	600元					
3	任风英	120元					
4	任风英	1000元					
5	周远碧	210元					
6	王庆	129元					
7	易诗琼	123元					
8	总销售额	0					

图7-27 SUBSTITUTE函数通俗语法

单元格的“元”是多余的，需要替换成空格，空格可以用""表示，替换成空格后直接求和，可以吗，不验证猜一下？

=SUM(SUBSTITUTE(B2:B7,"元",""))

这个是数组公式，用法也跟前面差不多，目测应该可以汇总。

但实际SUBSTITUTE这个函数属于文本函数，所以替换得到的数字也属于文本，在这里叫作文本数字。数字有两种类型，一种是文本数字，另一种是真正的数字，即数值。对数值可以求和，而文本不能实现求和。如账簿上的数字与墙上的数字是不同的，我们可以用前者进行各种分析，后者只能欣赏用。那有什么办法还原数字的本质呢？



利用VALUE函数可以将文本型转换成数值型。

```
=VALUE("600")
```

但一般情况下不会用这种方法，而是通过运算转换。

下面一起来了解“减负”运算。

在函数或公式中，运算过程会自动把文本转换为数值(一个隐含过程)，再与数值进行运算，负值运算(-)也是一种运算，能把文本转换成数值：

```
-"600"=-600
```

还记得负负得正吧？

```
-("600")=-(-600)=600
```

简写为：

```
--"600"=600
```

“--”可以把文本转换为数值，但它不是标准的转换方式，而是借用负运算的隐含功能。

其实负负运算称为减负运算更好，减去数字的负担，还原数字的本质。

如图7-28所示，直接在编辑栏选择SUBSTITUTE(B2:B7,"元","")后按F9键得到的是一组加了双引号的数字，我们知道对文本才加双引号，数字是不加双引号的，这就是不能求和的原因。而在编辑栏选择--SUBSTITUTE(B2:B7,"元","")后按F9键得到的是一组标准的数字，这种情况就能直接求和。




图7-28 F9键比较说明

综上，最终的数组公式为：

```
=SUM(--SUBSTITUTE(B2:B7,"元",""))
```

如果对字符提取三兄弟有印象的话，还可以这样设置数组公式，因为都是包含“元”，也可以利用总字符数-1这个特点来做。



=SUM(--LEFT(B2:B7,LEN(B2:B7)-1))

最后再来一个高端黑的，如图7-29所示，录入金额的时候，居然把姓名也写在一起。

	A	B	C	D
1	金额		总计	
2	陈生700			
3	笑看今朝20			
4	王小姐120			
5	卢10000			
6				

图7-29 含姓名求和

看到这个，卢子都头大了，奉劝大家一句，别把Excel当Word使用，否则后果很严重。

不过即使困难重重，SUM函数也能一一拿下，没有求不了和的数据，只有求不了和

的人。

我们知道LEN函数可以统计字符数，其实它还有一个兄弟叫LENB函数，用于统计字节数。汉字占2字节，数字占1字节。利用这个特点我们可以得出以下公式。

汉字的个数：

=LENB-LEN

数字的个数：

=LEN-汉字的个数=LEN-(LENB-LEN)=2*LEN-LENB

最终公式：

=SUM(--RIGHT(A2:A5,2*LEN(A2:A5)-LENB(A2:A5)))

2. 针对特殊布局的统计

如图7-30所示，根据每个人员12个月的销售明细，统计其中某些人员的销售数量。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1	姓名	1月	2月	3月	4月	5月	6月	7月	8月	9月	10月	11月	12月		姓名	销售数量
2	张三	15	84	47	72	51	10	60	62	79	91	78	34		周七	
3	李四	57	72	74	83	59	19	11	27	95	29	42	84		丽萍	
4	王五	41	88	74	5	94	42	79	59	14	44	91	71		海华	
5	马六	5	98	7	83	50	61	71	79	91	89	31	74		张三	
6	周七	63	8	30	3	88	61	13	73	12	61	69	30			
7	丽萍	51	39	10	25	79	24	74	20	27	41	23	91			
8	海华	20	32	43	20	13	19	66	6	79	19	68	53			
9	方海	30	92	79	51	95	44	63	59	8	23	52	82			
10	凯丽	75	22	70	28	54	81	62	68	40	49	29	94			
11	李阳峰	83	30	50	42	45	94	38	81	52	16	70	69			
12	蒋海坡	84	2	8	35	91	14	52	64	39	8	44	70			
13	蔡明波	60	44	93	91	59	10	72	7	26	59	38	79			
14	姜大雨	85	73	73	3	37	74	50	29	52	56	76	95			
15	马华	44	2	64	55	74	73	1	62	99	33	62	52			
16	呈浩歌	79	63	58	28	60	29	93	50	100	55	51	13			
17																

图7-30 统计销售数量

对于这种单条件统计，首先想到的是用SUMIF函数，可惜输入公式以后就发现统计出错，只是统计了1月份，如图7-31所示。




```
=SUMIF('1月'!A:A,A2,'1月'!B:B)+SUMIF('2月'!A:A,A2,'2月'!B:B)+...+SUMIF('11月'!A:A,A2,'11月'!B:B)+SUMIF('12月'!A:A,A2,'12月'!B:B)
```

对于按12个月分表格还好，如果按365天分表格，估计写公式写到哭。对于这种重复的事情，Excel总是有解决的办法。在汇总表输入数组公式，并向下填充公式就能解决。

```
=SUM(SUMIF(INDIRECT(ROW($1:$12)&"月!A:A"),A2,INDIRECT(ROW($1:$12)&"月!B:B")))
```

公式看起来很长，其实并不难理解，ROW(\$1:\$12)表示获取1到12的数字，条件区域是每个表的A列，也就是ROW(\$1:\$12)&"月!A:A"，因为这里是间接引用，所以借助INDIRECT函数。同理，求和区域是每个表的B列，也就是INDIRECT(ROW(\$1:\$12)&"月!B:B")。

如图7-33所示，SUMIF函数部分统计出来的是每个月的对应值，因为是分开的，所以必须要在最外面嵌套SUM函数将这部分的值加起来。




图7-33 用F9键解读SUMIF部分

有了这个公式，即使是分成365天的表格也可以很从容地搞定。

7.1.5 SUM函数优势尽显的一面

COUNTIF函数和SUMIF函数由于本身的一些限制，在统计的时候会出错，而借助SUM函数的数组公式却可以轻易解决，这就是SUM函数的优势。下面通过4个例子来详细说明。

(1) 如图7-34所示，根据水果明细表统计苹果出现的次数。




D2		:				=COUNTIF(A2:A10,"苹果")
	A	B	C	D	E	F
1	水果	数量				
2	苹果	32		4		
3	橙子	54				
4	桃子	75				
5	苹果	86				
6	桃子	20				
7	苹果	30				
8	桃子	10				
9	苹果	34				
10	桃子	23				
11						

图7-34 统计苹果出现的次数



COUNTIF函数第一参数除了不能用数组的形式，还可以嵌套绝大多数的函数。如图7-37所示，现在根据明细表求12月出现的次数。按理说使用下面的公式是可以的，但实际上不允许，也会提示公式警告消息框。

```
=COUNTIF(MONTH(A2:A10),10)
```

	A	B
1	日期	数量
2	2016-11-27	32
3	2016-11-28	54
4	2016-11-29	75
5	2016-11-30	86
6	2016-12-1	20
7	2016-12-2	30
8	2016-12-3	10
9	2016-12-4	34
10	2016-12-5	23
11		

图7-37 明细表

而对于SUM函数来说，就没有那么多要求，直接套用数组公式即可。

```
=SUM(--(MONTH(A2:A10)=12))
```

(2) 如图7-38所示，在输入身份证号码的时候，为了防止身份证号码重复输入，用COUNTIF函数进行辅助判断，但奇怪的事情就发生了，明明最后一个跟前面的所有身份证号码都不一样，但统计的时候就是出错。

Excel有个很奇怪的规定，就是数字超出15位，即使单元格设置为文本，后面都是当0处理，也就是说虽然后面两个身份证号码不一样，但会被认为一样处理，如图7-39所示。





D2	:				=COUNTIF(C\$2:C2,C2)
	A	B	C	D	E
1	姓名	证件类型	身份证号	次数	
2	陈**	身份证	445121198709055616	1	
3	郑**	身份证	440582199001191545	1	
4	李**	身份证	440508198305083916	1	
5	邱**	身份证	440509198209092819	1	
6	龚**	身份证	420821199007052028	1	
7	李**	身份证	420821199007052021	2	
8					
9					

图7-38 身份证出现次数统计出错

420821199007052028	420821199007052000
420821199007052021	420821199007052000

图7-39 超过15位数字部分当0处理

对于这种情况，借助SUM函数的数组公式可以解决。

```
=SUM(--(C$2:C2=C2))
```

(3) 如图7-40所示，在字符中包含有通配符(*)的时候，用SUMIF函数统计出错。


E2	=SUMIF(A:A,D2,B:B)					
	A	B	C	D	E	F
1	字符	数量		字符	数量	
2	A*A	10		A*A	80	
3	A*A*A	10		A*A*A	30	
4	A*A	10		A*A*A*A	10	
5	A*A	10				
6	A*A	10				
7	A*A	10				
8	A*A*A	10				
9	A*A*A*A	10				
10						

图7-40 含有通配符统计出错

通配符(*)就代表所有的意思，A*A不管中间包含什么字符，都被认为是一样的，也就是说在这里A*A包含了A*A*A和A*A*A*A，因而在统计的时候多统计了数量。



这时会采用SUM函数的数组公式解决。

=SUM((\$E\$2:\$E\$16=G2)*(\$C\$2:\$C\$16=H2)*\$D\$2:\$D\$16)

这确实是解决问题的一种方法，但往往对于初用数组公式的人常常会遗忘用Ctrl+Shift+Enter组合键来结束，而用Enter键直接结束，导致出错。

出于这个原因，能否用SUMPRODUCT函数来代替常用SUM函数的多条件求和公式呢？答案是肯定的。

把上面的公式稍做修改即可。

=SUMPRODUCT((\$E\$2:\$E\$16=G2)*(\$C\$2:\$C\$16=H2)*\$D\$2:\$D\$16)

而且直接按Enter键来结束，函数仍然可以正确计算结果。

原因在于：SUMPRODUCT函数本身就支持数组间运算（相同长度的数组相乘后再加总），因此不需要用数组公式组合键来结束。

对于多条件求和，最后一个“*”也可以用“,”代替。

=SUMPRODUCT((\$E\$2:\$E\$16=G2)*(\$C\$2:\$C\$16=H2),\$D\$2:\$D\$16)

但并不是所有情况下都可以用“,”代替“*”,如现在只是根据学部统计成绩。

=SUMPRODUCT((\$E\$2:\$E\$16=G2)*\$D\$2:\$D\$16)

用“,”代替“*”,会得到错误的结果,如图7-44所示。

H2	:		\sum		=SUMPRODUCT((\$E\$2:\$E\$16=G2),\$D\$2:\$D\$16)	
	A	B	C	D	E	F
1	姓名	年龄	性别	成绩	学部	
2	张三	22	女	86	财经	
3	李四	21	男	73	网络技术应用	
4	王五	18	男	90	计算机	
5	马六	19	女	100	财经	
6	周七	23	男	93	多媒体技术	
7	丽萍	17	女	76	网络技术应用	
8	海华	18	男	65	网络技术应用	
9	方海	20	女	92	财经	
10	凯丽	17	女	77	多媒体技术	
11	李阳峰	20	男	85	财经	
12	蒋海坡	19	男	90	网络技术应用	
13	蔡明波	22	男	95	多媒体技术	
14	姜大雨	19	男	87	计算机	
15	马华	21	女	99	多媒体技术	
16	呈浩歌	19	男	100	计算机	
17						

如果多个条件这样并列写入，系统默认用AND关系（即相乘）来运算，但是



SUMPRODUCT函数不支持逻辑值数组常量间相乘，因此一定要将逻辑值转换成数值才行。

将逻辑值转换成数值可以用--、*1、+0等。

=SUMPRODUCT(--(\$E\$2:\$E\$16=G2),\$D\$2:\$D\$16)

用“,”代替“*”有两个好处。

(1) 当数据源出现文本值的时候当作0处理，进行运算不会出错，而用“*”就会出错，如图7-45所示。

	A	B	C	D	E	F	G	H	I	J	K
1	姓名	年龄	性别	成绩	学部		学部	错误	正确		
2	张三	22	女	86	财经		财经	#VALUE!	263		
3	李四	21	男	73	网络技术应用		网络技术应用	#VALUE!	304		
4	王五	18	男	90	计算机		计算机	#VALUE!	277		
5	马六	19	女	缺考	财经		多媒体技术	#VALUE!	364		
6	周七	23	男	93	多媒体技术						
7	丽萍	17	女	76	网络技术应用		错误：				
8	海华	18	男	65	网络技术应用		=SUMPRODUCT((\$E\$2:\$E\$16=G2)*\$D\$2:\$D\$16)				
9	方海	20	女	92	财经						
10	凯丽	17	女	77	多媒体技术		正确：				
11	李阳峰	20	男	85	财经		=SUMPRODUCT(--(\$E\$2:\$E\$16=G2),\$D\$2:\$D\$16)				
12	蒋海坡	19	男	90	网络技术应用						
13	蔡明波	22	男	95	多媒体技术						
14	姜大雨	19	男	87	计算机						
15	马华	21	女	99	多媒体技术						
16	呈浩歌	19	男	100	计算机						
17											

图7-45 “,”代替“*”效果对比

(2) 运算速度更快。

最后再提醒读者一点：并不是所有用SUM数组公式的都可以用SUMPRODUCT函数来代替。

7.2 万金油公式组合的妙用

日常工作中，使用Excel常常会遇到以一对多查询。即使VLOOKUP、LOOKUP等函数用法练得炉火纯青，但仍然会对以一对多查询无从下手。

Excel大神会用VBA去解决，设置好条件，改改代码，需要什么查什么！

可是，VBA不是每个人都能快速掌握的，不会VBA的你，怎么办？



这时万金油公式组合就派上用场了，这个组合也叫筛选公式。由INDEX+SMALL+IF+ROW(COLUMN)函数组成，这就是解决一对多查询的一种通式，如果你能掌握，那么在Excel里基本上就没有什么查询实现不了的了。

7.2.1 按条件提取对应值

万金油公式组合很好用，同时也很难学，当初卢子花了几个月的时间才学会这个用法。

其实公式好比是牛，光看牛是不知道牛的内部结构的，只有操刀将牛分解才知道牛的内部构造。很多时候我们看到别人写的公式很长，不知道是什么意思，就可以用庖丁解牛法解读。

如图7-46所示，将左边的格式变成右边的形式。

	A	B	C	D	E	F	G	H	I	J	K
1	地市	区县		地市	区县						
2	汕头	金平区		汕头	金平区	龙湖区	濠江区	澄海区	潮阳区	潮南区	南澳县
3	汕头	龙湖区		潮州	湘桥区	枫溪区	潮安县	饶平县			
4	汕头	濠江区		揭阳	榕城区	东山区	揭东区	普宁市	惠来县	揭西县	揭阳试验区
5	汕头	澄海区									
6	汕头	潮阳区									
7	汕头	潮南区									
8	汕头	南澳县									
9	潮州	湘桥区									
10	潮州	枫溪区									
11	潮州	潮安县									
12	潮州	饶平县									
13	揭阳	榕城区									
14	揭阳	东山区									
15	揭阳	揭东区									
16	揭阳	普宁市									
17	揭阳	惠来县									
18	揭阳	揭西县									
19	揭阳	揭阳试验区									

图7-46 格式变身

提取不重复地市：

```
=INDEX(A:A,SMALL(IF(MATCH($A$2:$A$19,$A$2:$A$19,0)=ROW($A$2:$A$19)-1,ROW($A$2:$A$19),4^8),ROW(A1)))&"
```

根据不重复地市获取所有区县对应值：

```
=INDEX($B:$B,SMALL(IF($A$2:$A$19=$D2,ROW($A$2:$A$19),4^8),COLUMN(A1)))&"
```

这么长，还没把它大卸八块，自己就先晕倒了。



当初卢子看到这两条公式也吓晕了，不过后来转念一想，公式拆开后的每个函数我都会，组合起来应该也可以弄懂，不能被困难吓倒，要对自己有信心。

那我们就来庖丁解牛，呵呵。

先来看看1,2,3，如图7-47所示，庖丁解牛1。

=MATCH(\$A\$2:\$A\$19,\$A\$2:\$A\$19,0)，得到每个地市在数据源第一次出现的位置，如1。

=ROW(\$A\$2:\$A\$19)-1，获取1到N的序列号，如2。

=MATCH(\$A\$2:\$A\$19,\$A\$2:\$A\$19,0)=ROW(\$A\$2:\$A\$19)-1，将第一次出现的问题跟序号比较，如果一样就显示TRUE，否则显示FALSE，如3。

1	2	3
1	1	TRUE
1	2	FALSE
1	3	FALSE
1	4	FALSE
1	5	FALSE
1	6	FALSE
1	7	FALSE
8	8	TRUE
8	9	FALSE
8	10	FALSE
8	11	FALSE
12	12	TRUE
12	13	FALSE
12	14	FALSE
12	15	FALSE
12	16	FALSE
12	17	FALSE
12	18	FALSE

图7-47 庖丁解牛1

接着看4,5，如图7-48所示，庖丁解牛2。

为了便于解读，将MATCH(\$A\$2:\$A\$19,\$A\$2:\$A\$19,0)=ROW(\$A\$2:\$A\$19)-1设置为牛1。

=IF(牛1,ROW(\$A\$2:\$A\$19),4^8)，通过牛1知道，排位和序号相同就是TRUE，不同就是FALSE。通过IF将相同的显示本身的序号，不同的显示4^8即65536，Excel 2003版允许的最大行数，这一行通常是没有数据的，也可以将4^8改成任意一个比较大的数，如10000。最后获得由本身行号跟65536组成的区域，如4。

=SMALL(IF(牛1,ROW(\$A\$2:\$A\$19),4^8),ROW(A1))，SMALL(区域,N)，就是将数据升序排序，也就是说将第一次出现的地市的序号放在最前面，如5。

4	5
2	2
65536	9
65536	13
65536	65536
65536	65536
65536	65536
65536	65536
65536	65536
9	65536
65536	65536
65536	65536
65536	65536
13	65536
65536	65536
65536	65536
65536	65536
65536	65536
65536	65536
65536	65536
65536	65536

图7-48 庖丁解牛2



经过这两次庖丁解牛，已经完成了80%的工作，只需再解牛一次即可搞定。

最后看6,7，如图7-49所示，庖丁解牛3。

=INDEX(A:A,牛2)，获得序号的对应值，65536因为是空单元格，所以引用过来就是0，如6。

=INDEX(A:A,牛2)&""，将引用过来的0转变成空文本，这样看起来美观点，如7。

6	7
汕头	汕头
潮州	潮州
揭阳	揭阳
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	
0	

图7-49 庖丁解牛3

本来还想将公式大卸八块，现在才分成七块就搞定了，看来公式还不够长。

有了这次的剖解过程，下面这条公式理解起来就变得简单多了，重点看不同的地方即可。

```
=INDEX($B:$B,SMALL(IF($A$2:$A$19=$D2,ROW($A$2:$A$19),4^8),COLUMN(A1)))&""
```

\$A\$2:\$A\$19=\$D2就是区域\$A\$2:\$A\$19跟\$D2的比较，返回TRUE和FALSE。

IF(\$A\$2:\$A\$19=\$D2,ROW(\$A\$2:\$A\$19),4^8)让符合条件的显示本身行号，否则显示4^8。

SMALL(IF(\$A\$2:\$A\$19=\$D2,ROW(\$A\$2:\$A\$19),4^8),COLUMN(A1))，因为公式是向右拖拉，COLUMN(A1)可以水平获得序号，从而得到前N个最小值。

```
=INDEX($B:$B,SMALL(IF($A$2:$A$19=$D2,ROW($A$2:$A$19),4^8),COLUMN(A1)))&""
```

让符合条件的值显示出来，不符合的显示空。

对于高版本可以将公式略作改动进行容错。

地市的数组公式为：

```
=IFERROR(INDEX(A:A,SMALL(IF(MATCH($A$2:$A$19,$A$2:$A$19,0)=ROW($2:$19)-1,ROW($2:$19),ROW(A1))),""))
```

区县的数组公式为：



=IFERROR(INDEX(\$B:\$B,SMALL(IF(\$A\$2:\$A\$19=\$D2,ROW(\$2:\$19)),COLUMN(A1))),"

7.2.2 筛选符合条件的所有对应值

如图7-50所示，左边是明细表，要将品名为“汽水”的所有对应值筛选出来。

	A	B	D	E	F	G	H	I	J	K
1	日期	品名	数量	单价	金额	品名	汽水			
2	4-1	可乐	53	2	106					
3	4-1	可乐	120	2	240	日期	品名	数量	单价	金额
4	4-2	汽水	187	3	561					
5	4-3	可乐	254	2	508					
6	4-4	汽水	321	3	963					
7	4-5	可乐	388	2	776					
8	4-1	汽水	455	3	1365					
9	4-2	可乐	522	2	1044					
10	4-3	汽水	589	3	1767					
11	4-9	可乐	656	2	1312					
12	4-10	汽水	723	3	2169					
13	4-11	可乐	790	2	1580					
14	4-12	汽水	857	3	2571					
15	4-13	可乐	924	2	1848					
16	4-14	汽水	991	3	2973					

图7-50 筛选品名为“汽水”的所有对应值

按照万金油公式组合，我们很快就写出了下面的数组公式，但效果不是很完美，日期格式得到的是数字，如图7-51所示。

G4 {=INDEX(A:A,SMALL(IF(\$B\$2:\$B\$16=\$H\$1,ROW(\$2:\$16),4^8),ROW(\$2:\$16)))&"}

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	日期	品名	数量	单价	金额		品名	汽水					
2	4-1	可乐	53	2	106								
3	4-1	可乐	120	2	240		日期	品名	数量	单价	金额		
4	4-2	汽水	187	3	561		42462	汽水	187	3	561		
5	4-3	可乐	254	2	508		42464	汽水	321	3	963		
6	4-4	汽水	321	3	963		42461	汽水	455	3	1365		
7	4-5	可乐	388	2	776		42463	汽水	589	3	1767		
8	4-1	汽水	455	3	1365		42470	汽水	723	3	2169		
9	4-2	可乐	522	2	1044		42472	汽水	857	3	2571		
10	4-3	汽水	589	3	1767		42474	汽水	991	3	2973		
11	4-9	可乐	656	2	1312								
12	4-10	汽水	723	3	2169								
13	4-11	可乐	790	2	1580								
14	4-12	汽水	857	3	2571								
15	4-13	可乐	924	2	1848								
16	4-14	汽水	991	3	2973								
17													

(Note: In the original image, a red cloud-shaped callout points to the empty cell at G11, stating "不是日期格式" (Not a date format).)

图7-51 不显示日期格式

日期也是数字，因为公式在进行容错的时候后面有"&"，导致数字变成文本格式，文本格式的数字不能直接显示日期格式。

这时将公式略做改动，输入公式后将日期这一列的数字设置为日期格式就行了。



```
=IFERROR(INDEX(A:A,SMALL(IF($B$2:$B$16=$H$1,ROW($2:$16)),ROW(A1))),"
```

筛选到的数据按原来的顺序排列是理想状态，但很多时候，并不需要筛选出全部的数据，排列也与原来不一样。少了品名这一列，还有就是单价跟数量调换了位置，如图7-52所示。

	A	B	C	D	E	F	G	H	I	J
1	日期	品名	数量	单价	金额		品名	汽水		
2	4-1	可乐	53	2	106					
3	4-1	可乐	120	2	240		日期	单价	数量	金额
4	4-2	汽水	187	3	561					
5	4-3	可乐	254	2	508					
6	4-4	汽水	321	3	963					
7	4-5	可乐	388	2	776					
8	4-1	汽水	455	3	1365					
9	4-2	可乐	522	2	1044					
10	4-3	汽水	589	3	1767					
11	4-9	可乐	656	2	1312					
12	4-10	汽水	723	3	2169					
13	4-11	可乐	790	2	1580					
14	4-12	汽水	857	3	2571					
15	4-13	可乐	924	2	1848					
16	4-14	汽水	991	3	2973					

图7-52 不按原来顺序排列

要实现这种形式只需将公式略做改动就行，将INDEX函数的区域改成\$A:\$E，并增加INDEX函数的第三参数MATCH(G\$3,\$A\$1:\$E\$1,0)，也就是判断返回区域第几列。

```
=IFERROR(INDEX($A:$E,SMALL(IF($B$2:$B$16=$H$1,ROW($2:$16)),ROW(A1)),MATCH(G$3,$A$1:$E$1,0)),"")
```

下面依次查找满足条件的值。

我们知道查找对应值一般用VLOOKUP函数或者LOOKUP函数，但遗憾的是，并不是所有内容通过这两个函数都能查找到对应值。

如图7-53所示，依次查找学部对应的姓名。

	A	B	C	D	E	F	G	H
1	学部	姓名		学部	姓名	VLOOKUP	LOOKUP	
2	财经	张三		财经		张三	李阳峰	
3	网络技术应用	李四		网络技术应用		李四	蒋海坡	
4	计算机	王五		网络技术应用		李四	蒋海坡	
5	财经	马六		网络技术应用		李四	蒋海坡	
6	多媒体技术	周七		计算机		王五	呈浩歌	
7	网络技术应用	丽萍		计算机		王五	呈浩歌	
8	网络技术应用	海华						
9	财经	方海						
10	多媒体技术	凯丽						
11	财经	李阳峰						
12	网络技术应用	蒋海坡						
13	多媒体技术	蔡明波						
14	计算机	姜大雨						
15	多媒体技术	马华						
16	计算机	呈浩歌						
17								

图7-53 依次查找学部对应的姓名

用VLOOKUP函数不管学部出现几次，都只能查询到第一个对应值。

```
=VLOOKUP(D2,A:B,2,0)
```

而用LOOKUP函数不管学部出现几次，都只能查询到最后一个对应值。

```
=LOOKUP(1,0/(A$2:A$16=D2),B$2:B$16)
```

把中间这些对应值都忽略掉，如网络技术应用，第一个对应值是李四，第二个是丽萍，第



三个是海华。

怎么判断返回第几个对应值呢？

利用COUNTIF函数可以实现累计计数，如图7-54所示。

	D	E	F	G	H	I
1	学部	姓名	VLOOKUP	LOOKUP		
2	财经		张三	李阳峰	1	
3	网络技术应用		李四	蒋海波	1	
4	网络技术应用		李四	蒋海波	2	
5	网络技术应用		李四	蒋海波	3	
6	计算机		王五	呈浩歌	1	
7	计算机		王五	呈浩歌	2	
8						

图7-54 累计计数

如果考虑到容错处理，可以用：

```
=IFERROR(INDEX(B:B,SMALL(IF(A$2:A$16=D2,ROW($2:$16)),COUNTIF(D$2:D2,D2))),"
```

```
=COUNTIF(D$2:D2,D2)
```

D\$2:D2通过下拉会逐渐扩大区域，变成D\$2:D3、D\$2:D4…，区域不断变大，从而实现累计。

```
=INDEX(B:B,SMALL(IF(A$2:A$16=D2,ROW($2:$16)),COUNTIF(D$2:D2,D2)))
```

万金油公式组合最大的好处就是模式固定，只需改变条件或者返回第几个值两部分。只要理解这个固定模式，所有问题都能轻松解决。

7.3 小结与练习

数组公式的神奇之处，在于能够解决常规公式无法解决或者需要用辅助列才能解决的问题。虽然数组公式不是必须学会，但学会了至少能够给你提供另一种更好的解决思路。就如SUM函数的数组可取代SUMIF(SUMIFS)函数和COUNTIF(COUNTIFS)函数，并且可以实现它们所不能做到的效果。

1. 如图7-55所示，根据明细表计算每个学部的男女出现次数。
2. 如图7-56所示，计算A字头号店铺的销售量，用不少于两种方法完成。
3. 如图7-57所示，按指定次数重复文本。



	A	B	C	D	E	F	G	H	I	J
1	姓名	年龄	性别	成绩	学部		学部	男	女	
2	张三	22	女	86	财经		财经			
3	李四	21	男	73	网络技术应用		网络技术应用			
4	王五	18	男	90	计算机		计算机			
5	马六	19	女	100	财经		多媒体技术			
6	周七	23	男	93	多媒体技术					
7	丽萍	17	女	76	网络技术应用					
8	海华	18	男	65	网络技术应用					
9	方海	20	女	92	财经					
10	凯丽	17	女	77	多媒体技术					
11	李阳峰	20	男	85	财经					
12	蒋海坡	19	男	90	网络技术应用					
13	蔡明波	22	男	95	多媒体技术					
14	姜大雨	19	男	87	计算机					
15	马华	21	女	99	多媒体技术					
16	呈浩歌	19	男	100	计算机					
17										

图7-55 计算每个学部的男女出现次数

	A	B	C	D	E	F	G
1	店铺	商品	销售量		计算A字号店铺的销售量		
2	A001	T恤	101				
3	B001	西裤	102				
4	A002	衬衣	103				
5	B001	鞋子	104				
6	B002	T恤	105				
7	A002	西裤	106				
8	A003	衬衣	107				
9	A001	鞋子	108				
10							

图7-56 计算A字号店铺的销售量

	A	B	C	D	E
1	班级	次数		结果	
2	PS班	2		PS班	
3	Excel班	5		PS班	
4	Word班	3		Excel班	
5	PPT班	2		Excel班	
6				Excel班	
7				Excel班	
8				Excel班	
9				Word班	
10				Word班	
11				Word班	
12				PPT班	
13				PPT班	
14					
15					
16					

图7-57 按指定次数重复文本



第8章

惨了！走火入魔



当你学习到一定程度以后，发觉自己就像疯子一样，每天就是下载资料，一遍又一遍地研究公式，满脑子除了公式还是公式。卢子就曾经差点走火入魔，为了一个公式可以研究一两个月。整天想着公式，搞得无心上班，罪过！

8.1 从疯狂下载者变成真正学习者

在ExcelHome学习那段时间，是卢子最疯狂的时候。刚来到论坛时，简直就是进入天堂，看到很多人发帖子，帖子里面有很多精品资料，于是开始疯狂下载收集函数与公式相关资料。过了几年，卢子收集到相当全的函数资料。如图8-1所示，这个仅仅是冰山一角，实际是这个的多倍。



图8-1 函数资料

那时以下载资料最多为荣，时不时还跟朋友炫耀。你看，我这里资料那么多！以为有了这些就是高手。但猛然间发现，这些资料只是满满地占据了电脑硬盘，自己还是一无所知。我沉痛



地发现，我忽略了一个实质问题，那就是学习要靠自己的力量，而不是他人的力量，他人的帮助只能推动学习的步伐，但代替不了学习的全部过程。这世界上没有免费的午餐，这句话一点都不错，当我们为下载了几GB甚至几十GB的免费资料而沾沾自喜的时候，我们已经失去了宝贵的时间，那就是生命，人生能有几回搏？我们浪费掉一个小时，我们的生命中就失去一个小时。

本来我们一生中就做了很多没有意义的事情而浪费时间了，那么学习上就不要再浪费了，多下点功夫认真学点东西比在下载中自我陶醉有意义得多。掌握科学的学习方法，风雨无阻，有勇气和毅力，相信自己一定能做到，把自己真正从一个疯狂下载者变成学习者，那么你也一定会再雄心勃勃。

8.2 为公式消得人憔悴

卢子自从学会了几个数组公式以后，就一心想成为高手，所以凡是高手需要掌握的东西卢子都苦心专研。几年后，卢子才意识到这只是个骗局，让人走上一条不归路，越陷越深，最后不能自拔。

8.2.1 三大隐藏函数的用法

卢子经常会看到DATEDIF函数的用法，但帮助文件中没有这个函数的任何解释，这就奇怪了，怎么回事呢？

后来经过多方打听才知道，这个是隐藏函数，似乎Excel也有难言之隐。唉！家家都有本难念的经，没想到微软公司也一样，这么好用的函数居然被隐藏起来而不告知。

Excel中其实有三个隐藏函数，分别是DATEDIF、NUMBERSTRING和DATESTRING函数。下面通过一些案例来说明。

首先，一起来看看DATEDIF函数的介绍。



DATEDIF函数用于计算两个日期之间的天数、月数或年数。提供此函数是为了与Lotus 1-2-3 兼容。

函数语法：

=DATEDIF(起始日期,结束日期,返回类型)

参数3返回类型共有6种。

- (1) “Y” 时间段中的整年数。
- (2) “M” 时间段中的整月数。
- (3) “D” 时间段中的天数。
- (4) “MD” 日期中天数的差，忽略日期中的月和年。
- (5) “YM” 日期中月数的差，忽略日期中的日和年。
- (6) “YD” 日期中天数的差，忽略日期中的年。

这个介绍很抽象，下面通过实例详细说明。

有一个女生的微博上这样写：今年谁愿意陪我过光棍节，我让他明年过父亲节。

谁敢过这个父亲节，请举手？反正我是不敢。

假如现在是2012年，光棍节就是2012-11-11，明年的父亲节就是2013-6-16，将日期录入Excel中，如图8-2所示。

	A	B	C	D
1	光棍节	父亲节	相差几个月	
2	2012-11-11	2013-6-16	7	
3				

图8-2 日期相差月份

=DATEDIF(A2,B2,"M")

相差7个月，这个才是亮点！
如果想知道差了几天：

=DATEDIF(A2,B2,"D")

也才217天。这时你就会觉得Excel其实很重要，兄弟们好好学习吧！

这个函数用得最多的还是在计算工龄上，如图8-3所示。

	A	B	C	D	E
1	姓名	参加工作时间	截止日期	工龄	
2	邵广金	2005-7-4	2012-12-31	7年5个月	
3	杨承亮	2003-7-1	2012-12-31	9年5个月	
4	陈友益	1998-9-1	2012-12-31	14年3个月	
5	陈庆云	1999-8-5	2012-12-31	13年4个月	
6	陈友根	2006-7-2	2012-12-31	6年5个月	
7	邵松茂	2006-7-3	2012-12-31	6年5个月	
8					

图8-3 计算工龄

=DATEDIF(B2,C2,"Y")&"年"&DATEDIF(B2,C2,"YM")&"个月"

参数设置为“Y”就是算两日期相差的年份，“YM”就是算忽略日和年算月份，年月都有了，接着就用“&”将所有内容连接起来。

其次就是NUMBERSTRING函数，这个的使用频率远比不上DATEDIF函数，主要是因为函数名太长，很难记住，再有就是只有特定人群才会使用。

NUMBERSTRING函数是干什么用呢？一起来了解下：

NUMBERSTRING函数就是将小写数字转换成大写。

函数语法：

=NUMBERSTRING(数字,返回类型)

参数2返回类型共有3种，分别是1、2、3，简单易记。

如图8-4所示，依次返回3种形式的大写。



	A	B	C	D	E
1	数字	大写1	大写2	大写3	
2	34210	三万四千二百一十	叁万肆仟贰佰壹拾	三四二一〇	
3	4211	四千二百一十一	肆仟贰佰壹拾壹	四二一一	
4	2	二	贰	二	
5	10	一十	壹拾	一〇	
6	214	二百一十四	贰佰壹拾肆	二一四	
7					

图8-4 小写转大写

```
=NUMBERSTRING(A2,1)
=NUMBERSTRING(A2,2)
=NUMBERSTRING(A2,3)
```

微博上有一个有趣的段子：书桓，你走的第一天，想你；书桓，你走的第二天，想想你；书桓，你走的第三天，想想想想你；书桓，你走的第四天，想想想想想想你；书桓，你走的第五天，想想想想想想想想你；书桓，你走的第六天，想想想想想想想想想想你；书桓，你走的第七天，想想想想想想想想想想想想你；书桓，你走的第八天，想想想想想想想想想想想想想想你。

你；书桓，你走的第四天，想想想想想想想想你；书桓，你走的第五天，想想想想想想想想想想你；书桓，你走的第六天……

也许琼瑶就是利用Excel生成这些，如图8-5所示。

	A
1	书桓，你走的第一天，想你
2	书桓，你走的第二天，想想你
3	书桓，你走的第三天，想想想想你
4	书桓，你走的第四天，想想想想想想你
5	书桓，你走的第五天，想想想想想想想想你
6	书桓，你走的第六天，想想想想想想想想想想你
7	书桓，你走的第七天，想想想想想想想想想想想想你
8	书桓，你走的第八天，想想想想想想想想想想想想想想你
9	

图8-5 琼瑶段子

```
= "书桓，你走的第"&NUMBERSTRING(ROW(),1)&"天 "&REPT("想你",ROW())
```

使用NUMBERSTRING函数将小写的数字转换成大写，再利用REPT函数将“想你”重复N次。



知识扩展

其实NUMBERSTRING函数可以用TEXT函数取代。对于小写转大写，TEXT函数可以做到4种形式，如图8-6所示。

```
=TEXT(A2,"[DBNum1]")
=TEXT(A2,"[DBNum2]")
=TEXT(A2,"[DBNum1]0")
=TEXT(A2,"[DBNum2]0")
```

	A	B	C	D	E
1	数字	大写1	大写2	大写3	大写4
2	34210	三万四千二百一十	叁万肆仟贰佰壹拾	三四二一〇	叁肆贰壹零
3	4211	四千二百一十一	肆仟贰佰壹拾壹	四二一一	肆贰壹壹
4	2	二	贰	二	贰
5	10	一十	壹拾	一〇	壹零
6	214	二百一十四	贰佰壹拾肆	二一四	贰壹肆
7					

图8-6 小写转大写的4种形式

最后就是DATESTRING函数，因为使用的频率实在太低了，已经被绝大多数人遗忘。DATESTRING函数的作用就是将日期转变成文本日期的格式，如图8-7所示。

	A	B	C
1	日期	效果	公式
2	2016-10-8	16年10月08日	=DATESTRING(A2)
3	2016-10-9	16年10月09日	=DATESTRING(A3)
4	2016-10-10	16年10月10日	=DATESTRING(A4)
5	2016-10-11	16年10月11日	=DATESTRING(A5)
6	2016-10-12	16年10月12日	=DATESTRING(A6)
7	2016-10-13	16年10月13日	=DATESTRING(A7)
8	2016-10-14	16年10月14日	=DATESTRING(A8)
9	2016-10-15	16年10月15日	=DATESTRING(A9)
10	2016-10-16	16年10月16日	=DATESTRING(A10)
11			

图8-7 文本格式的日期

其实转换成文本格式的日期，TEXT函数也可以轻易做到。

```
=TEXT(A2,"yy年mm月dd日")
```

8.2.2 两大让你望而却步的高级函数

Excel存在两个高级函数，一个是MMULT函数，另一个是FREQUENCY函数。这两个函数有一些共同特点就是参数很简单，只有两个，但结合其他函数能变化无穷，让人捉摸不透。卢子当初学习这两个函数伤透了脑筋，不过听说学会了就晋升为高级，所以动力无限。

如图8-8所示，已知每个学生考试科目的成绩，现在要获取总分前5名。

	A	B	C	D	E	F	G
1	姓名	语文	数学	英语		总成绩前5名	
2	张三	54	73	69		227	
3	李四	62	13	78		196	
4	王五	78	39	79		196	
5	马六	10	77	13		188	
6	周七	7	84	81		172	
7	丽萍	40	91	96			
8	海华	62	87	39			
9	方海	50	33	53			
10							

图8-8 总成绩前5名

正常的办法是先用辅助列求总分，在E2单元格输入公式，并向下填充公式。

```
=SUM(B2:D2)
```

最后用LARGE函数求前N大。

```
=LARGE(E$2:E$9,ROW(A1))
```

也就是通过一个辅助列间接实现，现在不想用辅助列，要怎么才能一步到位？

```
=LARGE(MMULT(B2:D9,ROW(1:3)^0),ROW(1:5))
```

这个公式是多单元格数组，需要选择F2:F6区域，输入公式后按Ctrl+Shift+Enter组合键结束，后面涉及这个函数都需要用三键结束。

```
MMULT(B2:D9,ROW(1:3)^0)
```

这部分其实就相当于辅助列SUM函数，现在来了解下MMULT函数语法。

```
=MMULT(参数1,参数2)
```

参数1、参数2为要进行矩阵乘法运算的两个数组，可以是单元格区域、数组常量或引用。参数1的列数必须与参数2的行数相



前面知道SUM函数可以实现多条件求和，当数据比较庞大时运算效率比较低。而MMULT函数一般是多单元格数组公式，效率大大高于多个公式。

```
=MMULT(--(G2:G5=TRANSPOSE(E2:E16)),(C2:C16=H1:I1)*D2:D16)
```

G2:G5不能直接与E2:E16比较，需要转置后才可以比较，比较后得到的结果是逻辑值，也就是相同即TRUE，不同即FALSE。使用减负的作用就是将逻辑值转换成数值，只有数值才能参与运算。

```
--(G2:G5=TRANSPOSE(E2:E16))
```

参数1获得一个4行15列的矩阵区域。

```
(C2:C16=H1:I1)*D2:D16
```

参数2获得一个2列15行的矩阵区域。

也就是说，参数1列数15跟参数2行数15相同，这个要求很特殊，如果不同的话会出错。如下面这个公式将条件的顺序调换：

```
=MMULT((G2:G5=TRANSPOSE(E2:E16))*D2:D16,--(C2:C16=H1:I1))
```

输入公式后直接返回#N/A，所以这个函数没有SUM函数的数组灵活，其条件可以任意写。限制太多，这才是MMULT函数难学之处。

MMULT函数还有一个特殊用途，就是解方程。

如图8-11所示，是一个三元一次方程组，如何求出方程组的解？

将方程的系数和常量依次输入B2:E4单元格区域，选择C7:C9单元格区域，输入多单元格数组公式，按Ctrl+Shift+Enter组合键结束。结果如图8-12所示。

```
=MMULT(MINVERSE(B2:D4),E2:E4)
```

$$\begin{cases} x+y-z=6 \\ x-3y+2z=1 \\ 3x+2y-z=4 \end{cases}$$

图8-11 三元一次方程组

	A	B	C	D	E
1	方程	x系数	y系数	z系数	常量
2	$x+y-z=6$	1	1	-1	6
3	$x-3y+2z=1$	1	-3	2	1
4	$3x+2y-z=4$	3	2	-1	4
5					
6		求解	值		
7		x=	2.2		
8		y=	-6.4		
9		z=	-10.2		

图8-12 用MMULT函数解方程



本例根据数学中“克莱姆法则”，利用方程组系数的矩阵求解多元一次方程组。其中MINVERSE函数用于求出方程组的系数逆矩阵，MMULT函数将系数逆矩阵与方程组的常量数组相乘，得到方程组的解。

这个有点复杂，数学基础不好的基本看不懂，知道有这么一回事就行。

学完MMULT函数，再接着看第二个高级函数FREQUENCY。这个函数是干什么用的？不急，先看看例子。如图8-13所示，统计各区间销售额的人数。

	A	B	C	D	E	F	G	H
1	编号	姓名	部门	销售额		区间	人数	
2	XS15	杜月	销售(1)部	497200		<=400000万	2	
3	XS6	杜乐	销售(1)部	504000		400001-450000万	14	
4	SC25	许泽平	销售(3)部	443550		450001-480000万	14	
5	XS40	詹荣华	销售(3)部	225000		480001-500000万	6	
6	XS1	刘丽	销售(1)部	500001		>=500001	8	
7	XS19	马路刚	销售(2)部	478550				
8	XS29	卢红燕	销售(1)部	500000				
9	XS3	李诗	销售(2)部	468500				
10	XS10	刘志刚	销售(3)部	440000				
11	XS43	张小丽	销售(2)部	479000				
12	XS38	唐艳霞	销售(1)部	500500				
13	SC4	杜乐	销售(3)部	335500				
14	XS31	张恬	销售(3)部	428500				
15	XS8	李诗诗	销售(1)部	509550				

图8-13 统计各区间销售额的人数

我们知道COUNTIF、COUNTIFS函数可以用于条件计数，不过现在划分成5个区间，要写成下面的5条公式才能统计。

```
=COUNTIF(D:D,"<=400000")
=COUNTIFS(D:D,">=400001",D:D,"<=450000")
=COUNTIFS(D:D,">=450001",D:D,"<=480000")
=COUNTIFS(D:D,">=480001",D:D,"<=500000")
=COUNTIFS(D:D,">=500001")
```

这样写相当烦琐，如果区间再多几个还容易出错。这时就是FREQUENCY函数一显身手的好机会。

函数语法：

```
=FREQUENCY(数据源,分段点)
```

结果：以分段点为间隔，统计数据源值在各段出现的频数。

其中：数据源为对一行/一列单元格或一个连续的单元格区域的引用；分段点为对一行/一列单元格或一个连续的单元格区域的引用。



数据引用支持跨工作表、工作簿。

公式输入方法：

以多单元格数组方式输入，且必须是纵向数组；

所选单元格数比分段点个数大1，以统计数据源大于分段点最大值的频数。

看完帮助文件可以知道，这个只需设置分段点就可以统计人数。5个区间可以分成4个分段点：

```
{400000,450000,480000,500000}
```

分段点有了，公式就推出来了，选择单元格区域G2:G6，输入以下公式，按Ctrl+Shift+Enter组合键结束，这个与MMULT函数一样，也是多单元格数组公式。

```
=FREQUENCY(D2:D45,{400000,450000,480000,500000})
```

有了FREQUENCY函数，现在区间再多都不怕了。

FREQUENCY函数还有一个妙用，就是忽略重复值，数据排名的时候经常用到。

如图8-14所示，直接用LARGE函数求销售额前4名的时候出错。

	A	B	C	D	E	F	G
1	编号	姓名	部门	销售额(万元)		销售额前4名	
2	XS15	杜月	销售(1)部	50		51	
3	XS6	杜乐	销售(1)部	50		50	
4	SC25	许泽平	销售(3)部	44		50	
5	XS40	詹荣华	销售(3)部	23		50	
6	XS1	刘丽	销售(1)部	50			
7	XS19	马路刚	销售(2)部	48			
8	XS29	卢红燕	销售(1)部	50			
9	XS3	李诗	销售(2)部	47			
10	XS10	刘志刚	销售(3)部	44			
11	XS43	张小丽	销售(2)部	48			
12	XS38	唐艳霞	销售(1)部	50			
13	SC4	杜乐	销售(3)部	34			
14	XS31	张恬	销售(3)部	43			
15	XS8	李诗诗	销售(1)部	51			

图8-14 用LARGE函数求销售额前4名出错

```
=LARGE(D$2:D$15,ROW(A1))
```

销售额出现了重复，50万元的有好几个，在我们中国，不管重复多少个都是只算一个，第1名是51万元，第2名是50万元，第3名是48万元，第4名是47万元。这也许与中国人好面子有关，如果说销售额48万元是第7名，肯定有很多人不高兴。

解决方案：

```
=LARGE(IF(FREQUENCY(D2:D15,D2:D15),D2:D15),ROW(1:4))
```

公式有点长，拆分看一步步了解。

```
=FREQUENCY(D2:D15,D2:D15)
```

就是让非首次出现的返回0，如图8-15所示，50万元的第2次起都是返回0。



=IF(FREQUENCY(D2:D15,D2:D15),D2:D15)

就是让首次出现的返回数据本身，其他返回FALSE，如图8-16所示。

	A	B	C	D	E	H
1	编号	姓名	部门	销售额(万元)		
2	XS15	杜月	销售(1)部	50	5	
3	XS6	杜乐	销售(1)部	50	0	
4	SC25	许泽平	销售(3)部	44	2	
5	XS40	詹荣华	销售(3)部	23	1	
6	XS1	刘丽	销售(1)部	50	0	
7	XS19	马路刚	销售(2)部	48	2	
8	XS29	卢红燕	销售(1)部	50	0	
9	XS3	李诗	销售(2)部	47	1	
10	XS10	刘志刚	销售(3)部	44	0	
11	XS43	张小丽	销售(2)部	48	0	
12	XS38	唐艳霞	销售(1)部	50	0	
13	SC4	杜乐	销售(3)部	34	1	
14	XS31	张恬	销售(3)部	43	1	
15	XS8	李诗诗	销售(1)部	51	1	

图8-15 让非首次出现的返回0

	A	B	C	D	E	H
1	编号	姓名	部门	销售额(万元)		
2	XS15	杜月	销售(1)部	50	50	
3	XS6	杜乐	销售(1)部	50	FALSE	
4	SC25	许泽平	销售(3)部	44	44	
5	XS40	詹荣华	销售(3)部	23	23	
6	XS1	刘丽	销售(1)部	50	FALSE	
7	XS19	马路刚	销售(2)部	48	48	
8	XS29	卢红燕	销售(1)部	50	FALSE	
9	XS3	李诗	销售(2)部	47	47	
10	XS10	刘志刚	销售(3)部	44	FALSE	
11	XS43	张小丽	销售(2)部	48	FALSE	
12	XS38	唐艳霞	销售(1)部	50	FALSE	
13	SC4	杜乐	销售(3)部	34	34	
14	XS31	张恬	销售(3)部	43	43	
15	XS8	李诗诗	销售(1)部	51	51	

图8-16 让首次出现的返回数据本身

LARGE函数忽略文本，所以前N大都只是针对数字本身排名，如50、44等。

接着聊中国式排名，如图8-17所示。

在E2单元格输入公式，按Ctrl+Shift+Enter组合键结束，这个公式不同于前面的多单元格数组，它只是一个单元格的数组公式。

	A	B	C	D	E	F
1	编号	姓名	部门	销售额(万元)	中国式排名	
2	XS15	杜月	销售(1)部	50	2	
3	XS6	杜乐	销售(1)部	50	2	
4	SC25	许泽平	销售(3)部	44	5	
5	XS40	詹荣华	销售(3)部	23	8	
6	XS1	刘丽	销售(1)部	50	2	
7	XS19	马路刚	销售(2)部	48	3	
8	XS29	卢红燕	销售(1)部	50	2	
9	XS3	李诗	销售(2)部	47	4	
10	XS10	刘志刚	销售(3)部	44	5	
11	XS43	张小丽	销售(2)部	48	3	
12	XS38	唐艳霞	销售(1)部	50	2	
13	SC4	杜乐	销售(3)部	34	7	
14	XS31	张恬	销售(3)部	43	6	
15	XS8	李诗诗	销售(1)部	51	1	

图8-17 中国式排名

=COUNT(1/FREQUENCY(D\$2:D\$15,IF(D\$2:D\$15>=D2,D\$2:D\$15)))

首先使用IF函数判断D列的销售额是否大于等于当前值，如果是，则返回D列的销售额，否则返回FALSE，如IF(D\$2:D\$15>=D2,D\$2:D\$15)返回：

{50;50;FALSE;FALSE;50;FALSE;50;FALSE;FALSE;FALSE;50;FALSE;FALSE;51}

接着利用FREQUENCY函数的第2参数忽略逻辑值的特性，以大于等于当前值的D列销售额区间分段点统计D列销售额的分布频率。

{13;0;0;0;0;1;0}

最后使用COUNT(1/FREQUENCY)统计非0数据的个数，得出大于等于当前值销售额的不重复个数，则得出当前值的排名名次。

其实刚开始接触这两个高级函数还是有点难，但如果能够掌握它的特性，还是很好理解的，毕竟函数都是相通的。

8.2.3 与名称共存亡的宏表函数

宏表函数是个“老古董”，实际上是现在广泛使用的VBA的“前身”。虽然后来的各版本已经不再使用它，但还能支持。宏表函数很奇葩，必须得定义名称才能使用，脱离了名称就没有存在的价值。

宏表函数有很多，但实际用到的并不多，挑几个常用的学习下用法即可。

如图8-18所示，是一份出差伙食补贴费明细。伙食补贴费都是用表达式显示在单元格，现在要合计费用，该如何做？

	A	B	C	D	E
1	报销人	车号	货物送达地	伙食补贴费	合计
2	高吉	苏E36013	临沂	20+10+15	
3	马磊	苏E36013	临沂	10-20+40	
4	高吉	苏E36013	丰县	15*2	
5	谢金楼	苏E36013	丰县	15*4	
6					

图8-18 计算表达式

STEP 01 选择E2单元格，单击“公式”选项卡中的“定义名称”按钮，在“名称”文本框中输入“合计”，在“引用位置”编辑框输入下面的公式，最后单击“确定”按钮，如图8-19所示。

=EVALUATE(D2)




图8-19 创建名称

STEP 02 定义完名称后，只需在E2单元格输入公式，并向下方填充公式，即可自动统计表达式，如图8-20所示。

=合计



有些人喜欢将需要统计的数据用背景色填充，以为这样比较醒目，统计起来就方便。当然前提是你用计算器统计，如果用Excel那就是自找麻烦。

如图8-21所示，对需要统计的津贴用不同背景色填充，要分别统计两种颜色的津贴总数。

E2

:

✕

✓

f_x

=合计

	A	B	C	D	E
1	报销人	车号	货物送达地	伙食补贴费	合计
2	高吉	苏E36013	临沂	20+10+15	45
3	马磊	苏E36013	临沂	10-20+40	30
4	高吉	苏E36013	丰县	15*2	30
5	谢金楼	苏E36013	丰县	15*4	60
6					

图8-20 自动统计表达式

	A	B	C	D	E
1	工号	津贴		背景色	津贴
2	12051136	200			
3	12051137	100			
4	12051138	150			
5	12051139	200			
6	12051142	50			
7	12052260	50			
8	12052261	200			
9	12052262	350			
10	12052263	50			
11	12052264	100			
12	12052265	100			
13	12052266	150			
14	12052267	150			
15	25137771	150			
16	25137772	150			

图8-21 统计背景色

STEP 01 选择C2单元格，单击“公式”选项卡中的“定义名称”按钮，在“名称”文本框中输入“颜色”，在“引用位置”编辑框中输入下面的公式，最后单击“确定”按钮，如图8-22所示。

=GET.CELL(63,B2)




图8-22 定义名称

STEP 02 定义完名称后，只需在C2单元格输入公式，并向下填充公式，即可自动统计表达式，有背景色的就大于0，没有背景色的就是0，如图8-23所示。

C2	:	X	✓	fx	=颜色
	A	B	C	D	E
1	工号	津贴		背景色	津贴
2	12051136	200	0		
3	12051137	100	46		
4	12051138	150	46		
5	12051139	200	0		
6	12051142	50	10		
7	12052260	50	0		
8	12052261	200	10		
9	12052262	350	0		
10	12052263	50	0		
11	12052264	100	46		
12	12052265	100	0		
13	12052266	150	0		
14	12052267	150	10		
15	25137771	150	0		
16	25137772	150	0		
17					

图8-23 背景色对应值

橙色的背景色为46，所以公式可设置为：

```
=SUMIF(C:C,46,B:B)
```

绿色的背景色为10，所以公式可设置为：

```
=SUMIF(C:C,10,B:B)
```

如果背景色多的话，需要再次定义一个名称作为辅助列，获取D列的颜色对应值。

除了背景色，常用的还有一个字体颜色，参数1为24代表字体颜色。

```
=GET.CELL(24,B2)
```

GET.CELL参数1可以设置成1~66，共66种，有兴趣的读者可以去测试。

高版本的Excel可以按背景色进行筛选，再借助SUBTOTAL函数就可以按背景色筛选求和。

在C17单元格输入公式：

```
=SUBTOTAL(9,B2:B16)
```

如图8-24所示，筛选橙色的背景色，就可以统计出来。不过有一个局限性就是只能按筛选的颜色进行统计，不能分别统计各种颜色。

B17	:	X	✓	fx	=SUBTOTAL(9,B2:B16)
	A	B	C	D	E
1	工号	津贴			
3	12051137	100			
4	12051138	150			
11	12052264	100			
17		350			
18					

图8-24 筛选颜色统计

除了这两个宏表函数，还有一个比较常用，就是GET.WORKBOOK函数，可以利用这个函数创建目录。

在工作中常常遇到这样一种情况，一个工作簿中有很多个格式类似的工作表，比如每个项目单位都提供一个工作表，一共是几十个或上百个，这时要用鼠标单击其中的某个工作表查阅数据，会很费力，于是希望能够为这些工作表制作出一个带超链接的目录，只要在目录上找到这个表的名字并单击，就可以直接跳转到想要的工作表了。那么对多个工作表如何快速创建超链接目录？

只需四步就能轻松搞定。

STEP 01 新建一张表格，工作表名称改成“目录”。单击“公式”选项卡中的“定义名称”按钮，在“名称”文本框中输入“目录”，在“引用位置”编辑框中输入下面的公式，最后单击“确定”按钮，如图8-25所示。

```
=GET.WORKBOOK(1)
```






图8-25 定义名称



知识扩展

直接用=GET.WORKBOOK(1)虽然可以获取目录，但当工作表更改时不能自动更新，这样略有缺陷。这时可以做一个改进，在后面用&T(NOW())。NOW函数是随时更新的，T(NOW())就是代表随时更新的空文本，最终公式为：

=GET.WORKBOOK(1)&T(NOW())

T函数的作用就是将数字变成空文本，而文本保持不变。说到T函数就必须说一下N函数，N函数的作用就是将文本转换成0，而数字保持不变，如图8-26所示。

N函数有一个经典案例，就是获取累计金额，比SUM函数使用起来还简单，如图8-27所示。

	A	B	C
1	内容	T	N
2	卢子	卢子	0
3	123		123
4	22:41		0.945139
5	1月8日		42743
6	不加班	不加班	0
7			

图8-26 T、N函数的作用

	A	B	C	D
1	日期	金额	累计	
2	2016-5-8	45	45	
3	2016-5-9	65	110	
4	2016-5-10	12	122	
5	2016-5-11	23	145	
6	2016-5-12	69	214	
7	2016-5-13	70	284	
8	2016-5-14	94	378	
9				
10				

图8-27 累计金额

STEP 02 在A1单元格输入下面的公式，并向下复制填充公式，得到包含工作簿名称的目录，如图8-28所示。

=INDEX(目录,ROW(A1))

	A	B	C	D	E	F	G
1	[8.2.3 与名称共存的宏表函数 (目录).xlsm]目录						
2	[8.2.3 与名称共存的宏表函数 (目录).xlsm]精确查找						
3	[8.2.3 与名称共存的宏表函数 (目录).xlsm]屏蔽错误值						
4	[8.2.3 与名称共存的宏表函数 (目录).xlsm]按顺序返回多列对应值						
5	[8.2.3 与名称共存的宏表函数 (目录).xlsm]按不同顺序返回多列对应值						
6	[8.2.3 与名称共存的宏表函数 (目录).xlsm]逆序查找						
7	[8.2.3 与名称共存的宏表函数 (目录).xlsm]趣聊IF						
8	[8.2.3 与名称共存的宏表函数 (目录).xlsm]多条件查找						
9	[8.2.3 与名称共存的宏表函数 (目录).xlsm]通配符查找						
10	[8.2.3 与名称共存的宏表函数 (目录).xlsm]区间查找						
11	#REF!						
12	#REF!						
13	#REF!						
14							

图8-28 包含工作簿名称的目录

超出部分显示#REF!这种错误，显然不美观，将公式稍作改动，让错误值显示空白。

=IFERROR(INDEX(目录,ROW(A1)),"")

STEP 03 去掉工作簿名称，在B1单元格输入下面的公式，并向下复制填充公式。

=IF(A1="", "", MID(A1, FIND("]", A1) + 1, 99))

利用FIND函数获取“]”的位置+1位，也就是工作表名的起始位置，MID函数从工作表名的起始位置提取99位。99是一个比较大的字符，就是保证能够比工作表名的长度还长，这样可以获取整个工作表名。

现在目录做出来了，如果再进行超级链接就更完美，以后只需单击工作表名称就可以链接到相应的表格。

STEP 04 在C1单元格输入下面的公式，向下复制填充公式，并隐藏A列，将字体稍微

调整一下，效果如图8-29所示。

=IF(B1="", "", HYPERLINK("#"&B1&"!A1", ">>>单击打开<<<"))

	B	C
1	目录	>>>单击打开<<<
2	精确查找	>>>单击打开<<<
3	屏蔽错误值	>>>单击打开<<<
4	按顺序返回多列对应值	>>>单击打开<<<
5	按不同顺序返回多列对应值	>>>单击打开<<<
6	逆序查找	>>>单击打开<<<
7	趣聊IF	>>>单击打开<<<
8	多条件查找	>>>单击打开<<<
9	通配符查找	>>>单击打开<<<
10	区间查找	>>>单击打开<<<
11		

图8-29 创建带有超级链接的目录

用多条公式看起来比较明朗而已，如果水平高只需两步就可以做到。首先定义一个叫“目录”的名称，INDEX函数放在名称中为了后期缩短字符用。

=INDEX(GET.WORKBOOK(1),ROW(目录!D1))&T(NOW())

定义名称后在E1单元格输入公式，并向下复制，效果如图8-30所示。

=IFERROR(HYPERLINK(目录&"!A1",MID(目录,FIND("]",目录)+1,99)),"")

	D	E
1	目录	
2	精确查找	
3	屏蔽错误值	
4	按顺序返回多列对应值	
5	按不同顺序返回多列对应值	
6	逆序查找	
7	趣聊IF	
8	多条件查找	
9	通配符查找	
10	区间查找	
11		
12		

图8-30 重新生成的目录



这个GET.WORKBOOK函数有38种用法，只需将1种改成2~38种的任意一种即可，有兴趣的朋友可以逐一测试，看得到什么内容。

最后强调一点，宏表函数的后缀跟普通表格不一样，需要将表格另存为“Excel启用宏的工作簿 (*.xlm)”，如图8-31所示。




图8-31 Excel 启用宏的工作簿 (*.xlm)

同时需要在“开发工具”选项卡中将宏的安全性设置为“启用所有宏”，如图8-32所示，否则宏表函数不能使用。




图8-32 启用所有宏

8.3 小结与练习

当初为了成为别人眼中的高手，卢子吃尽了苦头。在研究高级函数的时候，卢子一个月瘦了7斤，本来就瘦，现在就更是皮包骨。虽然后来零零散散用了几回高级函数，但使用频率太低了。说真的，如果时光能倒流的话，我宁愿不成为高手，只愿做一个普通的Excel爱好者。“表”妹“表”姐们好好研究吧，这可比吃减肥药安全多了！一来可以成为高手，二来可以减肥，一举两得。

1. 将你电脑上一些暂时不会去看的Excel资料删除，好好思考你要的是什么。
2. 如图8-33所示，统计两个日期相隔的年月。

	A	B	C	D
1	开始日期	截止日期	期限	
2	2009-7-10	2012-1-1	2年05个月	
3	2010-3-1	2011-11-2	1年08个月	
4	2008-1-2	2013-4-1	5年02个月	
5	2009-7-8	2014-8-27	5年01个月	
6				
7				

图8-33 统计两个日期相隔的年月



第9章

回归正途

学习Excel究竟为了什么？难道是为了成为高手？其实80%以上的人学习Excel仅仅是为了提高工作效率，早做完不加班。卢子当初也是突然间醒悟过来，不过还不算太迟。从此卢子走上了另外一条路——实用为王！



9.1 深思，迷途知返

卢子学习Excel已有好几年时间，一天晚上卢子彻底地失眠了！躺在床上翻来覆去地思考这样一个问题：我到底是为了什么才学习函数与公式的？

回想2009年，卢子对函数与公式还是门外汉。一直以来，卢子的英语从来没有及格过，以至于后来毕业看到函数与公式都是敬而远之，从来不敢奢望能学好函数与公式。但因为一次意外，见证了VLOOKUP函数的神奇，才打算学习函数与公式。

不会英语这个借口用了两年，是时候该放弃借口，行动起来了！刚开始学习函数与公式，是有点吃力，不过一次看不懂，再看一次，如果不懂就看十次百次，直到看懂为止。在学习的过程中看到了一句话：不会英语照样可以学好函数与公式。在这个关键时刻看到这话，顿时信心十足。

一个月过去了，卢子只懂两三个函数的最基础用法，不过还是很欣慰。

两个月过去了，卢子懂的函数也不多，也就是四五个。

.....

一年过去了，卢子已经能熟练掌握几十个函数的运用，常用的函数嵌套也会一些。

两年过去了，卢子已经不满足函数嵌套，已经开始研究数组公式跟高级函数。这时内心有一个声音在说：我要成为高手！

就是因为这个雄心勃勃的声音，卢子付出了很多。衣带渐宽终不悔，为函数消得人憔悴，这个就是卢子真实的写照。同事不经意的一句话：“卢子，你最近好像瘦了。”不说还没在意，一称体重，吓了一跳，比吃减肥药还有效，瘦了好多斤。

这时，卢子才意识到自己陷入了一个恶性循环中，不断地钻研公式，越来越高深，直到迷



失自我。这样的学习状况已经超出了一个Excel爱好者的痴迷程度，只有专家们才需要如此钻研。难不成自己也想成为专家？

当初学习函数与公式，只是为了提高工作效率，早做完不加班，然后顺便帮助需要帮助的人，让他们也因此提高工作效率。这时卢子恍然大悟，原来自己错了，钻研公式并不是自己的初衷！

知错能改，还是好孩子。卢子第二天将电脑上所有与Excel有关的资料全部删除，整整删了一个晚上才全部删除干净，顿时资料盘可用空间增加了很多。删除后，卢子整个人都觉得轻松起来。听会歌儿，睡了一个好觉，从明天起，卢子将成为一个全新的卢子。

9.2 学以致用

删除资料的前几天，卢子心里感觉空荡荡的，好像少了点什么似的？下班后空闲的时间比较多，偶尔看电影、听歌还行，长时间肯定受不了。后来，卢子转念一想，怎么不把过去所学用在工作上，一来可以验证自己这几年的学习效果，二来下班后充实一点儿。卢子以前制作的表格其实挺落后的，很少用到公式，也就只有SUM函数和VLOOKUP函数。反正现在有的是时间，可以对所有表格进行优化处理，提高工作效率。

9.2.1 优化出差费用统计表

如图9-1所示，是2010年9月份出差费用明细表，其实就是流水账，把所有消费项目都记录下来。



	A	B	C	D	E
1	9月份出差费用明细表				
2					
3					陈锡卢
4					2010-9-18
5					
6	日付	类型	人数	人员	金额
7	9月6日	朝食(一天)	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	27.0
8	9月6日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	70.0
9	9月6日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	48.0
10	9月7日	朝食(一天)	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	28.0
11	9月7日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	78.0
12	9月7日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	71.0
13	9月8日	朝食(两天)	3	郑春琴、宋晓媛、陈锡卢	55.8
14	9月8日	昼食	3	郑春琴、宋晓媛、陈锡卢	49.0
15	9月8日	夕食	3	郑春琴、宋晓媛、陈锡卢	50.0
16	9月8日	杂费			30.0
17	9月8日	杂费			13.3

图9-1 出差费用明细表

粗略看了下表格，感觉都挺规范的，记录得很好，但实际上还是有一部分内容可以进行优化。

制表日期是采用快捷键Ctrl+;生成的，是静态日期，这份表格明显需要一个动态日期，每天记录的时候日期能够自行变动，而非静止不变。可以设置公式为：

```
=TODAY()
```

这样就能获取动态日期，这时如果能够显示星期几会更好，将公式更改为：

```
=TEXT(TODAY(),"e-m-d aaaa")
```

e-m-d代表年月日，e可以用yyyy表示，aaaa代表星期几。

人数是手工输入的，按理说也没问题，但手工输入难免会出差错。仔细观察人员，发现有个规律，人员都用顿号（、）隔开，也就是说人员比顿号（、）多一个，只要能统计顿号（、）的数量，人员的数量就推出来了。如果直接统计顿号（、）有点难，这时可以间接统计，公式为：

```
=LEN(D7)-LEN(SUBSTITUTE(D7," ",))
```



SUBSTITUTE函数就是将顿号（、）替换掉，LEN(SUBSTITUTE())就是统计替换掉后的字符数，这时总字符数减去替换掉的字符数就等于顿号（、）的数量。

人数=顿号（、）数+1，也就是：

```
=LEN(D7)-LEN(SUBSTITUTE(D7," "))+1
```

如图9-2所示，出现一个小问题，就是当人员列为空白时，会出现错误。

	A	B	C	D	E
1	9月份出差费用明细表				
2					
3					陈锡卢
4					2014-7-29 星期二
5					
6	日付	类型	人数	人员	金额
7	9月6日	朝食（一天）	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	27.0
8	9月6日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	70.0
9	9月6日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	48.0
10	9月7日	朝食（一天）	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	28.0
11	9月7日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	78.0
12	9月7日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	71.0
13	9月8日	朝食（两天）	3	郑春琴、宋晓媛、陈锡卢	55.8
14	9月8日	昼食	3	郑春琴、宋晓媛、陈锡卢	49.0
15	9月8日	夕食	3	郑春琴、宋晓媛、陈锡卢	50.0
16	9月8日	杂费	1		30.0
17	9月8日	杂费	1		13.3

图9-2 人员列空白时出错

当人员列为空白时，人数也显示空白才对，这时就得增加一个判断。

```
=IF(D7="", "", LEN(D7)-LEN(SUBSTITUTE(D7," "))+1)
```

设置好人数，再来看金额。公司规定：每餐每人的平均金额不得超过18元。既然如此，就得加一个警告，如果平均金额超过18元就显示红色字体、加粗显示。

- STEP 01** 如图9-3所示，选择金额区域E7:E40，单击“条件格式”按钮，选择“新建规则”选项。
- STEP 02** 如图9-4所示，在弹出的“新建格式规则”对话框中，单击“使用公式确定要设置格式的单元格”选项，在“为符合此公式的值设置格式”编辑框中输入下面的公式，并单击“格式”按钮。

```
=E7/C7>18
```




图9-3 新建规则




图9-4 设置公式

STEP 03 如图9-5所示，在弹出的“设置单元格格式”对话框中，对字体的字形进行加粗设置，颜色设置为红色，最后连续单击两次“确定”按钮。

经过了3步设置，最终效果如图9-6所示。





图9-5 设置字体格式

	A	B	C	D	E
1	9月份出差费用明细表				
2					
3					陈锡卢
4					2014-7-29 星期二
5					
6	日付	类型	人数	人员	金额
7	9月6日	朝食(一天)	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	27.0
8	9月6日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	70.0
9	9月6日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	48.0
10	9月7日	朝食(一天)	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	28.0
11	9月7日	昼食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	78.0
12	9月7日	夕食	4	郑春琴、宋晓媛、陈焕锐、陈锡卢	71.0
13	9月8日	朝食(两天)	3	郑春琴、宋晓媛、陈锡卢	55.8
14	9月8日	昼食	3	郑春琴、宋晓媛、陈锡卢	49.0
15	9月8日	夕食	3	郑春琴、宋晓媛、陈锡卢	50.0
16	9月8日	杂费			30.0
17	9月8日	杂费			13.3

图9-6 金额超出警告数值的效果图

截止到目前都是对数据源进行处理，其实最重要的还是对金额进行各项汇总，如图9-7所示。

	F	G	H	I	J	K	L
4							
5							
6		项目	金额		项目	金额	
7		食事代	2,270.0		应归还	0.0	
8		杂费	43.3		应领取	313.3	
9		总费用	2,313.3		核算	TRUE	
10		预支	2,000.0				
11		差额	-313.3				
12							

图9-7 金额汇总

“食事代”等同于中文的餐费，就是统计朝食、昼食、夕食三者的金额。卢子对日语一窍不通，这些名词都是抄袭领导的。

也就是说，只要含有“食”字，就是食事代，所以可以将公式设置为：

=SUMIF(B:B,"*食*",E:E)

“*”是通配符，代表所有字符。还有一个通配符“？”，代表一个字符。“*食*”的意思就是包含“食”。

接着看杂费，这个比较好理解，不需要转弯。

=SUMIF(B:B,"杂费",E:E)

总费用在这里只包含食事代跟杂费，所以只要将两者相加即可。

=SUM(H7:H8)

预支就是每次出差前根据出差的天数和人员初步预算的金额，向财务部先预支。

差额就是总费用跟预支的比较，正常的话差额为正数最好，当差额为负数而且金额

很大时，唯一的结果就是挨批。出差不是享受，要学会控制！

$$=H_{10}-H_9$$

STEP 01 如图9-8所示,对差额进行警告提示。单击“条件格式”按钮,选择“突出显示单元格规则”→“小于”命令。




图9-8 突出显示单元格规则

STEP 02 如图9-9所示，在弹出的“小于”对话框，输入0，单击“确定”按钮。



图9-9 小于0设置

经过设置后，只要差额为正数的话就不会出现警告。



应归还，就是差额为正数，如果为负数就不用还。

```
=IF(H11>0,H11,0)
```

应领取，就是差额为负数，这种情况出现次数越少越好。

```
=IF(H11<0,ABS(H11),0)
```

核算，就是对总费用、预算、应归还、应领取这4个项目进行核对，因为都是通过公式得到，一般不会有问题。

```
=H9+K7=H10+K8
```

9.2.2 优化检查日报

如图9-10所示，是产品的检查表，每天需要将前一天检查的产品明细发送给领导。因为序号是通过输入1，下拉生成。领导只需要看一天的检查情况，其他都不看，也就要隐藏掉。这样问题就来了，隐藏后前面的1~N条记录就没有，就从N+1行开始显示。而实际上需要达到的效果是隐藏掉的内容不记序号，不管隐藏掉多少行，都是从1开始显示，依次累加。

	A	B	C	D	E	F	G	H	I
1	不良记录统计汇总表								
2									
3	序号	检查日期	番号	日期印	检查数	良品数	不良数	良品率	
4									
5	1	111213	41321 (BR)	1112071213	133	130	3	98%	
6									
7	2	111214	41321 (BR)	1112071214	255	250	5	98%	
8									
9	3	111215	41322 (BK)	1112071215	62	60	2	97%	
10									
11	4	111215	93696 (PU)	1112091215	342	336	6	98%	
12									
13	5	111216	93696 (PU)	1112091216	344	336	8	98%	
14									

图9-10 检查表

有一个很笨的办法，就是每次隐藏记录后，重新输入1~N条记录，但这样未免太烦琐。天天重复地操作，显然费力不讨好。

后来与好友“无言的人”交流，发现有一个神奇的函数，可以对隐藏或者筛选的数据进行累加，这个函数就是SUBTOTAL。

打开帮助文件了解这个函数的语法，如图9-11所示，当参数1为3或者103时可以统计非空单元格的个数（COUNTA），为103时忽略隐藏值。

语法

```
SUBTOTAL(function_num,ref1,[ref2],...)
```

SUBTOTAL 函数语法具有以下参数：

- **Function_num** 必需。1 到 11（包含隐藏值）或 101 到 111（忽略隐藏值）之间的数字，用于指定使用何种函数在列表中进行分类汇总计算。

FUNCTION_NUM (包含隐藏值)	FUNCTION_NUM (忽略隐藏值)	函数
1	101	AVERAGE
2	102	COUNT
3	103	COUNTA
4	104	MAX
5	105	MIN
6	106	PRODUCT
7	107	STDEV
8	108	STDEVP
9	109	SUM
10	110	VAR
11	111	VARP

图9-11 SUBTOTAL函数语法

也就是说可以设置公式为：

```
=SUBTOTAL(103,$B$5:B5)
```

为了使表格看起来更美观，再加一个IF函数进行判断。

```
=IF(B5="", "", SUBTOTAL(103,$B$5:B5))
```

接下来看检查日报，如图9-12所示，与检查表的大部分内容是重复的，但比检查表更全面。

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	检查日期	番号	日期印	检查时间	实际时间	分钟	检查员	检查人数	工数	良品数	检查数	不良数	良品率
2	111213	41321(BR)	1112071213	15:00-22:00	6	360	天、刘、凤、卢	4	3.0	130	133	3	98%
3	111214	41321(BR)	1112071214	8:00-17:30	8	480	天、刘、凤、卢	4	4.0	250	255	5	98%
4	111215	41322(BK)	1112071215	8:00-10:00	2	120	天、刘、凤、卢	4	1.0	60	62	2	97%
5	111215	93696(PU)	1112091215	10:00-21:00	9	540	天、刘、凤、卢	4	4.5	336	342	6	98%
6	111216	93696(PU)	1112091216	8:00-17:30	8	480	天、刘、凤、卢	4	4.0	336	344	8	98%
7	111220	93696(PU)	1112091220	13:30-22:00	7.5	450	天、凤、卢	3	2.8	288	295	7	98%
8	111221	93696(PU)	1112091221	8:00-21:00	11	660	天、凤、卢	3	4.1	528	550	22	96%
9	111222	93696(PU)	1112091222	8:00-17:30	8	480	天、凤、卢	3	3.0	336	338	2	99%
10	111223	93696(PU)	1112091223	8:00-9:00	1	60	天、凤、卢	3	0.4	18	18	0	100%

图9-12 检查日报



刚开始的时候，卢子每天重复输入一遍，后来那段时间很忙，经常加班。本来下班就晚了，再重复输入这些内容很浪费时间，再说人本来就累，更容易出错。

仔细观察检查表，发现都是隔行引用内容，依次是第5行，第7行，第9行……，可以利用下面的公式获得奇数行数。

$=\text{ROW}(\text{A2}) * 2 + 1$

知道行数后，只需在最外面嵌套INDEX函数即可获取相应内容。

$=\text{INDEX}(\text{检查表!B:B}, \text{ROW}(\text{A2}) * 2 + 1)$

将公式右拉就获取番号跟日期印，用同样的办法设置良品数、检查数、不良数、良品率。良品数公式为：

$=\text{INDEX}(\text{检查表!E:E}, \text{ROW}(\text{E2}) * 2 + 1)$

如图9-13所示，通过复制公式，良品率没有显示百分比。

M2 : X ✓ fx =INDEX(检查表!H:H,ROW(H2)*2+1)					
I	J	K	L	M	N
1	工数	良品数	检查数	不良数	良品率
2	3.0	133	130	3	1
3	4.0	255	250	5	1
4	1.0	62	60	2	1
5	4.5	342	336	6	1
6	4.0	344	336	8	1
7	2.8	295	288	7	1
8	4.1	560	528	22	1

图9-13 良品率不显示百分比

这是个小问题，设置很简单。如图9-14所示，选择M列，单击“%”这个按钮即可。

通过前面的设置，重复录入的问题解决了，现在剩下一部分内容需要重新录入，如图9-15所示。




图9-14 设置单元格为百分比格式

	D	E	F	G	H	I
1	检查时间	实际时间	分钟	检查员	检查人数	工数
2	15:00-22:00	6	360	天、刘、凤、卢	4	3.0
3	8:00-17:30	8	480	天、刘、凤、卢	4	4.0
4	8:00-10:00	2	120	天、刘、凤、卢	4	1.0
5	10:00-21:00	9	540	天、刘、凤、卢	4	4.5
6	8:00-17:30	8	480	天、刘、凤、卢	4	4.0
7	13:30-22:00	7.5	450	天、凤、卢	3	2.8
8	8:00-21:00	11	660	天、凤、卢	3	4.1
9	8:00-17:30	8	480	天、凤、卢	3	3.0
10	8:00-9:00	1	60	天、凤、卢	3	0.4

图9-15 需要重新录入内容

检查时间、检查员必须手工录入，而实际时间虽然可以用公式解决，考虑到中间休息时间不确定，设置公式的难度比较大，所以采取手工录入。

分钟就是实际时间乘以60。

$=\text{E2} * 60$

求检查人数，套用出差费用统计表的公式：

$=\text{LEN}(\text{G2}) - \text{LEN}(\text{SUBSTITUTE}(\text{G2}, " ", "")) + 1$

工数就是实际时间乘以检查人数除以8小时。

=E2*H2/8

经过一系列设置，每天又能节省不少时间，多动脑思考还是有好处的。

9.2.3 优化月度总结表

很多人在月底和月初这几天会比较忙，忙啥呢？还不就是为了月度总结表，每月都有那么几天加班加点的苦日子。很多事只有自己亲身经历才知道，这时如果能够有三头六臂那就再好不过。

如图9-16所示，是月度总结表，别看只有那么几个数字，用计算器能按到你哭。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
8	52 期	4月	5月	6月	7月	8月	9月	10月	11月	12月	1月	2月	3月	最小数	最大数	合计	平均
9	检查回数	11	8	4	11	11	2	7	10	15	5	4	0	0	15	88	7.3
10	总人数	45	31	18	38	45	8	22	32	49	18	12	0	0	49	318	26.5
11	实绩时间	79	48	24	63	74	11	41	72	125	37	26	0	0	125	600	50
12	平均每回人数(人)	4	4	5	3	4	4	3	3	3	4	3	#####	#####	#####	#####	4
13	平均每回实绩时间(H)	7.2	6.0	6.0	5.7	6.7	5.5	5.9	7.2	8.3	7.4	6.5	#####	#####	#####	#####	6.8
15	检查批数	25	11	10	17	20	2	9	10	26	7	6	0	0	26	143	11.9
16	不合格批数	25	9	10	17	20	2	9	10	26	7	6	0	0	26	141	12
17	合格批数	0	2	0	0	0	0	0	0	0	0	0	0	0	2	4	0.16667
18	批合格率	0.0%	18.2%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	0.0%	18.2%	18.2%	3%
19	累计检查批数	25	36	46	63	83	85	94	104	130	137	143	143	——	——	——	——
20	累积不合格批	25	34	44	61	81	83	92	102	128	135	141	141	——	——	——	——
21	累积合格批数	0	2	2	2	2	2	2	2	2	2	2	2	——	——	——	——
22	累积批合格率	0.0%	5.6%	4.3%	3.2%	2.4%	2.4%	2.1%	1.9%	1.5%	1.5%	1.4%	1.4%	——	——	——	——
24	出货数(pcs)	8,025	2,696	1,340	3,287	4,650	278	389	900	4,381	2,255	1,121	0	0	8,025	29,322	2443.5
25	检查数(pcs)	8,235	2,244	1,371	3,364	4,697	278	447	1,783	4,476	2,298	1,238	0	0	8,235	30,431	2535.92
26	良品率	97%	120%	98%	98%	99%	100%	87%	50%	98%	98%	91%	#####	#####	#####	——	96%
28	出货检工程异常(件)	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
29	受入检查工程异常(件)	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0

图9-16 月度总结表

表格模板是“菜头”设计的，当初他设置这张表格还在吹牛，说已经设置了好多公式，以后统计就都智能化。如果真信他的话，卢子就等着加班吧。不过仔细一看，还真设置了不少公式。

最小值：

=MIN(B9:M9)

最大值：

=MAX(B9:M9)

合计：

=SUM(B9:M9)

平均：

=P9/12

除了这几个还有好多，卢子再列举几个



出来。

平均每回人数：

```
=IF(B9="", "", B10/B9)
```

平均每回实绩时间：

```
=IF(B9="", "", B11/B9)
```

合格批数：

```
=B15-B16
```

“菜头”也算卢子的老师，当初从他那里学习了很多。不过不能因为他是老师就放弃对表格进行改善，让表格更完美！说真的，也只能偷偷地说一句，这个表格的所有公式仅仅是糊弄外行，真正需要运用到公式的地方一个都没有设置。

卢子当初前几个月也老老实实在按这个表来做，没有对表格进行优化，每到月底就

敲计算器进行统计。还好那时的敲计算器水平还不错，如果以卢子现在的用计算器水平，效率至少得低一半。

后来卢子实在不忍心埋没自己的Excel操作技术，所以开始对表格中其他没有公式的地方设计公式。刚开始考虑的是动态数据源，跟好友“无言的人”交流了好长一段时间。最后将所有公式都设置好，虽然能够实现动态统计，但问题来了：计算速度比乌龟还慢，电脑都卡死。

这时即使再完美的公式都没有意义，表格上交给领导只有挨批。公式完美很重要，但计算速度同样重要，缺一不可。为了兼顾计算速度，卢子经过最后的思考，对公式做了一些小小的变动。

至于怎么变动，先看表，再慢慢说明。

如图9-17所示，这是明细表，月度总结表就是根据明细表统计而来。

	A	B	C	D	E	F	G	H	I
1	日期	分钟	小时	人数	番号	出货数	检查数	不良数	型号
2	4月1日	120	2.0	3	40065	142	142	0	H126
3	4月1日	540	9.0	admin:	40066	616	626	10	H126
4	4月2日	30	0.5	每一批只记入一次人数	40061	31	35	4	H126
5	4月2日	120	2.0		40066	128	149	21	H126
6	4月2日	75	1.3		40062	61	63	2	H126
7	4月6日	60	1.0	5	93657	197	201	4	R102
8	4月6日	360	6.0		93657	1,140	1,171	31	R102
9	4月7日	120	2.0	5	93657	263	268	5	R102
10	4月7日	360	6.0		43010	1,176	1,222	46	R102
11	4月8日	300	5.0	5	43010	819	830	11	R102
12	4月9日	240	4.0	4	43010	804	825	21	R102
13	4月12日	270	4.5	4	40061	299	311	12	H126
14	4月12日	60	1.0		40062	61	61	0	H126
15	4月13日	330	5.5	4	40062	389	396	7	H126
16	4月13日	150	2.5		40065	196	198	2	H126

图9-17 明细表

检查回数就是统计每个月检查的天数，本来是需要按日期统计不重复值。但卢子的表格设计得好，“人数”这里在每天第一次出现的时候记录，其他地方都空着。只要计算每月的人数出现次数即可。

考虑到日期会跨年，所以将明细表日期用TEXT函数转换成年月的形式。

```
=TEXT(明细表!$A$2:$A$885,"e-mm")
```

月度总结表的月份是通过自定义单元格格式获取，如4月，实际上是2014-4-1，所以也要转换成年月的形式。

```
=TEXT(B$8,"e-mm")
```

最终，检查回数公式为：

```
=SUMPRODUCT((TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm"))*(明细表!$D$2:$D$885>0))
```

后来听说SUMPRODUCT函数的条件用“,”隔开比用“*”运算速度要快，所以将公式进一步处理。

```
=SUMPRODUCT(--(TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm")),--(明细表!$D$2:$D$885>0))
```

明细表的区域使用：“明细表!\$A\$2:\$A\$885”是为了实现动态统计，因为一年的数据通常没有这么多行，当然为了保险起见，可以将区域改成更大，如“明细表!\$A\$2:\$A\$8888”。这样的区域就比使用公式OFFSET+COUNTA确定使用范围运算速度更快。

接着看总人数，其实就是将检查回数判断大于0的条件取消即可。

```
=SUMPRODUCT(--(TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm")),明细表!$D$2:$D$885)
```

检查批数：就是每月检查的时间大于0的次数。

```
=SUMPRODUCT(--(TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm")),--(明细表!$B$2:$B$885>0))
```

不合格批数：因为实施全数检查，正常的话等于检查批数。5月份有2批是抽检的，这个手工输入即可。

```
=B15
```

出货数：

```
=SUMPRODUCT(--(TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm")),明细表!$F$2:$F$885)
```



检查数:

```
=SUMPRODUCT(--(TEXT(B$8,"e-mm")=TEXT(明细表!$A$2:$A$885,"e-mm")),明细表!$G$2:$G$885)
```

没尝试的时候，卢子以为很难统计，没想到统计这些那么简单，仅仅用到SUMPRODUCT跟TEXT两个函数。

9.2.4 挑战月度不良统计表

成长是需要付出代价的，多从别人那里获取失败的经验，这样就会少走很多弯路。闭门造车是不可取的，终将付出惨重的代价！

卢子接手中山隆成那边的工作，中山隆成有事先设置好的表格模板，如图9-18所示。像这种二维表格经常会见到，其好处就是录入数据简单。那时没有考虑到以后要对数据源进行汇总分析，也就没去重新设置表格模板。虽然表格存在很多不合理，如合并单元格，但因贪图一时之便，后来害惨自己。

	A	B	C	D	E	F	G	H	I	J	K
1											
2	供应商：中山隆成（小天使）										
3											
4	番号	俗称	不良数	不良率	次数	序号	不良内容	等级	4-1	4-2	4-6
5			0	0.0%	0	1	内外箱印刷不良、破损、脏污	C			
6			0	0.0%	0	2	内外箱（产品）漏贴条形码、日期印	C			
7			10	0.2%	2	3	管件划伤、凹痕、坑洼	B			8
8			71	1.1%	8	4	防滑垫划伤、倾斜、裂	B			12
9			1	0.0%	1	5	布地破损、开线	B			
10			2	0.0%	2	6	带子破损、开线	B			1
11			9	0.1%	5	7	涂装不良	B			4
12	9		19	0.3%	5	8	脏污、油污、胶水残留	B			3
13	3		0	0.0%	0	9	金属钮扣不良	B			
14	6		0	0.0%	0	10	螺杆生锈	B			
15	5	R	1	0.0%	1	11	塑胶扣毛边、水口残留等	A			
16	7	1	11	0.2%	4	12	焊接部刮手、有孔	A			5
17		0	0	0.0%	0	13	管件毛刺残留	A			
18	.	~	1	0.0%	1	14	三段高度调节紧	A			

图9-18 中山隆成的表格模板

后来领导要卢子对不良记录进行统计，好查看每个月各类产品的出货检查情况。通过这张明细表进行汇总，得到这样一张表，如图9-19所示。

	A	B	C	D	E	F	G	H	I
1	Richell海外外注课 51 期 中山隆成（小天使） 殿 出货检查 不良总结								
2									
3									
4	俗称	H142			月份	6			
5									
6									
7		出货总数:			检查总数:	622			
8									
9	序号	不良描述			不良等级	不良数	不良率	影响度	
10	1	站立脚不良			C	8			
11	2	顶蓬不良（开线、破损、变形、脏污）			B	6			
12	3	漏部材			A	1			
13	4	涂装不良			C	1			
14	5	管件划伤、凹痕、坑洼等			C	1			
15	6								
16	7								

图9-19 汇总表

通过明细表获得这张汇总表，对当时的卢子来说难度不小，暂时为了应付领导利用各种转换方法，费了九牛二虎之力才勉强完成，效率极低。

为了寻求更方便快捷的方法，苦思一周无果。

在网络上各处求助，终于皇天不负有心人，在wangjguo44老师的帮助下完成了这项艰巨的任务，在这里对他老人家说一声：谢谢！

一起来看看老师的超级公式，至于你晕不晕我不知道，反正卢子已经晕了。

检查总数：

```
=SUM((LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(小天使!$G$5:$G$136="检查数")*(MONTH(小天使!$I$4:$DJ$4)=$F$4)*(小天使!$I$5:$DJ$136))
```

不良描述：

```
=INDEX(小天使!G:G,RIGHT(TEXT(LARGE(MMULT((LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(MONTH(小天使!$I$4:$DJ$4)=$F$4)*(小天使!$G$5:$G$136<>"检查数")*(小天使!$G$5:$G$136<>"其他")*小天使!$I$5:$DJ$136,ROW($1:$106)^0)+(LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(小天使!$G$5:$G$136<>"检查数")*(小天使!$G$5:$G$136<>"其他")*(小天使!$G$5:$G$136<>"")*0.1+ROW($5:$136)%%,ROW(A1)), "0.0000"),3)*1)
```



不良等级:

```
=INDEX(小天使!H:H,RIGHT(TEXT(LARGE(MMULT((LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(MONTH(小天使!$I$4:$DJ$4)=$F$4)*(小天使!$G$5:$G$136<>"检查数")*(小天使!$G$5:$G$136<>"其他")*小天使!$I$5:$DJ$136,ROW($1:$106)^0)+(LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(小天使!$G$5:$G$136<>"检查数")*(小天使!$G$5:$G$136<>"其他")*(小天使!$G$5:$G$136<>"")*0.1+ROW($5:$136)%%,ROW(A1)),0.0000"),3)*1)
```

不良数:

```
=LARGE(MMULT((LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(MONTH(小天使!$I$4:$DJ$4)=$F$4)*(小天使!$G$5:$G$136<>"检查数")*(小天使!$G$5:$G$136<>"其他")*小天使!$I$5:$DJ$136,ROW($1:$106)^0),ROW(A1))
```

合计:

```
=SUM((LOOKUP(ROW($5:$136),IF(小天使!$B$5:$B$136<>"",ROW($5:$136)),小天使!$B$5:$B$136)=$B$4)*(MONTH(小天使!$I$4:$DJ$4)=$F$4)*(小天使!$G$5:$G$136<>"检查数")*小天使!$I$5:$DJ$136)
```

为了进行容错处理，还使用了条件格式。

我的天啊，这些是人看的公式吗？简直就是神才能看得懂的公式。卢子也懒得去看，拿来就用，而且一用就是一年，期间都不敢再去看公式一眼。

一年后的某一天，卢子不甘心使用看不懂的公式，所以开始对数据源进行小改动，然后设置公式。如图9-20所示，添加一辅助列获取俗称。

	A	B	C	D	E	F	G	H
1								
2	供应商：中山隆成（小天使）							
3								
4	辅助列	序号	俗称	不良数	不良率	次数	序号	不良内容
11	R102	9 3 6 5 7 4 3 0 1	R 1 0 2	9	0.1%	5	7	涂装不良
12	R102			19	0.3%	5	8	脏污、油污、胶水残留
13	R102			0	0.0%	0	9	金属纽扣不良
14	R102			0	0.0%	0	10	螺杆生锈
15	R102			1	0.0%	1	11	塑胶扣毛边、水口残留等
16	R102			11	0.2%	4	12	焊接部刮手、有孔
17	R102			0	0.0%	0	13	管件毛刺残留
18	R102			1	0.0%	1	14	三段高度调节紧
19	R102			0	0.0%	0	15	高度调节管和靠背管嵌合不良
20	R102			1	0.0%	1	16	弹珠弹不出
21	R102	1	0.0%	1	17	螺杆活动不良（紧）		
22	R102	5	0.1%	3	18	漏装管塞、扣子等		
23	R102	0	0.0%	0	19	金属扣打歪		
24	R102	7	0.1%	3	20	其它		

图9-20 添加辅助列

辅助列公式为：

```
=LOOKUP("座",C$5:C5)
```

“座”是一个接近最大的文本，查找最后一个满足条件的文本就用“座”，查找最后满足条件的数字就用9E+307。

在不确定为文本或者数字的情况下，可以用通用公式：

```
=LOOKUP(1,0/(C$5:C5<>""),C$5:C5)
```

另外考虑到日期会跨年，在月度不良统计表增加一个年份作为筛选的条件，如图9-21所示。

	A	B	C	D	E	F	G	H	I
1	Richell海外外注课	52	期	中山隆成（小天使）	殿	出货检查	不良总结		
2									
3									
4	俗称	H126			月份	12		年份	2011
5									
6									
7		出货总数：	754		检查总数：	766			
8									
9	序号	不良描述			不良等级	不良数	不良率	影响度	
10	1	背肩带不良			B	6	0.8%	50.0%	
11	2	塑胶垫片裂开			C	4	0.5%	33.3%	
12	3	管件划伤、凹痕、坑洼			B	1	0.1%	8.3%	
13	4								
14	5								
15	6								
16	7								

图9-21 月度不良统计表（新）

现在设计公式就稍微简单点，毕竟可以不用去想合并单元格的事儿。

检查总数：

```
=SUMPRODUCT((I4&F4=TEXT(小天使!J4:NJ4,"em"))*(小天使!A5:A162=B4)*(小天使!H5:H162="检查数")*小天使!J5:NJ162)
```

不良描述使用多单元格数组，选择区域B10:B16输入公式，按Ctrl+Shift+Enter组合键结束。

```
=INDEX(小天使!H:H,MOD(LARGE(MMULT((I4&F4=TEXT(小天使!J4:NJ4,"em"))*(小天使!A5:A162=B4)*(小天使!H5:H162<>"检查数")*(小天使!H5:H162<>"其他")*小天使!J5:NJ162,ROW(1:365)^0)*10^4+(小天使!A5:A162=B4)*(小天使!H5:H162<>"检查数")*(小天使!H5:H162<>"其他")*(小天使!H5:H162<>"")*ROW(5:162),ROW(1:7)),10^4))
```



不良等级，直接用VLOOKUP函数进行查找。

```
=IFERROR(VLOOKUP(B10,小天使!H:I,2,),"")
```

不良数也是使用多单元格数组，选择区域G10:G16输入公式，按Ctrl+Shift+Enter组合键结束。

```
=LARGE(MMULT((I4&F4=TEXT(小天使!J4:NJ4,"em"))*(小天使!A5:A162=B4)*(小天使!H5:H162<>"检查数")*(小天使!H5:H162<>"其他")*小天使!J5:NJ162,ROW(1:365)^0),ROW(1:7))
```

合计，使用YEAR+MONTH两个函数其实作用跟TEXT函数一样，都是为了得到年月作为条件，可以互相转换。

```
=SUMPRODUCT((小天使!$A$5:$A$162=$B$4)*(MONTH(小天使!$J$4:$IY$4)=$F$4)*(YEAR(小天使!$J$4:$IY$4)=$I$4)*(小天使!$H$5:$H$162<>"检查数")*小天使!$J$5:$IY$162)
```

虽然字符减少一半，但还是很难！

这可怎么办，如果让别人看到这些公式，依然晕菜！

一天卢子跟好友“无言的人”聊到这个话题，公式虽然很好用，但对不规范的数据源使用公式还是很鸡肋。所以，考虑到怎么样将不规范的数据源转换成标准的数据源，这样处理就更简单。

好友“无言的人”提到这可以用VBA代码转换，几天后，他提供了下面一段代码。

```
Option Explicit
Public Sub 二维转一维()
    Dim Arr, Brr(), i As Byte, ii As Integer, T As Long
    Dim iR As Long, iC As Byte, iRC As Long, iTem As Long
    Arr = Sheets("小天使").Range("A4").CurrentRegion
    iR = UBound(Arr): iC = UBound(Arr, 2): iRC = iR * iC
    iTem = 1 '计数器
    ReDim Preserve Brr(1 To iRC, 1 To 9)

    For i = 9 To iC
        For ii = 2 To iR
            '
            If i = 113 And ii = 80 Then Stop
            Brr(iTem, 1) = iTem '序号
            If Arr(ii, 1) <> "" And Arr(ii, 1) <> "合计" Then
                Brr(iTem, 2) = Arr(ii, 1) '番号
                Brr(iTem, 3) = Arr(ii, 2) '俗称
            '
            ElseIf Arr(ii, 1) = "合计" Then
            "
                MsgBox Arr(ii, 1) & vbTab & ii
```

```

'      ii = ii + 1
'      Brr(iTem, 2) = Arr(ii, 1)
'      Brr(iTem, 3) = Arr(ii, 2) '俗称
Else
    Brr(iTem, 2) = Brr(iTem - 1, 2)
    Brr(iTem, 3) = Brr(iTem - 1, 3) '俗称
End If
Brr(iTem, 4) = Arr(ii, 7) '不良内容
Brr(iTem, 5) = Arr(ii, 8) '等级
Brr(iTem, 6) = Arr(1, i) '日期
If Arr(ii, 7) <> "检查数" Then Brr(iTem, 7) = Arr(ii, i) '不良数
If Arr(ii, i) <> "" And Arr(ii, 7) <> "检查数" Then Brr(iTem, 8) = 1 '不良数次数
If Arr(ii, 7) = "检查数" Then Brr(iTem, 9) = Arr(ii, i) '检查数
iTem = iTem + 1
Next ii
Next i
With Sheet4.Range("A2")
    .CurrentRegion.Clear
    .Resize(iRC, 9) = Brr
    .Offset(-1, 0).Resize(1, 9) = Array("序号", "番号", "俗称", "不良内容", "等级", "日期", "不良数", "不良
数次数", "检查数")
    .CurrentRegion.Columns.AutoFit
    .CurrentRegion.Borders.LineStyle = 1
    .CurrentRegion.Borders.ColorIndex = 3
End With
End Sub

```

转换后效果如图9-22所示，变成一维的标准表格，但有一点缺陷，就是存在大量空白行。

	A	B	C	D	E	F	G	H	I
1	序号	番号	俗称	不良内容	等级	日期	不良数	不良数次数	检查数
32	31	4040 4006	H126	布地开线、破损、脏污	C	2011-4-1			
33	32	4040 4006	H126	顶蓬不良（开线、破损、变形、脏污）	B	2011-4-1			
34	33	4040 4006	H126	车轮不良（划伤、毛边、凹痕）	C	2011-4-1			
35	34	4040 4006	H126	手把套不良（转动不良、粗细）	C	2011-4-1			
36	35	4040 4006	H126	菜篮不良	C	2011-4-1			
37	36	4040 4006	H126	车轮歪	B	2011-4-1			
38	37	4040 4006	H126	铆钉、螺丝刮手	A	2011-4-1	5	1	
39	38	4040 4006	H126	塑胶件毛边、水口残留	B	2011-4-1	3	1	
40	39	4040 4006	H126	管件刮手	A	2011-4-1			
41	40	4040 4006	H126	行走振动、不走直线，四脚不平	B	2011-4-1			
42	41	4040 4006	H126	折叠时，挂钩不良	B	2011-4-1			
43	42	4040 4006	H126	收车关节嵌合不良	B	2011-4-1			

图9-22 转换后的效果



VBA对于卢子而言还是有难度的，所以卢子采用一些基础操作法对表格进行再处理。

STEP 01 如图9-23所示，建立辅助列，判断不良数、不良数次数、检查数三列是不是都空白，然后筛选0，也就是三列都是空白的。

=COUNT(G2:I2)

	E	F	G	H	I	J	K
1	等级	日期	不良数	不良数次数	检查数	辅助列	
2	C	2011-4-1				0	
3	C	2011-4-1				0	
4	B	2011-4-1				0	
5	B	2011-4-1				0	
6	B	2011-4-1				0	
7	B	2011-4-1				0	
8	B	2011-4-1				0	
9	B	2011-4-1				0	
10	B	2011-4-1				0	
11	B	2011-4-1				0	

图9-23 筛选数字为0的项目

STEP 02 单击A2单元格，利用组合键Ctrl+Shift+End快速选中区域，然后删除工作表行，如图9-24所示。

STEP 03 如图9-25所示，选择辅助列，右击鼠标，在快捷菜单中选择“删除”命令。




图9-24 删除工作表行




图9-25 删除辅助列

STEP 04 如图9-26所示，对数据源进行美化处理。

	A	B	C	D	E	F	G	H
1	俗称	番号	不良内容	等级	日期	不良数	不良数次数	检查数
2	H126	4040 4006	铆钉、螺丝刮手	A	2011-4-1	5	1	
3	H126	4040 4006	塑胶件毛边、水口残留	B	2011-4-1	3	1	
4	H126	4040 4006	其他		2011-4-1	2	1	
5	H126	4040 4006	检查数		2011-4-1			768
6	H126	4040 4006	塑胶件毛边、水口残留	B	2011-4-2	1	1	
7	H126	4040 4006	手把管塞装不到位(脱落)	C	2011-4-2	26	1	
8	H126	4040 4006	检查数		2011-4-2			247
9	R102	93657 4301	管件划伤、凹痕、坑洼	B	2011-4-6	8	1	
10	R102	93657 4301	防滑垫划伤、倾斜、裂	B	2011-4-6	12	1	
11	R102	93657 4301	带子破损、开线	B	2011-4-6	1	1	
12	R102	93657 4301	涂装不良	B	2011-4-6	4	1	
13	R102	93657 4301	脏污、油污、胶水残留	B	2011-4-6	3	1	
14	R102	93657 4301	焊接部刮手、有孔	A	2011-4-6	5	1	
15	R102	93657 4301	漏装管塞、扣子等	A	2011-4-6	1	1	
16	R102	93657 4301	其他		2011-4-6	1	1	

图9-26 美化处理数据源

如此规范的数据源，使用数据透视表进行汇总很方便。数据透视表是Excel中的战斗机，分分钟搞定问题。

卢子考虑到以后都用这张表格录入数据，也就是数据源会动态更新，在使用数据透视表之前就先创建一个动态的数据源名称。

STEP 01 如图9-27所示，单击“公式”选项卡中的“定义名称”按钮，在“名称”文本框中输入“动态数据源”，在“引用位置”编辑框输入下面的公式，最后单击“确定”按钮。

=OFFSET(处理后!\$A\$1,,COUNTA(处理后!\$A:\$A),8)




图9-27 创建动态数据源名称



接下来就是看数据透视表大展神威的时刻。

STEP 02 如图9-28所示，单击“插入”选项卡中的“数据透视表”按钮，弹出“创建数据透视表”对话框，输入“表/区域”的名称为“动态数据源”，单击“确定”按钮。




图9-28 插入数据透视表

STEP 03 如图9-29所示，弹出“数据透视表字段”对话框，通过这些字段可以进行各种分析。




图9-29 数据透视表字段

在进行布局之前，先考虑下要做什么分析。我们是要统计各种俗称年月的各项不良数。现在日期是2011-4-1这种形式，并没有出现年月，难不成需要用函数提取年月？不是！数据透视表本身就提供了按日期组合的功能，可以按年份、季度、月份等进行分组。

STEP 04 如图9-30所示，将日期拉到行字段。

STEP 05 如图9-31所示，单击任意日期，右击鼠标后在快捷菜单中选择“创建组”命令。




图9-30 将日期拉到行字段




图9-31 创建组

STEP 06 如图9-32所示，弹出“组合”对话框，从中可以实现各种组合，我们选择按月跟年组合。组合后的效果如图9-33所示。



图9-32 按年月组合

行标签	
2011年	
4月	
5月	
6月	
7月	
8月	
9月	
10月	
11月	
12月	
2012年	
1月	
2月	
总计	

图9-33 组合后的效果



STEP 07 如图9-34所示，将需要分析的字段拖动到相应的区域。




图9-34 布局

STEP 08 这种布局看起来很乱，不良内容与不良等级应该并排才对。如图9-35所示，单击“设计”选项卡中的“报表布局”按钮，在下拉菜单中选择“以表格形式显示”命令。




图9-35 以表格形式显示

STEP 09 以表格形式显示后出现大量汇总，看起来依然很乱。如图9-36所示，右击鼠标后在快捷菜单中取消分类汇总“不良内容”的勾选。



STEP 10 经过前面几步的设置，看起来就顺眼多了，但还是有点问题。就是不良数以计数的形式统计，跟我们实际的要求和有差异。如图9-37所示，单击“计数项：不良数”单元格，右击鼠标，在快捷菜单中选择“值汇总依据”→“求和”命令。




图9-36 取消分类汇总




图9-37 更改值汇总依据

STEP 11 现在虽然能体现各种统计，但并不完美，不良数乱七八糟，没有进行排序怎么办呢？如图9-38所示，单击“不良项目”的下拉按钮，选择“其他排序选项”命令，在弹出的“排序”对话框，选中“降序排序”单选按钮，内容选择“求和项：不良数”。




图9-38 按不良数降序

STEP 12 降序排序看起来就清晰多了，但项目太多，一般只关注前8项不良项目，其他都忽略掉。如图9-39所示，单击不良项目的下拉按钮，选择“值筛选”→“前10项”命令，在弹出的“前10个筛选”对话框中，将最大10项改成最大8项，单击“确定”按钮。

STEP 13 到这一步可以说大功告成，需要查看哪种俗称、哪一年、哪一月只需筛选就可以。对于处女座的小卢而言，还差了一小步，那就是进一步美化表格。最终效果如图9-40所示。






图9-39 前8项不良数

	A	B	C	D	E	F
1						
2	年	2011年		月	11月	
3	俗称	H6284				
4						
5	不良内容	等级	不良数			
6	塑胶扣脱落、断裂	B	30			
7	其它		15			
8	前扶手脱开(有缝隙)	B	9			
9	刹车时,车轮转动	B	8			
10	前扶手超声波裂开	B	8			
11	漏贴纸	B	8			
12	顶蓬不良(开线、破损、变形、脏污)	B	7			
13	漏部材(铆钉盖、轮盖、管塞)	A	6			
14	总计		91			
15						

图9-40 最终效果

操作步骤看似很多,其实真正动手操作也就用几分钟,比用公式所花费的时间缩短好多。有的时候一条长公式就得考虑1~2小时才能完成。



知识扩展

当升级到Excel 2016的时候,你会发现非常神奇的一幕,在创建数据透视表时,将“日期”拉到行字段,不做任何处理,Excel就自动帮你组合好了,如图9-41所示。




图9-41 自动组合

如图9-42所示，单击折叠按钮(+)，发现与我们原来想要的组合效果差不多，自动实现按年、季度、月组合（多了一个季度），微软设计的软件实在太贴心了！

3	行标签	
4	2012年	
5	第四季	
6	10月	
7	11月	
8	12月	
9	2013年	
10	总计	
11		

图9-42 打开折叠按钮

9.2.5 月度数据核对

有这么一句话：小心驶得万年船。制作表格也一样，凡事多留个心眼儿。有两份月度数据表格：一份是以检查日报为主，一份是以不良统计表为主。因为是在不同表格录入，有可能因个人原因导致数据录入错误。为了更准确地统计数据，再核对一次两个月度表格是很有必要的。绝不能因为自己的疏忽让领导产生错误的抉择！

只要两个表格的月度检查数跟不良数一样，就表示录入正确。

如图9-43所示，是检查日报。

	A	B	C	D	E	F	G	H	I
1	日期	分钟	小时	人数	番号	出货数	检查数	不良数	俗称
2	4月1日	120	2.0	3	40065	142	142	0	H126
3	4月1日	540	9.0	admin:	40066	616	626	10	H126
4	4月2日	30	0.5	每一批只记入一次人数	40061	31	35	4	H126
5	4月2日	120	2.0		40066	128	149	21	H126
6	4月2日	75	1.3		40062	61	63	2	H126
7	4月6日	60	1.0	5	93657	197	201	4	R102
8	4月6日	360	6.0		93657	1,140	1,171	31	R102
9	4月7日	120	2.0	5	93657	263	268	5	R102
10	4月7日	360	6.0		43010	1,176	1,222	46	R102
11	4月8日	300	5.0	5	43010	819	830	11	R102
12	4月9日	240	4.0	4	43010	804	825	21	R102
13	4月12日	270	4.5	4	40061	299	311	12	H126
14	4月12日	60	1.0		40062	61	61	0	H126
15	4月13日	330	5.5	4	40062	389	396	7	H126
16	4月13日	150	2.5		40065	196	198	2	H126

图9-43 检查日报

如图9-44所示，是不良统计表。

如图9-45所示，是需要核对的内容。



	A	B	C	D	E	F	G	H
1	俗称	番号	不良内容	等级	日期	不良数	不良数次数	检查数
2	H126	4040 4006	铆钉、螺丝刮手	A	2011-4-1	5	1	
3	H126	4040 4006	塑胶件毛边、水口残留	B	2011-4-1	3	1	
4	H126	4040 4006	其他		2011-4-1	2	1	
5	H126	4040 4006	检查数		2011-4-1			768
6	H126	4040 4006	塑胶件毛边、水口残留	B	2011-4-2	1	1	
7	H126	4040 4006	手把管塞装不到位(脱落)	C	2011-4-2	26	1	
8	H126	4040 4006	检查数		2011-4-2			247
9	R102	93657 4301	管件划伤、凹痕、坑洼	B	2011-4-6	8	1	
10	R102	93657 4301	防滑垫划伤、倾斜、裂	B	2011-4-6	12	1	
11	R102	93657 4301	带子破损、开线	B	2011-4-6	1	1	
12	R102	93657 4301	涂装不良	B	2011-4-6	4	1	
13	R102	93657 4301	脏污、油污、胶水残留	B	2011-4-6	3	1	
14	R102	93657 4301	焊接部刮手、有孔	A	2011-4-6	5	1	
15	R102	93657 4301	漏装管塞、扣子等	A	2011-4-6	1	1	
16	R102	93657 4301	其他		2011-4-6	1	1	

图9-44 不良统计表

	A	B	C	D	E
1	年份	月份	检查数	不良数	
2	2011年	4月	OK	OK	
3	2011年	5月	OK	OK	
4	2011年	6月	OK	OK	
5	2011年	7月	OK	OK	
6	2011年	8月	OK	OK	
7	2011年	9月	OK	OK	
8	2011年	10月	OK	OK	
9	2011年	11月	OK	OK	
10	2011年	12月	OK	OK	
11	2012年	1月	NG	NG	
12	2012年	2月	OK	OK	
13	2012年	3月	OK	OK	
14					

图9-45 核对表

两个表都需要将日期转换成年月形式才能统计检查数。实现转换可以用TEXT函数，将第二参数设置为“e年m月”即可。

检查日报的不良数：

```
=SUMPRODUCT((TEXT(检查日报!A$2:A$999,"e年m月")=$A2&$B2)*检查日报!G$2:G$999)
```

不良统计表的不良数：

```
=SUMPRODUCT((TEXT(不良统计表!E$2:E$999,"e年m月")=$A2&$B2)*不良统计表!H$2:H$999)
```

检查数核对，也就是：

```
=IF(检查日报的不良数=不良统计表的不良数,"OK","NG")
```

将中文替换成公式就得出，虽然公式很长，但拆开看实际都很简单。

```
=IF(SUMPRODUCT((TEXT(检查日报!A$2:A$999,"e年m月")=$A2&$B2)*检查日报!G$2:G$999)=SUMPRODUCT((TEXT(不良统计表!E$2:E$999,"e年m月")=$A2&$B2)*不良统计表!H$2:H$999),"OK","NG")
```

用同样的方法，设置不良数的核对公式：

```
=IF(SUMPRODUCT((TEXT(检查日报!A$2:A$999,"e年m月")=$A2&$B2)*检查日报!H$2:H$999)=SUMPRODUCT((TEXT(不良统计表!E$2:E$999,"e年m月")=$A2&$B2)*不良统计表!F$2:F$999),"OK","NG")
```

还好进行了核对，里面有1个月的数据出错，看似出错概率不高，但做表格本来就需要100%的正确，不容许有任何差错！

1%的错误=100%的失败

出错了就得赶紧找原始报表查看寻找原因。

后话：这些表格模板设置完成不久，卢子就辞职了。在进行工作交接的时候交代了这些模板的使用方法，但不知道她们使用没有，反正自从卢子走后，没听过关于这些模板的任何问题。这显然不符合逻辑，每份模板在使用的过程中都会出现或多或少的问题，之所以没出现问题就是最大的问题。也许后来的同事压根儿就没有使用，还在用计算器搞定所有报告。

卢子能做的就是尽量将工作做到最好，能够优化的尽量优化。同时也尽量感染身边的人，让他们也使用Excel提高效率。

9.3 疑难汇集

卢子在学习Excel过程中，长期混在Excel论坛和交流群，在帮助别人的过程中，也总结



出一系列问题。帮助别人其实就是帮助自己，如果不是这样，卢子也不会知道那么多Excel技能，更不可能成为版主，甚至出书。

9.3.1 身份证问题汇集

经常有从事HR工作的朋友询问一些与身份证信息提取相关的问题，下面通过一些例子来简单总结一下与之相关的函数公式用法。

身份证的类型简单来讲有一代和二代之分，一代是15位号码，二代是18位号码，与一代的不同之处在于补足了四位年份日期并且在末尾添加了一个校验字符。二代身份证自2005年开始换发，一代可以同时通用。根据规定，在2013年1月1日起全面停止使用一代身份证。15位和18位身份证说明如图9-46所示。



图9-46 15位和18位身份证说明

因为15位身份证已经停用，现在重点说下18位身份证的号段含义。

前2位表示所属省份代码，3~6位表示所属城市和区县代码，7~14位表示出生日期，格式为YYYYMMDD，15~17位是个人顺序码，其中第17位可以标识性别，奇数表示男性，偶数表示女性，第18位是校验码，由前17位通过计算求得。

如图9-47所示，输入18位身份证显示4.45121E+17，怎么回事？

这个是科学记数法，E+17就是10的17次方。默认情况下，数字超过11位就显示科学记数法，数字超过15位后面都显示0，如图9-48所示。

	A	M
1	身份证	
2	4.45121E+17	?
3		
4		

图9-47 输入身份证的疑惑

	A	M	N	O
1	身份证			
2	4.45121E+11			
3	4.45121E+17			
4				
5				

图9-48 长数字的说明

如图9-49所示，选择A列，将单元格格式设置为文本，再重新输入身份证即可。




图9-49 设置单元格为文本格式

设置好单元格格式后，又有一个问题。如图9-50所示，如何才能保证身份证刚好是18位，同时还得符合身份证的所有条件，前17位必须是数字，第18位是字母X或者数字？



	A	M	N
1	身份证		
2	445121198709055616		
3	44542321233445555		
4	445423212334455589		
5	44542321233445558t		
6			

图9-50 防止出错设置

18位可以用：

=LEN(A2)=18

前17位是数字：

=ISNUMBER(-LEFT(A2,17))

第18位是数字或者字母X：

=OR(ISNUMBER(-RIGHT(A2)),RIGHT(A2)="X")

综合这3个条件就可以进行数据有效性设置。

如图9-51所示，选择区域A2:A5，切换到“数据”选项卡，单击“数据验证”（低版本叫数据有效性）按钮，在“允许”下拉列表框中选择“自定义”选项，在“公式”文本框输入下面的公式，最后单击“确定”按钮。

=AND(LEN(A2)=18,ISNUMBER(-LEFT(A2,17)),OR(ISNUMBER(-RIGHT(A2)),RIGHT(A2)="X"))




图9-51 设置数据验证

数据验证对于已经输入好的身份证无效，但可以把不满足条件的圈出来。如图9-52所示，设置好数据有效性后进行“圈释无效数据”。

将无效的数据重新更改即可，然后输入正确的身份证，以后如果输入不正确的话，会提示错误，直到输入正确为止，如图9-53所示。




图9-52 解释无效数据




图9-53 出错提示

关于身份证的准备工作已经做好，后面是获取各种信息的关键方法。

如图9-54所示，是一份详细的地区码。

	A	B
1	编码	所在地
2	001102	(县)
3	001201	(市辖区)
4	001202	(县)
5	001301	石家庄市
6	001302	唐山市
7	001303	秦皇岛市
8	001304	邯郸市
9	001305	邢台市
10	001306	保定市
11	001307	张家口市
12	001308	承德市
13	001309	沧州市
14	001310	廊坊市
15	001311	衡水市
16	001321	邯郸地区
17	001322	邢台地区
18	001323	石家庄地区

图9-54 地区码

通过地区码可以查询到省份和地区。

获取省份就是提取身份证前两位，然后根据地区码查询对应值。

```
=VLOOKUP(LEFT(A2,2),地区码!A:B,2,0)
```

获取地区就是提取身份证前6位，然后根据地区码查询对应值。

```
=VLOOKUP(LEFT(A2,6),地区码!A:B,2,0)
```

出生日期就是从身份证第7位开始提取8位，然后将提取到的数字转换成0-00-00这种形式。

```
=TEXT(MID(A2,7,8),"0-00-00")
```

性别就是第15~17位，奇数代表男性，偶数代表女性。

```
=IF(MOD(MID(A2,15,3),2),"男","女")
```

获取验证码太复杂了，这里不做说明。通过出生日期还可以获取一系列问题：周岁、生肖、生日提醒、星座等。现在就以这4个常见问题再进行说明。为了方便，出生日期就引用D2单元格。



周岁:

```
=DATEDIF(D2,NOW(),"y")
```

生肖:

```
=MID("鼠牛虎兔龙蛇马羊猴鸡狗猪",MOD(TEXT(D2,"[$-130000]e")-4,12)+1,1)
```

TEXT(D2,"[\$-130000]e")的作用就是将日期转换成阴历年份。

生日提醒:

```
=TEXT(TEXT(D2,"m-d")-TEXT(NOW(),"m-d"),"还有0天生日;;今天生日")
```

先将日期转变成月日再进行相减,大于0就显示还有多少天过生日,小于0就不显示,0代表今天生日。

星座:

```
=LOOKUP(--TEXT(D2,"m.dd"),{0,"摩羯座";1.21,"水瓶座";2.2,"双鱼座";3.21,"白羊座";4.21,"金牛座";5.22,"双子座";6.22,"巨蟹座";7.24,"狮子座";8.24,"处女座";9.24,"天秤座";10.24,"天蝎座";11.23,"射手座";12.23,"摩羯座"})
```

这个看着很长,其实很简单,只要获取星座对应表就可以。

经过了几步处理,最终效果如图9-55所示。

	A	B	C	D	E	F	G	H	I	J
1	身份证	省份	地区	出生日期	性别	周岁	生肖	生日提示	星座	
2	445121198709055616	广东省	潮州市潮安县	1987-09-05	男	26	兔	还有30天生日	处女座	
3	420821199007052028	湖北省	荆门市京山县	1990-07-05	女	24	马		巨蟹座	
4	362201198706125428	江西省	宜春地区宜春市	1987-06-12	女	27	兔		双子座	
5	42098219660306103X	湖北省	孝感市安陆市	1966-03-06	男	48	马		双鱼座	
6										

图9-55 身份证问题汇集



知识扩展

身份证的使用一般是与人事管理有关,这时还会涉及根据员工的入职日期获取工龄,再根据工龄获取年终奖。统计的截止日期为2016-12-31,入职不足一年的年终奖为半个月工资,满一年的为1个月工资,以后每增减一年增减1个月工资,如图9-56所示。



	A	B	C	D	E
1	身份证	入职日期	月薪	工龄	年终奖
2	445121198709055616	2007-7-10	5000		
3	420821199007052028	2012-3-5	2700		
4	362201198706125428	2015-1-3	5100		
5	42098219660306103X	2015-10-12	5000		
6	445121198709055621	2016-7-10	3000		
7					

图9-56 工龄及年终奖

工龄的公式为：

```
=DATEDIF(B2,"2016-12-31","Y")
```

年终奖的公式为：

```
=IF(D2<1,0.5,D2)*C2
```

9.3.2 字符提取之奇思妙想

使用字符提取大部分情况下都是因为将不同属性的内容放在一个单元格，违背了表格设计的初衷——一格一属性。也正是因为有大批这样的人，才催生了很多函数公式高手，根据混乱的表格，想出神奇的解决方案。凡事都有两面性，有利必有弊，我们就往好的方面想，通过了解这些知识熟练掌握函数的嵌套，同时锻炼思维。

如图9-57所示，这种就是最常见的，金额里含有单位。如何去除单位，只得到金额？

	A	B	C	D
1	产品	金额	效果	
2	马铃薯	200元	200	
3	西红柿	100元	100	
4	萝卜	23元	23	
5	菜心	120元	120	
6				

图9-57 只有一个单位

单位都是统一为“元”，这个很好处理，直接将单位替换掉就行。

```
=SUBSTITUTE(B2,"元",)
```

但有些时候会出现混合单位，如图9-58所示，kg跟斤混用。

	A	B	C	D
1	产品	重量	效果	
2	马铃薯	20斤	20	
3	西红柿	10斤	10	
4	萝卜	2kg	2	
5	菜心	1斤	1	
6				

图9-58 混合单位

这时LENB函数就派上用场，它与LEN函数很像，但略有差异。

LENB是统计字节数，汉字为2字节，字母跟数字为1字节。也就是说斤与kg其实都是2个字节数，所以提取左边总字节数-2即可。

```
=LEFT(B2,LENB(B2)-2)
```

接下来看如何分离姓名跟电话号码，如图9-59所示。

	A	B	C	D
1	联系方式	姓名	电话	
2	杨西粉13756214567	杨西粉	13756214567	
3	张萍07562356234	张萍	07562356234	
4	李丽拉15987654321	李丽拉	15987654321	
5	张姚文0102345678	张姚文	0102345678	
6	杨阳心语13112345678	杨阳心语	13112345678	
7				

图9-59 分离姓名跟电话号码

虽然姓名和电话字符数都不确定，但还



是有规律可循，即可以利用字节跟字符数两者的特点来获取长度。

汉字=总字节数-总字符数

=LEFT(A2,LENB(A2)-LEN(A2))

数字=总字符数-汉字=总字符数-(总字节数-总字符数)=2*总字符数-总字节数

=RIGHT(A2,2*LEN(A2)-LENB(A2))

还有更难解决的问题，如图9-60所示，在地址中提取数字。

	A	B	C
1	地址	提取数字	
2	成都市万年场横街235号	235	
3	成都市万年场横街214号	214	
4	成都市万年场北横街365号院	365	
5	成都市双林路78号大院	78	
6			

图9-60 在地址中提取数字

数字不在前后，而在中间，愁死个人儿。怎么判断数字的起始位置呢？

难题之二，如图9-61所示，获取最后一个电话号码。

	A	B	C
1	电话	最后一个电话	
2	TEL1 : 63791222TEL2 : 13456701811TEL3 : 13735556709	13735556709	
3	TEL : 13777368408	13777368408	
4	TEL : 13456881155	13456881155	
5	TEL : 63701628	63701628	
6	TEL1 : 13819137809TEL2 : 13516805096	13516805096	
7	TEL : 13906816693	13906816693	
8	TEL : 13208044738	13208044738	
9	TEL : 13858179666	13858179666	
10	TEL : 13093715558	13093715558	
11	TEL : 13968039919	13968039919	
12	TEL : 13805705802	13805705802	
13	TEL : 15005927603	15005927603	
14	TEL : 13615716958	13615716958	
15	TEL : 15988105806	15988105806	
16	TEL : 15957195876	15957195876	
17	TEL : 13175088432	13175088432	
18			

图9-61 获取最后一个电话号码

同样只能依靠字符跟字节这个特点来处理，可惜FIND函数不支持通配符，要不然用这个函数来查找很不错。这时SEARCHB函数就派上用场了，其作用与FIND函数差不多，但支持通配符。单字节可以用通配符问号(?)，通过查找问题的字节数从而知道数字的起始位置。

=SEARCHB("?",A2)

因为现在是使用字节数，MID函数就派不上用场，不过它的兄弟MIDB函数正好可以解决这个问题。

=MIDB(A2,SEARCHB("?",A2),2*LEN(A2)-LENB(A2))

函数与公式粗看很难学，因为内置的就有500多个函数，而其实函数很容易学，就如很多函数的用法都差不多。例如MID(MIDB)、LEN(LENB)、FIND(SEARCH、SEARCHB)等，学会一个函数就等于学会两个以上。

单元格中全部是单字节，不能借助字节数跟字符数的特点来提取。这时就是见证奇迹的时刻！

```
=-LOOKUP(1,-RIGHT(A2,ROW($1:$15)))
```

对ROW函数我们已经很熟悉了，它就是用于获取行号，ROW(\$1:\$15)表示获取1到15位。那么看看RIGHT函数的语法：

```
=RIGHT(文本,提取右边N位)
=RIGHT(A2,1)得到9
=RIGHT(A2,2)得到09
=RIGHT(A2,3)得到709
=RIGHT(A2,4)得到6709
...
=RIGHT(A2,11)得到13735556709
...
=RIGHT(A2,15)得到EL3 13735556709
=RIGHT(A2,ROW($1:$15))
```

也就是获取右边1到15位：

```
={"9";"09";"709";"6709";"56709";"556709";"5556709";"35556709";"735556709";"3735556709";"13735556709";" 13735556709";"3 13735556709";"L3 13735556709";"EL3 13735556709"}
=-RIGHT(A2,ROW($1:$15))
```

就是将数字变成负数，文本变成错误值：

```
={-9;-9;-709;-6709;-56709;-556709;-5556709;-35556709;-735556709;-3735556709;-13735556709;#VALUE!;#VALUE!;#VALUE!;#VALUE!}
```

根据以大欺小法的原则=LOOKUP(1,-RIGHT(A2,ROW(\$1:\$15)))会提取到最后一个数字-13735556709。

既然将数字变成负数，就得想办法将它复原，再加一个负号就可以，负负得正。

```
=-LOOKUP(1,-RIGHT(A2,ROW($1:$15)))
```

为什么要提取1到15位而不是提取1到更多呢？

Excel允许的最大数字刚好是15位，提取再多也没有意义，只要保证能提取到全部数字就行。



9.3.3 将查找进行到底

基本上任何人都会接触到查找，通过某某查找某某，很神奇的一个功能。

如图9-62所示，是泸州老窖的出货情况，想知道最后一天出货的产品及规格、数量，该如何查找？

	A	B	C	D	E	F
1	时间	物料编码	品种及规格	起止地	结费地	数量
2	2015-6-6	LJSJ-007	52老窖世家窖之藏	泸州-济南	济南	105
3	2015-6-7	LJSJ-009	52老窖世家窖之醇	泸州-济南	济南	135
4	2015-6-8	LJSJ-010	38老窖世家窖之醇	泸州-济南	济南	54
5	2015-6-9	LJSJ-008	38老窖世家窖之藏	泸州-济南	济南	49
6	2015-6-10	LJSJ-007	52老窖世家窖之藏	泸州-济南	济南	300
7	2015-6-11	LJSJ-009	52老窖世家窖之醇	泸州-济南	济南	900
8	2015-6-12	LJSJ-010	38老窖世家窖之醇	泸州-济南	济南	360
9	2015-6-13	LJSJ-008	38老窖世家窖之藏	泸州-济南	济南	140
10	2015-6-14	LJSJ-007	52老窖世家窖之藏	泸州-杭州	杭州	400
11	2015-6-15	LJSJ-009	52老窖世家窖之醇	泸州-杭州	杭州	1500
12	2015-6-16	LJSJ-010	38老窖世家窖之醇	泸州-杭州	杭州	201
13	2015-6-17	LJSJ-008	38老窖世家窖之藏	泸州-杭州	杭州	200
14						
15	最后一天：					
16	品种及规格	38老窖世家窖之藏				
17	数量	200				
18						

图9-62 查找最后一天的对应值

```
=LOOKUP("座",C2:C13)
=LOOKUP(9E+307,F2:F13)
```

这个9E+307与“座”是什么意思？

```
=LOOKUP(2000,{105;135;54;49;300;900;360;140;400;1500;201;200})
```

返回200。

```
=LOOKUP(5000,{105;135;54;49;300;900;360;140;400;1500;201;200})
```

返回200。

```
=LOOKUP(10000,{105;135;54;49;300;900;360;140;400;1500;201;200})
```

返回200。

通通都返回最后一个数字200，也就是说，LOOKUP函数查找到最后一个满足条件的值，在数字不确定的情况下，查找的值很大也能保证查找到的值的准确性。9E+307是一个很大的

数字，Excel允许最大的数字不能超过15位，而 $9E+307$ 是9乘以10的307次方，比最大值还要大，查找到最后一个值是相当保险的。而“座”是一个接近最大的文本，虽然还有比“座”大的文本，但正常情况不会出现，所以写“座”就能查找到最后一个文本。

如果想获取查找文本或者数字的通用公式，可以利用LOOKUP函数的经典查找模式。

```
=LOOKUP(1,0/(C2:C13<>""),C2:C13)
```

如图9-63所示，根据姓名或者工号查找工资，很多时候我们能记住的就是其中一个而已。

	A	B	C	D	E	F	G
1	姓名	工号	工资		姓名或工号	工资	
2	袁恒桂	龙001	4528		龙002	3228	
3	邵广福	龙002	3228		龙003	2952	
4	袁昌林	龙003	2952		邵广福	3228	
5	刘志银	龙004	3946		王永金	3357	
6	刘宝植	龙005	3298				
7	袁昌友	龙006	4738				
8	谢俭华	龙007	3792				
9	王永金	龙008	3357				
10	王永舟	龙009	4387				
11							

图9-63 根据姓名或者工号查询工资

常规法就是用VLOOKUP进行两次查找，第一次查找不到对应值，再进行一次查找。

```
=IFERROR(VLOOKUP(E2,B:C,2,),VLOOKUP(E2,A:C,3,))
```

当然我们也可以动点脑筋，姓名跟工号都是在一行，只要知道行号，再嵌套INDEX函数也可以获取对应值。可以让区域A\$2:B\$10和E2进行比较，只要两列有一列符合条件就显示本身行号，输入公式后，按Ctrl+Shift+Enter组合键结束。

```
=MAX((A$2:B$10=E2)*ROW($2:$10))
```

最后公式如下，按Ctrl+Shift+Enter组合键结束。

```
=INDEX(C:C,MAX((A$2:B$10=E2)*ROW($2:$10)))
```

如图9-64所示，是每个目的地的价钱明细，如何跨表查询各个区间的价钱？如图9-65所示。

	A	B	C	D	E	F	G	H
1	区间	价钱						
2	20吨以上	504						
3	10吨-20吨(含)	534						
4	3吨-10吨(含)	563						
5	3吨以下(含)	640						
6								
7								
8								

图9-64 目的地价钱明细



	A	B	C
1	目的地	池州	
2			
3	区间	价钱	
4	3吨以下(含)	643	
5	3吨-10吨(含)	566	
6	10吨-20吨(含)	536	
7	20吨以上	514	
8			

图9-65 查询表

第一想法肯定是判定目的地，然后进行查询，如池州。

```
=IF(B$1="池州",VLOOKUP(A4,池州!A:B,2,))
```

同理每个目的地就设置一个VLOOKUP函数，虽然这样可以解决问题，但非常烦琐，还容易出错。

例如B\$1="池州"，就在池州这个工作表查找，也就是引用这个工作表名。因为单元格是变动的，直接间接引用，INDIRECT函数对数据就是间接引用的。

```
=INDIRECT(B$1&"!A1")
```

就是间接引用池州这个工作表的A1单元格。

最终公式为：

```
=VLOOKUP(A4,INDIRECT(B$1&"!A:B"),2,)
```

关于查找还有一个很有趣的功能，就是查找图片。如图9-66所示，根据花名查找图片。




	A	B	C	D	E
1	图片	花名		花名	图片
2		菊花		八仙花	
3		八仙花			
4		郁金香			

图9-66 根据花名查找图片

STEP 01 使用名称引用动态图片，定义一个叫“花”的名称。如图9-67所示，单击“公式”选项卡中的“定义名称”按钮。在“名称”文本框输入“花”，在“引用位置”编辑框输入下面的公式，最后单击“确定”按钮。

```
=INDEX(Sheet1!$A:$A,MATCH(Sheet1!$D$2,Sheet1!$B:$B,))
```



图9-67 定义名称

STEP 02 如图9-68所示，选择D2单元格，单击“数据”选项卡中的“数据验证”按钮，弹出“数据验证”对话框。在“允许”下拉列表框中选择“序列”选项，在“来源”编辑框中引用下面的区域，单击“确定”按钮。




图9-68 设置序列



=B\$2:B\$4

STEP 03 如图9-69所示，插入任意一张图片，在编辑栏输入公式“=花”，按Enter键后图片就会自动改变，最后再稍微调整一下图片的位置及大小。



图9-69 插入图片并输入公式

9.3.4 认识各种各样的排名

排名看似很简单的一个问题，但实际上包含很多知识。有中国排名也有美国排名，中国排名又分很多种，如按条件排名、加权排名等。

如图9-70所示，是某学校各班级成绩，现在根据成绩进行有并列排名、无并列排名、中国式排名、加权排名、班级排名五种排名。

	A	B	C	D	E	F	G	H	I
1	班级	姓名	语文	总分	有并列排名	无并列排名	中国式排名	加权排名	班级排名
2	101	张三	114	382	1	1	1	1	1
3	101	王五	111	379	2	2	2	3	2
4	101	李四	103	303	11	11	9	11	4
5	101	闻一	115	324	8	8	6	8	3
6	102	张二	111	313	9	9	7	9	3
7	102	江八	102	359	6	6	5	7	2
8	102	张六	108	359	6	7	5	6	2
9	103	赵六	112	362	5	5	4	5	3
10	103	钱子	104	284	12	12	10	12	5
11	103	赵一	107	311	10	10	8	10	4
12	103	孙六	110	370	4	4	3	4	2
13	103	周末	120	379	2	3	2	2	1
14									

图9-70 五种排名方式

有并列排名其实就是默认的排名，也叫美式排名。

```
=RANK(D2,D$2:D$13)
```

RANK函数语法：

```
=RANK(单元格,区域,升降序)
```

第三参数为0或者省略就是降序，非0就是升序。其实降序就是正数排名，升序就是倒数排名。倒数排名可用：

```
=RANK(D2,D$2:D$13,1)
```

卢子以前的英语成绩很稳定，基本上都是倒数10名，这个公式很适合。

无并列排名，也就是成绩一样的情况下第一次出现的那个人就排前一位，后面出现的人就排后一位。这时座位靠前的同学就有优势了。现在需要判断的是每个成绩累计出现几次。

```
=COUNTIF(D$2:D2,D2)
```

知道了每个成绩的累计出现次数，只需加上原来的排名减去1即可。

```
=RANK(D2,D$2:D$13)+COUNTIF(D$2:D2,D2)-1
```

中国式排名在前面提过，这里就写个公式，不做说明。

```
=COUNT(1/FREQUENCY(D$2:D$13,IF(D$2:D$13>=D2,D$2:D$13)))
```

加权排名，主要排名依据是总分，但总分相同的时候看语文成绩。这种排名就比前面的座位靠前有优势更符合实际，要想排名靠前还得靠真正的实力。

```
=SUMPRODUCT(--(D$2:D$13+C$2:C$13%>=D2+C2%))
```

C\$2:C\$13%就是让语文成绩缩小，变成小数点后面的数，这样相加就不会影响到前面的总分排名。

班级排名，就是按班级给里面的学生进行排名，不同班级互相不影响。

```
=SUMPRODUCT((A$2:A$13=A2)*(D$2:D$13>=D2))
```



9.4 小结与练习

学好函数公式并不是为了成为高手，而是为了提高工作效率，在完成工作的情况下，顺便帮助有困难的人，让他们也能提高效率，告别加班！

1. 如图9-71所示，在杂乱无章的数据中提取数值。

	A	B	C
1	数据	效果	
2	nbe12.368mnbnbr	12.368	
3	冰箱5台	5	
4	mny2.358ny	2.358	
5	muy0000128nm	0000128	
6	n258cm	258	
7	21.3698	21.3698	
8	abc00015998.36km	00015998.36	
9	mkunbt2198.361kg	2198.361	
10	895人/月	895	
11			

图9-71 杂乱无章的数据提取数值

2. 如图9-72所示，把每个班级变成一个区域，现在如何根据多个区域进行排名？

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	班级	姓名	总分	排名	班级	姓名	总分	排名	班级	姓名	总分	排名	
2	101	张三	382	1	102	张二	313	9	103	赵六	362	5	
3	101	王五	379	2	102	江八	359	6	103	钱子	284	12	
4	101	李四	303	11	102	张六	359	6	103	赵一	311	10	
5	101	闻一	324	8					103	孙六	370	4	
6									103	周末	379	2	
7													

图9-72 多区域排序

3. 如图10-73所示，字符有大小写之分，如何区分大小写查找数量？

	A	B	C	D	E
1	字符	数量		字符	数量
2	AB	1		ab	2
3	ab	2		abc	3
4	abC	4			
5	abc	3			
6					

图9-73 区分大小写查找数量



第10章

转行做财务



说真的，卢子做梦都没想到自己会转行做财务，而且成为中国会计视野版主。做品质分析已经5年了，2012年某一天，估计是因为做一行太久乏味了，没有经过任何考虑，就直接写了辞职信。从想辞职到写完辞职信，这个过程还不到30分钟。这就是单身的好处，不用考虑太多，不想做就辞职。本来想从事数据分析，最后因学历问题被刷下来，有点遗憾。卢子回到老家潮州，迷茫了好一阵子，最后鬼使神差地被某HR叫去某厂做财务。说是财务，但小厂负责的工作内容就多，说句实话跟打杂没什么区别。从那以后，别人问我做什么，我的回答永远是——打杂。

10.1 我的前任是极品

前财务是个女的，去的时候听说她经常算错账目。卢子无意间也看了她算账，还停留在计算器的时代。将数据录入Excel中，然后就敲计算器，把结果输入到Excel中。数据少还可以应付，数据一多，出错在所难免。卢子是新人，也不好说什么，就任她去。不过她走得太匆忙，没交接就走了，留下一堆乱七八糟的表格给卢子。头大啊，这样也行？卢子硬着头皮上。

打开她留下的文件夹，如图10-1所示，出现这么几份表格。











名称	修改日期	类型	大小
 2012.06.24盘点清单 工作表.xls	2012-07-01 15:20	Microsoft Office...	20 KB
 2012.06.30盘点清单 工作表.xls	2012-07-02 11:13	Microsoft Office...	20 KB
 客户退货(工作).xls	2012-07-01 11:48	Microsoft Office...	23 KB
 老板购货清单 工作表.xls	2012-07-01 11:23	Microsoft Office...	20 KB
 连祥零售清单 工作表.xls	2012-07-01 10:38	Microsoft Office...	21 KB
 连祥批发清单 工作表.xls	2012-07-01 10:55	Microsoft Office...	26 KB
 每天成品入库清单 工作表.xls	2012-07-01 11:51	Microsoft Office...	31 KB
 外购配料 工作表.xls	2012-07-01 11:21	Microsoft Office...	23 KB
 外购猪肉工作表.xls	2012-07-01 13:39	Microsoft Office...	21 KB
 员工内购产品工作表.xls	2012-07-01 10:06	Microsoft Office...	25 KB

图10-1 各种表格

卢子打开了第一份表《2012.06.24盘点清单》，如图10-2所示。



	A	B	C	D	E	F	G	H	I
1	2012.06.24**食品厂仓库盘点清单					2012.06.24**食品厂退货盘点清单			
2	名称	数量	单价	合计		名称	数量	单价	合计
3	228g潮厨猪肉脯（黑椒）	6箱	12.8元	1536		228g潮厨猪肉脯（黑椒）	2箱+16包	12.8元	7168
4	228g潮厨猪肉脯（芝麻）	3箱	12.8元	768		228g潮厨猪肉脯（芝麻）	32包	12.8元	409.6
5	32g潮厨猪肉脯（芝麻）	2箱+25包	2.5元	363		190g连祥枣肠	8包	7.5元	60
6	32g潮厨猪肉脯（黑椒）	2箱+5包	2.5元	313		150g连祥腊肠	10包	6.5元	65
7	100g潮厨猪肉脯（黑椒）	2箱+8包	8元	448		2.5kg潮厨猪肉脯（小脯）	13.2斤	26元	343.2
8	100g潮厨猪肉脯（原味）	2箱+9包	8元	456		散装大脯	19斤	30元	570
9	138g潮厨猪肉松（原味）	3罐	12元	36		总额为：8615.8元			
10	138g潮厨猪肉松（紫菜味）	2罐	12元	24					
11	190g连祥枣肠	15包	7.5元	112.5		账目交接日期	2012.06.24		
12	150g连祥腊肠	2包	6.5元	13		经手人			
13	散装猪肉条	14箱+7斤	31元	4557		移交人			

图 10-2 2012.06.24 盘点清单

我的天啊！直接把卢子敲晕好了。

“数量”这里含有单位，而且还是不同单位，里面还有运算！

“单价”这里是针对哪个单位的价钱？

“合计”是怎么算出来的？

看了半天，卢子始终没反应过来。

卢子接着打开第二张表《客户退货》，如图10-3所示。

	A	B	C	D	E	F	G
1	2012年6月份客户退货清单						
2	日期	名称	数量	单价	总额	合计	客户
3	2012.06.25	2.5kg潮厨猪肉脯（小脯）	5.1斤	26元	133	133	花姐
4	2012.06.25	潮厨咸肉	1箱	7.5元	150元	150	枫春
5	2012.06.25	散装腊肠	80斤	18元	1440元	1440	潮厨腊味
6		190g连祥枣肠	4包	7.5元	30元		
7	2012.06.28	150g连祥腊肠	1包	6.5元	6.5元	530	金石海
8		190g 连祥枣肠	66包	7.3元	482元		
9		150g连祥腊肠	18包	6.3元	113元		
10		138g潮厨猪肉松	3罐	12元	36元		
11		散装大脯	1.4斤	30元	42元		
12		无印日期小脯	10斤	26元	260元		
13		腊肉	1包	8.5元	8.5元	3365	黄泽佳
14		散装小脯	26斤	26元	676元		
15		132g金钱猪肉脯	116包	7.5元	870元		

图 10-3 客户退货表



跟第一张表格差不多，多了一些合并单元格。
卢子打开第三张表《每天成品入库清单》，如图10-4所示。

	A	B	C
1	2012年6月份成品入库清单		
2	日期	名称	数量
3	2012.06.25	2.5kg潮厨猪肉脯（黑椒味）	19箱+5斤
4		2.5kg潮厨猪肉脯（原味）	9箱+250g
5		2.5kg潮厨猪肉脯（芝麻味）	20箱
6		150g优质香肠	300包
7		XO酱猪肉脯（5kg）	20箱
8	2012.06.26	儿童猪肉脯（5kg）	4箱
9		100g潮厨猪肉脯（小脯）	4箱
10		190g连祥腊肠	5箱
11	2012.06.27	150g优质香肠	13箱
12		138g潮厨猪肉松（紫菜味）	62罐
13		138g潮厨猪肉松（原味）	4箱+3罐
14		228g潮厨猪肉脯（芝麻）	2件
15		229g潮厨猪肉脯（芝麻）	2件

图10-4 每天成品入库清单

形式与上面的表格差不多，看到这里，卢子快崩溃了。接下来依次打开了所有表格，都是不规范的。
录入数据随心所欲，汇总表格痛不欲生。这样做的表格还停留在手工记账算账的年代，高效的Excel跟它们没有任何关系。唯一值得欣慰的是，她已经到了第三阶段，脱离了算筹、算盘，如图10-5所示。



图10-5 计算等级

好好学习，借助函数公式能够提升数据处理的效率和准确度。



10.2 表格大改革

遇到这样的前任，卢子能说什么呢？唯一能做的就是对所有表格进行大改革，改革的目的是让汇总数据变得随心所欲，录入数据随心所欲并没有用。

10.2.1 来料

如图10-6所示，是杂费来料表。不使用合并单元格，一个单元格只输入一种内容，如数量跟单价放在两列，价格的单位放在标题上。

	A	B	C	D	E	F
1	日期	名称	单位	数量	单价（元）	金额（元）
2	2012-9-1	华丰气	瓶	2	108	216
3	2012-9-1	木炭	斤	3127	1.8	5629
4	2012-9-1	酒	斤	100	4	400
5	2012-9-1	肠衣	斤	22	105	2310
6	2012-9-1	素仔	斤	60	14	840
7	2012-9-3	鸡蛋	件	2	215	430
8	2012-9-3	白糖	包	2	300	600
9	2012-9-3	膜卷	kg	311	22.4	6966
10	2012-9-6	不锈钢筛	个	100	58	5800
11	2012-9-8	鸡蛋	件	2	215	430
12	2012-9-10	潮思乐肉松彩盒	个	1020	2.1	2142
13	2012-9-11	酒	斤	100	4	400
14	2012-9-13	小口径	把	100	41	4100
15	2012-9-6	麻仁酱	斤	200	9	1800
16	2012-9-14	麻仁粉	斤	190	9	1710

图10-6 杂费来料表

这里对金额设计一个简单的公式：




```
=ROUND(D2*E2,)
```

这种表格如果以后要汇总的话，借助数据透视表就能轻松搞定，当然用函数汇总也不难，删除重复的名称，借助SUM函数进行汇总。

借助同样的方法，制作猪肉来料表，如图10-7所示。

猪肉在每个时间段都有对应价格，这里可以再设计一个猪肉价格表，作为引用单价，如图10-8所示。

	A	B	C	D	E	F
1	日期	名称	单位	数量	单价（元）	金额（元）
2	2012-9-1	赤肉	斤	137.5	8.3	1141
3	2012-9-1	白肉	斤	26.5	4.5	119
4	2012-9-1	肚肉	斤	12	8	96
5	2012-9-2	赤肉	斤	65	8.3	540
6	2012-9-3	赤肉	斤	83	8.3	689
7	2012-9-3	肚肉	斤	18.5	8	148
8	2012-9-4	赤肉	斤	201	8.3	1668
9	2012-9-4	肚肉	斤	24	8	192
10	2012-9-5	赤肉	斤	291	8.3	2415
11	2012-9-5	肚肉	斤	17	8	136
12	2012-9-6	赤肉	斤	186	8.3	1544
13	2012-9-6	肚肉	斤	17	8	136
14	2012-9-6	肚肉	斤	25.5	8	204
15	2012-9-7	肚肉	斤	33.5	8	268
16	2012-9-7	肚肉	斤	100.5	8	804

图10-7 猪肉来料表

	A	B	C	D	E
1	名称	2012-9-1	2012-9-10	2012-9-12	
2	白肉	4.5	4.5	4.5	
3	肚肉	8	8	9	
4	赤肉	8.3	8.7	9	
5					

图10-8 猪肉价格表

单价的公式为：

```
=VLOOKUP(B2,猪肉价格表!A:D,MATCH(A2,猪肉价格表!$1:$1),)
```

时间段属于模糊查找，MATCH函数第三参数省略，名称需要精确查找，所以VLOOKUP函数第四参数为0（注：利用“,”占位，其实等同于后面是0）。

细小的改变，使数据的准确性更高，终于搞定来料。

10.2.2 生产

如图10-9所示，生产表其实很简单，就是记录每天生产的产品数量而已。

但光有这份表肯定不够，还得知每种产品主要由哪些材料生产，这样每天好准备材料。如图10-10所示，是每种产品由哪些配料组成。



	A	B	C
1	日期	产品	生产量
2	2012-9-1	腊肠	550
3	2012-9-1	肉松	100
4	2012-9-2	腊肠	350
5	2012-9-2	肉脯	200
6	2012-9-2	腊肉	400
7	2012-9-3	肉脯	500
8	2012-9-4	肉脯	400
9	2012-9-4	肉松	100
10	2012-9-5	肉松	100
11	2012-9-6	肉脯	500
12	2012-9-6	肉松	100
13	2012-9-6	咸肉	488
14	2012-9-7	肉脯	500
15	2012-9-8	肉脯	500
16	2012-9-9	肉脯	300

图10-9 生产表

	A	B	C	D	E
1	产品	赤肉	肚肉	白肉	
2	腊肠	70%		30%	
3	肉松	100%			
4	肉脯	100%			
5	腊肉		100%		
6	咸肉		100%		
7	肉条	100%			
8					

图10-10 用料对应表

有了这份用料对应表，就知道每天生产的产品，需要准备什么东西，如图10-11所示。

	A	B	C	D	E	F
1	日期	产品	生产量	赤肉	肚肉	白肉
2	2012-9-1	腊肠	550	385	0	165
3	2012-9-1	肉松	100	100	0	0
4	2012-9-2	腊肠	350	245	0	105
5	2012-9-2	肉脯	200	200	0	0
6	2012-9-2	腊肉	400	0	400	0
7	2012-9-3	肉脯	500	500	0	0
8	2012-9-4	肉脯	400	400	0	0
9	2012-9-4	肉松	100	100	0	0
10	2012-9-5	肉松	100	100	0	0
11	2012-9-6	肉脯	500	500	0	0
12	2012-9-6	肉松	100	100	0	0
13	2012-9-6	咸肉	488	0	488	0
14	2012-9-7	肉脯	500	500	0	0
15	2012-9-8	肉脯	500	500	0	0
16	2012-9-9	肉脯	300	300	0	0

图10-11 查询用料

各种用料根据对应表可以设置公式为：

=C2*VLOOKUP(\$B2,用料对应表!\$A:\$D,COLUMN(B1),)



10.2.3 销售

来料和生产都比较简单，销售才是关键。

如图10-12所示，是产品出库单，是针对批发的。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	潮安县**食品厂														
2	产品出库单														
3	出库单号: 1x 2012-9-3 6单										日期: 2012年9月3日				
4	客 户: 阿佳														
5	编号	产品名称			规格	单位	数量	单价 (元)	金额(元)		备注				
6	005	228g潮厨猪肉脯(合味小脯)			1×20包	件	1	250	250						
7	001	2.5kg金牌猪肉脯(合味)			1×2.5kg	件	20	130	2600						
8	012	138g潮厨猪肉松(合味)			1×12罐	件	15	144	2160						
9															
10															
11															
12	合计						36	5010							
13	收货人:														
14	供应商: 潮安县**食品厂										电话: 0768-***				
15	地址: 潮安县**										传真: 0768-***				
16															

白联: 存根

红联: 客户

黄联: 财务

白联: 存根
红联: 客户
黄联: 财务

图10-12 产品出库单

如图10-13所示，是产品明细表。

	A	B	C	D	E	F	H
1	潮安县**食品厂产品明细						
2	编号	产品名称	规格	单位	单价	金额	备注
3	001	2.5kg金牌猪肉脯(合味)	1×2.5kg	件	130	元	
4	002	32g潮厨猪肉脯(合味)	1×60包	件	150	元	
5	003	100g潮厨猪肉脯(合味大脯)	1×20包	件	140	元	
6	004	100g潮厨猪肉脯(合味小脯)	1×20包	件	126	元	
7	005	228g潮厨猪肉脯(合味小脯)	1×20包	件	250	元	
8	006	5kg连祥猪肉脯	1×5kg	件	260	元	
9	007	150g优质腊肠	1×30包	件	195	元	
10	008	190g优质腊肠	1×20包	件	146	元	

图10-13 产品明细表

编号是唯一的，也就是说可以通过编号用VLOOKUP函数查询到产品名称、规格、单位、单价。

```
=IF($A6="", "", VLOOKUP($A6, 产品明细!$A$3:$AC$299, COLUMN(B1),))
```



金额的公式：

```
=IF(A6="", "", ROUND(I6*J6,))
```

“合计”那里就直接用SUM函数求和。

本来这样的公式并没有问题，但实际情况往往比预想更复杂。

(1) 有些时候会出产很多产品，出库单行数不够，插入行后用SUM函数直接求和会出错。这是卢子在某次核对数据的时候发现的，金额比实际少了几百元。只要涉及金额的，绝不容许有半点差错！

合计的金额最后改成：

```
=SUM(K6:INDEX(K:K,ROW()-1))
```

为了保险起见，合计的数量公式也改成：

```
=SUM(I6:INDEX(I:I,ROW()-1))
```

ROW()-1的作用就是返回合计上一行的行号，也就是说不管你插入多少行，永远都是返回上一行，在外面嵌套INDEX函数就得到上一行的对应值。

SUM(起始单元格:结束单元格)就是统计这个区域的和。

(2) 批发的也有小客户，某些产品购买的量不足1件。刚开始的时候，卢子针对这种情况就采用手工计算，但下回重新开单的时候，有的时候会忘记哪里是用手输入并没有公式，导致价钱出错。

这时先从产品明细表下手，获取散装产品的单位及价格，如图10-14所示。

	A	B	C	D	E	F	G	H	I	J	K
1											
2											
3											
4											
5											
6											
7											
8											
9											
10											

潮安县**食品厂产品明细										
编号	产品名称	规格	单位	单价	金额	备注	新单位	新价格		
001	2.5kg金锣猪肉脯(合味)	1×2.5kg	件	130	元		斤	26		
002	32g福厨猪肉脯(合味)	1×60包	件	150	元		包	2.5		
003	100g福厨猪肉脯(合味大桶)	1×20包	件	140	元		包	7		
004	100g福厨猪肉脯(合味小桶)	1×20包	件				包	6.3		
005	228g福厨猪肉脯(合味小桶)	1×20包	件				包	12.5		
006	5kg连洋猪肉脯	1×5kg	件	260	元		斤	26		
007	150g优品腊肠	1×30包	件	195	元		包	6.5		
008	190g优品腊肠	1×20包	件	146	元		包	7.3		

图10-14 散装产品的单位及价格

新单位很好提取，直接用：

```
=RIGHTB(C3,2)
```



卢子比较喜欢用“斤”作为单位，而不是用“kg”，所以将“kg”替换掉。

```
=SUBSTITUTE(RIGHTB(C3,2),"kg","斤")
```

新价格的获取就稍微有点难。

```
=IF(RIGHTB(C3,2)="kg",0.5,1)*(E3/LOOKUP(1,-MID(C3,3,ROW($1:$9))))
```

先把公式分成三部分，第一步是提取规格中的数量。

```
=LOOKUP(1,-MID(C3,3,ROW($1:$9)))
```

数字都是从第3位开始，提取连续的最大值。当然也可以将公式改成：

```
=SUBSTITUTE(MID(C3,3,9),RIGHTB(C3,2),)
```

第二步就用单价除以规格中的数字，得出新价格。

```
=E3/LOOKUP(1,-MID(C3,3,ROW($1:$9)))
```

本来到这一步已经结束，但因为前面将kg转换成斤，所以这里需要再次转换，让kg的数据乘以0.5，其他不变。

```
=IF(RIGHTB(C3,2)="kg",0.5,1)
```

产品明细表设置完成后，重新回到出库表。现在“单位”这里全部改成手工输入，不足1件就记产品的实际单位，如图10-15所示。

	A	B	C	D	F	G	H	I	J	K	L	M	N	O	
1	潮安县**食品厂														白联：存根 红联：客户 黄联：财务
2	产品出库单														
3	出库单号：lx		2012-9-3		6单						日期：2012年9月3日				
4	客 户：阿佳														
5	编号	产品名称		规格	单位	数量	单价 (元)	金额(元)		备注					
6	005	228g潮厨猪肉脯(合味小脯)		1×20包	件	1	250	250							
7	001	2.5kg金牌猪肉脯(合味)		散装	斤	20	26	520							
8	012	138g潮厨猪肉松(合味)		1×12罐	件	15	144	2160							
9															
10															
11															
12		合计				36		2930							
13								收货人：							
14	供应商：潮安县**食品厂														
15	地址：潮安县**														
16															

图10-15 单位记录



产品名称公式:

```
=IF(A6="", "", VLOOKUP(A6, 产品明细!A:N, 2,))
```

规格公式:

```
=IF(A6="", "", IF(H6<>"件", "散装", VLOOKUP(A6, 产品明细!A:N, 3,)))
```

单价公式:

```
=IF(A6="", "", VLOOKUP(A6, 产品明细!A:N, IF(H6<>"件", 11, 5),))
```

利用同样的方法，设置产品退货单。

出货跟退货的单据设置好后，接下来的问题就是如何将这些单据导入一个新的工作表，方便以后汇总。前几个星期，卢子尝试过重新录入和复制、粘贴法，不过效率很低。如果每天处理五六份表格就要搞半个钟头。

最后卢子跟好友“无言的人”商量了大半个月后，最终使用下面的VBA代码解决这个问题。以后只要选择需要汇总的工作簿，30秒不到就将所有表格数据汇总完，很方便。

Option Explicit

Public Arr(), Luj As String, ShD As String

Rem 明细汇总操作

Sub 明细汇总文件选择发退货()

Dim Tim As Long, TiS As Byte

TiS = InputBox("请输入你需提取的工作表是发货单 1 还" & Chr(10) & "是退货单 2 请输入 1 或 2 !" & _

Chr(10) & "1 代表发货单" & Chr(10) & "2 代表 退货单", Title:="汇总工作表名称选择", Default:=1)

If Val(TiS) <= 1 Then

ShD = "发货单"

ElseIf Val(TiS) > 1 And Val(TiS) <= 255 Then

ShD = "退货单"

Else

End

End If



```

Call NewSh      '可执行——判断新建工作表并填入标题
Call FileDialBookName '可执行——选择多个文件
Call MXHuiZ     '可执行——明细汇总操作 分为2步 1 打开已记录的文件路径下的Excel
文件 _
                2 执行工作表复制并将数据录入明细表中
Call dk         '返回明细表
Call tz         '调整列宽
End Sub

```

Rem 第一步：新建发货单明细表。

```

Private Sub NewSh()
Dim Sh As Worksheet, i%
Do While Sheets.Count = 1 And Sheets(1).Name <>"发货单明细表"
With Sheets.Add
.Move After:=Worksheets(Worksheets.Count)
.Name = "发货单明细表"
.Range(Cells(1, 1), Cells(1, 11)).Interior.Color = RGB(150, 50, 200)
.Range(Cells(1, 1), Cells(1, 11)).Font.Bold = True
.Cells(1, 1).Resize(1, 11) = Array("日期", "客户", "单号", "产品名称", _
"规格", "单位", "数量", "单价 元 ", "金额 元 ", "进货或退货")
End With
Loop
For Each Sh In Worksheets
If Sh.Name = "发货单明细表" Then Exit For
Next Sh
End Sub

```

Rem 第二步：获取选定文件夹中的Excel文件。

```

Private Sub FileDialBookName()
Dim WJj, WJ, i%

```




```

Set WJj = Application.FileDialog(msoFileDialogFilePicker)
With WJj
    .AllowMultiSelect = True
    .Filters.Clear
    .Filters.Add "Excel", "*.xls;*.xlsx", 1
    .Show
    For i = 1 To .SelectedItems.Count
        ReDim Preserve Arr(1 To i)
        Arr(i) = .SelectedItems(i)
    Next i
End With
End Sub

```

Rem 第三步：打开文件夹下的Excel文件。

```

Private Sub MXHuiZ()
Dim i%, ii%, Wb As Workbook, Sht As Worksheet, R As Integer

For i = 1 To UBound(Arr)
    Workbooks.Open Arr(i)
    With Workbooks(Dir(Arr(i)))

        .Sheets(ShD).UsedRange.Copy
        ThisWorkbook.Sheets(1).Activate
        [Q1] = Dir(Arr(i))
        [A1].PasteSpecial
        Call FHMXFZ
        Sheets(1).UsedRange.Clear

        Application.SendKeys "{ENTER}" '回车
        .Close False
    End With
Next i

End Sub

```



Rem 第四步：获取各工作簿明细→复制、粘贴对应取值，在第三步过程中执行。

```
Private Sub FHMXFZ()
Dim i As Integer, iR As Integer, ii As Integer, Mxr As Integer
iR = 0
Application.ScreenUpdating = False
With Sheets(1)
iR = .Cells(.Cells(Cells.Rows.Count, 2).End(xlUp).Row, 1).End(xlUp).Row - 5
End With
Mxr = Sheets("发货单明细表").UsedRange.Rows.Count
For ii = 1 To iR
With Sheets("发货单明细表")
Rem 日期 N3 vs A2开始
.Cells(Mxr + ii, 1) = Sheets(1).[N3] 'Format(Sheets(1).[N3], "yyyy年mm月dd日")
Rem 客户 A4 vs B2开始
.Cells(Mxr + ii, 2) = Mid(Sheets(1).[A4], 8, 5)
Rem 单号 F3 vs C2开始
.Cells(Mxr + ii, 3) = Sheets(1).[E3] & Sheets(1).[F3]
Rem 产品名称 B6 开始 vs D2开始
.Cells(Mxr + ii, 4) = Sheets(1).Cells(5 + ii, "B")
Rem 规格 G6 开始 vs E2开始
.Cells(Mxr + ii, 5) = Sheets(1).Cells(5 + ii, "G")
Rem 单位 H6 开始 vs F2开始
.Cells(Mxr + ii, 6) = Sheets(1).Cells(5 + ii, "H")
Rem 数量 I6 开始 vs H2开始
.Cells(Mxr + ii, 7) = Sheets(1).Cells(5 + ii, "I")
Rem 单价 元 J6 开始 vs I2开始
.Cells(Mxr + ii, 8) = Sheets(1).Cells(5 + ii, "J")
Rem 金额 元 K6 开始 vs J2开始
.Cells(Mxr + ii, 9) = Sheets(1).Cells(5 + ii, "K")
Rem 注明获取工作表是发货还是退货
.Cells(Mxr + ii, 10) = If(Sheets(1).[A2] = "产品出库单", "出货", "退货")
End With
Next ii

Application.ScreenUpdating = True
End Sub
```



```

Sub dk()
    Sheets(2).Select
End Sub

Sub tz()
    Columns("A:J").Select
    Columns("A:J").EntireColumn.AutoFit
    Range("a2").Select
    Range(Selection, Selection.End(xlToRight)).Select
    Range(Selection, Selection.End(xlDown)).Select
End Sub

```

STEP 01 如图10-16所示，运行代码，输入1或者2，1代表发货单，2代表退货单。




图10-16 选择发货或者退货

STEP 02 如图10-17所示，浏览到批发的各个工作表的位置，按住Ctrl键，选择有出货的工作表。




图10-17 选择工作表

导入后的效果如图10-18所示。

	A	B	C	D	E	F	G	H	I	J
1	日期	客户	单号	产品名称	规格	单位	数量	单价(元)	金额(元)	进货或退货
2	2012-9-10	阿辉	1单	散装腊肠	散装	斤	100	18	1800	出货
3	2012-9-10	阿兴	1单	2.5kg金牌猪肉脯(合味)	1×2.5kg	件	1	130	130	出货
4	2012-9-10	阿兴	1单	138g潮厨猪肉松(合味)	1×12罐	件	2	144	288	出货
5	2012-9-10	阿兴	1单	散装腊肠	散装	斤	20	18	360	出货
6	2012-9-10	大连	1单	200G连祥腊肉 咸肉	散装	斤	30	19	570	出货
7	2012-9-10	大连	1单	散装腊肠	散装	斤	30	18	540	出货
8										

图10-18 导入明细的效果

搞定批发，零售更是分分钟搞定的事情。

如图10-19所示，是零售明细表。如图10-20所示，是零售价格表。根据价格表可以获取产品的单位和单价。

	A	B	C	D	E	F	G
1	日期	客户	产品名称	单位	数量	单价(元)	金额(元)
2	2012-10-1	秀枝	小脯	斤	7	30	210
3	2012-10-2	惠玉	小脯	斤	1	30	30
4	2012-10-2	惠玉	肉松	斤	1	32	32
5	2012-10-4	秀枝	肉松	斤	1.5	32	48
6	2012-10-4	雪红	小脯	斤	1	30	30
7	2012-10-4	惜兰	小脯	斤	2.67	30	80
8	2012-10-4	惜兰	肉松	斤	1.56	32	50
9	2012-10-4	少荣	小脯	斤	6	30	180
10	2012-10-4	零售	小脯	斤	8	30	240
11	2012-10-5	零售	大脯	斤	4.68	32	150
12	2012-10-5	惠玉	小脯	斤	2	30	60
13	2012-10-5	零售	小脯	斤	2	30	60
14	2012-10-6	秀枝	小脯	斤	1	30	30
15	2012-10-7	美燕	小脯	斤	2	30	60
16	2012-10-7	零售	肉松	斤	2	32	64

图10-19 零售明细表

	A	B	C	D
1	产品名称	类型	单位	单价
2	小脯	肉制品	斤	30
3	大脯	肉制品	斤	32
4	肉松	肉制品	斤	32
5	肉条	肉制品	斤	35
6	腊肠	肉制品	斤	22
7	138g潮厨猪肉松(合味)	肉制品	罐	13
8	150g优质腊肠	肉制品	包	6.5
9	190g优质腊肠	肉制品	包	7.5
10	赤肉	猪肉	斤	10
11	肚肉	猪肉	斤	9
12	白肉	猪肉	斤	5
13	190g枣肠	肉制品	包	7

图10-20 零售价格表

VLOOKUP真是一个神奇的函数，在这里又派上用场了。

单位：

```
=VLOOKUP(C2,零售价格表!A:D,3,)
```

单价：

```
=VLOOKUP(C2,零售价格表!A:D,4,)
```

金额：

```
=ROUND(E2*F2,)
```



10.2.4 进销存

进销存分成半成品和完成品两大类，当初卢子没有进行交接就上手。期初数就不清楚了，后来又没去清点，导致库存有点小乱，这个也是小小的遗憾。很多事情不管怎么样，都尽量做到最好。

如图10-21所示，是半成品进销存。

	A	B	C	D	E
1	半成品进销存 (单位:斤)				
2	分类	进库量	生产量	库存量	
3	赤肉	12040	10350	1690	
4	肚肉	1202	280	922	
5	白肉	333	1650	-1317	
6					

图10-21 半成品进销存

如图10-22所示，就是猪肉来料表，进库量就是根据这个表汇总得出。

	A	B	C	D	E	F
1	日期	客户	名称	数量(斤)	单价(元)	金额(元)
2	2012-10-2	少华	赤肉	44.4	9	400
3	2012-10-3	老板	赤肉	103	9	927
4	2012-10-4	少华	白肉	14.5	5	73
5	2012-10-4	少华	赤肉	5.5	9	50
6	2012-10-4	少华	肚肉	12	9	108
7	2012-10-4	老板	赤肉	73	9	657
8	2012-10-4	秀枝	肚肉	78.5	8.5	667
9	2012-10-4	秀枝	赤肉	22.4	9.2	206
10	2012-10-5	老板	赤肉	105	9	945
11	2012-10-5	老板	肚肉	16	9	144
12	2012-10-5	少华	赤肉	20.8	9	187
13	2012-10-5	少华	肚肉	26.2	9	236
14	2012-10-5	秀枝	赤肉	5.8	9.2	53
15	2012-10-5	秀枝	肚肉	38.8	8.5	330
16	2012-10-5	其它	赤肉	15	9	135

图10-22 猪肉来料表

进库量公式为：

=ROUND(SUMIF(猪肉来料!C:C,A3,猪肉来料!D:D),)

如图10-23所示，是生产明细表，生产量

就是根据这个表汇总。

	A	B	C	D	E	F
1	日期	产品	生产量	赤肉	肚肉	白肉
2	2012-10-1	小脯	150	150	0	0
3	2012-10-4	腊肠	500	350	0	150
4	2012-10-5	腊肠	700	490	0	210
5	2012-10-5	小脯	200	200	0	0
6	2012-10-5	腊肉 咸肉	280	0	280	0
7	2012-10-6	小脯	400	400	0	0
8	2012-10-6	肉松	100	100	0	0
9	2012-10-7	小脯	100	100	0	0
10	2012-10-7	肉松	100	100	0	0
11	2012-10-8	肉松	100	100	0	0
12	2012-10-9	肉松	100	100	0	0
13	2012-10-9	小脯	500	500	0	0
14	2012-10-10	肉条	100	100	0	0
15	2012-10-11	小脯	600	600	0	0
16	2012-10-11	肉松	100	100	0	0

图10-23 生产明细表

生产量公式为：

=SUMPRODUCT((生产明细!\$D\$1:\$F\$1=A3)*
生产明细!\$D\$2:\$F\$900)

库存量公式为：

=B3-C3

没有期初数，库存量并不准确，仅供参考而已。

如图10-24所示，是完成品进销存。

	A	B	C	D	E	F
7	完成品进销存 (单位:斤)					
8	分类	进库量	批发量	零售量	库存量	
9	小脯	3690	3531	126	33	
10	大脯	90	302	14	-226	
11	腊肠	4950	2097	27	2826	
12	肉松	1620	453	43	1124	
13	肉条	450	81	3	366	
14	腊肉 咸肉	252	64	0	188	
15						

图10-24 完成品进销存

进库量根据生产明细表汇总，按产品损失率10%计算，也就是生产100斤产品，实际进库量只有90%。

进库量公式为：

=ROUND(SUMIF(生产明细!B:B,A9,生产明细!C:C)*90%,)

销售分成两大类：批发、零售。

如图10-25所示，是批发明细表。

	D	E	F	G	H	I	J
1	产品名称	规格	单位	数量	单价(元)	金额(元)	出货管理
2	2.5kg金牌猪肉脯(合味)	1×2.5kg	件	5	115	575	出货
3	5kg连祥猪肉脯	1×5kg	件	10	230	2300	出货
4	2.5kg金牌猪肉脯(合味)	1×2.5kg	件	25	130	3250	出货
5	228g潮厨猪肉脯(合味小脯)	1×20包	件	7	250	1750	出货
6	2.5kg金牌猪肉脯(合味)	散装	斤	25	26	650	出货
7	散装大脯	散装	斤	10	31	310	出货
8	2.5kg金牌猪肉脯(合味)	散装	斤	20	26	520	出货
9	100g潮厨猪肉脯(合味大脯)	1×20包	件	3	140	420	出货
10	138g潮厨猪肉松(合味)	1×12罐	件	3	144	432	出货
11	190g优质枣肠	散装	包	14	7.3	102	退货
12	138g潮厨猪肉松(合味)	散装	罐	1	12	12	退货
13	200G连祥腊肉 咸肉	散装	包	8	8	64	退货
14	228g潮厨猪肉脯(合味小脯)	散装	包	13	12.5	163	退货
15	2.5kg金牌猪肉脯(合味)	散装	斤	9.3	26	242	退货
16	散装大脯	散装	斤	8	31	248	退货

图10-25 批发明细表

产品名称都是详细的产品，需要进行分类处理，比如将2.5kg金牌猪肉脯(合味)转换成小脯，只有这样才能进行下一步统计。

如图10-26所示，是产品清单。

	A	B	C	D	E	F
1	编号	产品名称	规格	单位	分类	含量(斤)
2	001	2.5kg金牌猪肉脯(合味)	1×2.5kg	件	小脯	5
3	002	32g潮厨猪肉脯(合味)	1×60包	件	小脯	3.84
4	003	100g潮厨猪肉脯(合味大脯)	1×20包	件	小脯	4
5	004	100g潮厨猪肉脯(合味小脯)	1×20包	件	大脯	4
6	005	228g潮厨猪肉脯(合味小脯)	1×20包	件	小脯	9.12
7	006	5kg连祥猪肉脯	1×5kg	件	小脯	10
8	007	150g优质腊肠	1×30包	件	腊肠	9
9	008	190g优质腊肠	1×20包	件	腊肠	7.6
10	009	190g优质枣肠	1×20包	件	腊肠	7.6
11	010	400g优质腊肠	1×20包	件	腊肠	16
12	011	500g优质枣肠	1×20包	件	腊肠	20
13	012	138g潮厨猪肉松(合味)	1×12罐	件	肉松	3.312
14	013	散装大脯	1×10kg	件	大脯	20
15	014	散装腊肠	1×5kg	件	腊肠	10
16	015	散装肉松	1×5kg	件	肉松	10

图10-26 产品清单



根据产品清单，就能查找到每个产品的详细分类。

```
=VLOOKUP(D2,产品清单!B:E,4,)
```

有了分类，还需要获取各种规格的实际含量，只有统一单位才能汇总。根据产品名称、规格、单位三个条件查找含量。实现查找一般用VLOOKUP函数或者LOOKUP函数，但这里是查询数量，可以借助SUMIFS函数多条件查询。

```
=SUMIFS(产品清单!F:F,产品清单!B:B,批发明细!D2,产品清单!C:C,E2,产品清单!D:D,F2)
```

最后将规格的含量乘以数量得到每种产品的出货重量。

```
=SUMIFS(产品清单!F:F,产品清单!B:B,批发明细!D2,产品清单!C:C,E2,产品清单!D:D,F2)*G2
```

现在分类和重量单位统一，汇总就变得非常简单。

```
=ROUND(SUMIFS(批发明细!K:K,批发明细!L:L,A9,批发明细!J:J,"出货"),)
```

批发分为出货和退货，但对退货不做考虑。

如图10-27所示，是零售明细表。

	A	B	C	D	E	F	G
1	日期	客户	产品名称	单位	数量	单价(元)	金额(元)
2	2012-10-1	秀枝	小脯	斤	7	30	210
3	2012-10-2	惠玉	小脯	斤	1	30	30
4	2012-10-2	惠玉	肉松	斤	1	32	32
5	2012-10-4	秀枝	肉松	斤	1.5	32	48
6	2012-10-4	雪红	小脯	斤	1	30	30
7	2012-10-4	惜兰	小脯	斤	2.67	30	80
8	2012-10-4	惜兰	肉松	斤	1.56	32	50
9	2012-10-4	少荣	小脯	斤	6	30	180
10	2012-10-4	零售	小脯	斤	8	30	240
11	2012-10-5	零售	大脯	斤	4.68	32	150
12	2012-10-5	惠玉	小脯	斤	2	30	60
13	2012-10-5	零售	小脯	斤	2	30	60
14	2012-10-6	秀枝	小脯	斤	1	30	30
15	2012-10-7	美燕	小脯	斤	2	30	60
16	2012-10-7	零售	肉松	斤	2	32	64

图10-27 零售明细表

零售的量不是很多，不用考虑太复杂，将数量全部当作“斤”处理。

```
=ROUND(SUMIF(零售明细!C:C,A9,零售明细!E:E),)
```



库存量公式为：

=B9-C9-D9

当然这样的库存量都是理论上的，与实际数据会存在差异，最好能够隔一个月清点库存，以得到更准确的数据。

10.3 左手系统，右手Excel

Excel并不是单打独斗的，结合其他软件如Access或者系统威力更强大。1+1>2就体现在这里。

10.3.1 巧借考勤系统

刚开始记录考勤都是靠手工录入，毕竟人数也不多，操作不是很麻烦。如图10-28所示，就是原始的考勤录入表格。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		12月份考勤汇总表													
2															
3		方献		金顺		丽真		明果		少荣		万芳		锡卢	
4	日期	上午	下午	上午	下午	上午	下午	上午	下午	上午	下午	上午	下午	上午	下午
26	12-22			√											
27	12-24			√	√	√	√	√	√	√	√	√	√	√	√
28	12-25			√	√	√	√	√	√	√	√	√	√	√	√
29	12-26					√	√	√	√	√	√	√	√	√	√
30	12-27	√	√	√	√	√	√	√	√	√	√	√	√	√	√
31	12-28	√		√		√	√	√	√	√	√	√	√	√	√
32	12-29	√		√	√	√	√	√	√	√	√	√	√	√	√
33	12-30	√		√	√	√	√	√	√	√	√	√	√	√	√
34	12-31	√		√	√	√	√	√	√	√	√	√	√	√	√
35	加班														
36	合计	10.5		25.5		28		28		27.5		28		28	

图10-28 手工录入的考勤表



手工录入的时候难免会出现差错,比如记录人漏记一天半天的。次数多了就会产生争议,记录人说记录没问题,某人说少记录,记得不准。一来二去,就伤了感情。

后来,老板大手一拍:买个指纹打卡机,以后根据打卡记录统计就不会出差错。刚开始的前两个月,很多人都不习惯,偶尔还是会漏打卡,到第三个月起就养成了习惯。这样借助打卡记录,可以获取更准确的数据,再多人的考勤都可以处理。

每到月底,卢子只需将打卡记录导入到U盘,再将U盘数据导入到考勤系统中,最后转换成Excel格式即可。

如图10-29所示,是考勤管理系统的操作界面。




图10-29 考勤管理系统操作界面

STEP 01 如图10-30所示,打开考勤管理系统,选择“数据”菜单中的“USB闪存管理”命令。在弹出的“U盘数据类型选择”对话框,保持默认设置不变,单击“确认”按钮。






图10-30 USB管理

STEP 02 如图10-31所示，在USB闪存管理的操作界面中，选择“导入记录数据至电脑”选项，单击“从U盘导入记录”按钮，就会逐步将U盘数据导入到电脑，数据量比较大，速度会稍微慢点。




图10-31 将记录导入电脑

STEP 03 如图10-32所示，将数据导入到电脑后就可以进行各种查询。如查询2014年7月份考勤，就设置起始日期为2014-7-1，结束日期为2014-7-31，就会出现这个时间段的所有明细。






图10-32 查询考勤记录

STEP 04 如图10-33所示，单击“导出”按钮，选择需要导出的字段，单击“确认”按钮。



图10-33 导出考勤记录

STEP 05 如图10-34所示，将Excel文件另存在桌面。




图10-34 将记录另存在桌面

如图10-35所示，就是导出的考勤记录。

这样的数据并不能直接使用，还需要进一步加工才可以。

	A	B	C	D
1	姓名	日期时间		
2	锡卢	2014-7-1 上午 09:02:15		
3	锡卢	2014-7-2 上午 08:38:34		
4	锡卢	2014-7-3 上午 08:39:29		
5	锡卢	2014-7-4 上午 08:40:03		
6	锡卢	2014-7-5 上午 08:40:21		
7	锡卢	2014-7-5 下午 07:37:55		
8	锡卢	2014-7-6 上午 08:38:19		
9	锡卢	2014-7-7 上午 08:49:37		
10	锡卢	2014-7-8 上午 08:39:50		
11	锡卢	2014-7-9 上午 08:41:06		
12	锡卢	2014-7-10 上午 08:45:37		
13	锡卢	2014-7-11 上午 08:45:57		
14	锡卢	2014-7-12 上午 08:42:07		
15	锡卢	2014-7-14 上午 08:38:49		
16	锡卢	2014-7-15 上午 08:39:44		
17	锡卢	2014-7-16 上午 08:45:35		
18	锡卢	2014-7-17 上午 08:40:28		
19	锡卢	2014-7-18 上午 08:47:10		
20	锡卢	2014-7-19 上午 08:38:52		

图10-35 导出的考勤记录



10.3.2 考勤数据再整理

日期、时间合并在一起，不利于统计数据，需要借助分列将数据分开。

STEP 01 如图10-36所示，选择B列，然后单击“数据”选项卡中的“分列”按钮，并在弹出的“文本分列向导”对话框中保持默认设置不变，单击“下一步”按钮。




图 10-36 文本分列向导 1

STEP 02 如图10-37所示，在“分割符号”选项组选中“空格”复选框，单击“完成”按钮。




图 10-37 文本分列向导 2



分列后的效果如图10-38所示。

分列后的字段名有的没有，需要补全，然后进行简单的美化。如图10-39所示，就是美化后的效果。仔细观察，会发现一个问题，时间段只有上午和下午，没有晚上。这明显不符合实际情况，里面有两个时间——下午7:37:55和下午7:41:11，应该叫晚上才更合适。

	A	B	C	D
1	姓名	日期时间		
2	锡卢	2014-7-1	上午	9:02:15
3	锡卢	2014-7-2	上午	8:38:34
4	锡卢	2014-7-3	上午	8:39:29
5	锡卢	2014-7-4	上午	8:40:03
6	锡卢	2014-7-5	上午	8:40:21
7	锡卢	2014-7-5	下午	7:37:55
8	锡卢	2014-7-6	上午	8:38:19
9	锡卢	2014-7-7	上午	8:49:37
10	锡卢	2014-7-8	上午	8:39:50
11	锡卢	2014-7-9	上午	8:41:06
12	锡卢	2014-7-10	上午	8:45:37
13	锡卢	2014-7-11	上午	8:45:57
14	锡卢	2014-7-12	上午	8:42:07
15	锡卢	2014-7-14	上午	8:38:49
16	锡卢	2014-7-15	上午	8:39:44
17	锡卢	2014-7-16	上午	8:45:35
18	锡卢	2014-7-17	上午	8:40:28
19	锡卢	2014-7-18	上午	8:47:10

图10-38 分列后的效果

	A	B	C	D
1	姓名	日期	时间段	时间
7	锡卢	2014-7-5	下午	7:37:55
32	锡卢	2014-7-30	下午	7:41:11
57	明果	2014-7-23	下午	1:36:31
58	明果	2014-7-23	下午	1:36:34
59	明果	2014-7-23	下午	1:36:37
61	明果	2014-7-24	下午	1:30:15
63	明果	2014-7-25	下午	1:33:17
65	明果	2014-7-26	下午	12:36:59
67	明果	2014-7-27	下午	1:32:29
69	明果	2014-7-28	下午	1:30:24
72	明果	2014-7-30	下午	1:33:20
74	明果	2014-7-31	下午	6:16:13

图10-39 默认时间段和实际不符

正常晚上加班的打卡时间是7点到8点，所以将这个时间段更改为晚上，其他不变。

```
=IF(AND(C2="下午",D2<--"08:00:00",D2>--"07:00:00"),"晚上",C2)
```

打卡记录难免会出现重复，比如下午1:36:31打了一次，1:36:34又打一次。若有这种重复，直接汇总就会出现错误，这时需要进行判断是不是第一次打卡的记录，若是，其他记录就直接忽略掉。

卢子所在的单位是小公司，要求没那么严格，只要上午、下午、晚上有打卡就行，时间上没硬性规定，晚点上班也没事。

如图10-40所示，根据姓名、日期、标准时间段三个条件判断是不是第一次打卡。

	A	B	C	D	E	F
1	姓名	日期	时间段	时间	标准时间段	首次打卡
2	锡卢	2014-7-1	上午	9:02:15	上午	1
3	锡卢	2014-7-2	上午	8:38:34	上午	1
4	锡卢	2014-7-3	上午	8:39:29	上午	1
5	锡卢	2014-7-4	上午	8:40:03	上午	1
6	锡卢	2014-7-5	上午	8:40:21	上午	1
7	锡卢	2014-7-5	下午	7:37:55	晚上	1
8	锡卢	2014-7-6	上午	8:38:19	上午	1
9	锡卢	2014-7-7	上午	8:49:37	上午	1
10	锡卢	2014-7-8	上午	8:39:50	上午	1
11	锡卢	2014-7-9	上午	8:41:06	上午	1
12	锡卢	2014-7-10	上午	8:45:37	上午	1
13	锡卢	2014-7-11	上午	8:45:57	上午	1
14	锡卢	2014-7-12	上午	8:42:07	上午	1
15	锡卢	2014-7-14	上午	8:38:49	上午	1
16	锡卢	2014-7-15	上午	8:39:44	上午	1

图10-40 判断是否为首次打卡




```
=IF(COUNTIFS(A$2:A2,A2,B$2:B2,B2,E$2:E2,E2)=1,1,"")
```

让首次打卡记录显示1，非首次显示空白，这样一来要统计就方便很多。

10.3.3 统计工资

前期的准备工作已经完成，接下来就是统计工资。

如图10-41所示，就是2014年7月份工资表。

上班天数是根据考勤表汇总得到，因为时间段是以半天计算，所以最后面乘以0.5。

```
=SUMIF(考勤记录!A:A,A3,考勤记录!F:F)*0.5
```

日薪是根据日薪对应表查询引用数据过来的。如图10-42所示，就是日薪对应表。按日薪统计的都是比较低级的人员，中级人员是按月薪统计，高级人员是按年薪统计。

	A	B	C	D	E	F
1	2014年7月份工资表					
2	姓名	上班天数	日薪	基本工资	回扣额	实际工资
3	锡卢					
4	明果					
5	万芳					
6	惠玉					
7	丽真					
8	秋爱					
9	雪红					
10	巧娇					
11	美燕					
12	惜兰					
13	少荣					
14	金顺					
15	方献					
16	永清					
17	起林					
18	春梅					

图10-41 2014年7月工资表

	A	B	C
1	姓名	日薪	
2	巧娇	74	
3	惜兰	112	
4	春梅	93	
5	明果	89	
6	雪红	48	
7	秋爱	110	
8	少荣	81	
9	丽真	69	
10	永清	54	
11	锡卢	49	
12	方献	60	
13	金顺	48	
14	万芳	46	
15	惠玉	85	
16	起林	116	
17	美燕	74	
18			

图10-42 日薪对应表

```
=VLOOKUP(A3,日薪对应表!A:B,2,)
```

基本工资就是上班天数乘以日薪。

```
=ROUND(B3*C3,)
```

回扣额是根据零售明细表获取，在销售的时候有说过，客户每购买一斤产品，可获得2元的回扣。如图10-43所示，就是零售明细表。

	A	B	C	D	E	F	G	H	I	J	K	L
1	日期	客户	产品名称	类型	单位	数量	单价	特殊扣减 (增加)	金额	回扣额	收款人	单号
2	2014-7-3	惜兰	肉松	肉制品	斤	0.5	32		16	1	少荣	001
3	2014-7-3	零售	小脯	肉制品	斤	2	30		60	0	少荣	002
4	2014-7-3	惜兰	肉松	肉制品	斤	1	32		32	2	少荣	003
5	2014-7-5	惜兰	腊肠	肉制品	斤	2	22		44	4	少荣	004
6	2014-7-5	金顺	肉松	肉制品	斤	1	32		32	2	少荣	005
7	2014-7-6	永清	肉松	肉制品	斤	0.5	32		16	1	少荣	006
8	2014-7-7	金顺	小脯	肉制品	斤	1	30		30	2	少荣	007
9	2014-7-7	金顺	肉松	肉制品	斤	2	32		64	4	少荣	007
10	2014-7-8	金顺	大脯	肉制品	斤	2	35		70	4	少荣	008
11	2014-7-8	零售	肉松	肉制品	斤	1.5	32		48	0	少荣	009
12	2014-7-8	零售	小脯	肉制品	斤	1	30		30	0	少荣	009
13	2014-7-11	惠玉	小脯	肉制品	斤	2	30		60	4	少荣	010
14	2014-7-11	惠玉	肉松	肉制品	斤	1	32		32	2	少荣	010
15	2014-7-11	永清	小脯	肉制品	斤	1	30		30	2	少荣	011
16	2014-7-11	惜兰	肉松	肉制品	斤	0.5	32		16	1	少荣	012

图10-43 零售明细表

=SUMIF(零售明细表!B:B,A3,零售明细表!J:J)

实际工资就是基本工资加上回扣额。

=D3+E3

10.3.4 制作工资条

工资对于每个人而言都是隐私，不能直接将统计好的工资表给所有工人看，需要给他们每人发放的仅仅是属于他们自己的工资条而已。

STEP 01 如图10-44所示，将汇总好的工资表复制、粘贴成值到新表，然后添加一个辅助列，输入1，然后下拉到16，再将输入的序号复制到G19单元格中。

	A	B	C	D	E	F	G	H
1	2014年7月份工资表							
2	姓名	上班天数	日薪	基本工资	回扣额	实际工资	辅助列	
3	锡卢	16	112	1792	0	1792	1	
4	明果	19.5	109	2126	0	2126	2	
5	万芳	30	98	2940	6	2946	3	
6	惠玉	32.5	90	2925	0	2925	4	
7	丽真	32	61	1952	26	1978	5	
8	秋爱	30.5	65	1983	2	1985	6	
9	雪红	31.5	77	2426	6	2432	7	
10	巧娇	30	74	2220	3	2223	8	
11	美燕	30.5	44	1342	8	1350	9	
12	惜兰	30	58	1740	0	1740	10	
13	少荣	29	46	1334	17	1351	11	
14	金顺	0.5	73	37	3	40	12	
15	方献	0.5	113	57	28	85	13	
16	永清	28	44	1232	0	1232	14	
17	起林	31.5	107	3371	5	3376	15	
18	春梅	32.5	56	1820	0	1820	16	
19								1
20								2
21								3
22								4

图10-44 添加辅助列



STEP 02 如图10-45所示，复制表头到工资表下面。

18	春梅	32.5	56	1820	0	1820	16
19	姓名	上班天数	日薪	基本工资	回扣额	实际工资	1
20	姓名	上班天数	日薪	基本工资	回扣额	实际工资	2
21	姓名	上班天数	日薪	基本工资	回扣额	实际工资	3
22	姓名	上班天数	日薪	基本工资	回扣额	实际工资	4
23	姓名	上班天数	日薪	基本工资	回扣额	实际工资	5
24	姓名	上班天数	日薪	基本工资	回扣额	实际工资	6
25	姓名	上班天数	日薪	基本工资	回扣额	实际工资	7
26	姓名	上班天数	日薪	基本工资	回扣额	实际工资	8
27	姓名	上班天数	日薪	基本工资	回扣额	实际工资	9
28	姓名	上班天数	日薪	基本工资	回扣额	实际工资	10
29	姓名	上班天数	日薪	基本工资	回扣额	实际工资	11
30	姓名	上班天数	日薪	基本工资	回扣额	实际工资	12
31	姓名	上班天数	日薪	基本工资	回扣额	实际工资	13
32	姓名	上班天数	日薪	基本工资	回扣额	实际工资	14
33	姓名	上班天数	日薪	基本工资	回扣额	实际工资	15
34	姓名	上班天数	日薪	基本工资	回扣额	实际工资	

图10-45 复制表头

STEP 03 选择G列，单击“数据”选项卡中的“升序”按钮，排序后将没用的辅助列删除，就得到了工资条，最终效果如图10-46所示。

	A	B	C	D	E	F
1	2014年7月份工资表					
2	姓名	上班天数	日薪	基本工资	回扣额	实际工资
3	锡卢	16	112	1792	0	1792
4	姓名	上班天数	日薪	基本工资	回扣额	实际工资
5	明果	19.5	109	2126	0	2126
6	姓名	上班天数	日薪	基本工资	回扣额	实际工资
7	万芳	30	98	2940	6	2946
8	姓名	上班天数	日薪	基本工资	回扣额	实际工资
9	惠玉	32.5	90	2925	0	2925
10	姓名	上班天数	日薪	基本工资	回扣额	实际工资
11	丽真	32	61	1952	26	1978
12	姓名	上班天数	日薪	基本工资	回扣额	实际工资
13	秋爱	30.5	65	1983	2	1985
14	姓名	上班天数	日薪	基本工资	回扣额	实际工资
15	雪红	31.5	77	2426	6	2432

图10-46 工资条





知识扩展

对于单行标题的工资条制作就比较容易，但很多时候会出现双行标题的工资条，这种工作条的制作难度就比较大。

如图10-47所示，这是3月工资薪金明细表，要制作双行标题工资条。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1	3月工资薪金明细表																
2	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份		扣除部份					实发金额
3										补工资	电脑补助	个税	罚款	迟到早退	借支	社保	
4	1	天丽华	业务部	31	31	5000		5000	5000							-266.88	4733.12
5	2	姚映全	业务部	31	31	10000		10000	10000							0	10000
6	3	叶应波	业务部	31	31	4500	100	4600	4600		100					-270.88	4429.12
7	4	尹连菊	技术部	31	31	3200	100	3300	3300							-270.88	3029.12
8	5	尹连琼	市场部	31	31	5800	100	6100	6100		100					-270.88	5929.12
9	6	尹思平	技术部	31	31	4500	100	4600	4600		100					-270.88	4429.12
10	7	永开虎	技术部	31	31	2700	100	3000	3000							-270.88	2729.12
11	8	余洪菊	项目部	31	31	4000	100	4100	4100							-266.88	3833.12
12	9	余兰美	项目部	31	31	5000	100	5100	5100							-270.88	4829.12
13	10	余文翠	项目部	31	31	5000	100	5100	5100		100					-270.88	4929.12
14	11	余学利	项目部	31	31	3700	100	3800	3800							-270.88	3529.12
15	12	俞廷宝	项目部	31	31	4500	100	4600	4600		100					-270.88	4429.12
16	13	袁树存	项目部	31	31	4500	100	4600	4600		100					-270.88	4429.12
17	14	曾建忠	项目部	31	31	3000	100	3100	3100							0	3100

图10-47 双行表头工资表

STEP 01 如图10-48所示，将表头复制、粘贴在表格下面的空白位置。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
28																	
29	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份		扣除部份					实发金额
30										补工资	电脑补助	个税	罚款	迟到早退	借支	社保	
31																	

图10-48 粘贴表头

STEP 02 编号是唯一的，可以借助它用VLOOKUP函数进行引用。在A31单元格输入1，并在B31单元格输入公式，向右复制。

```
=VLOOKUP($A31,$A$4:$Q$22,COLUMN(),0)
```

STEP 03 添加边框，如图10-49所示。



文件

开始

插入

页面布局

公式

数据

审阅

视图

开发工具

剪贴板

剪切的

复制

粘贴

格式刷

微软雅黑

10

A⁺ A⁻

B

I

U

图10-49 添加边框

STEP 04 将编号下拉会自动增加，而公式是根据编号进行引用数据，也就是下拉就能获取其他人员的工资条。如图10-50所示，选择区域A29:Q32，然后一直往下拉。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
26																	
27																	
28																	
29	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份		扣除部份				实发金额	
30										补工资	电脑补助	个税	罚款	迟到早退	借支	社保	
31	1	天丽华	业务部	31	31	5000	0	5000	5000	0	0	0	0	0	0	-266.88	4733.12
32																	
33																	
34																	

图10-50 下拉数据生成工资条

STEP 05 因为行数不好把握，如图10-51所示，有多余的再选择删除即可。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	
100																		
101	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份		扣除部份				实发金额		
102										补工资	电脑补助	个税	罚款	迟到早退	借支	社保		
103	19	张炳全	项目部	31	31	2800	100	2900	2900	0	0	0	0	0	0	-270.67	2629.33	
104																		
105	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份		扣除部份				实发金额		
106										补工资	电脑补助	个税	罚款	迟到早退	借支	社保		
107	20	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	#N/A	###	#N/A	###	#N/A	#N/A	
108																		
109																		

图10-51 工资条生成后的效果

如果有新员工进来，就在工资明细表添加，记得一定要写上编号。添加完成后，如图10-52所示，选择区域A101:Q104后向下复制，即可生成新员工的工资条。如果人员没有变动可以不做

处理。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
98	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补工资	电脑补助	个税	罚款	迟到早退	借支	社保	实发金额
99	18	张炳富	项目部	31	31	3100	100	3200	3200	0	0	0	0	0	0	0	3200
100																	
101	编号	姓名	部门	应出勤天数	实出勤天数	基本工资	全勤奖	本月工资	出勤工资	补贴部份			扣除部份				实发金额
102										补工资	电脑补助	个税	罚款	迟到早退	借支	社保	
103	19	张炳金	项目部	31	31	2800	100	2900	2900	0	0	0	0	0	0	-270.67	2629.33
104																	
105																	

图10-52 添加数据的处理

10.4 小结与练习

Excel如果能够处理好各个表格之间的关系，再联合系统威力将无穷大！在学习Excel之前如果能够学点数据库的理论，会使数据处理变得更简单。

1. 上班时间为8:00，如何根据上班打卡时间判断是否迟到？如图10-53所示。

	A	B
1	上班打卡时间	迟到/正常
2	7:59	正常
3	8:01	迟到
4	8:30	迟到
5	7:58	正常
6	8:00	正常
7	7:30	正常
8	8:10	迟到
9		

图10-53 判断打卡时间是否迟到



2. 如图10-54所示，用SUM函数对工资进行求和，发现合计多了一分钱，怎么回事？

B9 : =SUM(B2:B8)					
	A	B	C	D	
1	人员	工资			
2	俞廷宝	5000.50			
3	袁树存	3000.50			
4	曾建忠	2000.00			
5	曾永祥	2500.50			
6	张炳聪	2800.00			
7	张炳芬	3200.50			
8	张炳富	2000.00			
9	合计	20502.01			
10					

图10-54 合计多了1分钱

3. 全书已经学完，利用自己所学的函数知识，对工作上的表格进行改进。



写在后面的话

函数非常精彩，但也不要过度迷恋函数。函数虽然可以解决非常多的问题，但是并非万能，它不过是Excel中的一个功能而已。Excel除了函数还包含基础操作、数据透视表、图表、SQL、VBA，每个功能都非常强大，但也有自身的局限性。合理地利用每个功能自身的特点才是正道。

比如对十几万行数据多角度汇总分析，虽然用函数也可以做到，但是非常卡，而且设置公式很不方便，这时就要借助数据透视表。

又如多个工作簿合并，用函数非常难做到，这时就要借助SQL或者VBA。

最后，别傻傻地用函数去实现它做不到的事，你让函数画个图表看看！

用人之短，天下无可用人；用人之长，天下皆是可用之人。在Excel中也一样，用每个功能的长处，每个功能都非常好用。

